
2.7 La forme spéciale lambda . . . . . . . . . . . . . . . . . . . . . . . . . . . . 33

4.3.1 Récursion sur les nombres . . . . . . . . . . . . . . . . . . . . . . . . 68

10.6.2 Trois réalisations concrètes du type ensemble . . . . . . . . . . . . . 209
10.6.3 Fonctions d’interface . . . . . . . . . . . . . . . . . . . . . . . . . . . 210
10.6.4 Version abstraite des opérations de base . . . . . . . . . . . . . . . . 211

12 Instructions altérantes et vecteurs 245
12.1 Introduction . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 245
12.2 L’instruction d’affectation . . . . . . . . . . . . . . . . . . . . . . . . . . . . 247
12.3 Altération de structures . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 248
12.4 Les vecteurs . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 249

12.4.1 Pourquoi des vecteurs? . . . . . . . . . . . . . . . . . . . . . . . . . . 249
12.4.2 Les primitives vectorielles . . . . . . . . . . . . . . . . . . . . . . . . 250
12.4.3 Tri par insertion . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 251
12.4.4 Retournement d’une liste et d’un vecteur . . . . . . . . . . . . . . . 254

1 INTRODUCTION 7

1 Introduction

Ce livre se veut un support à l’apprentissage de la programmation. Il existe plusieurs
grands styles de programmation, parfois appelés “paradigmes” de programmation. On
distingue notamment les styles impératif, fonctionnel, logique, orienté objet. Ce texte
développe les bases du style fonctionnel de programmation et utilise le langage Scheme; il
s’adresse au lecteur n’ayant aucune connaissance de programmation fonctionnelle. Il (ou
elle) peut avoir ou ne pas avoir d’expérience antérieure de la programmation impérative
(le style le plus répandu), dans un langage tel que Fortran, Pascal ou C.

Comme son nom l’indique, la programmation fonctionnelle a pour concept de base la
fonction, notion mathématique fondamentale bien connue du lecteur, mais sur laquelle il
n’est pas inutile de revenir. En particulier, on verra jusqu’à quel point on peut assimiler
l’activité mathématique consistant à définir une fonction et l’activité informatique
consistant à programmer une fonction. On présentera à ce propos la notion de récursivité,
centrale en programmation fonctionnelle.

Dans ce livre, on insiste sur les principes de la programmation fonctionnelle mais ils
n’occupent guère d’espace, car ils sont peu nombreux, simples et clairs. Il est nécessaire
de les assimiler parfaitement pour comprendre les programmes en profondeur, et surtout
pour pouvoir programmer soi-même.1 Ce livre se veut à orientation pratique; son but est
d’aider le lecteur à apprendre à programmer et à concrétiser cet apprentissage par une
résolution simple et fiable de problèmes variés, impliquant l’emploi correct d’un langage
de programmation, Scheme (en fait, un sous-ensemble réduit de Scheme).

Ce texte ne constitue pas un recueil de problèmes et de programmes.2 Les exercices
sont donc relativement peu nombreux, mais ils sont très importants. Il servent d’abord
à présenter une technique particulière ou à démontrer concrètement son utilité, mais ils
sont aussi et surtout destinés à convaincre le lecteur d’un fait largement méconnu : la
programmation est essentiellement une technique, ou un ensemble de techniques. Bien
programmer requiert toujours compétence, rigueur et soin; ce n’est qu’occasionnellement
que le programmeur doit en outre faire preuve de créativité, d’astuce, voire même de
sens esthétique. Dans nos exercices corrigés, la démarche de résolution a donc autant
d’importance que le programme produit. Nous nous sommes efforcés d’apporter des
solutions complètes et détaillées aux problèmes traités; en particulier, chaque fonction
auxiliaire est spécifiée; nous pensons en effet que la documentation doit toujours
accompagner le programme. Notons enfin que la programmation s’apprend surtout . . .
en programmant. Le lecteur ne pourra tirer un bénéfice optimal de cet ouvrage que s’il

1C’est la raison pour laquelle certains points sont présentés deux fois. La première présentation donne
l’idée, immédiatement mise en œuvre; la seconde situe le concept dans un contexte plus large, permettant de
mieux apprécier son importance. Par contre, ce livre n’est pas un manuel de référence du langage Scheme

et de nombreux aspects importants du langage ne sont évoqués que de manière indirecte et occasionnelle.
Par exemple, nous ne mentionnons pas explicitement que le nombre 3/4 en Scheme peut s’écrire 3/4, 0.75,
75e-2, .75+0i, etc., ni que la première écriture seule permet un calcul “exact”, ni enfin que cette écriture
fractionnaire n’est pas admise sur tous les systèmes Scheme. Ces aspects ont leur importance mais leur
traitement n’aurait pas contribué à l’objectif premier de cet ouvrage.

2Le lecteur consultera utilement [6], où il trouvera de nombreux problèmes supplémentaires.

écrit ses propres programmes et expérimente largement sur machine.3

1.1 De la fonction mathématique à la fonction programmée

Considérons la définition suivante :

La factorielle d’un entier naturel n
est le produit des entiers de 1 à n.

Le mathématicien admettra la définition en observant qu’elle repose sur des notions
antérieurement admises (entier naturel, ensemble, produit d’un ensemble de nombres) et
que ces notions sont correctement combinées.4

L’algorithmicien suivra une démarche analogue, mais ira plus loin. En effet, une
définition mathématiquement acceptable d’une fonction f (ici, la fonction qui à tout
entier naturel n associe sa factorielle) n’implique pas automatiquement l’existence d’un
processus d’évaluation qui, au départ de l’expression syntaxique f(x), calcule la valeur

de cette expression, c’est-à-dire l’image de x par f . Si de tels algorithmes existent, ils
peuvent n’être pas d’égal intérêt pour l’algorithmicien, qui s’interrogera notamment sur
leur efficacité. Pour le mathématicien, les fonctions

g : n 7→
n

∑

i=1

i2

et

h : n 7→ n(n + 1)(2n + 1)

6

sont égales, parce que, pour tout entier naturel n, les images de n par g et par h sont égales.
L’algorithmicien notera une différence importante entre g et h : la définition de g suggère
que le calcul de l’image de 100 par g, c’est-à-dire le processus d’évaluation de g(100),
requiert 100 multiplications et 99 additions tandis que la définition de h suggère que le
calcul de l’image de 100 par h requiert seulement trois multiplications, deux additions et
une division. Dès que l’algorithmicien connâıt l’égalité mathématique de g et h, il peut
cesser de s’intéresser à g.

Le programmeur ira encore un peu plus loin, puisqu’il doit coder l’algorithme dans
un certain langage de programmation. Les primitives (opérations fournies par le système)
varient d’un langage à l’autre; en outre, dans un langage fixé, il arrive que plusieurs

3Des systèmes Scheme performants et gratuits existent en abondance; on peut les
télécharger via Internet. Un point d’entrée intéressant vers des références multiples est
http://www.cs.indiana.edu/scheme-repository/home.html .

4En particulier, les cas n = 0 et n = 1 ne sont pas exclus; on convient que “l’ensemble des entiers de 1 à
0” est l’ensemble vide et que “l’ensemble des entiers de 1 à 1” est le singleton {1}; le produit de chacun de
ces ensembles vaut 1. (Toute autre convention conduirait à des difficultés.) Notons en outre que la notion
de produit d’un ensemble (fini) de nombres n’est pas primitive, mais dérivée de la notion de produit de
deux nombres, opération associative et commutative.

solutions soient envisageables pour un même algorithme; le programmeur choisit en
fonction de plusieurs facteurs (efficacité, clarté, et même goûts personnels).

Une définition donnée en langage naturel, même claire et précise, devra être formalisée.
Le mathématicien écrira simplement

n! =def 1 ∗ 2 ∗ · · · ∗ n ,

les points de suspension ayant ici une signification parfaitement claire.5

L’algorithmicien et le programmeur n’admettront pas les points de suspension si
leur signification n’est pas formellement spécifiée.6 En programmation fonctionnelle, la
technique privilégiée pour l’élimination des points de suspension est la récursivité. Pour
représenter de manière finie le tableau infini

0! = 1 ,
1! = 1 ∗ 1 ,
2! = 2 ∗ (1 ∗ 1) ,
3! = 3 ∗ (2 ∗ (1 ∗ 1)) ,
...

...

on observe qu’il peut se récrire en

0! = 1 ,
1! = 1 ∗ 0! ,
2! = 2 ∗ 1! ,
3! = 3 ∗ 2! ,
...

...

ce qui suggère l’écriture plus concise suivante :

n! =def [ if n = 0 then 1 else n ∗ (n − 1)! ] .

On voit en particulier que, d’après cette définition,

4! = 4 ∗ 3! = 4 ∗ (3 ∗ 2!) = 4 ∗ (3 ∗ (2 ∗ 1!)) = 4 ∗ (3 ∗ (2 ∗ (1 ∗ 0!))) = 4 ∗ (3 ∗ (2 ∗ (1 ∗ 1))) .

La mâıtrise de la récursivité est la première et la principale difficulté rencontrée lors de
l’apprentissage de la programmation fonctionnelle.7 Par ailleurs, l’approche fonctionnelle
préserve entièrement le principal atout de la notion mathématique de fonction, à savoir
la facilité avec laquelle des fonctions se composent et se combinent pour donner lieu à de
nouvelles fonctions. Cela permet l’élaboration de fonctions très compliquées au moyen d’un
petit nombre de mécanismes élémentaires produisant des fonctions à partir de fonctions
primitives, données au départ, et/ou de fonctions antérieurement construites.

5Cette écriture est acceptable même pour n = 0 et n = 1; on a 0! = 1 et 1! = 1.
6De plus, ils souhaiteront peut-être lever le non-déterminisme et préciser l’ordre dans lequel les

multiplications seront effectuées. C’est obligatoire pour le programmeur, sauf s’il utilise un langage non
déterministe.

7C’est quasi la seule si, comme dans ce livre, on se limite à la programmation fonctionnelle élémentaire;
cette simplicité est un avantage décisif par rapport à d’autres styles de programmation.

1.2 La récursivité

Le concept de récursivité étant essentiel, il est intéressant de l’aborder dans un cadre
général, avant même de considérer son rôle particulier en programmation fonctionnelle.
Le point important de cette courte étude est de permettre au lecteur de déterminer si une
définition récursive est bien une définition, au sens habituel du terme. Nous verrons en
effet qu’un texte ayant la syntaxe d’une définition récursive peut très bien ne rien définir
du tout !

1.2.1 Principe et exemples

Une définition de fonction telle que

h(x, y) =def

√

x2 + y2

présuppose l’existence des fonctions d’addition, d’élévation au carré et d’extraction de
racine carrée. Plus précisément, on admettra la validité d’une égalité telle que

h(3, 4) = 5

en se basant sur des égalités concernant les trois fonctions auxiliaires déjà citées, en
l’occurrence

32 = 9 , 42 = 16 , 9 + 16 = 25 ,
√

25 = 5 .

On le souligne, les égalités justificatives concernent seulement les fonctions auxiliaires,
prédéfinies, et non pas la fonction h elle-même, nouvellement définie. On note également
que le nom h intervient seulement à gauche du symbole de définition “=def ”.

Considérons alors, sur le domaine des entiers naturels, la définition classique de la suite
de Fibonacci :

fib(n) =def [ if n < 2 then n else fib(n − 1) + fib(n − 2) ] .

Cette définition est récursive parce que le nom de l’entité que l’on définit, à savoir la
fonction fib, intervient non seulement à gauche du symbole “=def ”, mais aussi dans
l’expression définissante, à droite de ce même symbole. On admet immédiatement les
égalités

fib(0) = 0 , fib(1) = 1 ,

parce que fib(n) est explicitement spécifié égal à n si n < 2. Sur base de ces deux égalités
évidentes, et des égalités supplémentaires

2 − 1 = 1 , 2 − 2 = 0 , 1 + 0 = 1 ,

on admet ensuite l’égalité fib(2) = 1 ou, plus explicitement, la châıne d’égalités

fib(2) = fib(2 − 1) + fib(2 − 2) = fib(1) + fib(0) = 1 + 0 = 1 ;

de proche en proche, on constate que l’expression fib(n) a une valeur bien précise pour tout
entier naturel n. La différence essentielle entre la définition (non récursive) de la fonction h
et celle (récursive) de la fonction fib est l’intervention, dans les égalités justificatives, de
la fonction définie fib elle-même. On conçoit immédiatement le danger de la récursivité,
assimilable à un “cercle vicieux”. Ce danger existe : des “définitions” telles que

α(x) =def α(x) ,
β(x) =def β(x + 1) ,
γ(x) =def γ(x) + 1 ,

sont clairement à déconseiller ! Nous verrons comment éviter le risque de cercle vicieux.

Beaucoup de relations de récurrence classiques de l’arithmétique se transforment
aisément en définitions récursives de fonctions. Par exemple, la fonction cb, qui à tout
couple d’entiers naturels (n, p) tels que p ≤ n associe le coefficient binomial correspondant,8

donne lieu à la relation de récurrence bien connue

cb(n, p) = cb(n − 1, p − 1) + cb(n − 1, p) ,

valable à condition que p soit strictement compris entre 0 et n. Cette relation peut servir
à définir la fonction cb; il suffit de préciser, en outre, la valeur de cb(n, p) quand p vaut 0
ou n. On obtient ainsi une (première) définition récursive de la fonction cb :

cb1(n, p) =def [ if n=p or p=0 then 1 else cb1(n−1, p−1) + cb1(n−1, p) ] .

On voit notamment que

cb1(3, 1) = cb1(2, 0) + cb1(2, 1) = 1 + cb1(1, 0) + cb1(1, 1) = 1 + 1 + 1 = 3 .

Une autre relation de récurrence bien connue est

cb(n, p) = [n ∗ cb(n − 1, p − 1)]/p ,

valable si p n’est pas nul; il suffit de préciser la valeur de cb(n, 0) pour obtenir une seconde
définition récursive :

cb2(n, p) =def [ if p = 0 then 1 else [n ∗ cb2(n − 1, p − 1)]/p ] .

On voit notamment que

cb2(4, 2) = [4 ∗ cb2(3, 1)]/2 = [4 ∗ 3 ∗ cb2(2, 0)/1]/2 = (4 ∗ 3 ∗ 1)/2 = 6 .

Remarque. Les fonctions cb1 et cb2 sont mathématiquement égales, puisqu’elles
contiennent exactement les mêmes couples. Pour l’informaticien, elles sont cependant
très différentes. En particulier, comme nous le verrons plus loin, cb1 est inefficace mais
cb2 est efficace. La différence est considérable.9

8c’est-à-dire le coefficient de xp dans le développement du binôme (1 + x)n.
9Par exemple, l’évaluation de cb1(20, 10) = 184 756 prend beaucoup plus de temps que celle

de cb2(200, 100) = 90548514656103281165404177077484163874504589675413336841320; le lecteur verra
comment le vérifier expérimentalement au chapitre suivant.

1.2.2 Domaine de validité

Les définitions récursives des fonctions f , cb1 et cb2 sont acceptables parce que la
détermination de la valeur associée à un point quelconque de leur domaine exige une
suite finie de calculs, puisque les développements sont finis. Une définition telle que

g(n) =def [ if n < 2 then n else g(n − 1) + g(n + 2) ] ,

qui ressemble à la définition de la fonction de Fibonacci fib donnée plus haut, n’a pas cette
propriété :

g(2) = g(1) + g(4) = 1 + g(3) + g(6) = 1 + g(2) + g(5) + g(5) + g(8) = · · ·

Même si, par extraordinaire, il existait une fonction g vérifiant l’égalité

g(n) = [ if n < 2 then n else g(n − 1) + g(n + 2) ]

pour tout entier naturel n, on ne considérerait pas la définition précédente comme une
définition (récursive) acceptable, notamment parce que l’expression g(2) donne lieu à un
développement infini.

Pour préciser notre notion d’acceptabilité, considérons un cas célèbre (l’expression
even?(n) est vraie si n est pair et fausse sinon) :

t(n) =def if n = 1 then 1
else if even?(n) then t(n/2)
else t(3n + 1) .

On pense que ce texte définit la fonction t qui à tout entier strictement positif associe le
nombre 1 . . . mais cela n’a pas été démontré.10 On a par exemple

t(52)= t(26)= t(13)= t(40)= t(20)= t(10)= t(5)= t(16)= t(8)= t(4)= t(2)= t(1)=1 .

1.2.3 Terminaison, totalité

En pratique, il semble peu réaliste de développer une étude mathématique parfois difficile
chaque fois qu’une définition récursive d’une fonction f sur le domaine des entiers naturels
est envisagée, dans le seul but de prouver que le calcul de f(n) se termine pour tout n,
c’est-à-dire que la fonction f est partout définie, ou encore totale, sur le domaine N.

10A priori, la fonction t divise l’ensemble des naturels en deux parties. La première contient tout nombre
pour lequel le calcul se termine et la seconde tout nombre pour lequel le calcul ne se termine pas. Si n
appartient à la première partie, on a t(n) = 1; sinon, t(n) n’est pas défini. La conjecture qui, à ce jour,
n’a pas été démontrée, est que la seconde partie est vide.

Heureusement, dans beaucoup de cas, la conclusion est évidente. Considérons la définition
suivante, d’apparence compliquée :

h(n) =def if n = 0 then 1
else if n = 1 then 5
else if even?(n) then h(n/2) + h(n − 2)
else h((n−3)/2) + h(n − 1) ∗ h((n+1)/2) .

La valeur de h(n) est susceptible de dépendre des valeurs de h(n/2), h(n−1), h((n−3)/2),
h(n− 2) et h((n+1)/2), mais on observe que, chaque fois que h(n) dépend de h(m),
l’argument m est un entier naturel, strictement inférieur à m.

En conclusion, la fonction h est totale sur le domaine N et il est possible de calculer
h(n), quel que soit l’entier naturel n; un moyen simple, et raisonnablement efficace, consiste
à calculer successivement h(0), h(1), . . . , h(n). En particulier, on a :

n 0 1 2 3 4 5 6 7 8 9
h(n) 1 5 6 37 12 449 49 594 61 27426

Notons que, en pratique, la généralisation consistant à permettre que f(n) dépende
non seulement de f(n − 1) mais aussi de n’importe quel f(i) où i = 0, . . . , n − 1, est
rarement nécessaire; au contraire, on observe que le cas particulier où f(n) est défini en
terme de f(n−1) seulement (si n > 0) suffit le plus souvent. C’est vrai notamment dans
le cas des nombres de Fibonacci, moyennant une petite astuce. En effet, plutôt que de
définir immédiatement la fonction fib, on peut définir la fonction fib2 comme suit :

fib2 (0) =def (0, 1) ,
fib2 (n + 1) =def (b, a + b) , où (a, b) = fib2 (n) .

On observe que fib2 (n) dépend de fib2 (n− 1) (si n > 0) mais pas de fib2 (n− 2). On peut
alors démontrer que, pour tout naturel n, on a

fib2 (n) = (fib(n),fib(n+1)) .

On procède par récurrence. Pour n = 0, c’est vrai par construction. Si n > 0, on peut
supposer que la propriété est vraie pour n−1 et donc que fib2 (n−1) = (fib(n−1),fib(n)).
Par construction on a fib2 (n) = (fib(n),fib(n − 1) + fib(n)), c’est-à-dire fib2 (n) =
(fib(n),fib(n + 1)) .

1.2.4 Récurrence simple, récurrence complète

Le principe de récurrence vient d’être utilisé pour démontrer une certaine relation existant
entre deux fonctions définies récursivement. Plus généralement, il sera largement utilisé
par la suite pour démontrer les propriétés des programmes Scheme. Il est donc utile de
revenir sur ce principe ici. L’idée de base est que tout naturel est, soit 0, soit le successeur
d’un autre naturel, que l’on nomme son prédécesseur. Le principe de récurrence est un

mécanisme de raisonnement qui, en logique formelle, se représente par la règle d’inférence11

suivante :
P (0) , ∀n > 0 [P (n − 1) ⇒ P (n)]

∀n P (n)

Cette règle est la traduction formelle d’un énoncé bien connu :

Si la propriété P est vraie du nombre naturel 0,
si dès qu’elle est vraie d’un naturel n − 1, elle l’est aussi de n,
alors elle est vraie de tout naturel n.

Intuitivement, le principe de récurrence est évident car les prémisses peuvent se récrire en

P (0) , P (0) ⇒ P (1) , P (1) ⇒ P (2) , . . .

dont on déduit de proche en proche

P (0) , P (1) , P (2) , . . .

c’est-à-dire la conclusion. On justifie plus formellement le principe de récurrence sans
aucune difficulté. Supposons, par l’absurde, qu’une certaine propriété P vérifie les deux
prémisses mais pas la conclusion. L’ensemble des naturels pour lesquels la propriété est
fausse ne serait pas vide et admettrait donc un minimum m. Cependant, m ne peut pas
être 0 à cause de la première prémisse; il ne peut non plus être le successeur de m− 1, car
on aurait P (m− 1) sans avoir P (m), ce qui contredirait la seconde prémisse. On a obtenu
une contradiction : le nombre m ne peut donc pas exister.

Le principe de récurrence présenté ci-dessus est en fait le principe de récurrence simple;
le principe de récurrence complète prévoit que, pour établir P (n), on peut disposer de
P (n − 1) mais aussi de P (n − 2), . . . , P (0). La récurrence complète semble donc plus
puissante que la récurrence simple, mais elle lui est en fait équivalente. Cette variante
s’énonce classiquement comme suit :

Si dès que la propriété P est vraie des nombres naturels 0, . . . , n − 1
elle l’est aussi de n,
alors elle est vraie de tout naturel n.

La règle d’inférence formelle correspondante est

∀n [(∀m<n P (m)) ⇒ P (n)]

∀n P (n)

Intuitivement, le principe de récurrence complète est évident car la prémisse peut se récrire
en

P (0) , P (0) ⇒ P (1) , (P (0) ∧ P (1)) ⇒ P (2) , . . .

11Une règle d’inférence est une notation du type P1,...,Pm

C
; les Pi sont des énoncés nommés prémisses;

C est un énoncé nommé conclusion. La règle est correcte si la conclusion est conséquence logique des
prémisses, c’est-à-dire si, dans toute situation où chaque prémisse est vraie, la conclusion est également
vraie. Rappelons aussi que “a ⇒ b” est “a implique b” ou “si a alors b”; “a ∧ b” est “a et b”; “∀n Φ” est
“pour tout n, Φ”.

dont on déduit de proche en proche

P (0) , P (1) , P (2) , . . .

c’est-à-dire la conclusion. La preuve formelle est quasi la même que pour la récurrence
simple.

1.2.5 Notation lambda

Dans une écriture du type “X =def A”, le terme défini est normalement X, tandis
que l’expression définissante est A. Cette constatation semble évidente, mais n’est pas
parfaitement respectée dans les définitions fonctionnelles données plus haut (qu’elles soient
récursives ou non). En effet, nous avons écrit “f(n) =def A”, alors que le terme défini
était f et non f(n). Une meilleure écriture, parfois utilisée par les mathématiciens, est

f =def [n 7→ A] ;

dans le présent contexte, on utilisera plutôt la “notation lambda”12 et on écrira

f =def λn.A .

L’expression définissante est une “forme lambda”,13 dont A est le “corps”. On écrira, par
exemple,

h =def λxy.
√

x2 + y2 ,

ce qui se lit “h est la fonction qui, à tous x, y, associe
√

x2 + y2”. De même, la définition
classique de la fonction de Fibonacci se récrit

fib =def λn. [ if n < 2 then n else fib(n − 1) + fib(n − 2) ] .

1.2.6 Aspects opérationnels de la récursivité

Nous verrons que toutes les définitions fonctionnelles données plus haut se transposent
immédiatement en Scheme.

Une définition acceptable donne lieu à une procédure qui se termine toujours.14

Cependant, l’efficacité de la procédure peut grandement varier et, dans certains cas, se
révéler inacceptable. Considérons d’abord le programme

12Ce nom fait référence au “lambda-calcul”, un formalisme abstrait proposé par A. Church en 1936 pour
étudier les notions d’algorithme et de calculabilité.

13On dit aussi, d’après l’anglais, “lambda-forme”, ce que l’on écrira dans la suite, “λ-forme”.
14Nous emploierons souvent les mots “fonction” et “procédure” de manière interchangeable. Le mot

“procédure” insiste sur le fait que la fonction (au sens mathématique du terme) se double d’un programme,
c’est-à-dire d’un outil permettant le calcul des couples de cette fonction. A la notion mathématique
de “fonction définie pour la valeur x de l’argument” correspond la notion informatique de “procédure
se terminant pour la valeur x de l’argument”. La notion d’efficacité (en programmation) concerne la
procédure et non la fonction mathématique sous-jacente. Deux procédures très différentes peuvent avoir
même fonction mathématique sous-jacente. Par exemple, il existe de nombreux algorithmes pour trier une
liste d’entiers mais ils partagent tous la même fonction mathématique sous-jacente.

(define fib

(lambda (n)

(if (< n 2)

n

(+ (fib (- n 1)) (fib (- n 2))))))

Même sans connâıtre le langage Scheme, on devine que ce programme est une
transcription naturelle immédiate de la définition de la suite de Fibonacci. Il se termine
toujours, mais on constate que le temps d’exécution est anormalement long. La raison
est simple; pour évaluer (fib 33), par exemple, le système évalue séparément (fib 32)

et (fib 31); il s’agit clairement d’un gaspillage, puisque tout calcul fait dans le cadre
de l’évaluation de (fib 31) est aussi fait dans le cadre de l’évaluation de (fib 32). On
peut facilement vérifier que (fib 31) est calculé deux fois, (fib 30) trois fois, (fib 29)

cinq fois, etc. On peut éviter ce gaspillage de ressources en utilisant plutôt la fonction
fib2 introduite au paragraphe précédent.15 L’évaluation de fib2 (33) en (3524578, 5702887)
est maintenant immédiate; bien plus, l’évaluation de fib2 (3300) reste quasi instantanée,
quoiqu’elle produise deux nombres de 691 chiffres. Un programme d’efficacité comparable
s’obtient en transposant en Scheme la définition suivante :

fib it(n, a, b) =def [if n = 0 then a else fib it(n − 1, b, a + b)] .

En effet, on démontre facilement que, pour tous nombres de Fibonacci consécutifs fib(i)
et fib(i+1), on a

fib it(n,fib(i),fib(i+1)) = fib(i+n)

et donc, en particulier
fib it(n, 0, 1) = fib(n) ,

pour tout naturel n. Le temps de calcul sera quasi proportionnel à n,16 alors qu’avec le
programme précédent il était proportionnel à fib(n).

Quoique cela ne soit guère utile, on peut encore faire mieux, en notant que

(

fib(n)
fib(n + 1)

)

=

(

0 1
1 1

)n (

0
1

)

.

En effet, que M soit un nombre ou une matrice, on peut calculer Mn en un temps
proportionnel à log n, en adaptant l’algorithme classique dit “exponentielle rapide” :

exp(M,n) =def if n = 0 then I (matrice unité)
else if even?(n) then [exp(M,n/2)]2

else M ∗ exp(M,n − 1) .

15Nous verrons la transposition en Scheme plus loin.
16Si on tient compte du fait que l’addition de deux nombres ne prend pas un temps constant, mais un

temps proportionnel à la longueur (ou au logarithme) de ces nombres, le temps de calcul est est à peu près
proportionnel à n log n.

On notera que cette solution très efficace est récursive, de même que la solution näıve, très
inefficace, donnée plus haut. Contrairement à une opinion tenace, la récursivité n’est pas
incompatible avec l’efficacité.

Dans la mesure où un programme est, avant tout, une entité de commande pour
une machine, la notion d’efficacité est très importante. Nous avons déjà mentionné au
paragraphe 1.2.1 que les programmes cb1 et cb2, quoique réalisant tous deux la même
fonction mathématique, différaient radicalement dans les processus de calcul mis en œuvre.
Le lecteur peut immédiatement observer que la définition cb1 est inefficace parce qu’elle
implique des recalculs (exactement comme la version näıve du programme de Fibonacci),
alors que la version cb2 n’en implique pas et est efficace.

1.3 Structures de données

Les données constituent un aspect essentiel de la programmation et des langages de
programmation. D’une part, les caractéristiques des données d’un problème influencent
largement la manière dont le programmeur tentera de résoudre le problème. Nous venons
de voir que, si l’une des données d’un problème est un entier naturel n, une stratégie
souvent féconde consiste à considérer d’abord le cas n = 0, puis à essayer de réduire le cas n
au cas n−1 si n > 0. D’autre part, les mécanismes de structuration, de représentation et de
manipulation de données qu’offrent les différents langages de programmation déterminent,
dans une large mesure, l’adéquation de ces langages aux problèmes mettant en jeu ces
données.

Les exemples considérés jusqu’ici étaient numériques parce que le type de donnée
“entier naturel” est à la fois le plus familier et le plus important. Néanmoins, d’autres
types de données importants existent. D’une part, certaines données élémentaires ne sont
pas numériques, mais plutôt symboliques. Il est naturel, par exemple, de représenter les
données “couleurs de l’arc-en-ciel” par les symboles “violet”, “indigo”, “bleu”, “vert”,
“jaune”, “orange” et “rouge”, plutôt que par les nombres 0, 1, 2, 3, 4, 5 et 6. D’autre
part, et ceci est plus important en pratique, les données élémentaires, numériques ou
symboliques, apparaissent souvent sous forme d’un groupe de données bien structuré
plutôt que sous forme de nombres ou de symboles isolés. Par exemple, une donnée de
type “matrice 2 × 2” comporte quatre composants numériques; on peut imaginer qu’une
donnée de type “personne” comporterait deux composants symboliques (nom et prénom)
et un composant numérique (année de naissance). Une donnée de type “famille” sera un
arbre (généalogique) dont les nœuds sont étiquetés par des données de type “personne”.
Toute donnée formée de composants est dite structurée.

Avant de programmer, c’est-à-dire de manipuler des données, il est utile de s’interroger
sur la notion même de donnée. Dans ce paragraphe, nous décrivons d’abord trois types
de données, à la fois importants et représentatifs; nous esquissons une technique abstraite
et axiomatique de description des types de données qui donnera lieu dans la suite à un
schéma général pour les programmes manipulant ces données. Nous montrons ensuite
comment, à partir de types ainsi définis, on peut dériver des variantes et cas particuliers
utiles.

1.3.1 Les entiers naturels

Les entiers naturels constituent un type de donnée bien connu et il peut parâıtre artificiel
et vain de vouloir les définir. Toutefois, il est utile de les présenter ici d’une manière qui
met en évidence l’intérêt du principe de récurrence dans le raisonnement à propos des
entiers naturels et aussi dans la construction des programmes qui les manipulent. On se
donne d’abord deux règles de construction :

• 0 est un naturel.

• Si n est un naturel, alors s(n) est un naturel.

La fonction à un argument s (pour “successeur”) est le constructeur du type “entier
naturel”. Des deux règles de construction, on déduit que 0, s(0), s(s(0)), etc., sont des
naturels (que l’on peut noter plus économiquement 0, 1, 2, etc.). On admet aussi que
tout naturel peut s’obtenir de la sorte, et ceci d’une seule manière. On se donne enfin
une fonction p (pour “prédécesseur”), inverse de s, qui à tout naturel non nul s(n) associe
le naturel n. L’ensemble des entiers naturels, présenté de la sorte, est le domaine des
naturels, que nous noterons classiquement N. On dit aussi que 0 est un élément de base de
ce domaine (en fait, le seul); en outre, tout naturel n est un composant direct du naturel
s(n) (en fait, le seul) d’où, si m est un naturel non nul, p(m) est son composant direct.
Un composant d’un naturel non nul s(n) est soit le composant direct n de ce naturel, soit
un composant (direct ou non) de n.17

Cette présentation permet de reformuler les principes de récurrence comme suit :

(Récurrence simple.)
Si la propriété P est vraie du naturel de base 0
et si dès qu’elle est vraie du composant direct d’un naturel,
elle l’est aussi du naturel lui-même,
alors elle est vraie de tout naturel.

(Récurrence complète.)
Si dès que la propriété P est vraie des composants d’un naturel,
elle l’est aussi du naturel lui-même,
alors elle est vraie de tout naturel.

Cette nouvelle formulation suggère que la notion de composant est la clef du principe de
récurrence, et que ce dernier pourrait se généraliser à tout domaine structuré, c’est-à-dire
à tout ensemble dont les éléments se partitionnent en objets de base et en objets composés
au moyen des objets de base.18 C’est bien le cas et cela permet d’étendre à des données
non numériques la technique de définition récursive de fonctions. Nous ne traitons pas ici

17La notion de composant est ici définie récursivement. On pourrait aussi dire que m est un composant
de n si n s’obtient en appliquant un certain nombre de fois le constructeur s au naturel m; plus simplement,
m intervient dans la construction de n.

18En tant que type de donnée structuré, le domaine N est structuré. Le zéro est l’unique élément de base,
les fonctions successeur et prédécesseur sont respectivement le constructeur et l’accesseur du domaine.

cette question dans toute sa généralité, mais la suite de ce paragraphe introduit quelques
domaines structurés importants.

1.3.2 Les arbres binaires

Considérons un ensemble non vide E. On associe à E un domaine structuré dont les
éléments sont appelés E-arbres binaires, ou simplement arbres binaires. On se donne deux
règles de construction :

• Tout élément de E est un arbre binaire.

• Si A et B sont des arbres binaires, alors c(A,B) est un arbre binaire.

Les arbres binaires les plus simples sont les feuilles, éléments de E. Les autres arbres sont
dits arbres composés. La fonction à deux arguments c est le constructeur du type “arbre
binaire”. Un objet est un arbre binaire si et seulement si cet objet peut être construit
au moyen des deux règles de construction, appliquées un nombre fini de fois. Si c’est
le cas, le mode de construction est unique; des arbres construits de manières distinctes
sont toujours distincts. Par exemple, pour tous x, y, z ∈ E, les deux arbres c(c(x, y), z),
c(z, c(x, y)) sont distincts (même si x, y et z sont égaux); d’autre part, les deux arbres
c(c(x, y), z) et c(c(x′, y′), z′) sont égaux si et seulement si x = x′, y = y′ et z = z′.

On se donne aussi deux fonctions a et d à un argument, appelées respectivement
fils gauche et fils droit qui, à tout arbre composé c(α, β) associent respectivement ses
composants directs α (aussi appelé fils gauche) et β (aussi appelé fils droit); ces deux
fonctions sont les accesseurs.19 Les composants d’un arbre binaire composé sont ses
composants directs ou les composants de ceux-ci. Un arbre binaire α est un composant
d’un arbre binaire β si α intervient dans la construction de β.
Remarque. Les éléments de E peuvent être variés (des nombres, des lettres, des symboles
quelconques) mais ils ne peuvent être des arbres binaires; plus spécifiquement, ils ne
peuvent être le résultat de l’application du constructeur c à des arguments. Dans la
suite, nous appelons atome tout objet de ce type. On notera que les accesseurs a et d ne
peuvent prendre pour argument un atome.

Les principes de récurrence se généralisent aux arbres binaires (ainsi qu’aux autres
domaines structurés dont quelques exemples supplémentaires sont introduits ci-après);
on les appelle alors principes d’induction. Dans ces énoncés, le symbole P désigne une
propriété, susceptible d’être vraie ou fausse.

(Induction simple sur les arbres binaires.)
Si P est vrai de toute feuille, élément de E
et si dès que P est vrai des deux fils d’un E-arbre binaire composé,
P l’est aussi du E-arbre binaire composé lui-même,
alors P est vrai de tout E-arbre binaire.

19Le constructeur d’arbre et les accesseurs ont été nommés respectivement c, a et d; notons déjà que
leurs homologues Scheme se nomment cons, car et cdr.

(Induction complète sur les arbres binaires.)
Si dès que P est vrai des composants d’un E-arbre binaire,
P l’est aussi du E-arbre binaire lui-même,
alors P est vrai de tout E-arbre binaire.

Figure 1: Représentations des arbres binaires x, c(c(x, y), z) et c(x, c(y, z))

Remarque. Les arbres binaires doivent leur nom à leur représentation habituelle (cf.
Figure 1); c’est de là aussi que provient la terminologie usuelle : les composants directs
sont appelés “enfants” ou “fils”; les composants sont aussi appelés “descendants” ou “sous-
arbres”. Par rapport à un arbre donné, l’arbre lui-même est la racine; la racine et les
sous-arbres composés forment les nœuds internes. On notera que les nœuds internes, au
contraire des feuilles, ne sont pas étiquetés; il est parfois commode de considérer que chaque
nœud interne est implicitement étiqueté par le sous-arbre dont il est la représentation.20

Remarque. La notion d’arbre binaire se généralise en celle d’arbre n-aire (tout nœud
interne a exactement n fils; il y a un constructeur à n arguments et aussi n accesseurs. En
ce sens, les entiers naturels ne sont rien d’autre que les {0}-arbres unaires; le constructeur
(unaire) est la fonction successeur et les entiers naturels sont 0, s(0), s(s(0)), etc. L’unique
accesseur est la fonction prédécesseur. Les arbres n-aires sont appelés arbres à degré fixe;
chaque nœud interne a exactement n fils.

1.3.3 Les listes

Intuitivement, une liste est une suite totalement ordonnée, comportant un nombre
quelconque d’objets appelés éléments de la liste. Ce nombre d’objet est la longueur de la
liste. On pourrait formaliser la notion de liste en créant un domaine dont le constructeur
admettrait un nombre quelconque d’arguments mais il serait difficile de s’accommoder
d’un nombre variable d’accesseurs. On utilisera donc une autre tactique pour formaliser
le domaine des listes. Le constructeur prend deux arguments, un objet α et une liste β;
il produit une liste dont le premier élément est α et dont les autres éléments forment la
liste β; on dispose, pour commencer la construction des listes, de la liste vide notée [ ]
et d’un ensemble non vide E d’éléments de base. Réciproquement, à toute liste non vide

20Dans la figure 1, la représentation du milieu comporte trois étiquettes explicites de feuilles; les étiquettes
implicites des nœuds internes sont c(c(x, y), z) pour la racine et c(x, y) pour l’autre nœud, fils gauche de
la racine.

on peut appliquer deux accesseurs, le premier fournissant la tête, c’est-à-dire le premier
élément de la liste, et le second renvoyant le reste, c’est-à-dire la liste des éléments restants.
Comme dans le cas des arbres binaires, le constructeur et les deux accesseurs seront notés
respectivement c, a et d. Une liste dont les trois éléments sont, dans l’ordre, x, y et z, sera
donc notée c(x, c(y, c(z, [ ]))). On devine que cette notation devient très incommode pour
les longues listes, aussi on utilisera souvent la notation abrégée [x, y, z]. Des notations
mixtes, telles c(x, [y, z]) et c(x, c(y, [z]) sont également admises.
Remarque. Il est important de distinguer les termes “élément” et “composant”. La liste
vide n’a ni élément ni composant. Une liste non vide a exactement deux composants
directs, la tête qui est un élément, et le reste qui est une liste. La liste [x, y, z] admet les
trois éléments x, y et z et les deux composants direct x et [y, z]; elle admet en outre les
composants indirects y, [z], z et [ ]. La liste [x] admet le seul élément x; sa tête est x et
son reste est [ ]; il n’y a pas de composant indirect.

1.3.4 Listes plates, listes profondes

La notion de liste qui vient d’être introduite donne lieu à un domaine structuré à condition
que l’on spécifie l’ensemble des objets qui peuvent être éléments d’une liste.

Etant donné un ensemble de base E non vide, dont les éléments sont des atomes, nous
considérerons deux domaines de listes, notés respectivement LE

−
et LE.

• Les éléments des listes de LE
−

appartiennent à E (et ne sont donc pas eux-mêmes
des listes). L’ensemble LE

−
est le domaine des E-listes plates.

• Les éléments des listes de LE appartiennent à l’ensemble E ∪ LE. L’ensemble LE

est le domaine des E-listes.

On note l’inclusion LE
−
⊂ LE; cette inclusion est stricte. On appelle E-liste profonde une

E-liste qui n’est pas plate.

Pour illustrer ces définitions, considérons le cas particulier E = {0, 1}. La seule E-liste
de longueur nulle est la liste vide, qui est une liste plate. Il existe deux listes plates de
longueur 1, qui sont [0] et [1], et quatre listes plates de longueur 2, qui sont [0,0], [0,1],
[1,0], [1,1]. Le monde des listes profondes est beaucoup plus touffu, comme le montrent
quelques exemples simples : [[ ]], [[0]], [[0,1,[0,1]]], [0,[ ]], [[0],0], [[[ ]],[0,[1],[0,[1]]]]. Les trois
premières listes sont de longueur 1, les trois dernières sont de longueur 2. Notons aussi
trois points importants, qui paraphrasent les définitions :

• [ ] est une E-liste et une E-liste plate; sa longueur est 0.

• Si α est un élément de E et si β est une E-liste plate de longueur n,
alors c(α, β) est une E-liste plate de longueur n+1.

• Si α est un élément de E ou une E-liste et si β est une E-liste de longueur n, alors
c(α, β) est une E-liste de longueur n+1.

Remarque. En termes algébriques, les deux domaines de listes se définissent de manière
concise; ce sont les solutions minimales des équations ensemblistes LE

−
= {[ ]} ∪ c(E,LE

−
)

et LE = {[ ]} ∪ c(E ∪ LE ,LE), respectivement.

On peut décrire une liste non vide en termes de ses éléments ou en termes de ses
composants. La seconde solution met en évidence le fait que toute E-liste est aussi un
(E∪{[ ]})-arbre binaire (la réciproque est fausse). Ces deux représentations sont illustrées
à la figure 2.

[[w, x, y], z]

Figure 2: Représentations de la liste c(c(w, c(x, c(y, [ ]))), c(z, [ ]))

Remarque. Comme dans le cas des arbres binaires, seules les feuilles sont explicitement
étiquetées. Il peut être utile d’attribuer une étiquette implicite à chaque nœud interne.
Dans la figure 2, l’étiquette de la racine est la liste elle-même; l’étiquette de son fils gauche
est la liste c(w, c(x, c(y, [ ]))), c’est-à-dire la liste [w, x, y]; l’étiquette du fils droit de la
racine est c(z, [ ]), c’est-à-dire la liste [z].
Remarque. La présentation que nous avons adoptée ici assimile les listes à des arbres
binaires particuliers. Cela n’empêche pas que les listes (profondes) soient souvent utilisées
pour modéliser en Scheme d’autres types d’arbres. Nous reviendrons sur ce point au
paragraphe 8.4.

L’intérêt de présenter les listes comme des domaines structurés est de pouvoir raisonner
par induction à leur sujet. On a d’abord les principes d’induction suivants :

(Induction superficielle simple sur les listes.)
Si P est vrai de la liste vide
et si dès que P est vrai du reste d’une liste non vide,
P l’est aussi de la liste elle-même,
alors P est vrai de toute liste.

(Induction superficielle complète sur les listes.)
Si dès que P est vrai des suffixes propres 21 d’une liste,

21Un suffixe propre d’une liste non vide est le reste de cette liste, ou un suffixe propre de ce reste; l’unique

P l’est aussi de la liste elle-même,
alors P est vrai de toute liste.

L’induction superficielle est spécialement adaptée aux listes plates. Dans certains cas, on
l’utilise aussi pour les listes. Cependant, pour les listes (quelconques), on utilise souvent
des principes un peu plus forts; l’idée est la suivante : dans la partie inductive du principe,
l’hypothèse est que la propriété est vraie non seulement pour le second composant, qui est
toujours une liste, mais aussi pour le premier, quand il s’agit d’une liste. Le symbole P
dénote ici une propriété de liste, c’est-à-dire un prédicat à un argument tel que, pour toute
liste ℓ, P (ℓ) est vrai ou faux. On a alors

(Induction profonde simple sur les listes.)
Si P est vrai de la liste vide;
si pour toute liste non vide dont la tête est atomique,
dès que P est vrai du reste de la liste,
P l’est aussi de la liste elle-même;
si pour toute liste non vide dont la tête est une liste,
dès que P est vrai de la tête et du reste de la liste,
P l’est aussi de la liste elle-même;
alors P est vrai de toute liste.

(Induction profonde complète sur les listes.)
Si dès que P est vrai de la liste vide
et des composants non atomiques d’une liste,
P l’est aussi de la liste elle-même,
alors P est vrai de toute liste.

L’expression de ces principes est un peu lourde parce que P (ℓ) n’a de sens que si ℓ est
une liste. Si on convient que P (x) est d’office vrai si x n’est pas une liste, on obtient des
énoncés plus simples :

(Induction profonde simple sur les listes.)
Si P est vrai de la liste vide;
si dès que P est vrai de la tête et du reste d’une liste non vide,
P l’est aussi de la liste elle-même;
alors P est vrai de toute liste.

(Induction profonde complète sur les listes.)
Si dès que P est vrai des composants d’une liste,
P l’est aussi de la liste elle-même,
alors P est vrai de toute liste.

1.4 Apprendre à programmer avec Scheme

Programmer n’est sans doute pas, en soi, une activité plus difficile que beaucoup d’autres.
Pourtant, on se plaint (depuis au moins trente ans !) de la qualité insuffisante des

suffixe impropre d’une liste, vide ou non, est la liste elle-même.

programmes et de la productivité insuffisante des programmeurs. Nous n’allons pas ici
nous engager dans ce thème maintes fois débattu, mais seulement donner quelques idées
qui nous ont guidé dans la rédaction de ce texte.

Un premier point est que la programmation, comme beaucoup d’activités, requiert
une certaine somme de connaissances préliminaires, sans lesquelles il n’est même pas
vraiment possible de commencer à pratiquer. Le double choix du style fonctionnel et du
langage Scheme réduit, à notre avis, cette quantité de connaissances prérequises. D’une
part, il suffit de mâıtriser le petit monde des expressions arithmétiques élémentaires pour
comprendre le style fonctionnel. D’autre part, même si le langage Scheme comporte un
certain nombre de notions et de mécanismes difficiles, un sous-ensemble relativement réduit
du langage, restreint à des constructions élémentaires, suffit à résoudre d’une manière
satisfaisante un grand nombre de problèmes intéressants; dans le même ordre d’idée, la
syntaxe de Scheme est extrêmement simple.

Un deuxième point est que la programmation, c’est-à-dire la construction d’une
solution informatique à un problème, requiert une certaine créativité. L’approche que
nous suivons ici ne supprime pas cette nécessité, mais rend plus aisée sa satisfaction.
Plus concrètement, nous pensons que l’élaboration d’un programme peut se faire
méthodiquement et que seules certaines étapes requièrent de la créativité. Nous attachons
un soin tout particulier, dans la suite, à circonscrire la partie créative de la résolution d’un
problème.

Un troisième point est que la programmation requiert une certaine discipline . . . et
que l’efficience des outils actuels pourrait malencontreusement induire le programmeur à
recourir à la “méthode” dite “essais et erreurs”. On croit qu’un essai ne coûte rien : c’est
faux, il coûte du temps !22 De plus, un programme auquel on arrive au terme d’une
longue série d’essais, d’erreurs et de corrections a peu de chances d’être entièrement
correct, et encore moins de chances d’être raisonnablement “propre”. Il ne s’agit pas
ici de recommander une approche “puriste”, où l’apprentissage de la programmation
serait vu comme une activité d’écriture, sans contact avec la machine; simplement, la
programmation fait partie des “arts de l’ingénieur”, qui requièrent, en dosage approprié,
réflexion et expérimentation.

22Il coûte du temps d’ordinateur, ce qui, de nos jours, n’est pas grave, mais aussi du temps humain,
ressource coûteuse et limitée . . .

2 Les bases de Scheme

2.1 Principe de l’interprète

Le langage Scheme peut être interprété ou compilé; nous utiliserons surtout l’interprète.
L’interaction entre l’interprète et l’utilisateur est en principe très simple; c’est une séquence
de quatre étapes, répétée aussi longtemps que l’utilisateur le souhaite :23

• L’utilisateur introduit une expression, c’est-à-dire un texte Scheme respectant
certaines règles;

• Le système lit l’expression;

• Le système évalue l’expression;

• Le système affiche la valeur de l’expression (ou un message approprié si l’expression
n’a pas de valeur).

Le rôle essentiel de l’interprète consiste donc à évaluer les expressions qui lui sont soumises,
c’est-à-dire à en calculer la valeur. L’interprète Scheme, que nous appellerons souvent “le
système”, se comporte un peu comme une calculatrice de poche; une machine de ce type
accepte des expressions arithmétiques et produit leurs valeurs numériques.

2.2 Les expressions

On distingue trois sortes d’expressions : les expressions simples, les combinaisons et les
formes spéciales.

• Une expression simple est une constante (numérique ou non) ou une variable.
Signalons déjà que la valeur d’une constante est cette constante, tandis que la valeur
d’une variable peut être un objet quelconque.

• Une combinaison est une liste; nous appellerons le premier élément foncteur

ou opérateur ; les autres éléments sont les opérandes.24 Les éléments d’une
combinaison sont eux-mêmes des expressions (expressions simples, combinaisons,
formes spéciales); la valeur du premier élément doit être une fonction, les valeurs des
suivants étant alors des arguments pour cette fonction.25

• Une forme spéciale est une liste dont le premier élément est un mot-clef spécifique;
le nombre et la nature des autres éléments varient selon le type de la forme spéciale.

23Cette séquence indéfiniment répétée s’appelle la boucle read-eval-print.
24Cette terminologie n’est pas standardisée.
25Cette terminologie n’est pas standardisée. Nous souhaitons souligner la différence entre, d’une part, les

éléments d’une combinaison (foncteur et opérandes) et les valeurs de ces éléments (fonction et arguments).
En pratique, l’usage est moins strict, et on parlera parfois de fonctions et d’arguments en place de foncteurs
et d’opérandes.

Syntaxiquement, les expressions simples sont des (représentations de) nombres ou
des identificateurs alphanumériques (les mots-clefs sont interdits, de même que certains
caractères, les parenthèses notamment). Les formes, c’est-à-dire les combinaisons et les
formes spéciales, sont des listes.26

Avant d’introduire des définitions plus précises concernant notamment quelques formes
spéciales importantes, nous donnons quelques exemples d’expressions simples et de
combinaisons. Dans les exemples suivants, le symbole ==> sépare l’expression à évaluer,
introduite par l’utilisateur, de la valeur correspondante, produite et affichée par le système.
Ce symbole, conventionnel, varie d’un système à l’autre; il inclut souvent un passage à la
ligne.

3 ==> 3 + ==> # procedure

8/5 ==> 8/5 pi ==> 3.14159

-3.14 ==> -3.14 x ==> Error - unbound variable x

Les six évaluations ci-dessus concernent des expressions simples. Les constantes (ici, des
nombres) s’évaluent en elles-mêmes. Un identificateur est vu par l’interprète comme une
variable, à laquelle une valeur est liée. L’interprète renvoie cette valeur, si elle existe; sinon,
un message signale “unbound variable” (variable non liée à une valeur). Le symbole + est
“pré-lié” (lié par le système) à la fonction d’addition; les fonctions ne sont pas affichables
explicitement, mais l’interprète signale que la valeur de “+” est une fonction. Enfin, on a
supposé ici que la variable pi avait été liée précédemment à la valeur numérique 3.14159,
tandis que la variable x n’est pas liée. Nous considérons maintenant des combinaisons.

(+ 2 5) ==> 7 (/ 2 0) ==> Error - 0 within proc /

(+ (* 3 -2) 2) ==> -4 (/ 8 3.0) ==> 2.66666666

Les combinaisons évaluées ci-dessus sont des formes arithmétiques. Pour évaluer
l’expression (+ 2 5), on évalue d’abord les trois sous-expressions +, 2 et 5, puis on applique
la première valeur (la fonction d’addition) aux deux suivantes (les valeurs 2 et 5), ce qui
donne la valeur de l’expression, soit 7.
Remarque. Si on essaie d’évaluer les listes (2 3) et (x 3), on obtient

(2 3) ==> Error - object 2 not applicable

(x 3) ==> Error - unbound variable x

Dans les deux cas, le système détecte que, si expression il y a, elle ne peut être simple,
puisqu’il y a une parenthèse initiale annonçant une liste. Le premier élément de cette
liste n’étant pas un mot-clef, le système suppose que la liste est une combinaison. Dans le
premier cas, le système tente d’appliquer 2 à l’argument 3, ce qui est impossible, 2 n’étant
pas une fonction. Dans le second cas, le système reconnâıt que le symbole x n’a pas de
valeur, n’est pas lié à une valeur.27

26Le mot “forme” est parfois employé comme synonyme de “expression (évaluable)”.
27Une telle liaison pourrait être le fait du système lui-même, comme pour le symbole +, ou de l’utilisateur

(cf. § 2.3).

Remarque. La notation Scheme est plus homogène que les notations mathématiques
usuelles, où les foncteurs s’emploient de manière préfixée (comme dans “sin x”), infixée
(comme dans “x+y”) ou postfixée (comme dans “n!”), ou encore font l’objet de conventions
d’écriture spéciales (comme dans “ex” et “

∫ x

x0
f(u)du” ).

Remarque. La structure arborescente des expressions se prête à une définition dans le
formalisme BNF.28 En se restreignant aux expressions arithmétiques, on a

〈A-expression〉 ::= 〈nombre〉 | 〈variable〉 | 〈A-combinaison〉

〈A-combinaison〉 ::= ( 〈A-op〉 [ 〈A-expression〉 ]∗ )

〈A-op〉 ::= + | - | * | /

On voit qu’une combinaison arithmétique, ou forme arithmétique, est un assemblage
composé (dans l’ordre), d’une parenthèse ouvrante, d’un opérateur arithmétique, d’un
certain nombre (0, 1 ou plus) d’expressions arithmétiques et d’une parenthèse fermante.

2.3 La forme spéciale define

La forme spéciale de mot-clef define est utilisée pour établir une liaison entre une variable
et une valeur. L’évaluation de la forme spéciale (define symb e) ne produit pas de valeur

mais a pour effet la liaison de la valeur de l’expression e à la variable symb. Dans la suite,
nous supposerons que le système affiche “...” en réponse à l’évaluation d’une forme
spéciale de mot-clef define. La session suivante illustre le rôle d’une telle forme :29

pi ==> 3.14159

(define inv_pi (/ 1 pi)) ==> ...

inv_pi ==> 0.318310155

Remarque. On pourrait créer une nouvelle liaison pour pi, ce qui rendrait la précédente
inaccessible mais ne modifierait en rien la liaison relative à inv_pi :30

(define pi 4.0) ==> ...

pi ==> 4.0

inv_pi ==> 0.318310155

28Pour “Backus-Naur Form”; le lecteur non familier avec cette notation peut passer immédiatement au
paragraphe suivant.

29L’expression (define symb e) est une forme spéciale dont define est le mot-clef. Néanmoins, on
parlera, par commodité, de “la forme spéciale define”, plutôt que de “la forme spéciale de mot-clef
define”. De même, on dit “forme +” pour une forme arithmétique dont l’opérateur est l’addition.

30On verra plus loin un moyen d’assurer la propagation automatique sur inv a d’une redéfinition de la
constante a.

2.4 Les symboles et leur double statut

Les textes que l’on peut introduire au clavier et soumettre au système Scheme sont
composés de “mots” (au sens large), séparés les uns des autres par des caractères spéciaux
tels l’espacement, le saut à la ligne, les parenthèses, etc.31 Certains de ces “mots” ont un
rôle particulier, notamment les mots-clefs, tels define et if, les constantes numériques
(2, -4/3, 3.5-2.3i, 6.02e23, etc.), les constantes booléennes #t et #f, . . . . La plupart
des autres “mots”, tels indigo, prks9tc, iso_9000 et 127-bis, sont des symboles.32

Les symboles peuvent avoir deux rôles bien distincts. Tout d’abord, ils peuvent être des
constantes lexicales (par opposition à d’autres types de constantes comme les nombres et
les valeurs booléennes). Ces constantes lexicales n’ont pour le système aucune signification
particulière, mais elles ont généralement une signification pour l’utilisateur.33 Par défaut
cependant, le système n’attribue pas à un symbole le statut de constante lexicale, mais
celui de variable. Comme nous l’avons vu au paragraphe précédent, la signification d’une
variable, ou plus exactement sa valeur, est l’objet qui lui est lié, s’il existe.34 Pour que le
système attribue à un symbole le statut de constante lexicale, on utilise le caractère spécial
’ (lire “quote”) : la valeur d’une expression constituée de ce caractère suivi immédiatement
d’un symbole est ce symbole. Ceci est illustré par la session suivante :

pi ==> 3.14159 ’pi ==> pi

pi.+2 ==> Error - unbound variable pi.+2 ’pi.+2 ==> pi.+2

indigo ==> Error - unbound variable indigo ’indigo ==> indigo

Remarque. Le langage Scheme a une syntaxe simple et systématique. Pour ne pas créer
une quatrième catégorie syntaxique (en plus des expressions simples, des combinaisons et
des formes spéciales), une expression telle que ’pi a le statut de forme spéciale; c’est en
fait une abréviation de la forme spéciale (quote pi). Nous reviendrons plus loin sur la
forme spéciale de mot-clef quote.

Remarque. Le problème de la “citation” n’est pas inhérent à Scheme, ni même aux
langages de programmation en général, comme le montre le petit dialogue suivant :

Q. Dites-moi votre nom !
R. Paul Dupont.

31Nous ne donnerons pas ici de manière exhaustive les règles compliquées de l’analyse syntaxique effectuée
par le système.

32Certains “mots” sont exclus pour des raisons techniques, notamment ceux qui comportent un caractère
spécial comme un séparateur. On n’essaiera pas ici d’être précis et exhaustif; notons cependant que l’usage
des caractères ( ) [ ] { } ; , " ’ ‘ # -+ dans les symboles est soumis à des restrictions ou totalement
interdit. Par exemple, aucun symbole ne peut contenir de parenthèse; d’autre part, 2/b et 3#a sont des
symboles mais 2/3 et #3 n’en sont pas.

33Nous avons déjà mentionné un exemple : il est plus naturel de représenter les données “couleurs de
l’arc-en-ciel” par les symboles “violet”, “indigo”, “bleu”, “vert”, “jaune”, “orange” et “rouge”, plutôt que
par les nombres 0, 1, 2, 3, 4, 5 et 6.

34Puisqu’en tant que constantes lexicales les symboles n’ont a priori aucune signification, il est naturel
de les utiliser comme variables, c’est-à-dire de leur attribuer une signification, ou plutôt une valeur, au gré
du système ou de l’utilisateur.

Q. Dites-moi “votre nom” !
R. Votre nom.

2.5 Les listes

Une liste est une suite ordonnée et finie d’objets; le nombre d’objets est la longueur et
les objets eux-mêmes sont les éléments de la liste. L’ensemble des expressions composées
est inclus dans l’ensemble des listes, qui est lui-même inclus dans l’ensemble des valeurs.
Syntaxiquement, une liste est représentée en Scheme par la suite de ses éléments35 séparés
par des blancs, entourée d’une paire de parenthèses. La seule liste de longueur 0 est la
liste vide (). Comme on l’a vu au chapitre précédent, le premier élément d’une liste non
vide est sa tête; les autres éléments forment le reste (qui est aussi une liste, éventuellement
vide). La tête et le reste sont les deux composants directs d’une liste non vide.36 Voici
trois exemples de listes qui sont aussi des expressions évaluables :

(- 3) , (+ 3 5) , (define pi 3.14159) .

Par contre, les listes

(0 1 2 3) , (5 (+ (d 4)) ()) , ((1 a b) (2 c d) (3 e f))

ne sont pas des expressions évaluables. Enfin, la liste

(x y z)

n’est une expression évaluable que si x est une variable liée à une fonction admettant
comme arguments les valeurs de y et z.

Nous avons vu que, par défaut, l’interprète attribue à un symbole le statut de variable;
la forme spéciale de mot-clef quote permet de modifier la règle. De la même manière,
lorsque l’on évalue, une liste non vide a par défaut le statut d’expression :

(2 3) ==> Error - object 2 not applicable

(x 1 2) ==> Error - unbound variable x

(+ 2 3) ==> 5

On déroge à la règle en utilisant quote, ce qu’illustre la session suivante :

’(2 3) ==> (2 3)

’(x 1 2) ==> (x 1 2)

(quote (2 3)) ==> (2 3)

(quote (x 1 2)) ==> (x 1 2)

35. . . des représentations de ses éléments . . .
36Rappelons que l’on ne doit pas confondre les composants directs et les éléments d’une liste. Les deux

composants directs de la liste (a b c) sont a et (b c), tandis que ses trois éléments sont a, b et c. Les
composants indirects de (a b c) sont b, (c), c et ().

On est souvent amené à considérer des listes dont les éléments sont soumis à certaines
restrictions. En particulier, étant donné un ensemble non vide E d’objets qui ne sont pas
des listes,37 on définit formellement le domaine structuré LE des listes (ou des E-listes)
comme indiqué au paragraphe 1.3.4. L’unique élément de base est la liste vide, notée (). Le
constructeur est noté cons et les deux accesseurs sont notés, pour des raisons historiques,
car et cdr. Rappelons encore que la liste vide n’admet ni composant ni élément; les
éléments d’une liste non vide sont des composants de cette liste (direct pour le premier,
indirects pour les suivants), mais certains composants d’une liste non vide n’en sont pas
des éléments.38 Il est cependant naturel et commode de construire une liste à partir de
ses éléments plutôt qu’à partir de ses deux composants directs; un constructeur auxiliaire
nommé list concrétise cette possibilité; il prend un nombre quelconque d’arguments, qui
sont les éléments de la liste-résultat.39

Soient α,α1, . . . , αn des expressions quelconques et β une expression dont la valeur est
une liste. En notant [[e]] la valeur d’une expression e, on a les règles suivantes :

• [[(cons α β)]] est une liste dont la tête est [[α]] et le reste est [[β]]; on a par exemple
[[(cons 0 (cons 1 ’()))]] = (0 1); la tête est 0 et le reste est la liste (1). Toute
liste non vide est valeur d’une forme cons.

• [[(list α1 ... αn)]] est la liste dont les éléments sont (dans l’ordre)
[[α1]], . . . , [[αn]]. Toute liste est valeur d’une forme list.

Les fonctions cons,40 à deux arguments, et list, à nombre quelconque d’arguments, sont
injectives; deux listes non vides sont égales si elles ont même tête et même reste; deux
listes sont égales si leurs éléments sont deux à deux égaux. Notons que l’expression

(list α1 α2 ...αn)

équivaut à l’expression

(cons α1 (cons α2 ... (cons αn ’()) ... )) ;

la valeur commune est une liste dont les n éléments sont les valeurs des expressions
α1, . . . , αn .

La notation pointée est souvent utilisée pour mettre en évidence la décomposition
d’une liste en sa tête et son reste. Si u est une liste non vide dont la tête est représentée
par x et dont le reste est représenté par ℓ, alors u admet la représentation (x . ℓ).

Les huit écritures
37Plus précisément, qui ne sont pas la liste vide ou le résultat de l’application du constructeur à des

arguments.
38Rappelons aussi, avec les notations de Scheme, l’exemple qui terminait le paragraphe 1.3.3. La liste

(x y z) admet les trois éléments x, y et z et les deux composants direct x et (yz); elle admet en outre les
composants indirects y, (z), z et (). La liste (x) admet le seul élément x; sa tête est x et son reste est ();
il n’y a pas de composant indirect.

39Pour être plus précis, cons, car, cdr et list sont des variables pré-liées par le système aux fonctions
primitives, constructeurs et accesseurs du type “listes”.

40Par abus de langage, on dit “la fonction f” plutôt que “la fonction liée à la variable f”, exactement
comme “Paul Durand” se dit pour “la personne dont le nom est Paul Durand”.

(0 1 2), (0 1 . (2)), (0 . (1 . (2))), (0 1 . (2 . ())),

(0 . (1 2)), (0 1 2 . ()), (0 . (1 2 . ())), (0 . (1 . (2 . ())))

représentent toutes la même liste de longueur 3, dont les éléments sont 0, 1 et 2.

Remarque. Nous avons vu au paragraphe 1.3.4 que les E-listes sont des cas particuliers
de (E ∪ {[ ]})-arbres binaires. La notation pointée met ce fait en évidence, comme
on le voit notamment en se reportant à la figure 2 : l’arbre binaire représenté
sur cette figure correspond à la liste ((w x y) z), dont la notation pointée est
((w . (x . (y . ()))) . (z . ())) . Naturellement, la notation pointée sert à
représenter non seulement les listes mais aussi les arbres binaires quelconques; l’égalité

[[(cons α β)]] = ([[α]] . [[β]])

restera vraie même si [[β]] n’est pas une liste; les deux membres de l’égalités sont des
paires pointées. On utilise fréquemment en Scheme les expressions symboliques, c’est-à-
dire les (E ∪ {[ ]})-arbres binaires, où E est l’ensemble des symboles. Le chapitre 8 leur
est consacré.
Remarque. L’usage de quote s’étend à la notation pointée; on a par exemple

(cons 2 (cons 3 ’())) ==> (2 3)

(cons 2 ’(3)) ==> (2 3)

’(2 3) ==> (2 3)

’(2 . (3)) ==> (2 3)

’(2 . (3 . ())) ==> (2 3)

(list 2 3) ==> (2 3)

(list . (2 3)) ==> (2 3)

L’interprète utilise l’écriture la plus concise pour l’affichage.

Dans le domaine des listes comme dans tout domaine structuré, les accesseurs
permettent de construire les fonctions inverses des constructeurs et réciproquement. Si
[[l]] est une liste non vide, sa tête est l’objet [[(car l)]]; son reste est la liste [[(cdr l)]].
De même, [[x]] et [[l]] sont respectivement la tête et le reste de la liste [[(cons x l)]]. On
a donc les égalités suivantes :

• [[l]] = [[(cons (car l) (cdr l))]] , si [[l]] est une liste non vide;

• [[(car (cons a l))]] = [[a]] et [[(cdr (cons a l))]] = [[l]] ,
si [[a]] est un objet quelconque et si [[l]] est une liste.

Ces égalités peuvent servir d’axiomes dans un système formel de raisonnement sur les
structures de listes.

Voici quelques exemples, illustrant l’usage des constructeurs list et cons et des
accesseurs car et cdr :

(define pi 3.14159) ==> ...

(define l (list 3 ’5 ’(6 7) ’a3 ’define pi)) ==> ...

l ==> (3 5 (6 7) a3 define 3.14159)

(cons (car l) (cdr l)) ==>

(3 5 (6 7) a3 define 3.14159)

(car l) ==> 3

(cdr l) ==> (5 (6 7) a3 define 3.14159)

(cdr (car (cdr (cdr l)))) ==> (7)

(cdaddr l) ==> (7)

(car ’()) ==> Error

(cons ’(a b) ’(6 7)) ==> ((a b) 6 7)

(list ’(a b) ’(6 7)) ==> ((a b) (6 7))

On notera les abréviations admises pour les enchâınements de car et cdr; en particulier,
les fonctions cadr, caddr et cadddr, renvoyant respectivement les deuxième, troisième et
quatrième éléments d’une liste, sont d’emploi fréquent.

2.6 Booléens, prédicats, forme spéciale if

Les booléens sont les constantes logiques : #t pour true (vrai) et #f pour false (faux).
Comme toutes les constantes, elles s’évaluent en elles-mêmes.41 Une expression est vraie

si sa valeur est #t, et fausse si sa valeur est #f.

Les prédicats sont des fonctions prenant leurs valeurs dans les booléens. Beaucoup de
prédicats numériques42 (<, =, zero?, . . . ) ou non numériques (null?, equal?, . . . ) sont
prédéfinis. Les reconnaisseurs sont des prédicats à un argument, associés aux différentes
catégories d’objets; l’argument est un objet quelconque, la valeur retournée est #t si cet
objet appartient à la catégorie en question et #f sinon. La table de la figure 3 illustre les
reconnaisseurs prédéfinis les plus importants.

Reconnaisseur
Objet(s)

reconnu(s)
Expression vraie Expression fausse

number? nombres (number? 4.5+.9e-3i) (number? ’(1))

list? listes (list? ’(a . ())) (list? ’(() . b))

null? liste vide (null? (list)) (null? ’(()))

procedure? fonctions (procedure? +) (procedure? ’+)

symbol? symboles (symbol? ’+) (symbol? +)

boolean? booléens (boolean? ’#f) (boolean? ’faux)

Figure 3: Quelques reconnaisseurs

Des prédicats d’égalité existent pour différents types d’objets. On a en particulier “=”

41Les expressions #t et ’#t ont toutes deux pour valeur #t.
42dont les arguments sont des nombres.

pour les nombres, “eq?” pour les symboles, “eqv?” pour les objets atomiques (nombres,
symboles, booléens) et “equal?” pour les listes. Voilà quelques exemples :

(= (/ 3 2) 3/2 1.5 150e-2 1.5+0i) ==> #t

(eq? (list) ’()) ==> #t

(equal? ’(a b) (cons ’a (cons ’b ’()))) ==> #t

Remarque. Les prédicats d’égalité sont binaires, sauf “=” qui admet un nombre quelconque
d’arguments.

Soient e0, e1 et e2 trois expressions. La forme spéciale (if e0 e1 e2) a pour valeur
[[e2]] si [[e0]] = #f et [[e1]] sinon. La forme spéciale if permet donc l’évaluation
conditionnelle, qui est un mécanisme essentiel en programmation. Voici quelques exemples
d’emploi de ce mécanisme :

(if #t 1 2) ==> 1

(if #f 1 2) ==> 2

(if ’any_symbol 1 2) ==> 1

(if 5 1 2) ==> 1

(if (/ 5 0) 1 2) ==> Error - divide by zero

(if #t 1 (/ 2 0)) ==> 1

Remarque. La condition [[e0]] d’une forme if est considérée comme vraie dès qu’elle diffère
de #f. Elle peut donc être d’un type non booléen (nombre ou liste, par exemple). Cette
possibilité est parfois utilisée dans les programmes.43

2.7 La forme spéciale lambda

Le langage Scheme permet l’utilisation de la notation lambda (cf. § 1.2.5). Dans cette
notation, la fonction x 7→ x ∗ x qui à tout nombre associe son carré s’écrit λx . x ∗ x . La
variable x n’a qu’un rôle de marque-place et peut être renommée; λx . x ∗ x et λy . y ∗ y
définissent naturellement la même fonction.

2.7.1 Définition

En Scheme, le mot-clef lambda caractérise les formes spéciales permettant de définir les
fonctions. La valeur de l’expression

(lambda (x) (* x x))

est fonctionnelle; il s’agit de la fonction λx . x ∗ x. Les fonctions définies par l’utilisateur
s’utilisent de la même manière que les fonctions du système; on a par exemple

43Certains systèmes assimilent “#f” et “()” (liste vide); dans la suite, nous refusons cette assimilation
mais, pour éviter que la valeur de la forme (if e0 e1 e2) puisse varier selon le système utilisé, une règle
de bonne pratique est de s’interdire le cas où [[e0]] est la liste vide.

((lambda (x) (* x x)) (+ 2 3)) ==> 25

((lambda (x y z) (+ (* x y) z)) 4 5 6) ==> 26

Naturellement, il est souvent indiqué de donner un nom à une fonction :44

square ==> Error - unbound variable square

(define square (lambda (x) (* x x))) ==> ...

square ==> # procedure

(square (+ 2 3)) ==> 25

Syntaxiquement, une forme spéciale lambda est une liste à trois éléments : le mot-
clef lambda, la liste des paramètres et le corps de la forme lambda. Ce type de forme
est omniprésent en Scheme. On écrit souvent “λ-forme” (lire “lambda forme”) au lieu
de “forme lambda”. La liste des paramètres est une liste de variables; le corps est une
expression quelconque. Il faut bien observer le rôle particulier des paramètres, et ne pas
confondre

1. la valeur d’une λ-forme;
2. la valeur d’une combinaison dont le premier élément est cette λ-forme.

La première valeur est une fonction; la seconde valeur résulte de l’application de cette
fonction à des arguments qui sont les valeurs des éléments restants de la combinaison.
Ceci est illustré dans les sessions suivantes.

pi ==> 3.14159

y ==> Error - unbound variable y

Au départ, la variable pi est liée, la variable y ne l’est pas.

(lambda (y) (* y y)) ==> # procedure

(lambda (pi) (* pi pi)) ==> # procedure

Les deux λ-formes sont équivalentes; le nom du paramètre est indifférent. Les valeurs
de ces λ-formes étant —comme toujours — fonctionnelles, elles ne sont pas affichables; il
s’agit, dans les deux cas, de la fonction carré, ce qui est illustré par les deux évaluations
suivantes.

((lambda (y) (* y y)) 12) ==> 144

((lambda (pi) (* pi pi)) 12) ==> 144

Le processus d’évaluation des combinaisons de ce type provoque une liaison entre le
paramètre (ici y ou pi) et l’argument (ici 12), mais cette liaison est active pendant le
processus d’application seulement; nous reviendrons sur ce processus plus loin.

44Rappelons que “donner le nom var à un objet” signifie créer une liaison entre la variable var et l’objet.
Tant que cette liaison est en vigueur, la valeur de la variable var est l’objet. Nous (re)voyons ici que cet
objet, cette valeur, peut être non seulement un nombre ou une liste, mais aussi une fonction. Le langage
Scheme traite de manière homogène les différents types d’objets. Nous reviendrons plus loin sur ce point
important.

2.7.2 Le modèle de substitution

Le concept de λ-forme est central en programmation fonctionnelle; ce concept et celui
de récursivité sont la programmation fonctionnelle. En fait, l’idée sous-jacente semble
élémentaire et résumable en une égalité :

(λx, y.M) (ex, ey) = M [(x, y)/(ex, ey)] .

Si M est une expression, M [(x, y)/(ex, ey)] désigne l’expression obtenue en substituant ex et
ey à x et y dans M , c’est-à-dire en remplaçant toutes les occurrences de x et y par ex et ey,
respectivement. L’égalité ci-dessus signifie donc que pour appliquer la fonction (λx, y.M)
aux arguments ex, ey, il suffirait d’évaluer le corps M dans lequel les arguments ex et ey

sont substitués aux paramètres x et y. La substitution des arguments aux paramètres
dans le corps d’une λ-forme s’appelle une réduction.
Remarque. Le modèle de substitution brièvement introduit ici est simple, mais ne traduit
pas parfaitement la sémantique du système Scheme. Une variante plus fidèle consiste à
substituer au paramètres x et y les valeurs des opérandes ex et ey.

Considérons deux évaluations en Scheme :

((lambda (x y) (* x (- y 1))) 12 24) ==> 276

((lambda (x y) (* x (- y 1))) 3 (square 4)) ==> 45

Le premier exemple se comprend immédiatement; on a

[[((lambda (x y) (* x (- y 1))) 12 24)]] = [[(* 12 (- 24 1))]]

Le second exemple tient compte de la définition de square donnée plus haut. L’explication
de cet exemple met en jeu des réductions, mais aussi une substitution de nature un peu
différente : la fonction carré a été substituée à la variable square lors du processus
d’évaluation. La technique de calcul que nous venons de présenter, appelée modèle de

substitution, permet d’expliquer le résultat obtenu, mais une question se pose, qui concerne
l’ordre dans lequel les calculs intermédiaires se font. La valeur 45, résultat de l’évaluation
ci-dessus, peut être vue comme la valeur de l’expression45(* 3 (- (square 4) 1))
obtenue en réduisant la première λ-forme, ou comme la valeur de l’expression((lambda (x y) (* x (- y 1))) 3 16)
obtenue en réduisant d’abord la seconde λ-forme, qui a permis la définition de square.
On observe que le résultat final est le même quel que soit l’ordre choisi. On pourrait
démontrer que, si on reste dans le monde des λ-formes, cette indépendance vis-à-vis de
l’ordre des réductions est la règle.46 Nous nous contentons de le vérifier ici sur un exemple
plus significatif. L’expression

45Nous utilisons une police de caractères différente pour noter ces expressions, qui ne sont pas le résultat
d’une évaluation faite par le système Scheme.

46Plus précisément, si une séquence de réductions conduit à une valeur, toute séquence de réductions
qui se termine conduit à la même valeur.

((lambda (w) (* 2 w))

((lambda (v) (- 9 ((lambda (u) (+ 3 u)) v))) 5))

comporte trois λ-formes; en effectuant une réduction, on arrive donc à l’une des trois
expressions suivantes :(* 2 ((lambda (v) (- 9 ((lambda (u) (+ 3 u)) v))) 5))((lambda (w) (* 2 w)) (- 9 ((lambda (u) (+ 3 u)) 5)))((lambda (w) (* 2 w)) ((lambda (v) (- 9 (+ 3 v))) 5))
En effectuant une réduction supplémentaire, on arrive à l’une des trois expressions
suivantes :(* 2 (- 9 ((lambda (u) (+ 3 u)) 5)))(* 2 ((lambda (v) (- 9 (+ 3 v))) 5))((lambda (w) (* 2 w)) (- 9 (+ 3 5)))
En effectuant une réduction supplémentaire ou, dans le troisième cas, une addition, on
arrive à l’une des deux expressions(* 2 (- 9 (+ 3 5)))((lambda (w) (* 2 w)) (- 9 8))
Il est clair que la valeur finale sera 2, quel que soit l’ordre des opérations. Par contre,
l’efficacité du processus d’évaluation peut dépendre de l’ordre des opérations. L’exemple
de la forme (square (+ 3 4)) suffit à le montrer. Si on applique d’abord la fonction carré
(valeur de la λ-forme liée à square), la valeur cherchée est celle de (* (+ 3 4) (+ 3 4)),
tandis que si on applique d’abord la fonction d’addition, la valeur cherchée est celle de
(square 7). Dans le premier cas, la somme de 3 et 4 sera calculée deux fois, dans le
second cas elle ne sera calculée qu’une fois.

Le système Scheme impose que, dans l’évaluation d’une expression telle que
((lambda (x y) M) ex ey), les expressions ex et ey soient évaluées avant d’être
substituées à x et y dans l’expression M .47 Plus généralement, l’évaluation de (f a b)

requiert d’abord la détermination des valeurs [[f]], [[a]] et [[b]] avant l’application de la
première aux deux dernières. Par contre, l’ordre dans lequel les trois valeurs [[f]], [[a]]
et [[b]] sont calculées est libre. Ce dernier point montre bien que la forme spéciale if

ne pourrait pas être une fonction : l’ordre d’évaluation des expressions e0, e1 et e2 dans
(if e0 e1 e2) n’est pas libre. Comme on l’a vu au paragraphe précédent, l’expression e0 est
évaluée d’abord; ensuite, en fonction de la valeur obtenue, l’expression e1 ou l’expression
e2 est évaluée.

Une deuxième question se pose si on considère une définition plus compliquée telle que

47Notons que la stratégie inverse, dans laquelle les opérandes seraient substitués aux paramètres avant
d’être évalués, est parfaitement concevable et présenterait certains avantages. D’ailleurs, même si le système
Scheme utilise la première stratégie, dite “ordre applicatif”, des mécanismes existent qui permettent
d’utiliser la seconde, dite “ordre normal”, dans certains cas. Ces questions sortent du cadre de ce livre
introductif.

(define f

(lambda (a b c) ; 1

((lambda (y) ; 2

(((lambda (y) ; 3

(lambda (x) ; 4

(* c x y)))

a)

((lambda (x) ; 5

(+ a x y))

b)))

c)))

Cette définition comporte cinq λ-formes, numérotées de 1 à 5. On observe que x et y

apparaissent dans deux listes de paramètres différentes. Cela n’empêche pas le modèle
de substitution de fonctionner, comme le montre l’exemple suivant. Le système Scheme

donne

(f 2 3 4) ==> 72

Le modèle de substitution donne le même résultat; nous le montrons ici en utilisant pour
les réductions l’ordre normal, c’est-à-dire en réduisant les formes plus extérieures d’abord :(f 2 3 4)== ((lambda (y)(((lambda (y)(lambda (x)(* 4 x y)))2)((lambda (x)(+ 2 x y))3)))4)== (((lambda (y)(lambda (x)(* 4 x y)))2)((lambda (x)(+ 2 x 4))3))== ((lambda (x)(* 4 x 2))(+ 2 3 4))== (* 4 9 2)==> 72

Plus généralement, si on pose

(define g

(lambda (a b c)

(* c (+ a b c) a)))

on peut montrer que [[(f a b c)]] = [[(g a b c)]], quels que soient les nombres a= [[a]],
b= [[b]] et c =[[c]].

2.7.3 Exemples de calcul par le modèle de substitution

Le modèle de substitution permet un calcul aisé de la valeur de l’expression suivante :

((lambda (a) (* a a))

((lambda (a b) (a (* b b)))

(lambda (a) (* a a))

(* 2 2)))

Pour rendre l’expression plus lisible, la première étape consiste en un renommage des
variables.

((lambda (x) (* x x))

((lambda (a b) (a (* b b))) (lambda (y) (* y y)) (* 2 2)))

((lambda (x) (* x x))

((lambda (a b) (a (* b b))) (lambda (y) (* y y)) 4))

((lambda (x) (* x x)) ((lambda (y) (* y y)) (* 4 4)))

((lambda (x) (* x x)) ((lambda (y) (* y y)) 16))

((lambda (x) (* x x)) (* 16 16))

((lambda (x) (* x x)) 256)

(* 256 256)

65536

L’exemple suivant se traite de même :

((lambda (x y z) (x y z))

(lambda (x y) (* 2 x y))

((lambda (x y) (+ x y)) (+ 1 2) (+ 3 4))

((lambda (x) (+ 3 x)) 0))

((lambda (a b c) (a b c))

(lambda (x y) (* 2 x y))

((lambda (u v) (+ u v)) 3 7)

((lambda (z) (+ 3 z)) 0))

((lambda (a b c) (a b c))

(lambda (x y) (* 2 x y)) (+ 3 7) (+ 3 0))

((lambda (a b c) (a b c)) (lambda (x y) (* 2 x y)) 10 3)

((lambda (x y) (* 2 x y)) 10 3)

(* 2 10 3)

60

2.7.4 Exemples de définitions de procédures

Les formes spéciales define, if et lambda, ainsi que les fonctions arithmétiques primitives,
permettent la programmation de nombreuses fonctions utiles. Nous avons déjà rencontré
(sous d’autres noms) les fonctions

(define square_area

(lambda (c) (* c c)))

(define circle_area

(lambda (r) (* pi r r)))

La deuxième définition présuppose que la variable pi ait été liée à (une approximation de)
la constante mathématique π, par exemple 3.14159.
Remarque. Un programme Scheme est un ensemble de formes define. Les objets
définis sont le plus souvent des fonctions, mais il s’agit parfois d’objets non fonctionnels,
comme des nombres “importants”, auxquels on souhaite donner un nom. Les variables
correspondantes sont dites “globales”. Une convention courante consiste à utiliser pour
les variables globales des symboles commençant et se terminant par “*”; on écrirait alors

(define *pi* 3.14159)

(define square_area (lambda (r) (* *pi* r r)))

Il est préférable d’introduire des variables globales seulement pour des objets auxquels
on se réfère fréquemment, à divers endroits du programme. Une autre possibilité est de
faire de π non pas une constante numérique, mais une fonction numérique sans argument;
on écrirait alors

(define pi (lambda () 3.14159))

(define circle_area (lambda (r) (* 2 (pi) r)))

Notons l’usage des parenthèses : la valeur numérique [[(pi)]] est l’application de la
fonction [[tt pi]].

L’usage des fonctions sans arguments a un avantage qu’illustre la session suivante :

(define pi 3.14)

(define inv_pi (/ 1 pi))

inv_pi ==> 0.31847

(define pi 3.14159)

inv_pi ==> 0.31847

Comme on l’a vu au paragraphe 2.3, la redéfinition de pi n’a pas eu d’effet sur inv_pi.
Par contre, on a



2 LES BASES DE SCHEME 40

(define pi (lambda () 3.14))

(define inv_pi (lambda () (/ 1 (pi))))

(inv_pi) ==> 0.31847

(define pi (lambda () 3.14159))

(inv_pi) ==> 0.31831

La redéfinition de pi a eu l’effet approprié sur inv_pi.

D’autres procédures simples permettent de calculer les surfaces de diverses formes
géométriques. On a par exemple

(define lateral_cone_area

(lambda (r a) (* *pi* r a)))

(define base_cone_area

(lambda (r) (* *pi* r r)))

(define cone_area

(lambda (r a)

(+ (lateral_cone_area r a) (base_cone_area r))))

Remarque. On pouvait aussi écrire

(define base_cone_area

(lambda (r) (circle_area r)))

ou, plus simplement,

(define base_cone_area circle_area)

On peut aussi, par exemple, écrire une fonction prenant comme argument une liste de
trois nombres et renvoyant la somme de leurs racines carrées :

(define sum_3_sqrt

(lambda (l)

(+ (sqrt (car l)) (sqrt (cadr l)) (sqrt (caddr l)))))

On peut généraliser cette fonction en remplaçant la fonction prédéfinie sqrt par une
fonction numérique unaire quelconque :

(define sum_3_f

(lambda (f l) (+ (f (car l)) (f (cadr l)) (f (caddr l)))))

On a

(sum_3_f (lambda (x) (* 2 x)) ’(1 2 3)) ==> 12

(sum_3_sqrt ’(25 36 49)) ==> 18

(sum_3_f sqrt ’(25 36 49)) ==> 18

Par contre, il semble difficile de généraliser sum_3_f en la fonction +_map, admettant
comme second argument une liste de longueur quelconque. Pour cela et pour la plupart
des problèmes courants, on devra recourir à la récursivité (cf. § 4).

2.7.5 La forme lambda généralisée

Nous avons déjà rencontré les fonctions prédéfinies +, * et list, qui admettent un nombre
quelconque d’arguments. Il en est de même des prédicats arithmétiques <, <=, =, > et >=.
Pour permettre à l’utilisateur de définir de telles fonctions, le langage Scheme propose
une version généralisée de la forme spéciale lambda; au lieu d’avoir une liste de paramètres
de longueur fixée, telle (x y z), on utilise un simple identificateur comme paramètre. La
valeur de ((lambda v E) e1 . . . en) s’obtient en évaluant E, la valeur du paramètre v

étant la liste des valeurs [[e1]],. . . ,[[en]] .

La fonction list, prédéfinie dans le langage, pourrait être définie comme suit :

(define list (lambda v v))

On a en effet

((lambda v v) 1 2 3 4) ==> (1 2 3 4)

On peut aussi créer des fonctions de projection, telles

(define proj_1 (lambda v (car v)))

(define proj_3 (lambda v (caddr v)))

On a alors

(proj_1 11 12 13 14) ==> 11

(proj_3 11 12 13 14) ==> 13

On veillera à distinguer (lambda v e) et (lambda (v) e). En particulier, proj_1 et car
ne sont pas équivalents. Rappelons que, si [[f]] est une fonction à trois arguments alors f
et (lambda (x y z) (f x y z)) ont pour valeurs des fonctions égales.

Enfin, la notation pointée introduite au paragraphe 2.5 permet de définir des fonctions
dont le nombre minimal d’arguments est fixé. Par exemple, la fonction proj_3 ne peut
être appliquée sans erreur que s’il y a au moins trois arguments; on peut souligner ce fait
en la redéfinissant comme suit :

(define proj_3 (lambda (x1 x2 x3 . v) x3))

Certaines fonctions prédéfinies, telles la soustraction et la division, admettent au moins
un argument; on notera leur comportement particulier :

(- 1) ==> -1

(- 1 2) ==> -1

(- 1 2 3) ==> -4

(/ 4) ==> 1/4

(/ 4 3) ==> 4/3

(/ 4 3 2) ==> 2/3

2.7.6 Statut “première classe” des procédures

Les procédures, valeurs des λ-formes, héritent de toutes les propriétés essentielles des
valeurs usuelles comme les nombres. En particulier, on peut définir une procédure
admettant d’autres procédures comme arguments, et renvoyant une procédure comme
résultat. De même, on peut lier une procédure à un symbole, en utilisant define.
Ceci évoque les mathématiques (en analyse moderne, les domaines de fonctions ont le
même “statut” que les domaines de nombres), mais contraste avec d’autres langages de
programmation usuels, qui ne permettent pas, par exemple, de lier une procédure à une
variable (sauf naturellement au moment où la procédure est définie). Nous savons déjà
que cette limitation n’existe pas en Scheme :

(define square (lambda (x) (* x x))) ==> ...

square ==> # procedure

(square (+ 4 1)) ==> 25

(define power2 square) ==> ...

power2 ==> # procedure

(power2 (+ 4 1)) ==> 25

Remarquons aussi l’analogie de traitement de l’opérateur et des opérandes lors de
l’évaluation d’une combinaison telle que (square (+ 4 1)); on commence par évaluer
(peut-être en parallèle) l’opérateur et l’opérande, ce qui donne respectivement la valeur
(fonction unaire) v0 = λx.x2 et la valeur (numérique) v1 = 5. On applique alors v0 à v1,
ce qui revient à évaluer la forme (* x x) où x est lié à 5.

Conceptuellement, la valeur-procédure v0 associée à la variable square pourrait être
la table (infinie) des couples (n, n2). Une telle table n’est pas affichable, et Scheme

affichera simplement “# procedure”. Concrètement, une valeur-procédure est soit une
fonction primitive, soit le résultat de l’évaluation d’une λ-forme; ce résultat s’appelle une
fermeture et comporte les informations nécessaires au système pour évaluer l’application
de la valeur-procédure à des arguments.

Un aspect plus spectaculaire du principe consistant à traiter les valeurs fonctionnelles
comme les valeurs numériques est la facilité avec laquelle, en Scheme, on écrit des
procédures dites “d’ordre supérieur”, dont le domaine et le codomaine comportent
eux-mêmes des procédures, éventuellement d’ordre supérieur. Un exemple simple est
l’opérateur mathématique de composition fonctionnelle. Il prend comme arguments une
fonction f de D2 dans D3 et une fonction g de D1 dans D2 et leur associe une troisième
fonction f ◦ g : x → f(g(x)) de D1 dans D3.

48 Ceci est illustré dans la session ci-dessous.

(define compose

(lambda (f g)

(lambda (x) (f (g x))))) ==> ...

48Quand on évalue (f ◦g)(x), c’est-à-dire f(g(x)), on applique d’abord g, puis f . Pour cette raison, f ◦g
se lit souvent “g rond f” ou “f après g”.

(compose car cdr) ==> # procedure

((compose car cdr) ’(1 2 3 4)) ==> 2

((compose (compose car cdr) cdr) ’(1 2 3 4)) ==> 3

((compose (compose car cdr) (compose cdr cdr)) ’(1 2 3 4)) ==> 4

Un exemple plus élaboré est l’opérateur de dérivation qui à toute fonction dérivable
de R dans R associe une fonction de R dans R. Pour éviter l’intervention de la notion de
limite, nous fixons un incrément dx, vu comme argument supplémentaire de l’opérateur.
On définit alors

(define deriv

(lambda (f dx)

(lambda (x)

(/ (- (f (+ x dx)) (f x)) dx))))

La valeur fonctionnelle v1 = [[(deriv f dx)]] est une bonne approximation de la dérivée
de la valeur fonctionnelle v0 = [[f]]. Par exemple, si v0 est la fonction carré et si
[[dx]] = 0.00001, alors la fonction v1 est

x −→ (x + 0.00001)2 − x2

0.00001

ou encore la fonction
x −→ 2x + 0.00001 ,

qui est une très bonne approximation de la fonction double, dérivée de la fonction carré.
On a par exemple :

((deriv square 0.00001) 10) ==> 20.00000999942131

Le modèle de substitution donne((deriv square 0.00001) 10)(((lambda (f dx)(lambda (x) (/ (- (f (+ x dx)) (f x)) dx))))square0.00001)10)((lambda (x)(/ (- (square (+ x 0.00001)) (square x)) 0.00001))10)(/ (- (square (+ 10 0.00001)) (square 10)) 0.00001)

Remarque. On n’a pas utilisé ici strictement l’ordre normal.

Remarque. Les algorithmes de calcul arithmétique en nombres décimaux souffrent souvent
d’erreurs d’arrondis; c’est pourquoi on a obtenu le résultat 20.00000999942131 au lieu de
20.00001. L’erreur d’arrondi ne doit pas être confondue avec l’erreur systématique liée
au choix de la valeur de dx; sans cette erreur systématique, la réponse attendue aurait été
20 et non 20.00001.

3 Règles d’évaluation

Dans ce chapitre nous synthétisons les règles relatives à l’évaluation des expressions. Le
modèle de substitution introduit au chapitre précédent ramène l’évaluation des expressions
à l’évaluation des expressions simples, constantes et variables, et à l’application de
fonctions prédéfinies à des arguments évalués. Ces mécanismes fondamentaux sont très
simples : la valeur d’une constante est elle-même, la valeur d’une variable est l’objet qui lui
est lié et le résultat de l’application d’une fonction prédéfinie à des arguments est obtenu
en exécutant le code relatif à cette fonction primitive.

L’introduction de la forme spéciale essentielle lambda fait apparâıtre la notion de
réduction, ce qui complique la situation. En effet, l’opération de substitution d’un
argument [[e]] à un paramètre [[x]] doit pouvoir se faire même si la variable x est liée
à une autre valeur v. On ne peut donc plus considérer que l’objet lié à une variable
est unique. Le problème fondamental consistant à déterminer l’objet lié à une variable
dans une situation donnée devient complexe si de nombreuses λ-formes, éventuellement
imbriquées, sont présentes dans le programme que l’on exécute. Ce chapitre a aussi pour
but d’introduire la notion d’environnement, utilisée pour gérer le problème du lien entre
une variable et une valeur, et le modèle de calcul lié à cette notion.

3.1 Résumé des règles

Le chapitre précédent contenait une définition précise de la syntaxe des expressions, ainsi
que quelques indications sur la manière de les évaluer. Nous revenons ici plus précisément
sur le processus d’évaluation des expressions.

Rappelons d’abord que les règles d’évaluation des expressions peuvent être récursives :
l’évaluation d’une expression peut requérir l’évaluation de sous-expressions. Nous avons
vu aussi que le processus d’évaluation renvoie en général une valeur; il peut aussi avoir
un effet (outre l’affichage de la valeur), comme la création d’une liaison dans le cas d’une
forme define.

Le processus d’évaluation comporte une série de règles; la règle à appliquer pour une
expression donnée dépend exclusivement du type de l’expression et donc, de sa syntaxe.
Pour le processus d’évaluation, les nombres, les booléens et les variables sont des cas de
base, tandis que les combinaisons sont des cas inductifs. Les formes spéciales sont des cas
de base ou des cas inductifs, cela dépend du type de la forme, déterminé par le mot-clef.

Voilà d’abord les règles relatives aux cas de base.

• L’évaluation d’un nombre donne ce nombre.

• L’évaluation d’un booléen donne ce booléen.

• L’évaluation d’un symbole donne la valeur liée à ce symbole, si elle existe.

• L’évaluation d’une forme spéciale quote donne l’expression citée, non évaluée.

• L’évaluation d’une forme spéciale lambda donne la valeur fonctionnelle associée; cette
valeur est une fermeture, objet à trois composants dont la nature sera précisée sous
peu.

Voici une session donnant quelques exemples :

(define pi 3.14159) ==> ...

’pi ==> pi

pi ==> 3.14159

#f ==> #f

twice ==> Error - unbound variable twice

(define twice (lambda (x) (* 2 x))) ==> ...

twice ==> # procedure ...

(twice 2) ==> 4

(twice pi) ==> 6.28318

car ==> # procedure ...

(lambda (x) (+ x y)) ==> # procedure ...

xxx ==> Error - unbound variable xxx

’xxx ==> xxx

’(a b) ==> (a b)

’(+ 3 5) ==> (+ 3 5)

(quote a) ==> a

Dans le cas inductif, l’évaluation d’une expression requiert l’évaluation préalable de
sous-expressions. On a les règles suivantes :

• L’évaluation d’une combinaison (e0 e1 . . . en) s’effectue comme suit :

1. Les ei sont évalués (soient vi les valeurs);
v0 doit être une fonction dont le domaine contient (v1, . . . , vn).

2. La fonction v0 est appliquée à (v1, . . . , vn).

3. Le résultat de l’application est la valeur de la combinaison.

Remarque. Si l’un des vi n’existe pas ou si v0 n’est pas applicable, l’évaluation
s’arrête sans fournir de valeur.

• L’évaluation de la forme spéciale (define symb e) s’effectue comme suit :

1. L’expression e est évaluée; soit v sa valeur.

2. La valeur v est liée à la variable symb.

3. Aucune valeur n’est produite.

• L’évaluation de la forme spéciale (if e0 e1 e2) s’effectue comme suit :

1. L’expression e0 est évaluée; soit v0 sa valeur.

2. Si v0 est #f, e2 est évalué, sinon e1 est évalué;
dans les deux cas, soit v la valeur produite.

3. La valeur de l’expression conditionnelle est v.

Remarque. Si v0 ou v n’existe pas, l’évaluation s’arrête sans produire de valeur.

Remarque. Une valeur fonctionnelle n’est pas affichable explicitement; un message
informatif (qui n’est pas un message d’erreur) remplace cet affichage.

Remarque. La forme spéciale if présente une particularité importante : une des sous-
expressions n’est pas évaluée. En fait, e0 est toujours évaluée mais, des deux sous-
expressions e1 et e2, une seule sera évaluée : la seconde si [[e0]] est #f, la première sinon.
Cette particularité est intéressante en pratique, parce que cela permet à l’expression non
évaluée de n’avoir pas de valeur, sans que cela provoque une erreur. Nous verrons au
chapitre suivant que ceci est mis à profit lors de la définition d’une fonction récursive.
Un exemple d’une telle définition, la fonction fib, apparaissait d’ailleurs déjà au premier
chapitre; le code comportait l’expression conditionnelle

(if (< n 2) n (+ (fib (- n 1)) (fib (- n 2))))

Quand la sous-expression

(< n 2)

est vraie, la sous-expression

(+ (fib (- n 1)) (fib (- n 2)))

n’est pas évaluée.

Rappel. La valeur [[e0]] est la condition.49 En général, la condition est booléenne, mais cela
n’est pas obligatoire : Une condition qui n’est pas fausse est assimilée à une condition vraie.
On a par exemple [[(if 1 2 3)]] = 2. Il faut cependant que la condition existe, sinon la
forme conditionnelle n’a pas de valeur. L’évaluation de l’expression (if (/ 1 0) 2 3)

donnera lieu à une erreur.

3.2 Mode d’application et environnements

3.2.1 Introduction et exemple

La valeur de (e0 e1 ... en) est le résultat de l’application de la fonction [[e0]] aux
arguments [[e1]], . . . , [[en]]. Si [[e0]] est une fonction primitive, le système exécute le
code interne associé à cette primitive pour produire ce résultat. Si par contre [[e0]] est
[[(lambda (x1 ... xn) M)]], un processus spécial est exécuté; il consiste en l’évaluation
du corps M de la λ-forme, étant admis que les variables x1, . . . , xn sont liées aux valeurs
[[e1]], . . . , [[en]], respectivement. La liaison relative aux paramètres est prise en compte

49En pratique, et par abus de langage, l’expression e0 elle-même est parfois appelée “condition”.

lors de cette évaluation seulement; dans le cadre de l’évaluation d’une autre expression,
elle sera invisible. Avant de décrire ce processus de manière plus précise, nous donnons
un petit exemple.

(define y 9) ==> ...

x ==> Error - unbound variable x

y ==> 9

((lambda (x y) (+ 2 x y)) 3 7) ==> 12

x ==> Error - unbound variable x

y ==> 9

Dans cette session, les liaisons de x et y à 3 et 7 ne sont actives que lors de l’évaluation
correspondant à la quatrième ligne de cette session, après quoi ces liaisons deviennent
inaccessibles, comme le montrent les deux dernières lignes de la session. Une réévaluation
ultérieure de la forme ((lambda (x y) (+ 2 x y)) 3 7) créera des liaisons analogues
qui deviendront à leur tour inaccessibles. Par contre, dans la session

x ==> Error - unbound variable x

y ==> 9

(define foo

((lambda (x y) (lambda (u) (+ u x y))) 3 7)) ==> ...

x ==> Error - unbound variable x

y ==> 9

(foo 8) ==> 18

x ==> Error - unbound variable x

y ==> 9

(foo 20) ==> 30

(foo y) ==> 19

les liaisons de x et y à 3 et 7, créée lors de la définition de foo, sont actives chaque fois
que la fonction [[foo]] est appliquée à un argument. En dehors de telles applications,
ces liaisons sont inaccessibles. Observons que [[foo]] est la fonction x 7→ x + 10; lors
de l’évaluation de (foo y), le système tient compte de deux liaisons relatives à y. Cet
exemple montre que plusieurs liaisons relatives à une même variable peuvent coexister à
un moment donné.

3.2.2 Notion d’environnement

En l’absence du mécanisme des λ-formes, le problème de la liaison entre variables et valeurs
peut être géré par une simple table, ou cadre, qui est un ensemble de paires variable-valeur.
Ce cadre global comporte les liaisons relatives aux variables prédéfinies et à celles qui ont
fait l’objet d’un define.

Le processus d’application d’une fermeture à des arguments suscite la création d’un
nouveau cadre, composé des paires paramètre-argument. Lors de l’évaluation du corps de

la procédure (compris dans la fermeture) la valeur d’une variable est recherchée d’abord
dans le nouveau cadre; elle s’y trouvera si la variable est un paramètre.

D’une manière générale, toute évaluation de variable a lieu dans un environnement,
c’est-à-dire un pointeur vers une structure dont les nœuds sont des cadres. Quand le
processus d’application d’une fermeture commence, un nouvel environnement est créé à
partir de l’environnement courant. Ce nouvel environnement comporte un pointeur vers le
nouveau cadre, qui lui-même pointe vers l’ancien environnement.50 Lorsque le processus
d’application est terminé, l’ancien environnement est restauré. On conçoit que, lors de
l’évaluation d’une expression comportant de nombreuses λ-formes, des environnements
compliqués puissent être créés. L’environnement de départ, dans lequel l’utilisateur se
trouve quand il entre en session, est l’environnement global, comportant seulement le
cadre global. Ce cadre comporte les liaisons prédéfinies et celles crées par l’utilisateur au
moyen d’un define.

3.2.3 Les fermetures et le processus d’application

On a mentionné au paragraphe 2.7.6 que la valeur d’une λ-forme s’appelle une fermeture

et comporte les informations nécessaires au système pour évaluer toute forme impliquant
l’application de la valeur-procédure à des arguments. On peut maintenant préciser que ces
informations comportent trois composants : la liste des paramètres, le corps de la λ-forme,
et l’environnement de définition, c’est-à-dire l’environnement dans lequel la λ-forme a été
évaluée. C’est dans cet environnement que les valeurs correspondant aux variables non
locales du corps de la λ-forme seront recherchées, chaque fois que la valeur-procédure sera
appliquée à des arguments.

On peut maintenant décrire plus finement le processus d’application, qui est un
mécanisme essentiel de l’évaluateur. Rappelons d’abord qu’une fonction (mathématique)
est un ensemble de couples de valeurs; la première est un élément du domaine, la seconde
l’image correspondante. Appliquer une fonction (à un nombre adéquat d’arguments de
types appropriés) est produire l’image associée à cette suite d’arguments. En Scheme,
pour les fonctions prédéfinies (liées à +, cons, etc.) le processus d’application est
l’exécution du code correspondant à la fonction. Le cas des fonctions définies par
l’utilisateur, au moyen de la forme spéciale lambda, requiert une définition supplémentaire :

• Appliquer la fermeture v0 = [[(lambda (x1 ... xn) M)]] à des arguments
v1,. . . ,vn consiste d’abord à créer un nouvel environnement, où les valeurs de
x1,. . . ,xn sont respectivement v1,. . . ,vn, puis à évaluer la forme M dans cet
environnement, les valeurs des variables libres éventuellement présentes dans M
étant recherchées dans l’environnement inclus dans la fermeture v0).

Illustrons d’abord cette règle au moyen d’un exemple élémentaire :

(define add_3 (lambda (u) (+ 3 u))) ==> ...

(add_3 (* 2 4)) ==> 11

50On verra plus loin que si E′ pointe vers E′′, E′ est l’environnement de contrôle de E′′.

Ces expressions sont évaluées dans l’environnement global E0, comportant le seul cadre
global C0. Dans ce cadre se trouvent les liaisons entre les variables prédéfinies (telles + et
*) et leurs valeurs. La première évaluation concerne une forme spéciale define. La valeur
de la λ-forme est la fermeture [(u);(+ 3 u);E0] (E0 étant l’environnement de définition
de la fermeture); elle est liée à la variable add_3, dans le cadre global.

La seconde évaluation concerne une combinaison dont l’opérateur est add_3 et
l’opérande est (* 2 4).

La première étape consiste en l’évaluation dans E0 de l’opérateur et de l’opérande.
L’évaluation de l’opérateur donne la valeur qui lui est líee dans E0. L’évaluation de
l’opérande consiste en l’évaluation des composants *, 2 et 4 (la valeur de * est trouvée dans
l’environnement courant E0) et en l’application de [[*]] à 2 et 4, c’est-à-dire en l’exécution
du code effectuant la multiplication de deux nombres; le résultat produit est 8. La seconde
étape consiste en l’application de la fermeture [(u);(+ 3 u);E0] à la valeur 8. Cette
application débute par la création d’un nouvel environnement E1, et donc d’un nouveau
cadre C1 (voir figure 4). Dans ce cadre est placée une liaison entre le paramètre u de la
fermeture et l’argument 8. Enfin, le corps (+ u 3) est évalué dans l’environnement E1;
la valeur de la constante 3 est 3; la valeur de la variable locale u est trouvée dans le cadre
local C1 et la valeur de la variable non locale + est cherchée et trouvée dans l’environnement

d’accès, qui est l’environnement E0 compris dans la fermeture. L’application de [[+]]
à 3 et 8 donne 11, valeur qui est passée à l’environnement de contrôle de E1, c’est-à-
dire à l’environnement E0 à partir duquel E1 a été créé, et auquel on revient après que
la valeur à produire dans E1 l’a été. Cette valeur 11 est le résultat de l’évaluation de
((lambda (u) (+ 3 u)) (* 2 4)) dans E0.

C0

* : ...

+ : ...

add 3 : [(u);(+ 3 u);E0]

C1

u : 8
-

E0

E1

Figure 4: Environnements

L’exemple de l’expression

((lambda (w) (* 2 w))

((lambda (v) (- 9 ((lambda (u) (+ 3 u)) v))) x))

déjà évoquée lors de la présentation du modèle de substitution (§ 2.7.2), sera une deuxième
illustration des principes et des notations utilisées dans le modèle des environnements.
Nous supposons que cette expression est évaluée dans l’environnement global E0, où la
variable x est liée à la valeur 5 suite à l’évaluation de (define x 5). L’expression à
évaluer est une combinaison. La valeur de l’opérateur est une fermeture w0 comportant
l’environnement E0. L’unique opérande est aussi une combinaison, dont l’opérateur a

C0

+ : ...

* : ...

x : 5

C1

v : 5

C2

u : 5

C3

w : 1

- -

-

E0

E1

E3

E2

Figure 5: Modèle des environnements

pour valeur une fermeture v0 comportant l’environnement E0 et dont l’opérande x a
pour valeur 5. Un environnement E1 est créé, où v a pour valeur 5 et dans lequel v0

est appliqué à [[v]]; le résultat de cette application est la valeur dans E1 du corps
(- 9 ((lambda (u) (+ 3 u)) v)) contenu dans la fermeture v0. Pour obtenir la valeur
de cette combinaison, il faut notamment évaluer son deuxième opérande, qui résultera
de l’application d’une fermeture u0 (comportant l’environnement E1) à la valeur 5, ce
qui implique de créer un environnement E2, où u a pour valeur 5 et dans lequel le corps
(+ 3 u) est évalué, ce qui donne la valeur 8. La valeur produite dans E1 est donc 1

(9 − 8). La valeur à produire dans E0 résulte de l’application à 1 de la fermeture w0.
Pour obtenir cette valeur, un environnement E3 est créé, où w a pour valeur 1 et dans
lequel le corps (* 2 w) est évalué, ce qui donne 2, valeur de l’expression de départ dans
l’environnement E0.

La figure 5 représente les divers environnements. Chaque environnement est un
pointeur vers un cadre qui porte son nom (cette convention restera d’application dans tous
les exemples ultérieurs). Du cadre E0 partent deux flèches parce que les évaluations à faire
dans l’environnement E0 impliquaient l’application de deux fermetures à des arguments.
On observe que les évaluations à faire dans un environnement Ei donné impliquent parfois
des valeurs à chercher dans d’autres cadres que le cadre Ei. En fait, l’environnement E2

comporte non seulement le cadre E2 mais aussi les cadres E1 et E0; c’est dans E0 qu’est
obtenue la valeur liée à la variable +, nécessaire pour l’évaluation dans E2 de la forme
(+ 3 u). Nous approfondissons ce point au paragraphe suivant.

3.3 Portée des variables

3.3.1 Variables globales, variables locales, variables libres

Supposons un environnement dont le premier cadre comporte une liaison entre la variable x
et la valeur v. La valeur de x dans cet environnement est naturellement la valeur v. Si ce
premier cadre ne comporte pas de liaison relative à la variable x, la valeur éventuelle de x

doit se trouver ailleurs. Nous commençons l’étude de cette question par l’exemple simple :

;; Exemple 1

(define square (lambda (x) (* x x))) ==> ...

((lambda (x) (+ 9 (square (+ x 1)))) 3) ==> 25

Le cadre global C0 comporte des liaisons relatives aux variables prédéfinies + et *;
la première ligne de la session a pour effet d’ajouter à ce cadre une liaison entre
la variable square et sa valeur, la fermeture [(x);(* x x);E0]. La seconde ligne
de la session est l’évaluation d’une combinaison dont l’opérateur a pour valeur la
fermeture [(x);(+ 9 (square (+ x 1)));E0]; l’application de cette fermeture suscite
la création d’un environnement E1, dont le premier cadre C1 lie x à 3. Ensuite, le corps
(+ 9 (square (+ x 1)) est évalué dans cet environnement. La valeur de x est trouvée
localement, c’est-à-dire dans le cadre C1, tandis que les valeurs des variables + et square
sont trouvées dans l’environnement de définition, enclos dans la fermeture, c’est-à-dire E0.
La suite du processus requiert l’évaluation de (square (+ x 1)) dans l’environnement E1;
[[square]] étant une fermeture, il y a création d’un nouvel environnement E2, dont le
premier cadre C2 lie x à la valeur de (+ x 1) dans E1, c’est-à-dire 4. L’évaluation de
(* x x) dans E2 produit la valeur 16. Enfin, une application de la fonction primitive [[+]]
donne le résultat final 25. Les environnements concernés par cette session sont représentés
à la figure 6. On notera que E0 comporte le seul cadre C0, E1 comporte les cadres C1 et
C0, et E2 comporte les cadres C2, C1 et C0.

C0

* : ...

+ : ...

square: [(x);(* x x);E0]

C1

x : 3

C2

x : 4
- -

E0

E1E2

Figure 6: Environnements et variable globale

Dans cet exemple, deux sortes de variables apparaissent : les variables locales, dont
la valeur dans un environnement donné est trouvée dans le cadre de même indice
que cet environnement, et les variables globales, dont les valeurs sont trouvées dans
l’environnement global.

Il existe aussi des variables libres, dont la valeur sera trouvée dans un cadre
intermédiaire. Un exemple simple de variable libre apparâıt dans la session suivante :

;; Exemple 2

((lambda (a) ((lambda (x) (+ a (* x 2))) 2)) 3) ==> 7

Avec les notations de la figure 7, évaluer l’expression complète dans l’environ-
nement global E0 implique la création de E1, où a est lié à 3; l’évaluation de
((lambda (x) (+ a (* x 2))) 2) dans l’environnement E1 suscite à son tour la création
de E2, où x est lié à 2. Il y a alors l’évaluation du corps (+ a (* x 2)) dans E2; dans
cette expression (et dans cet environnement), la variable a est libre.

C0

* : ...

+ : ...

C1

a : 3

C2

x : 2
- -

E0
E1E2

Figure 7: Environnements et variable libre

3.3.2 Conflits de noms

Dans l’exemple 1, la variable x est liée dans le cadre C1 et dans le cadre C2.
Cela ne pose pas de problème : la liaison C1 est utilisée pour déterminer la valeur
de x dans l’environnement E1, puisque, dans cet environnement, on n’a pas accès
au cadre E2; d’autre part, la liaison C2 est utilisée pour déterminer la valeur de x

dans l’environnement E2 : une occurrence d’une variable définie localement est toujours
considérée comme locale. Il n’y a donc jamais de conflit entre une variable locale et une
variable non locale. Par contre, il peut y avoir conflit entre deux variables libres, ou entre
une variable libre et une variable globale, comme le montre l’exemple suivant :

;; Exemple 3

(define pi 3.14159) ==> ...

(define circ (lambda (r) (* 2 pi r))) ==> ...

((lambda (pa pe pi po pu)

(+ (circ pa) (circ pe) (circ pi) (circ po) (circ pu)))

1 2 4 8 16) ==> 194.77858

L’évaluation de l’expression ((lambda ...) 1 2 4 8 16) dans l’environnement global E0

implique la création d’un nouvel environnement E1 (voir figure 8) et l’évaluation du corps
de la λ-forme, la combinaison (+ (circ pa) ... (circ pu)), dans cet environnement.
Les six éléments de cette combinaison (un opérateur et cinq opérandes) sont évalués dans le
même environnement E1. Considérons l’opérande (circ pi); son évaluation requiert celle
de circ, variable globale dont la valeur est la fermeture [(r);(* 2 pi r);E0], et celle de
pi, variable locale liée à 4. L’application de la fermeture [[circ]] à 4 requiert la création
de l’environnement E2 et l’évaluation de la forme (* 2 pi r) dans cet environnement.
Il faut donc connâıtre la valeur de pi, que l’on peut trouver soit dans l’environnement

d’application (de [[circ]] à 4, donc E1), soit dans l’environnement de définition (de cette
même fonction, donc E0).

Dans le langage Scheme, c’est toujours l’environnement de définition qui est utilisé en
pareil cas; on dit que Scheme utilise la règle de portée lexicale, ou statique des variables,
par opposition à la règle de portée dynamique des variables, adoptée dans certains autres
langages. Dans l’exemple qui nous occupe, c’est donc la valeur [[pi]] = 3.14 qui sera

C0

+ : ...

pi : 3.14159

circ : [...]

C1

pa : 1

pe : 2

pi : 4

po : 8

pu : 16

C2

r : 4
- -

E0

E1

E2

Figure 8: Environnements et conflit de nom

utilisée lors de l’évaluation de (* 2 pi r), et non la valeur 4. Concrètement, lors de la
création de l’environnement E2, le système note que les valeurs des variables libres sont
à chercher non dans l’environnement de contrôle E1 (à partir duquel E2 est créé) mais
dans l’environnement mentionné dans la fermeture [[circ]], qui est l’environnement de
définition de [[circ]], c’est-à-dire E0. Par rapport à E2, E0 est l’environnement d’accès.

Le fait, lors de l’application d’une fonction à ses arguments, de privilégier
l’environnement de définition a une conséquence générale très importante : la fonction
liée à la variable circ lors de l’évaluation (dans l’environnement global) de la forme
(define circ ...) est fixée une fois pour toutes; il suffit de lire le texte du programme,
c’est-à-dire le fichier contenant les définitions relatives à pi et à circ pour connâıtre
cette fonction. Si la règle de portée dynamique était d’application, la fonction [[circ]]
correspondant à ces deux définitions serait la même que précédemment . . . sauf dans les
cas ou, volontairement ou non, cette fonction serait appliquée dans un environnement où
pi a une autre valeur. Notons aussi qu’avec la règle de portée statique, les deux expressions

((lambda (pa pe pi po pu)

(+ (circ pa) (circ pe) (circ pi) (circ po) (circ pu)))

1 2 4 8 16)

et

((lambda (a e i o u)

(+ (circ a) (circ e) (circ i) (circ o) (circ u)))

1 2 4 8 16)

ont toujours même valeur, ce qui semble naturel.51 Observons enfin que, d’une manière
générale, la seule façon efficace qu’ait l’esprit humain d’appréhender une entité complexe,
qu’il s’agisse d’un programme, d’un moteur de voiture, d’un circuit électronique ou d’une
phrase de Cicéron, est de pouvoir décomposer cette entité en composants ayant chacun
leur signification propre, et tels que la signification du tout puisse s’obtenir à partir de la
signification des composants. Dans le cas d’un programme écrit dans un langage à portée
dynamique, cette décomposition est plus difficile à réaliser, puisque la signification d’une
définition, ou d’un groupe de définitions, peut dépendre d’éléments extérieurs à ce groupe.

51Avec la règle de portée dynamique, nous venons de voir que ce n’était pas le cas.

Remarque. On peut trouver des cas où la règle de portée dynamique serait commode.
Considérons les définitions suivantes :(define generi ir (lambda (r) (* 2 pi r))) ==> ...(define real ir (lambda (pi) ir)) ==> ...

Dans un système à portée dynamique, on pourrait calculer des circonférences en
utilisant selon les circonstances diverses approximations de pi, et on aurait((real ir 3.14) 1) ==> 6.28
et((real ir 3.1416) 1) ==> 6.2832
Pour obtenir le même effet en Scheme, on pourrait redéfinir à chaque fois la variable
globale pi, mais on a déjà mentionné le danger inhérent aux redéfinitions globales des
variables. De manière moins critiquable, on utiliserait en Scheme le code suivant :

(define mk_real_circ

(lambda (pi)

(lambda (r) (* 2 pi r))))

On aurait alors

((mk_real_circ 3.14) 1) ==> 6.28

((mk_real_circ 3.1416) 1) ==> 6.2832

La différence essentielle est qu’en portée lexicale, toute dépendance est matérialisée
par un élément syntaxique, lexical. En Scheme, la valeur de l’expression
((lambda (pi) circ) x) est indépendante de la valeur de x, puisque l’expression circ

ne comporte pas d’occurrence du paramètre pi. En fait, dès que x et circ ont une valeur,
on a toujours

[[((lambda (pi) circ) x)]] = circ

3.3.3 Exemple supplémentaire

Un exemple supplémentaire va permettre de mieux voir les implications de la règle de
portée statique et lexicale des variables. Tous les environnements dont il est question dans
ce paragraphe sont représentés à la figure 9. L’exemple commence par les deux définitions
suivantes :

(define add_7

((lambda (a f)

(lambda (x) (f a x)))

7

+))

(define add_8

(lambda (x)

((lambda (a f) (f a x))

8

+)))

L’évaluation de la première forme define dans l’environnement global E0 commence par
l’évaluation de la combinaison ((lambda (a f) (lambda (x) (f a x))) 7 +)

La valeur de l’opérateur est la fermeture [(a f);(lambda (x) (f a x));E0], qui
doit être appliquée aux valeurs des opérandes, soient 7 et +; l’environnement E1 est créé,
où a et f sont liés à 7 et +. Le corps (lambda (x) (f a x)) est évalué dans E1, ce qui
donne la fermeture [(x);(f a x);E1]; enfin, cette valeur est liée à la variable add_7 dans
l’environnement E0.

L’évaluation de la deuxième forme define dans l’environnement global E0 commence
par l’évaluation de la λ-forme

(lambda (x) ((lambda (a f) (f a x)) 8 +))

Sa valeur est la fermeture
[(x);((lambda (a f) (f a x)) 8 +));E0];
cette valeur est liée à la variable add_8 dans l’environnement E0.

Considérons ensuite la session suivante :

(define f *) ==> ...

(define a 0) ==> ...

(add_7 100) ==> 107

(add_8 100) ==> 108

Les évaluations des deux formes define conduisent à lier dans le cadre C0 les variables f
et a aux valeurs [[*]] et 0, respectivement (cf. figure 9). La troisième forme à évaluer est
une combinaison. La valeur de l’opérateur dans E0 est la fermeture [(x);(f a x);E1],
la valeur de l’opérande est 100. L’application de la fermeture à la valeur 100 consiste à
créer un environnement E2, dont l’environnement d’accès est E1 (et l’environnement de
contrôle est E0) où x est lié à 100 et dans lequel (f a x) est évalué; les valeurs de f et a
sont trouvées dans l’environnement d’accès E1 et celle de x dans l’environnement local E2.
La valeur de f, c’est-à-dire [[+]], est appliquée aux valeurs de a et x, c’est-à-dire 7 et 100,
ce qui donne la valeur finale 107 (cf. figure 9). On observe que les liaisons globales relatives
à f et a n’ont pas eu d’effet sur cette valeur finale.

La quatrième forme à évaluer est une combinaison. La valeur dans E0 de l’opérateur
est la fermeture [(x);((lambda (a f) (f a x)) 8 +));E0]; la valeur de l’opérande
est 100. L’application de la fermeture à la valeur 100 consiste à créer un environnement E3,
dont l’environnement de contrôle et d’accès est E0; la variable x est liée à 100 dans C3 et
le corps ((lambda (a f) (f a x)) 8 +) est évalué dans E3. Il s’agit d’une combinaison
dont l’opérateur est une λ-forme dont la valeur est la fermeture [(a f);(f a x);E3]; les
valeurs des opérandes sont 8 et [[+]]. L’application de la fermeture aux deux arguments
consiste en la création d’un environnement E4 où a et f sont liés à 8 et [[+]], respectivement,
et dans lequel le corps (f a x) est évalué. Les valeurs de f et a sont trouvées localement,

la valeur de x est trouvée dans l’environnement d’accès E3. L’application de [[+]] à 8 et 100
donne la valeur finale 108.

C0

add 7 : [..E1]

add 8 : [..E0]

a : 0

f : ...

* : ...

+ : ...

C1

a : 7

f : [[+]]
C2

x : 100

C3

x : 100

C4

a : 8

f : [[+]]

E1

-

-

- -

E0

E3
E4

E2

Figure 9: Evaluation de (add 7 100) et (add 8 100)

Ces exemples mettent en évidence le concept de variable libre; le lecteur est invité à
justifier de la même manière la session suivante :

((lambda (a f) (add_8 100)) 3 -) ==> 108

((lambda (a f) (add_7 100)) 4 -) ==> 107

3.3.4 Inférence statique et lexicale des liaisons

La règle de portée statique de Scheme est dite lexicale parce qu’il est possible d’inférer
les liaisons sur base du seul texte de la forme à évaluer et des définitions concernées par
cette évaluation; nous précisons ici comment cette inférence se fait.

Les paramètres d’une λ-forme sont liés à des valeurs lorsque la fermeture, valeur de
la λ-forme, est appliquée. Seules les occurrences des paramètres dans le corps de la λ-
forme sont concernées par ces liaisons; les occurrences extérieures sont des occurrences de
variables distinctes, même si elles portent le même nom. Le parenthésage rigoureux des
expressions en Scheme permet en effet d’utiliser le même nom de variable dans des zones
distinctes, sans que l’expression devienne ambiguë. Notons aussi que le corps d’une λ-
forme peut contenir d’autres λ-formes, dont les paramètres ne portent pas nécessairement
des noms distincts. C’est la λ-forme la plus interne qui prime, comme le montraient déjà
les exemples donnés plus haut. La règle de portée des variables détermine la portion du
programme dans laquelle une liaison variable-valeur est valide. Nous illustrons cette règle
au moyen d’un exemple.
Une expression, à évaluer dans l’environnement global, est représentée à la figure 10.
Cette expression comporte plusieurs occurrences de x. Quatre liaisons relatives à x sont
susceptibles d’entrer en ligne de compte. Il y a d’abord les trois liaisons dues aux trois

Figure 10: Portée

λ-formes que comporte l’expression. En ce qui concerne une quatrième liaison pertinente
éventuelle, on distingue deux cas.

1. Une liaison existe pour x dans l’environnement global, suite à l’évaluation d’une
forme (define x ...). Cette liaison est aussi à considérer.

2. Dans le cas contraire, certaines occurrences de x dans l’expression pourraient n’avoir
pas de valeur.

L’expression de la figure 10 est une combinaison à deux éléments; le premier est la λ-forme
(lambda (x) (* 2 x)), que nous notons Λ1. Le second élément de la combinaison est lui-
même une combinaison à deux éléments; le second élément est x tandis que le premier est
la λ-forme (lambda (x) (- 9 ((lambda (x) (+ 3 x)) x))), que nous notons Λ2. Le
corps de Λ2 est une combinaison arithmétique dont le dernier élément est une combinaison
dont le foncteur est la λ-forme (lambda (x) (+ 3 x)), que nous notons Λ3. Nous pouvons
maintenant décrire les trois λ-formes.

• L’unique paramètre de Λ1 est identifié par l’exposant 1 (cf. figure 10); l’occurrence
repérée par l’exposant 2 se trouve dans le corps de Λ1 et aussi dans sa portée.

• L’unique paramètre de Λ2 est identifié par l’exposant 3; les occurrences 5 et 6 se
trouvent dans le corps de Λ2 mais seule l’occurrence 6 se trouve dans la portée de
Λ2.

• L’unique paramètre de Λ3 est identifié par l’exposant 4; l’occurrence 5 se trouve dans
le corps de Λ3 et aussi dans sa portée.

L’occurrence 7 ne se trouve dans le corps d’aucune λ-forme; c’est une variable globale (ou
une occurrence globale de l’identificateur x).

Une occurrence d’une variable peut se trouver dans le corps de plusieurs λ-formes
imbriquées mais elle se trouve dans la portée d’une seule, la λ-forme la plus interne qui
contient cette occurrence. On a vu qu’en cas d’application d’une λ-forme à des arguments,
les paramètres étaient liés à ces arguments, puis le corps de la λ-forme était évalué. Ce
corps contient souvent une ou plusieurs occurrences des paramètres, mais seules celles qui
se trouvent dans la portée de la λ-forme sont concernées par ces liaisons.

3.3.5 Deux exercices

Pour s’assurer de sa bonne compréhension du modèle des environnements, le lecteur est
invité à évaluer la forme

((lambda (x) (* x x))

((lambda (a b) (a (* b b)))

(lambda (a) (* a a))

(* 2 2)))

selon ce modèle (la valeur cherchée est 65 536).

De même, le lecteur pourra déterminer que la valeur de la forme

((lambda (a b c) (a b c))

(lambda (x y) (* 2 x y))

((lambda (x y) (+ x y)) (+ 1 2) (+ 3 4))

((lambda (x) (+ 3 x)) 0))

est 60.

3.3.6 Occurrences libres, occurrences liées

Les subtilités inhérentes aux notions de variable et de portée de variable paraissent moins
rebutantes si on observe qu’elles ne sont pas spécifiques à la programmation, comme le
montrent les exemples suivants. Considérons l’expression mathématique E définie par

E =def 1 −
∫ x

0
sin(u) du .

Par analogie avec Scheme, nous pouvons dire que 0 et 1 sont des constantes numériques,
tandis que “sin” est une variable globale, préliée à la fonction trigonométrique sinus.52 Le
symbole — bizarre — “

∫

. . . d . . .”, tel qu’il apparâıt dans la sous-expression “
∫ x

0 sin(u) du”,
est aussi une variable globale, prélié à l’opération d’intégration, c’est-à-dire à une fonction à
trois arguments : les deux limites d’intégration et la fonction à intégrer. Si on rationalisait
l’écriture mathématique, on écrirait

∫

(a, b, f)

plutôt que
∫ b

a

f(u) du .

Revenons à l’expression mathématique E définie plus haut. Par analogie avec la
programmation, il est naturel de considérer que la valeur de E n’existe que dans un

52Le mathématicien dira plutôt que “sin” est une constante fonctionnelle, parce qu’il exclut d’emblée
la possibilité théorique de renommer la variable; une variable qui fait l’objet d’une liaison permanente et
intangible est assimilable à une constante.

environnement où x a une valeur, le fait que u ait une valeur ou non étant non pertinent.
L’usage mathématique permet naturellement cela mais il est plus laxiste que Scheme : si
x n’a pas de valeur, on considère que l’expression E est une abréviation de l’expression
λx.E; sa valeur est donc une fonction de R dans R. En fait, on ne peut jamais confondre
une λ-forme (lambda (x) e) et le corps e de cette forme.

En ce qui concerne les possibilités de renommage de variables, il y a une forte
analogie entre le langage Scheme et la logique classique. La valeur de vérité de la
formule ∀xP (x, y) dépend de la valeur de la “variable libre” y (et de l’interprétation
de la constante prédicative P ) mais pas de la valeur de la “variable liée” x qui peut
être renommée : ∀xP (x, y) et ∀z P (z, y) sont des formules logiquement équivalentes.53 Le
même phénomène s’observe en Scheme. Pour évaluer l’expression ((lambda (x) (+ x

y)) 5) dans l’environnement global E0, il sera nécessaire que y ait une valeur dans cet
environnement courant; par contre, la valeur éventuelle de x dans E0 est sans importance.
On peut d’ailleurs remplacer les deux occurrences de x par z, par exemple.54

Remarque. Même si y n’est pas lié dans l’environnement global, l’évaluation de

(define f (lambda (x) (+ x y)))

ne provoque pas d’erreur; par contre, l’évaluation subséquente de (f 5), par exemple,
provoquera une erreur.

Remarque. On observe aussi que la valeur de

((lambda (u) (+ u 5)) (* x 3))

dans l’environnement global dépend de celle de x dans cet environnement mais que la
valeur de

(lambda (x) ((lambda (u) (+ u 5)) (* x 3)))

n’en dépend pas.

Soit une λ-forme du type (lambda (...x...) E). Toute occurrence de x dans le corps
E est dite liée. Par extension, une occurrence de x est liée dans une expression α si cette
expression comporte une sous-expression (forme lambda) dans laquelle l’occurrence en
question est liée.

Exemple. Dans l’expression

((lambda (x) (- 9 ((lambda (x) (+ 3 x)) x))) x)

^ ^

53Par contre, on ne peut pas récrire ∀x P (x, y) en ∀y P (y, y); il y aurait capture de la variable en second
argument y, libre dans la première expression et liée dans la seconde.

54Sur le plan du renommage des variables, Scheme, vu sa politique stricte de parenthésage et de notation
préfixée, permet certaines écritures que l’usage mathématique interdirait. Par exemple, on peut écrire
((lambda (x) (+ x 5)) x) au lieu de ((lambda (u) (+ u 5)) x) alors qu’il n’est pas permis d’écrire
R x

0
sin(x) dx au lieu de

R x

0
sin(u) du.

les deux premières occurrences pointées de x sont liées, la dernière occurrence de x est non
liée. Dans l’expression

(* 9 ((lambda (x) (+ 3 x)) x) x)

^

l’occurrence pointée de x est liée, les deux occurrences suivantes de x sont non liées. Les
variables liées sont habituelles en logique, et aussi en mathématique (indice de sommation,
variable d’intégration).

3.4 Procédures eval et apply

Le processus d’évaluation d’une expression Scheme est récursif. Il y a des cas de base
(constantes, variables, formes quote) et des cas inductifs. Pour ces derniers, l’évaluation
de l’expression passe par l’évaluation de sous-expressions et aussi, parfois, par le processus
d’application d’une valeur fonctionnelle à ses arguments. Le processus d’application lui-
même peut requérir de nouvelles évaluations. C’est le cas si la fonction à appliquer est
“λ-définie”; il y a lieu alors d’évaluer le corps de la λ-forme dans un nouvel environnement,
enrichi des liaisons entre paramètres formels et paramètres réels.

On peut donc décrire précisément le système autour de deux procédures mutuellement
récursives eval et apply. La première prend comme arguments une expression à évaluer
et un environnement. La seconde prend comme arguments une valeur fonctionnelle et une
liste d’arguments évalués appropriés (en nombre et en type). En fait, on peut construire
le système sur cette base.

On peut aussi, pour s’assurer que l’on comprend bien les fonctionnalités du système,
le reconstruire sur base de fonctions eval et apply, qui peuvent être programmées dans
un langage quelconque . . . y compris Scheme lui-même. Ce genre d’exercice permet aussi
de créer des variantes du vrai système Scheme et de les expérimenter. Cette démarche
requiert que l’on représente les expressions Scheme à évaluer par des structures de données
du langage dans lequel on a programmé eval et apply. Si on utilise pour cela le langage
Scheme lui-même, on utilisera naturellement la représentation habituelle des expressions
en Scheme; on pourra aussi représenter les environnements par des listes de liaisons,
chaque liaison étant une liste (ou une paire) variable-valeur. Par exemple, on devra avoir

(eval ’(+ 3 4) env) ==> 7

et aussi

(eval (list ’+ 3 4) env) ==> 7

puisque l’on a

(+ 3 4) ==> 7

Construire en Scheme ce couple de fonctions eval et apply (et, bien sûr, une série
de fonctions auxiliaires) permet d’obtenir un “méta-évaluateur” ou “évaluateur méta-
circulaire”. C’est une tâche bien moins lourde que la construction d’un évaluateur ordinaire

(écrit par exemple en langage C ou en langage d’assemblage), mais qui sort néanmoins du
cadre de ce texte introductif. Il faut cependant noter que les systèmes Scheme fournissent
une procédure apply.55 La procédure apply fournie par le système est telle que, si [[proc]]
est une procédure à n arguments, et si [[x1]],. . . ,[[xn]] sont des arguments appropriés, alors

[[(apply proc (list x1 ... xn))]] = [[(proc x1 ... xn)]] .

Cette procédure peut être utile même en programmation élémentaire. Supposons par
exemple que l’on veuille construire une fonction qui associe à toute liste de nombres la
somme de ses éléments. On peut écrire

(define +_list (lambda (l) (apply + l)))

On notera la différence entre “+” et “+_list” :

(+ 1 2 3 4) ==> 10

(+_list ’(1 2 3 4)) ==> 10

(+ ’(1 2 3 4)) ==> Error

Cela se généralise à tout opérateur admettant un nombre quelconque d’arguments; si nous
définissons

(define op_list (lambda (op l) (apply op l)))

nous pouvons par exemple calculer la somme et le produit d’une liste de nombres :

(+ 1 2 3 4) ==> 10

(op_list + ’(1 2 3 4)) ==> 10

(* 1 2 3 4) ==> 24

(op_list * ’(1 2 3 4)) ==> 24

On peut aussi procéder autrement et définir l’opérateur fonctionnel gen_op_list qui
transforme une fonction à nombre quelconque d’arguments x1, . . . , xn en la fonction qui
prend comme seul argument la liste correspondante (x1, . . . , xn) :

(define gen_op_list

(lambda (op)

(lambda (l) (apply op l))))

La transformation inverse se programme facilement, en utilisant la forme lambda

généralisée :

(define inv_gen_op_list

(lambda (op_list)

(lambda v (op_list v))))

55Les systèmes Scheme fournissent une procédure eval, plus ou moins générale, mais nous ne décrirons
pas cela ici.

On a alors

(op_list + ’(1 2 3 4)) ==> 10

(op_list * ’(1 2 3 4)) ==> 24

((gen_op_list +) ’(1 2 3 4)) ==> 10

((gen_op_list *) ’(1 2 3 4)) ==> 24

((inv_gen_op_list +_list) 1 2 3 4) ==> 10

((inv_gen_op_list (gen_op_list *)) 1 2 3 4) ==> 24

((inv_gen_op_list (gen_op_list proj_1)) 1 2 3 4) ==> 1

3.5 Autres formes conditionnelles

3.5.1 La forme spéciale cond

La forme spéciale cond généralise la forme spéciale if. La syntaxe habituelle de la
forme cond est (cond (c1 e1) ... (cn en)) . Les (ci ei) sont des clauses. Le processus
d’évaluation est le suivant :

• Les conditions ci sont évaluées en séquence jusqu’à ce que une valeur [[ck]] distincte
de #f soit produite;

• L’expression correspondante ek est évaluée; la valeur produite [[ek]] est la valeur de
la forme spéciale.

La valeur de la forme cond est non définie si toutes les valeurs [[ci]] s’identifient à #f. Une
pratique fréquente pour éviter systématiquement ce type de problème consiste à choisir
comme dernière condition cn une expression dont la valeur est toujours #t. Dans ce
contexte, le symbole spécial else peut être utilisé. La forme (cond (e0 e1) (else e2))

est donc équivalente à la forme (if e0 e1 e2).

Voici quelques exemples :

(cond ((> 0 1) (/ 2 0))

((> 1 2) (/ 0 1))

((> 2 0) (/ 1 2))) ==> 1/2

(cond ((> 3 5) hi) (else 4)) ==> 4

(cond (’hi 3) (else 4)) ==> 3

(cond ((> 0 1) 5)) ==> ...

(cond) ==> ...

3.5.2 La fonction not

La valeur de la forme (not e) est #t si la valeur de e est #f; elle est #f sinon. On
peut voir (not e) comme l’abréviation de (if e #f #t). A titre d’exemple, signalons
l’existence de deux reconnaisseurs très utiles, pair? et atom?. Le premier reconnâıt tout
objet construit au moyen de cons, notamment les listes non vides (et d’autres objets plus
généraux introduits plus loin); le second reconnâıt les objets qui ne sont pas construits au
moyen de cons, notamment les nombres, les symboles et la liste vide. Certains systèmes
prédéfinissent les deux reconnaisseurs, d’autres imposent à l’utilisateur de dériver le second
du premier, en évaluant le code

(define atom?

(lambda (x) (not (pair? x))))

Remarque. Avec cette définition, [[(atom? x)]] = [[#t]] signifie que les fonctions [[car]] et
[[cdr]] ne peuvent s’appliquer à [[x]]. C’est le cas des objets “atomiques” au sens usuel du
terme, mais aussi des fonctions, et d’autres objets composés comme les vecteurs introduits
plus loin.

3.5.3 Les formes spéciales and et or

La valeur de la forme spéciale (and e1 ...en) s’obtient comme suit. Les ei sont évalués
dans l’ordre et la première valeur fausse est retournée; les valeurs suivantes ne sont pas
calculées. S’il n’y a pas de valeur fausse, la dernière valeur calculée est retournée. La
valeur de (and) est #t.

La valeur de la forme spéciale (or e1 ...en) s’obtient comme suit. Les ei sont évalués
dans l’ordre et la première valeur non fausse est retournée; les valeurs suivantes ne sont
pas calculées. Si toutes les valeurs sont fausses, #f est retourné; la valeur de (or) est #f.

Les exemples qui suivent illustrent l’usage des formes and et or

(and (= 2 2) (< 2 1) (/ 2 0) (+ 2 3)) ==> #f

(and (= 2 2) (+ 2 3) (/ 2 0) (< 2 1)) ==> Error - Div by zero

(and (= 2 2) (< 2 1) (/ 2 0) (+ 2 3)) ==> #f

(and (= 2 2) (+ 2 3)) ==> 5

(and (+ 2 3) (= 2 2)) ==> #t

(if (and (= 2 2) (+ 2 3)) ’hi ’ha) ==> hi

(or (< 2 1) (= 2 2) (/ 2 0) (+ 2 3)) ==> #t

(or (< 2 1) (+ 2 3) (/ 2 0) (= 2 2)) ==> 5

(or (< 2 1) (/ 2 0) (+ 2 3) (= 2 2)) ==> Error - Div by zero

(or (= 2 2) (+ 2 3) (< 2 1)) ==> #t

(or #f (< 2 1) #f) ==> #f

(if (or (+ 2 3) 1 2) ’hi ’ha) ==> hi

3.5.4 Relations entre if et cond

La forme if est un cas particulier de la forme cond; plus précisément, les formes (cond

(e0 e1) (else e2)) et (if e0 e1 e2) ont même valeur.
On peut aussi, dans certaines conditions, ramener une forme cond à une série de

formes if embôıtées; il suffit d’utiliser l’équivalence mentionnée ci-dessus pour démontrer
par récurrence sur n que la forme

(cond (c0 e0) (c1 e1) ... (cn−1 en−1) (else en))

a même valeur que la forme
(if c0 e0 (if c1 e1 (if ...(if cn−1 en−1 en)...))).

4 Procédures récursives

4.1 Préliminaires

La notion de procédure est cruciale en programmation parce qu’elle permet de résoudre
élégamment de gros problèmes en les ramenant à des problèmes plus simples. De même,
la notion de fonction est cruciale en mathématique parce qu’elle permet de construire et
de nommer des objets complexes en combinant des objets plus simples. La définition

hypo =def λx, y :
√

x2 + y2

ainsi que le programme

(define hypo

(lambda (x y)

(sqrt (+ (square x) (square y)))))

sont des exemples classiques. La définition (opérationnelle) de la longueur de l’hypoténuse
suppose les définitions préalables de l’addition, de l’élévation au carré et de l’extraction
de la racine carrée.

Remarque. L’ordre dans lequel on définit les procédures n’a pas d’importance. On peut
définir hypo avant de définir square, quoique le corps de la λ-forme liée à hypo contienne
des occurrences de square. C’est seulement lors de l’application de hypo qu’une valeur
sera cherchée pour la variable square, dans l’environnement de définition de hypo.

L’idée de la récursivité est d’utiliser, dans la définition d’un objet relativement
complexe, non seulement des objets antérieurement définis, mais aussi l’objet à définir
lui-même. Le risque de “cercle vicieux” existe, mais n’est pas inévitable. Nous avons déjà
introduit la récursivité dans le chapitre introductif. Avant d’y revenir plus concrètement
ici, nous allons montrer que l’idée de récursivité, loin d’être “moderne”, n’est qu’un cas
particulier de la notion classique d’équation.

4.2 Récursivité et équations

On peut faire une analogie avec les égalités et les équations. Dans un contexte où f , a et
b sont définis, on peut définir x par l’égalité

x = f(a, b) .

Par contre, l’égalité
x = f(a, x)

ne définit pas nécessairement un et un seul objet x; même si c’est le cas, le procédé de
calcul de x peut ne pas exister.

Un procédé de calcul parfois utilisé pour résoudre l’équation x = f(a, x) consiste à
construire un certain nombre de termes de la suite

x0, x1 = f(a, x0), x2 = f(a, x1), . . . , xn+1 = f(a, xn), . . .

Dans certains cas, cette suite converge vers une limite x et, par passage à la limite, on
déduit x = f(a, x) de xn+1 = f(a, xn). Cette approche est simple dans son principe mais
parfois délicate dans son application : le choix de x0 n’est pas évident et la convergence
de la suite n’est en général pas garantie. Ce procédé est néanmoins souvent utile. Il
fournit rapidement, par exemple, une bonne approximation de la solution de l’équation
x = cos x.56

Remarque. Quand on considère l’équation x = f(a, x), on ne prétend généralement
pas “définir” l’objet x mais, plus modestement, un ensemble d’objets, peut-être vide,
composé des solutions de l’équation. Pour qu’il y ait définition, on doit avoir démontré
que l’ensemble comporte un et un seul élément, c’est-à-dire l’existence et l’unicité de la
solution.

Le procédé d’approximation peut aussi être utilisé pour calculer des fonctions plutôt
que des nombres, et donc pour résoudre des équations fonctionnelles du type

f = Φ(g, f) ,

où la fonctionnelle Φ est une fonction qui, à toutes fonctions g et f de types appropriés
associe une fonction de même type que f .

L’équation différentielle y′ = f(x, y) (avec y(x0) = y0) peut s’écrire

y(x) = y0 +

∫ x

x0

f(t, y(t)) dt ,

ce qui permet souvent une résolution approchée. Un exemple classique est l’équation
y′ = y, avec la condition initiale y(0) = 1. Les approximations successives sont définies
par la suite

y0 =def λx . 1 ,

yn+1 =def λx .

(

1 +

∫ x

0
yn(t) dt

)

.

Ce sont les développements finis de MacLaurin; on a successivement :

y0(x) = 1 ,

y1(x) = 1 +

∫ x

0
1 dt = 1 + x ,

y2(x) = 1 +

∫ x

0
(1 + t) dt = 1 + x + x2/2

et, plus généralement,

yn(x) =
n

∑

i=0

xi/i! .

56Une calculatrice de poche, en mode “radians”, fournit la séquence suivante :
0, 1, 0.5403, 0.8576, 0.6543, 0.7935, 0.7014, 0.7640, 0.7221, 0.7504, 0.7314, 0.7442, 0.7356, . . . ;
pour amortir les oscillations, on pose xn+1 = (xn + cos xn)/2, ce qui donne :
0, 0.5, 0.6888, 0.7304, 0.7377, 0.7389, 0.7390, . . . , 0.73908513321516 . . .

En programmation, on doit associer à toute construction un procédé de calcul clair et
précis; on souhaite naturellement qu’il soit aussi raisonnablement efficace.57 Le mécanisme
de définition récursive de fonction respecte en général ces conditions. Une instance du
schéma f = Φ(g, f) que nous avons déjà rencontrée est

fact = λn . if n = 0 then 1 else n ∗ fact(n − 1) .

On définit fact en termes d’autres fonctions déjà définies (addition, comparaison,
soustraction, multiplication), et de la fonction fact elle-même. Il est intéressant de voir que
cette écriture donne lieu à une suite d’approximations. Le point de départ est la fonction
vide fact0, c’est-à-dire la fonction qui ne comporte aucun couple.58 On a successivement :

fact0 = λn .⊥ ,
fact1 = λn . if n = 0 then 1 else ⊥ ,
fact2 = λn . if n = 0 then 1 else if n − 1 = 0 then 1 else ⊥ ,
. . .

On démontre facilement par récurrence que, pour tout naturel m on a

factm = λn . if 0 ≤ n < m then n! else ⊥ .

On observe qu’il s’agit bien d’une suite d’approximations, chaque élément étant meilleur
que le précédent. Ce type d’approximation est très particulier. Un élément de cette
suite ne fournit que des réponses exactes, c’est-à-dire des factorielles, mais seulement pour
certains éléments du domaine; améliorer une approximation signifie ici étendre le domaine
de cette approximation. On remarque aussi que l’extension du domaine se fait de proche
en proche. Dans le cas présent, si on peut calculer fact(x), on sait que l’on pourra calculer
fact(x + 1) au moyen de l’approximation suivante.

En pratique, nous nous limiterons à ce type bien particulier de schéma fonctionnel
récursif : l’évaluation de f(x) nécessitera la détermination préalable d’un ensemble de
valeurs f(y1), . . . , f(yn) où les y1, . . . , yn sont, en un certain sens, “plus simples” que x.

Si le domaine de la fonction f à définir est tel que tout élément n’admet qu’un ensemble
fini d’éléments “plus simples”, on conçoit que le risque de “tourner en rond” pourra être
évité.

4.3 Quelques exemples

4.3.1 Récursion sur les nombres

Nous donnons d’abord les programmes correspondant aux fonctions définies récursivement
dans le chapitre introductif.

57En mathématique, toute définition de fonction doit être claire et précise, mais le fait que la fonction f
soit correctement définie sur le domaine Z ne signifie pas que l’on soit capable de calculer f(r) pour tout
entier r, ni même pour aucun d’entre eux.

58On dit souvent “la fonction définie nulle part”. Si l’ensemble de référence est N, on écrira fact0(n) = ⊥,
ce qui signifie que la valeur de fact0(n) n’est pas définie. On a, pour toute fonction g, g(. . . ,⊥, . . .) = ⊥ :
une combinaison dont un argument est non défini est elle-même non définie.

La fonction factorielle est définie sur le domaine N par l’égalité

n! = [ if n = 0 then 1 else n ∗ (n − 1)! ] .

L’exploitation de cette définition est évidente; on a, par exemple,

3! = 3 ∗ 2! = 3 ∗ 2 ∗ 1! = 3 ∗ 2 ∗ 1 ∗ 0! = 3 ∗ 2 ∗ 1 ∗ 1 = 6 .

Le cas de la suite fib des nombres de Fibonacci est analogue et déjà connu (cf. § 1.2.1).
Voici un nouvel exemple d’exploitation de la définition :

fib(0) = 0 ,
fib(1) = 1 ,
fib(2) = 1 + 0 = 1 ,
fib(3) = 1 + 1 = 2 ,
fib(4) = 2 + 1 = 3 ,
fib(5) = 3 + 2 = 5 ,
fib(6) = 5 + 3 = 8 ,
. . .

Dans les deux cas, “plus simple” s’identifie à “plus petit”. Sauf dans les cas de base, le
calcul de n! repose sur celui de (n−1)! et le calcul de fib(n) sur ceux de fib(n − 1) et
fib(n − 2).

Ceci se traduit aisément en Scheme :

(define fact

(lambda (n)

(if (zero? n) 1 (* n (fact (- n 1))))))

(define fib

(lambda (n)

(if (< n 2) n (+ (fib (- n 1)) (fib (- n 2))))))

Ces deux programmes permettent respectivement le calcul de la factorielle d’un entier
naturel n et le calcul du nième nombre de Fibonacci. Ils ne sont pas destinés à être
utilisés avec un argument non naturel et donc rien n’est spécifié pour ce cas, dans lequel
le résultat peut même ne pas exister.59

Nous avons introduit au paragraphe 1.2.1 les trois fonctions mathématiquement égales
cb1 et cb2, permettant le calcul des coefficients binomiaux; les programmes Scheme

correspondants s’obtiennent immédiatement. On a par exemple

59Si on appelle le programme fact avec un argument entier négatif, l’exécution ne se termine pas : on
dit qu’il y a régression infinie. Concrètement, l’exécution crée une suite du genre (−4)! = (−4) ∗ (−5)! =
(−4) ∗ (−5) ∗ (−6)! = · · · , ce qui n’a évidemment ni sens ni intérêt. Si on appelle le programme fib avec
un argument entier négatif −n, le résultat est −n, ce qui n’a pas d’intérêt non plus.

(define cb1

(lambda (n p)

(if (or (= n p) (= p 0))

1

(+ (cb1 (- n 1) (- p 1))

(cb1 (- n 1) p)))))

4.3.2 Récursion sur les listes

La récursion s’applique très souvent aux listes. Ecrivons par exemple un programme qui
ajoute un nombre [[n]] à tous les éléments d’une liste [[l]] de nombres et renvoie la liste
des résultats obtenus :

(define add_to_all

(lambda (n l)

(if (null? l)

’()

(cons (+ n (car l))

(add_to_all n (cdr l))))))

(add_to_all 2 ’(5 3 8)) ==> (7 5 10)

Une variante intéressante permet de traiter des listes comportant des sous-listes de niveau
quelconque. Dans ce cas, la récursion opère non seulement sur le reste d’une liste non
vide, mais aussi sur la tête, si celle-ci n’est pas un nombre et est donc une liste. On peut
utiliser le code suivant :

(define add_to_all*

(lambda (n l)

(cond ((null? l) ’())

((number? (car l))

(cons (+ n (car l))

(add_to_all* n (cdr l))))

(else (cons (add_to_all* n (car l))

(add_to_all* n (cdr l)))))))

(add_to_all* 2 ’(5 3 12)) ==> (7 5 14)

(add_to_all* 2 ’((5 (() 3)) ((12))))

==> ((7 (() 5)) ((14)))

La valeur de (member x l) est le premier suffixe de [[l]] qui commence par le
symbole [[x]], et #f si ce symbole n’appartient pas à la liste :

(define member

(lambda (x l)

(if (null? l)

’#f

(if (eq? x (car l))

l

(member x (cdr l))))))

On a donc :

(member ’x ’(a x b x d)) ==> (x b x d)

(member ’x ’(a x)) ==> (x)

(member ’y ’(a x)) ==> #f

Remarque. La définition est en pratique inutile, car member est prédéfinie en Scheme.

Une table est une collection de paires dont le premier composant est un symbole et
le second une entité se rapportant à ce symbole (une description, une liste de propriétés,
etc.). On peut représenter une table par une liste de paires ou de listes dont la tête est un
symbole et le reste est l’entité qui s’y rapporte. Dans la (représentation de) table

(define *table*

’((Dubois Pierre 1942) (Dupont Jean 1964) (Durand Paul 1980)))

les symboles sont des noms de personnes et les entités associées sont des listes comportant
le prénom et l’année de naissance de ces personnes. On définit une fonction [[assq]]
permettant de retrouver ce qui concerne un symbole donné dans une table :

(define assq

(lambda (symb table)

(if (null? table)

#f

(if (eq? symb (caar table))

(car table)

(assq symb (cdr table))))))

Notons l’emploi de caar, introduit au paragraphe 2.5. On a :

(assq ’Dupont *table*) ==> (Dupont jean 1964)

(assq ’Martin *table*) ==> #f

Remarque. La fonction assq est prédéfinie en Scheme, de même que ses variantes assv et
assoc, dans lesquelles le prédicat eq? est remplacé par eqv? et equal?, respectivement.

Considérons encore le cas du produit scalaire de deux vecteurs représentés par deux
listes de nombres de même longueur :

(define dot_product

(lambda (u v)

(if (null? u)

0

(+ (* (car u) (car v))

(dot_product (cdr u) (cdr v))))))

On a

(dot_product ’(1 2 3) ’(4 5 6)) ==> 32

Remarque. Le programme ne se comporte pas correctement si les deux listes sont de
longueurs différentes :

(dot_product ’(1 2 3) ’(4 5)) ==> Error

(dot_product ’(1 2 3) ’(4 5 6 7)) ==> 32

L’usage de la procédure prédéfinie error permet d’attirer l’attention de l’utilisateur sur
le problème :

(define dot_product

(lambda (u v)

(if (null? u)

(if (null? v)

0

(error "second vector too long" v))

(if (null? v)

(error "first vector too long" u)

(+ (* (car u) (car v))

(dot_product (cdr u) (cdr v)))))))

On a alors :

(dot_product ’(1 2 3) ’(4 5))

==> Error - first vector too long (3)

(dot_product ’(1 2 3) ’(4 5 6 7))

==> Error - second vector too long (7)

4.3.3 Schémas de récursion

On peut observer que, souvent, la définition récursive d’une fonction f dont un argument n
est un entier naturel consiste d’une part à donner directement la valeur f(0) et, d’autre
part, à exprimer f(n) en fonction de f(n− 1) (et de n et des autres arguments éventuels)
si n > 0. De même, très fréquemment, la définition récursive d’une fonction g dont un
argument ℓ est une liste consiste d’une part à donner directement la valeur g(ℓ) si ℓ est
vide et, d’autre part, à l’exprimer en fonction de g(ℓ′) (et de ℓ et des autres arguments
éventuels) si ℓ n’est pas vide et si ℓ′ est le reste de ℓ.

Ce mode de définition récursive permet de résoudre facilement la plupart des problèmes
usuels, même si des solutions plus complexes sont parfois nécessaires. Nous verrons au
chapitre suivant qu’il ne s’agit pas seulement d’un schéma de pensée, mais d’un cadre de
programmation très commode.

4.4 Le double rôle de define

Comme on l’a vu au paragraphe 3.1, évaluer (define symb exp) consiste à lier à la
variable symb la valeur de exp; le rôle de la forme define est donc de donner un nom
(symb) à une valeur (celle de exp). Dans le cas d’une définition récursive, rien ne change
techniquement; la valeur de exp est une fermeture, liée à symb. Du point de vue de
l’utilisateur cependant, le rôle de la forme define est alors double; il consiste non seulement
à nommer une fonction mais aussi à la définir. Avant l’évaluation de la forme spéciale
(définition non récursive)

(define square

(lambda (n) (* n n)))

il est déjà possible d’évaluer une combinaison du type

((lambda (n) (* n n)) 5)

Par contre, avant l’évaluation de la forme spéciale (définition récursive)

(define fact

(lambda (n) (if (zero? n) 1 (* n (fact (- n 1))))))

l’évaluation de la combinaison

((lambda (n) (if (zero? n) 1 (* n (fact (- n 1))))) 5)

provoque une erreur.

4.5 Le processus de calcul récursif

D’après les règles habituelles, on voit que le processus de calcul associé à la définition
récursive de la factorielle est relativement “encombrant”. On a par exemple, d’après le
modèle de substitution

(fact 3)(if (zero? 3) 1 (* 3 (fat (- 3 1))))(* 3 (fat (- 3 1)))(* 3 (fat 2))...(* 3 (* 2 (fat 1)))...(* 3 (* 2 (* 1 (fat 0))))

6

On conçoit que ce type de calcul requiert un espace-mémoire dont la taille dépend (ici,
linéairement) de la valeur de l’argument.

On note aussi que l’introduction de la récursivité n’exige pas de mécanisme particulier
pour appliquer une fonction à des arguments : les règles déjà introduites suffisent. En
particulier, l’évaluation de (fact 3) dans l’environnement global E0 suscite la structure
d’environnements de la figure 11; le tableau explicite les expressions évaluées dans chaque
environnement et les valeurs produites. La fermeture liée à fact dans l’environnement E0

indique que E0 est l’environnement de définition de [[fact]].

C0

* : ...

+ : ...

fact : ...

C1

n : 3

C2

n : 2

C3

n : 1

C4

n : 0
- - - -

E0

E1E2E3E4

E0 (fact 3) 6

E1 (* n (fact (- n 1))) 6

E2 (* n (fact (- n 1))) 2

E3 (* n (fact (- n 1))) 1

E4 1 1

Figure 11: Environnements, application de la fonction fact

Il convient d’être attentif au processus de calcul. Celui-ci peut être anormalement long,
en temps ou en espace;60 il peut même être infini, comme le montrent les trois exemples
(näıfs) qui suivent :

(define f (lambda (x) (f x)))

(define g (lambda (x) (g (+ x 1))))

(define h (lambda (x) (+ (h x) 1)))

Sur le plan syntaxique, ces définitions sont parfaitement correctes; il est pourtant clair
que l’évaluation des formes (f 0), (g 1) et (h 2) ne donnera rien . . . sauf un gaspillage
de ressources ! Le même phénomène de non-terminaison s’était produit pour (fact -4);
la procédure fact est intéressante, mais l’argument -4 est inapproprié. L’utilisateur peut

60Un processus consommant beaucoup d’espace mémoire consommera aussi, inévitablement, beaucoup
de temps, puisque chaque case mémoire devra être créée et/ou visitée; l’inverse n’est pas nécessairement
vrai.

éviter ce risque en vérifiant que tout appel pour des valeurs données donne lieu à un nombre
fini d’appels subséquents, pour des valeurs “plus simples”. Des schémas de programmes
récursifs existent, qui garantissent cette propriété de finitude, ou de terminaison. Un autre
exemple de processus infini est engendré par la définition

f(0) = 1 , f(x) = 2f(x/2) si x > 0 ,

dans le domaine des rationnels positifs. On a par exemple

f(1) = 2f(1/2) = 4f(1/4) = · · · = 1024f(1/1024) = · · ·

La non-terminaison d’un processus récursif est le risque le plus grave . . . mais pas
nécessairement le plus pernicieux. Nous avons déjà signalé que le calcul de (fib n) se
terminait toujours, quel que soit n ∈ N. Examinons néanmoins ce processus de plus près.
Pour n = 9, on a :

(fib 9)(+ (fib 8) (fib 7))(+ (+ (fib 7) (fib 6)) (fib 7))...
Nous avons déjà évoqué ce problème au paragraphe 1.2.6. On démontre facilement par
récurrence que le temps de calcul de (fib n) est proportionnel à la valeur calculée, ce
qui est inacceptable : on a en effet fib(n) ≃ 1.6n.61 Dans le cas présent, on peut améliorer
l’efficacité du calcul en utilisant un autre algorithme, moins näıf, ou en forçant l’évaluateur
à mémoriser et à réutiliser les résultats intermédiaires, plutôt qu’à les recalculer. La
première approche a été évoquée dans le chapitre introductif, où les bases mathématiques
d’algorithmes de calcul en des temps proportionnel à n (fonction fib it), puis à log n, ont
été présentés. Voici le programme Scheme correspondant au premier de ces algorithmes :

(define fib_a

(lambda (n a b)

(if (zero? n)

a

(fib_a (- n 1) b (+ a b)))))

On démontre facilement, par récurrence sur n, la propriété P (n) suivante.

Si i est un nombre naturel quelconque et
si n, a et b ont pour valeurs respectives n, fib(i) et fib(i+1),
alors (fib_a n a b) a pour valeur fib(n+i).
En particulier, (fib_a n 0 1) a pour valeur fib(n).

61On considère souvent qu’un algorithme est efficace quand le temps d’exécution est borné par une
fonction polynomiale en la taille des données. Ici la donnée est un nombre n, dont la taille est log n;
l’algorithme est donc doublement exponentiel en la taille des données. Notons aussi qu’approximer le temps
de calcul par le nombre d’opérations est un peu abusif puisque les opérations arithmétiques prennent plus
de temps pour des grands nombres que pour des petits.

Nous reviendrons plus loin sur ce type de récursivité terminale ou dégénérée, qui
se reconnâıt par une simple analyse de la syntaxe du programme. Lors de l’évaluation
d’un appel, il y a un seul appel récursif, et cet appel est la dernière action de
l’évaluation.62 Cette analyse permet à l’interprète d’exécuter le programme rapidement
et sans consommation inutile de mémoire. Le place mémoire requise par l’exécution du
processus est en effet constante. On a par exemple :
(fib a 9 0 1)(fib a 8 1 1)(fib a 7 1 2)(fib a 6 2 3)(fib a 5 3 5)(fib a 4 5 8)(fib a 3 8 13)(fib a 2 13 21)(fib a 1 21 34)(fib a 0 34 55)
34

Nous généraliserons plus loin la technique consistant à améliorer un processus de calcul
par l’adjonction d’accumulateurs, c’est-à-dire d’arguments supplémentaires (comme a et
b pour fib_a) permettant de mémoriser des résultats intermédiaires, et parfois de rendre
dégénérée la récursivité. Le temps d’exécution est ici proportionnel à n. Nous verrons
plus loin le programme permettant de rendre ce temps proportionnel à log n.

4.6 Récursivité croisée

Le caractère récursif d’une définition peut être indirect; l’expression définissant une
fonction f peut contenir des appels à une fonction g, elle-même définie en termes de
la fonction f . Un exemple classique très simple est le suivant :

(define even?

(lambda (n)

(if (zero? n) #t (odd? (- n 1)))))

(define odd?

(lambda (n)

(if (zero? n) #f (even? (- n 1)))))

Les prédicats even? (“pair?”) et odd? (“impair?”) sont définis l’un en fonction de l’autre.
Voici un exemple du processus de calcul correspondant :

62Dans le cas de la factorielle, il y a bien unicité de l’appel récursif, mais, lors de l’évaluation de la
forme (* n (fact (- n 1))), cet appel était l’avant-dernière action, la dernière étant la multiplication
de (fact (- n 1)) par n. La récursivité n’était donc pas terminale.

(odd? 4)(even? 3)(odd? 2)(even? 1)(odd? 0)
#f

Rappel. L’ordre dans lequel on définit les procédures n’a pas d’importance. La seule
exigence (naturelle) est que les procédures doivent être définies avant d’être appliquées à
des arguments. Dans le cas de la récursivité croisée, on évaluera toutes les formes define
avant d’appliquer l’une des procédures définies.

Remarque. Un bel exemple de récursivité croisée est fourni par l’évaluateur lui-même.
Nous avons signalé au paragraphe 3.4 qu’il pouvait être organisé autour de deux
procédures, traditionnellement nommées eval et apply. La procédure eval s’appelle elle-
même, puisque l’évaluation d’une expression composée requiert en général l’évaluation
de sous-expressions; elle appelle aussi apply, lors de l’évaluation d’une combinaison. De
même, dans le cas où une fonction à appliquer est définie par une λ-forme, la fonction
apply appelle la fonction eval, puisque le résultat de l’application est en fait la valeur du
corps de la λ-forme, dans un certain environnement.

5 Récursivité structurelle

Le système Scheme “accepte” toute définition récursive syntaxiquement correcte, même
si la procédure associée donne lieu à des calculs infinis. L’utilisateur doit savoir si, dans
un domaine donné, le calcul se terminera toujours. Cette tâche de vérification peut être
fastidieuse mais, pour les domaines usuels, des schémas existent dont le respect garantit
d’office la terminaison. Les plus utiles de ces schémas sont les schémas structurels, basés
sur la manière dont les objets du domaine de calcul sont construits. Dans ce cadre, le
processus d’évaluation réduit le calcul de f(v) au calcul de f(v1), . . . , f(vn) où les vi sont
des composants directs de v. Cette technique est sûre tant que l’on se limite aux domaines
dont les objets ont un nombre fini de composants (directs ou non), clairement identifiés.

5.1 Récursivité structurelle sur le domaine des naturels

Conceptuellement, les naturels sont construits à partir de 0 et de la fonction successeur. Le
naturel 0 n’a pas de composant; c’est l’unique naturel élémentaire. Tout autre naturel a,
par définition, un composant direct, qui est son prédécesseur (cf. § 1.3.1). En conséquence,
dans le cas des naturels, la récursivité structurelle consiste à définir f(0) comme une
constante, et f(n), avec n > 0, comme la valeur d’une expression impliquant un appel à
f(n− 1) seulement. On peut définir de la sorte des fonctions à plusieurs arguments, mais
la récursion “porte” sur un seul de ceux-ci. Voici le schéma de base :

(define F

(lambda (n u1 ... um)

(if (zero? n)

(G u1 ... um)

(H (F (- n 1) (K1 n u1 ... um) ... (Km n u1 ... um))

n

u1 ... um))))

Les fonctions G, H et K1, . . . , Km sont supposées déjà définies. On observe que le calcul
de (F 0 u1 ... um) n’implique pas d’appel récursif, tandis que le calcul de (F n u1

... um) implique un appel récursif qui est du type (F (- n 1) ...), si n n’est pas nul;
ce dernier implique un appel du type (F (- n 2) ...), et ainsi de suite jusqu’à un appel
du type (F 0 ...). Le nombre [[m]] d’arguments additionnels (en plus de l’argument sur
lequel porte la récursion) est généralement petit. Les cas où il n’y a qu’un argument
additionnel et où il n’y en a aucun sont spécialement fréquents; nous en verrons plusieurs
exemples. Dans le cas [[m]] = 0, le schéma prend une forme très simple :

(define F

(lambda (n)

(if (zero? n)

c

(H (F (- n 1)) n))))

Voici immédiatement quelques exemples d’application élémentaire de la récursivité
structurelle sur les naturels.

(define harmonic-sum

(lambda (n)

(if (zero? n)

0

(+ (/ 1 n) (harmonic-sum (- n 1))))))

Si [[n]] = n, alors [[(harmonic-sum n)]] =
∑n

i=1
1
i
.

(define mult

(lambda (n u)

(if (zero? n)

0

(+ u (mult (- n 1) u)))))

Si [[n]] = n et [[u]] = u, alors [[(mult n u)]] = n ∗ u .

(define fact

(lambda (n)

(if (zero? n)

1

(* n (fact (- n 1))))))

Si [[n]] = n, alors [[(fact n)]] = n! .

(define cbin

(lambda (n u)

(if (zero? n)

1

(/ (* (cbin (- n 1) (- u 1)) u) n))))

Si [[n]] = n et [[u]] = u, avec 0 ≤ n ≤ u alors [[(cbin n u)]] est le coefficient de xn dans
le développement du binôme (1 + x)u .

Dans chaque cas, le calcul de [[(F n)]] ou de [[(F n u)]] implique une châıne de [[n]]
appels récursifs successifs.

Les schémas sont d’abord des schémas de pensée; ils suggèrent d’exprimer f(n) en
termes d’expressions indépendantes de f , mais aussi en terme de f(n − 1), si n > 0. Les
schémas imposent en outre la structure du programme : le programmeur doit seulement
définir les fonctions G, H et K. Au lieu d’écrire directement

(define cbin

(lambda (n u)

(if (zero? n)

1

(/ (* (cbin (- n 1) (- u 1)) u) n))))

on pourrait récrire le schéma

(define F

(lambda (n u)

(if (zero? n)

(G u)

(H (F (- n 1) (K n u)) n u))))

et l’accompagner de définitions auxiliaires :

(define G (lambda (u) 1))

(define K (lambda (n u) (- u 1)))

(define H (lambda (r n u) (/ (* r u) n)))

(define cbin F)

Remarque. On peut aussi utiliser une variante FF du schéma, où les fonctions auxiliaires
apparaissent en arguments. On a, par exemple

(define FF

(lambda (G H K)

(lambda (n u)

(if (zero? n)

(G u)

(H ((FF G H K) (- n 1) (K n u)) n u)))))

(define cbin

(FF (lambda (u) 1)

(lambda (r n u) (/ (* r u) n))

(lambda (n u) (- u 1))))

Une autre possibilité est la suivante :

(define FFF

(lambda (G H K n u)

(if (zero? n)

(G u)

(H (FFF G H K (- n 1) (K n u)) n u)))))

(define cbin

(lambda (n u)

(FFF (lambda (u) 1)

(lambda (r n u) (/ (* r u) n))

(lambda (n u) (- u 1))

n

u)))



5 RÉCURSIVITÉ STRUCTURELLE 81

5.2 Les listes

Nous avons déjà mentionné la correspondance naturelle entre les E-listes et les arbres dont
les feuilles sont (étiquetées par) des éléments de E. Un exemple concret est le domaine
LAl des listes littérales, où Al est l’ensemble des lettres de l’alphabet (Fig. 12). Rappelons
que seules les feuilles sont explicitement étiquetées, les nœuds internes ayant une étiquette
implicite.63

cb d g

e f

(a (b c d) ((e f) g))

(b c d) ((e f) g)a

(e f)

Figure 12: Arbre et liste littérale

5.3 Récursivité superficielle sur les listes

5.3.1 Le schéma de récursion superficielle

Le principe d’induction superficielle sur les listes donne immédiatement lieu au schéma de
récursion superficielle sur les listes :

(define F

(lambda (l u1 ... um)

(if (null? l)

(G u1 ... um)

(H (F (cdr l) (K1 l u1 ... um) ... (Km l u1 ... um))

l

u1 ... um))))

Les fonctions G, H et K1, . . . , Km sont supposées déjà définies. Calculer la valeur de
l’expression (F l ...) n’implique pas d’appel récursif quand [[l]] est la liste vide; sinon,

63Les arbres dont les nœuds internes sont étiquetés (indépendamment des feuilles) forment un autre type
de donnée, auquel pourrait correspondre un autre type de liste.

l’évaluation implique une châıne d’appels récursifs dont la longueur est celle de la liste [[l]].
Comme dans le domaine des nombres naturels, le cas particulier où il n’y a pas d’argument
additionnel est d’emploi fréquent :

(define F

(lambda (l)

(if (null? l)

c

(H (F (cdr l)) l))))

5.3.2 Exemples élémentaires

Donnons immédiatement quelques exemples d’emploi du schéma de récursion superficielle :

(define length

(lambda (l)

(if (null? l)

0

(+ 1 (length (cdr l))))))

(define append

(lambda (l v)

(if (null? l)

v

(cons (car l) (append (cdr l) v)))))

(define reverse

(lambda (l)

(if (null? l)

’()

(append (reverse (cdr l))

(list (car l))))))

(define map

(lambda (f1 l)

(if (null? l)

’()

(cons (f1 (car l)) (map f1 (cdr l))))))

Observons que la fonction append permet de concaténer deux listes, tandis que la fonction
reverse (qui utilise la fonction append) permet de retourner une liste.64

Notons aussi que la procédure map admet comme premier argument une procédure (à
un argument), qui est appliquée à tous les éléments de la liste qui constitue le deuxième
argument. Cette fonction map s’obtient en instanciant le schéma

64Nous employons ici une tournure de langage fréquente et commode, mais abusive : en programmation
fonctionnelle, on ne “retourne” pas une liste : on construit une autre liste, version retournée de la première.

(define F

(lambda (l u)

(if (null? l)

(G u)

(H (F (cdr l) (K l u))

l

u))))

par les paramètres fonctionnels suivants :

(define G (lambda (u) ’()))

(define H (lambda (r l u) (cons (u (car l)) r)))

(define K (lambda (l u) u))

(define map (lambda (f1 l) (F l f1)))

Remarque. La fonction map prédéfinie en Scheme est plus générale; outre l’égalité [[(map f (list a1 ... an))]]
= [[(list (f a1) ... (f an))]], on a aussi, par exemple, [[(map g (list a1 ... an) (list b1 ... bn))]]
= [[(list (g a1 b1) ... (f an bn))]]. On a notamment

(map + ’(1 2 3) ’(10 20 30)) ==> (11 22 33)

(map + ’(1 2) ’(10 20) ’(100 200)) ==> (111 222)

D’autres fonctions prédéfinies existent; signalons notamment append qui réalise la
concaténation d’un nombre quelconque de listes :

(append) ==> ()

(append ’(a b c)) ==> (a b c)

(append ’(a b) ’() ’(c d) ’(e)) ==> (a b c d e)

Remarque. La fonction reverse permet de comprendre en quoi le schéma de récursion qui
vient d’être présenté est “superficiel”. La figure 13 comporte à gauche une liste littérale
[[l]] et à droite la liste [[(reverse l)]]; on voit que le retournement est “superficiel” : il
ne concerne que les branches du premier niveau dans l’arbre.

5.3.3 Les listes sans répétition

Les listes sans répétition sont souvent utilisées pour représenter des ensembles. Nous
reviendrons plus longuement sur les ensembles au paragraphe 10.6; seuls quelques exemples
simples sont donnés ici. La fonction add_elem ajoute un élément dans une liste, s’il ne s’y
trouve pas déjà. Elle utilise la fonction auxiliaire member (§ 4.3.2) :

(define add_elem

(lambda (x l)

(if (member x l) l (cons x l)))))

g

e f

((e f) g)

(e f)cb d

(b c d)

(a (b c d) ((e f) g))

g

e f

((e f) g)

(e f) cb d

a

(((e f) g) (b c d) a)

(b c d)a

Figure 13: Retournement superficiel d’une liste littérale

La fonction union réalise la réunion de deux listes sans répétition :

(define union

(lambda (u v)

(if (null? u) v (add_elem (car u) (union (cdr u) v)))))

Sur le même modèle, on a aussi les fonctions d’intersection et de différence :

(define inter

(lambda (u v)

(if (null? u)

’()

(if (member (car u) v)

(add_elem (car u) (inter (cdr u) v))

(inter (cdr u) v)))))

(define diff

(lambda (u v)

(if (null? u)

’()

(if (member (car u) v)

(diff (cdr u) v)

(add_elem (car u) (diff (cdr u) v))))))

On a par exemple

(define *s1* ’(a c d f h))

(define *s2* ’(a b e f g))

(union *s1* *s2*) ==> (c d h a b e f g)

(inter *s1* *s2*) ==> (a f)

(diff *s1* *s2*) ==> (c d h)

Remarque. Les deux dernières fonctions peuvent se récrire en utilisant cond; on a par
exemple

(define diff

(lambda (u v)

(cond ((null? u) ’())

((member (car u) v) (diff (cdr u) v))

(else (add_elem (car u) (diff (cdr u) v))))))

5.3.4 Le tri par insertion et l’ordre lexicographique

Une application classique dans le domaine des listes est celle du tri. Elle est pertinente
dans le cas de toute liste dont les éléments forment un ensemble totalement ordonné.
En programmation fonctionnelle, on ne développera pas d’algorithmes de tri spécifiques
à tel ou tel ensemble ordonné : il sera plus commode de considérer que seul le prédicat
représentant cet ordre sera spécifique; on le passera comme argument supplémentaire
(fonctionnel) à un algorithme de tri qui sera générique, c’est-à-dire valable pour des
domaines variés (listes de nombres, de lettres, de mots, de phrases, d’arbres, etc.). On a
le programme suivant :

(define sort

(lambda (l comp) ;; comp: argument procedural

(if (null? l)

’()

(insert (car l) (sort (cdr l) comp) comp))))

Remarque. Le point-virgule (éventuellement répété) introduit un commentaire, qui se
termine en fin de ligne.

Le schéma habituel s’applique sans surprise. La version triée d’une liste vide est la liste
vide. La version triée d’une liste non vide l est le résultat de l’insertion à sa place de la tête
(car l) dans la version triée du reste (cdr l) de la liste. La fonction auxiliaire insert

prend trois arguments : un objet x, à insérer dans une liste triée l, l’ordre des objets étant
spécifié par le prédicat binaire comp. Ici aussi, l’application du schéma récursif est aisée :

(define insert

(lambda (x l comp)

(if (null? l)

(list x)

(if (comp x (car l))

(cons x l)

(cons (car l)

(insert x (cdr l) comp))))))

Le cas de base est, comme d’habitude, évident. Le cas inductif concerne l’insertion d’un
objet dans une liste triée non vide. Il y a deux sous-cas, dont seulement un donne lieu à
un appel récursif : si x précède le premier élément de la liste l, la tête du résultat est x et
le reste est l. Sinon, la tête du résultat est la tête de l, tandis que son reste est le résultat

de l’insertion de x dans le reste de l. En fait, ce code est suffisamment simple pour se
passer de commentaires ! Voici quelques essais :

(insert 3 ’(0 2 3 3 5 7 8 9) <) ==> (0 2 3 3 3 5 7 8 9)

(sort ’(8 3 5 7 2 3 9 0) <=) ==> (0 2 3 3 5 7 8 9)

(sort ’(8 3 5 7 2 3 9 0) >=) ==> (9 8 7 5 3 3 2 0)

Vérifions que insert, par exemple, est bien une instance du schéma de récursion
superficielle. Dans le cas de deux arguments additionnels, ce schéma est

(define F

(lambda (l u1 u2)

(if (null? l)

(G u1 u2)

(H (F (cdr l) (K1 l u1 u2) (K2 l u1 u2))

l

u1

u2))))

Les instances correspondantes sont

(define G (lambda (u1 u2) (list u1)))

(define H

(lambda (r l u1 u2)

(if (u2 u1 (car l))

(cons u1 l)

(cons (car l) r))))

(define K1 (lambda (l u1 u2) u1))

(define K2 (lambda (l u1 u2) u2))

(define insert (lambda (x l comp) (F l x comp)))

Vérifions maintenant que le même programme, appelé avec un prédicat de comparaison
lexicale, permet de trier des listes de mots représentés par des châınes de caractères (les
châınes de caractères constituent un type de données primitif en Scheme). On a

(sort ’("bb" "ac" "acb") string<=?) ==> ("ac" "acb" "bb")

Etant donné un ensemble E, on note E∗ l’ensemble des suites d’éléments de E. Si E
est totalement ordonné par une relation notée �, on peut toujours définir sur E∗ un ordre
total noté �∗ et qui étende l’ordre �.65 On introduit d’abord deux notations. Si α est

65Deux éléments a, b ∈ E sont aussi éléments de E∗; le fait que �∗ étende � signifie que l’on a a �∗ b si
et seulement si on a a � b.

une suite non vide, hd(α) désigne le premier élément de α et tl(α) désigne la suite α
privée de son premier élément.66 De plus, on pose a ≺ b =def (a � b ∧ a 6= b) et
a ≺∗ b =def (a �∗ b ∧ a 6= b). On définit alors �∗ comme suit. Soient α et β deux suites
distinctes.

• α �∗ α.

• α ≺∗ β ou β ≺∗ α.

• Si α est vide, alors α ≺∗ β.

• Si α et β sont non vides et si hd(α) ≺ hd(β),
alors α ≺∗ β.

• Si α et β sont non vides et si hd(α) = hd(β),
alors α ≺∗ β si et seulement si tl(α) ≺∗ tl(β).

L’ordre total �∗ est l’extension lexicographique de l’ordre total �.67

On définit maintenant une procédure lex, prenant comme arguments (procéduraux)
une relation d’ordre total strict sur un certain domaine E et la relation d’égalité sur ce
domaine. Cette procédure renvoie la version lexicographique (ordre total strict) sur le
domaine E∗.

(define lex

(lambda (str-ord iden) ;; arg proceduraux

(lambda (u v)

(cond

((null? u) #t)

((null? v) #f)

((str-ord (car u) (car v)) #t)

((iden (car u) (car v))

((lex str-ord iden) (cdr u) (cdr v)))

(else #f)))))

Définissons deux cas particuliers fréquents. Le premier concerne les listes de nombres, le
second les listes de châınes de caractères :

(define numlex (lex < =))

(define alpha (lex string<=? string=?))

On peut maintenant trier des listes de nombres ou de mots :

66Les symboles hd et tl abrègent les mots “head” et “tail”.
67Si � est l’ordre alphabétique sur les 26 lettres de l’alphabet latin, sa version lexicographique est l’ordre

du dictionnaire, donc l’ordre lexicographique usuel, pour les langues utilisant l’alphabet latin.

(sort ’((2 3) (1 4) (2 3 2) (1 3)) numlex)

==> ((1 3) (1 4) (2 3) (2 3 2))

(sort ’(("acb" "ac") ("ac" "acb") ("ac")) alpha)

==> (("ac") ("ac" "acb") ("acb" "ac"))

Observons à nouveau l’économie de code rendue possible par une approche réellement
fonctionnelle de la programmation, et en particulier par l’usage de procédures admettant
des arguments procéduraux et/ou renvoyant un résultat procédural.

5.3.5 Récursivité sur les suites

Certaines fonctions prédéfinies admettent un nombre illimité d’arguments et la forme
lambda généralisée permet à l’utilisateur de créer lui-même de telles fonctions. La suite
des arguments s’apparentant à une liste, il est naturel de penser que les fonctions à nombre
illimité d’arguments peuvent être définies récursivement.

Etant donné une fonction f : D2 → D, on définit une extension f+ de f sur le domaine
⋃

i>0 Di en posant

f+(x1) = x1 ,
f+(x1, x2) = f(x1, x2) ,
f+(x1, x2, x3) = f(x1, f(x2, x3)) ,
. . .

On devine la récursivité sous-jacente, qui se résume à l’égalité

f+(x0, x1, . . . , xn) = f(x0, f
+(x1, . . . , xn)) .

Si de plus la fonction binaire f admet un neutre à gauche e, c’est-à-dire un élément de
D tel que f(e, x) = x pour tout x ∈ D, on peut créer l’extension f∗ de f+ en posant
f∗() = e. En Scheme, la fonction apply et la forme lambda généralisée permettent de
définir ces extensions comme suit :

(define f*

(lambda v

(if (null? v)

e

(f (car v) (apply f* (cdr v))))))

(define f+

(lambda (x . v)

(if (null? v)

x

(f x (apply f+ v)))))

On a admis ici l’égalité e = [[e]].

On peut aussi définir directement les opérateurs d’extensions *_ext et +_ext; ici, pour
le premier opérateur, le premier argument est la fonction binaire à étendre et le second
est son neutre à gauche. On a :

(define *_ext

(lambda (f e)

(lambda v

(if (null? v)

e

(f (car v) (apply (*_ext f e) (cdr v)))))))

(define +_ext

(lambda (f)

(lambda (x . v)

(if (null? v)

x

(f x (apply (+_ext f) v))))))

Ces deux fonctions sont a priori étonnantes et méritent quelques commentaires; nous
raisonnons ici sur la seconde. En dépit des apparences, la fonction +_ext n’est pas
définie récursivement. Son argument est une fonction et nous n’avons pas muni le
domaine des fonctions d’une structure permettant de dire qu’une fonction est “plus simple”
qu’une autre. En fait, c’est la valeur fonctionnelle retournée par +_ext qui est définie
récursivement. Cela apparâıt mieux si on nomme cette fonction; nous verrons au chapitre 9
comment on peut donner des noms locaux à certains objets. Il est intéressant d’observer
que, sur le plan syntaxique, ces définitions semblent faire un usage vicieux de la récursivité;
par exemple, (+_ext f) est défini en fonction de (+_ext f) et non d’un terme (+_ext g),
où [[g]] serait “plus simple” que [[f]]. Notons à ce propos que la complexité du processus
d’évaluation de la forme (+_ext f) ne dépend pas de f; par contre, la complexité du
processus d’évaluation de la forme (apply (+_ext f) v) dépend de v; c’est donc la valeur
de v (une liste) qui doit devenir plus simple (c’est-à-dire plus courte) à chaque appel.

Pour illustrer l’usage des opérateurs d’extension, rappelons la définition de l’opérateur
de composition fonctionnelle :

(define compose

(lambda (f g)

(lambda (x) (f (g x))))) ==> ...

On a par exemple

((compose car cdr) ’(1 2 3 4)) ==> 2

((compose (compose car cdr) (compose cdr cdr)) ’(1 2 3 4)) ==> 4

On voit que pour composer quatre fonctions, trois interventions de l’opérateur compose

sont nécessaires. Les opérateurs d’extension permettent de simplifier l’écriture. On a :

(define id (lambda (x) x)) ;; transformation identique

(define compose* (*_ext compose id))

(define compose+ (+_ext compose))

((compose*) ’x) ==> x

((compose+ cdr) ’(1 2 3 4)) ==> (2 3 4)

((compose* car cdr) ’(1 2 3 4)) ==> 2

((compose+ car cdr cdr cdr) ’(1 2 3 4)) ==> 4

Un autre exemple intéressant est celui de la fonction exponentielle expt, prédéfinie en
Scheme : on a

(expt 2 10) ==> 1024

Observons que la fonction exponentielle admet 1 comme neutre à droite (on a x1 = x pour
tout x) mais n’admet pas de neutre à gauche; on utilisera donc exclusivement l’extension
(+_ext expt). Notons aussi que l’exponentielle n’est pas associative. On a

((+_ext expt) 3) ==> 3

((+_ext expt) 3 2) ==> 9

((+_ext expt) 2 3 2) ==> 512

((+_ext expt) 2 3 2 1) ==> 512

((+_ext expt) 2 3 2 1 5) ==> 512

On aurait pu aussi définir directement expt+ :

(define expt+

(lambda (x . v)

(if (null? v)

x

(expt x (apply expt+ v)))))

(expt+ 2 3 2 1 5) ==> 512

On pouvait encore utiliser une fonction auxiliaire et écrire

(define expt_list

(lambda (l)

(if (null? (cdr l))

(car l)

(expt (car l) (expt_list (cdr l))))))

(define expt+ (inv_gen_op_list expt_list))

On utilise ici l’opérateur inv_gen_op_list introduit au paragraphe 3.4.

Remarque. La récursivité sur les suites est intéressante, mais nous voyons ici qu’elle
n’est pas indispensable : une suite d’arguments peut toujours être remplacée par une liste
d’arguments. Au point de vue des performances, ce remplacement est préférable.

A titre d’exemple supplémentaire, créons la fonction de tri numsort_* qui renvoie la
liste de ses arguments numériques triés par ordre croissant. Un moyen simple passe par
l’emploi de la fonction générique sort introduite au paragraphe 5.3.4. On a

(sort ’(8 3 5 7 2 3 9 0) <=) ==> (0 2 3 3 5 7 8 9)

(define <=sort_* (lambda (l) (sort l <=))) ==> ...

(define numsort_* (lambda v (<=sort_* v))) ==> ...

(numsort_* 8 3 5 7 2 3 9 0) ==> (0 2 3 3 5 7 8 9)

On pouvait aussi utiliser l’opérateur inv_gen_op_list

(define numsort_* (inv_gen_op_list <=sort_*))

On pouvait enfin écrire immédiatement

(define numsort_* (lambda v (sort v <=))

Cela suggère la généralisation suivante :

(define sort_*

(lambda (comp . v) (sort v comp))) ==> ...

(sort_* <= 8 3 5 7 2 3 9 0) ==> (0 2 3 3 5 7 8 9)

(sort_* >= 8 3 5 7 2 3 9 0) ==> (9 8 7 5 3 3 2 0)

5.4 Récursivité profonde sur les listes et les arbres

La récursivité profonde ne s’applique qu’à des listes dont les éléments sont des atomes
ou des listes du même type.68 Concrètement, le schéma superficiel est modifié comme
suit : un appel (F l ...) peut donner lieu non seulement à un appel récursif du type
(F (cdr l) ...) mais aussi à un appel similaire portant sur le car, du type (F (car

l) ...), à condition, naturellement, que [[(car l)]] soit une liste. On observera que ceci
est calqué sur le principe d’induction profonde introduit au paragraphe 1.3.4. Voilà le
schéma général :

(define F

(lambda (l u1 ... um)

(cond ((null? l) (G u1 ... um))

68La récursivité profonde s’applique donc aux liste littérales introduites au paragraphe précédent.

((atom? (car l))

(H (F (cdr l) (K1 l u1 ... um) ... (Km l u1 ... um))

l

u1 ... um))

(else ;; (car l) est une liste

(J (F (car l) (A1 l u1 ... um) ... (Am l u1 ... um))

(F (cdr l) (D1 l u1 ... um) ... (Dm l u1 ... um))

l

u1 ... um)))))

Comme d’habitude, notons que le nombre m d’arguments additionnels est généralement
faible; le cas où il est nul donne lieu à la version simplifiée suivante :

(define F

(lambda (l)

(cond ((null? l) c)

((atom? (car l)) (H (F (cdr l)) l))

(else (J (F (car l)) (F (cdr l)) l)))))

Remarque. On exclut (provisoirement) de rencontrer dans la liste l des objets qui ne
soient ni des listes ni des atomes.

A titre de premier exemple, voici une fonction qui calcule la frondaison d’une liste
littérale, c’est-à-dire la liste des feuilles de l’arbre sous-jacent :

(define flat_list

(lambda (l)

(cond ((null? l) ’())

((atom? (car l))

(if (null? (car l))

(flat_list (cdr l))

(cons (car l) (flat_list (cdr l)))))

(else

(append (flat_list (car l))

(flat_list (cdr l)))))))

Voici quelques exemples d’utilisation :

(flat_list ’a) ==> Error - 5 passed as argument to car

(flat_list ’()) ==> ()

(flat_list ’(a (b c) ((((d))) e) b)) ==> (a b c d e b)

Les listes considérées ici ont pour éléments des objets atomiques d’un certain type (par
exemple, des lettres) et aussi d’autres listes. Le parallèle entre principes d’induction et
schéma de récursion permet d’étendre le domaine LAl des listes littérales en le domaine LAl

+

des arbres littéraux. Les schémas précédents s’adaptent facilement. Dans le cas simplifié,
le schéma

(define F

(lambda (l)

(cond ((null? l) c)

((atom? (car l)) (H (F (cdr l)) l))

(else (J (F (car l)) (F (cdr l)) l)))))

devient

(define F

(lambda (l)

(cond ((null? l) c)

((atom? l) (G l))

(else (J (F (car l)) (F (cdr l)) l)))))

Le programme de calcul de la frondaison d’un arbre littéral est :

(define flat_tree

(lambda (l)

(cond ((null? l) ’())

((atom? l) (list l))

(else

(append (flat_tree (car l))

(flat_tree (cdr l)))))))

Ce programme admet maintenant les arguments atomiques; pour le reste, son
comportement ne change pas :

(flat_tree ’a) ==> (a)

(flat_tree ’()) ==> ()

(flat_tree ’(a (b c) ((((d))) e) b)) ==> (a b c d e b)

Voilà enfin le programme de retournement profond d’un arbre littéral :

(define deeprev

(lambda (l)

(cond ((null? l) ’())

((atom? l) l)

(else

(append (deeprev (cdr l))

(list (deeprev (car l))))))))

L’appel récursif porte sur le car et sur le cdr. On observera le comportement du
programme à la figure 14.

cb d

(b c d)

(a (b c d) ((e f) g))

c

a a(g (f e))

g

e f

((e f) g)

(e f) g (f e)

f e

(d c b)

d b

((g (f e)) (d c b) a)

Figure 14: Retournement profond d’une liste littérale

5.5 Remarque sur les schémas de programmes

Suivre un schéma “à la lettre”, c’est-à-dire se limiter strictement à instancier les paramètres
qu’il contient, rend la programmation particulièrement méthodique et sûre. On peut
cependant, pour diverses raisons, garder “l’esprit” d’un schéma sans respecter la lettre (sa
syntaxe précise). Un exemple classique est le programme flat_tree_a, version équivalente
mais plus efficace de flat_tree :

(define flat_tree_a

(lambda (l u)

(cond ((null? l) u)

((atom? l) (cons l u))

(else

(flat_tree_a

(car l)

(flat_tree_a (cdr l) u))))))

On montre facilement (par induction sur la structure de l) que les valeurs de
(flat_tree_a l u) et de (append (flat_tree l) u) sont égales pour toutes listes l et
u; en particulier, (flat_tree_a l ’()) et (flat_tree l) ont même valeur. Par contre,
on peut noter que la fonction flat_tree_a n’est pas une instance du schéma, même si
elle s’en inspire nettement.

5.6 Récursivité structurelle complète et mixte

Les schémas de récursion introduits jusqu’ici correspondent à des principes d’induction
simple. Le principe de la récursivité structurelle est que la structure du programme est
calquée sur celle des données. On exprime f(x, . . .) en termes des éléments de l’ensemble
{f(y, . . .) : y ≺ x}, où y ≺ x signifie que y est un composant direct de x. On peut
naturellement généraliser cela de diverses manières :

• Admettre aussi les composants indirects :
(- n 1), mais aussi (- n 2), (/ n 2), . . .
(cdr l), mais aussi (cddr l), . . .

• Admettre plusieurs appels récursifs.

• Admettre les appels imbriqués.

La première généralisation correspond au passage de l’induction simple à l’induction
complète. Pour toutes ces généralisations, la terminaison reste garantie . . . mais pas
l’efficacité, comme nous avons eu plusieurs fois l’occasion de l’observer. Notons aussi
qu’admettre les composants indirects peut légèrement compliquer le test d’existence du
composant, toujours indispensable. Il convient d’éviter les erreurs grossières, telles les
évaluations de (f (- n 2)) avec n ≤ 1, et de (f (cddr l)) où l comporte moins de deux
éléments. Nous avons déjà rencontré des exemples d’emploi de la récursivité structurelle
complète, notamment le programme näıf fib (voir §§ 1.2.6 et 4.3) et l’exponentielle rapide
exp, dont voici le code :

(define exp

(lambda (m n)

(cond ((zero? n) 1)

((even? n) (exp (* m m) (/ n 2)))

((odd? n) (* m (exp m (- n 1)))))))

Le principe de la récursivité structurelle mixte est de faire porter la récursivité
sur plusieurs arguments. On peut par exemple exprimer f(x1, x2, . . .) en termes de
{f(y1, x2, . . .), f(x1, y2, . . .), f(y1, y2, . . .) : y1 ≺ x1 ∧ y2 ≺ x2}. L’induction porte sur
plusieurs arguments; si f(a, b) dépend de f(c, d), alors c ≺ a ∧ d � b ou c � a ∧ d ≺ b.
La terminaison reste garantie. L’exemple le plus classique est sans doute la version näıve
de calcul du plus grand commun diviseurs de deux entiers strictement positifs par la règle
d’Euclide :

(define gcd

(lambda (x y)

(cond ((= x y) x)

((> x y) (gcd (- x y) y))

((< x y) (gcd x (- y x))))))

Un moyen simple de se convaincre que l’évaluation de (gcd x y) se termine toujours si
les valeurs des opérandes sont des entiers strictement positifs est d’observer que tout appel
récursif subséquent se fait avec des arguments dont la somme est strictement inférieure
à la somme des arguments initiaux. Une démarche analogue s’applique au programme
récursif suivant :

(define enum

(lambda (p q) (if (> p q) ’() (cons p (enum (+ p 1) q)))))

En effet, (enum p q) s’évalue sans appel récursif si [[p]] > [[q]]; sinon, tout appel récursif
subséquent se fait avec des arguments dont la différence est strictement inférieure à la
différence des arguments initiaux.

5.7 La séparation fonctionnelle

On peut “dérécursiver” le schéma classique

fact =def λn . [if n = 0 then 1 else n ∗ fact(n − 1)]

en l’écriture

factm =def λn . [if n = 0 then 1 else n ∗ factm−1(n − 1)] .

La récursivité revient (sous forme dégénérée) pour définir globalement la famille des
fonctions factm :

(define f

(lambda (m c)

(if (zero? m)

c

(f (- m 1)

(lambda (n)

(if (= n 0) 1 (* n (c (- n 1)))))))))

La fonctionnelle69 f admet un paramètre numérique m et un paramètre fonctionnel c;
quand leurs valeurs respectives sont m et factp, la valeur de (f m c) est factp+m; en
particulier on a :

(define fact0 ’emptyfunction)

(define fact (lambda (n) ((f (+ n 1) fact0) n)))

On observe que factm a pour domaine {0, 1, . . . ,m − 1}. Sur ce domaine, on a
factm(n) = n! . On a par exemple

(define fact8 (f 8 fact0))

(fact8 7) ==> 5040

(fact8 8) ==> Error

(fact 8) ==> 40320

On a séparé

• le calcul de la fonction factp

• l’application de cette fonction à un argument.

69On donne parfois ce nom aux fonctions “d’ordre supérieur” dont l’un des arguments, ou le résultat,
est de nature fonctionnelle.

On a fact(n) = factp(n) si p > n; on choisit p = n + 1.
La séparation fonctionnelle n’apporte rien dans le cas très simple de la fonction

factorielle mais, dans le cadre général de la définition récursive de fonctions, cette technique
sera parfois très utile ! Plus généralement, l’introduction de paramètres fonctionnels et/ou
de fonctionnelles auxiliaires définies récursivement est une technique importante.

Examinons de plus près le processus de calcul lié aux instances du principe de
séparation fonctionnelle :

(fact 8)((f 9 fat0) 8)((f 8 (lambda (n) (if (= n 0) 1 (* n (fat0 (- n 1)))))) 8)((f 8 fat1) 8)...((f 0 fat9) 8)(fat9 8)(* 8 (* ... 1))...(* 8 5040)
40320

La partie fonctionnelle du développement s’achève avant le début du calcul arithmétique
proprement dit, c’est-à-dire avant les multiplications.

5.7.1 Application : le double comptage

Le problème du double comptage, quoique très simple, permet une première illustration
de la technique de séparation fonctionnelle. On souhaite parcourir une liste et dénombrer
d’une part les éléments possédant une certaine caractéristique et, d’autre part, ceux ne la
possédant pas. On peut par exemple dénombrer les “a” et les “non-a” dans une liste de
lettres, et fournir la liste des deux résultats :

(count ’(a b a c d a c) ’a) ==> (3 4)

;; 3 occurrences de "a", 4 autres occurrences

La solution élémentaire consiste à utiliser deux fonctions auxiliaires, une fonction c-yes

pour compter les “a”, une fonction c-no pour compter les “non-a”, et à regrouper les deux
résultats partiels dans une liste :

(define count0 (lambda (l s) (list (c-yes l s) (c-no l s))))

(define c-yes

(lambda (l s)

(cond ((null? l) 0)

((eq? (car l) s) (+ 1 (c-yes (cdr l) s)))

(else (c-yes (cdr l) s)))))

(define c-no

(lambda (l s)

(cond ((null? l) 0)

((eq? (car l) s) (c-no (cdr l) s))

(else (+ 1 (c-no (cdr l) s))))))

L’inconvénient est que la liste l est parcourue deux fois lors de l’évaluation de
(count l s).

Un “remède” aussi näıf que catastrophique consiste en l’utilisation du schéma
habituel :70

(define count1

(lambda (l s)

(if (null? l)

(list 0 0)

(if (eq? (car l) s)

(list (1+ (car (count1 (cdr l) s)))

(cadr (count1 (cdr l) s)))

(list (car (count1 (cdr l) s))

(1+ (cadr (count1 (cdr l) s))))))))

L’inefficacité est catastrophique, parce que chaque appel avec l non vide donne lieu à deux
appels avec (cdr l) : le temps est donc exponentiel en la longueur de la liste. On peut
remédier à cela simplement, par la séparation fonctionnelle :

(define c2

(lambda (l s c)

(if (null? l)

c

(if (eq? (car l) s)

(c2 (cdr l)

s

(lambda (u)

(c (list (1+ (car u)) (cadr u)))))

(c2 (cdr l)

s

(lambda (u)

(c (list (car u) (1+ (cadr u))))))))))

(define id (lambda (v) v))

(define count2 (lambda (l s) ((c2 l s id) ’(0 0))))

70Notons à cette occasion les fonctions primitives 1+ et -1+; les expressions (1+ a) et (-1+ a) sont
équivalentes aux expressions (+ a 1) et (- a 1), respectivement.

Si ℓ est la longueur de la liste l, les temps d’exécution de (count0 l) et de (count2 l)

sont proportionnels à ℓ. Celui de (count1 l) est proportionnel à 2ℓ. Une bonne
compréhension de ce qui précède passe par une spécification claire de la fonction auxiliaire
c2 :

Si [[s]] et [[l]] sont respectivement un objet et une liste d’objets,
si [[c]] est une fonction qui à toute liste de deux nombres (n m)
associe la liste de deux nombres (n+ca m+cd),
alors [[(c2 l s c)]] est la fonction qui à toute liste de deux nombres (n m)
associe la liste (n+ca+ℓa m+cd+ℓd),
où ℓa est le nombre d’éléments de [[l]] égaux à [[s]] et
où ℓd est le nombre d’éléments de [[l]] distincts de [[s]].

On verra d’autres solutions pour ce problème au chapitre suivant.

5.7.2 Application : le produit d’une liste de nombres

Un autre problème élémentaire donnant lieu à l’application de la technique de séparation
fonctionnelle est celui du produit d’une liste de nombres. Le schéma habituel donne la
solution suivante :

(define prodlist0

(lambda (l)

(if (null? l) 1 (* (car l) (prodlist0 (cdr l)))))))

Si un des éléments de l est nul, le résultat est 0 . . . et toutes les multiplications ont été
effectuées en vain. Un raffinement simple consiste à s’arrêter dès qu’un 0 est rencontré :

(define prodlist1

(lambda (l)

(cond ((null? l) 1)

((zero? (car l)) 0)

(else (* (car l) (prodlist1 (cdr l)))))))

Si cependant le facteur 0 ne survient qu’en fin de liste, on aura quand même effectué
de nombreuses multiplications inutiles. En fait, la question est, si un facteur est nul,
comment éviter toutes les multiplications ? La séparation fonctionnelle permet de résoudre
ce problème simplement :

(define pl2_c

(lambda (l c)

(cond ((null? l) c)

((zero? (car l)) (lambda (v) 0))

(else (pl2_c (cdr l)

(lambda (u) (* (car l) (c u))))))))

(define prodlist2 (lambda (l) ((pl2_c l id) 1)))

Ici aussi, une bonne compréhension du comportement de la fonction principale passe par
une spécification claire de la fonction auxiliaire pl2_c :

Si [[l]] est une liste de nombres, si kc est un nombre
et si [[c]] est la fonction qui à tout nombre n associe le produit kcn,
alors la fonction [[(pl2_c l c)]] associe à tout n le produit ℓkcn,
où ℓ est le produit des éléments de [[l]].

On verra une autre solution au chapitre 7, basée sur un principe différent.

6 Conception de programme

Au terme de ces premiers chapitres, nous avons rencontré les principaux mécanismes de
Scheme. Ceux-ci suffisent à résoudre un grand nombre de problèmes. Cependant, sauf
peut-être indirectement, par le biais d’exemples, nous n’avons pas abordé le point essentiel
de l’apprentissage de la programmation, qui est Comment concevoir un programme ? .
Dans ce paragraphe, nous donnons quelques indications.

6.1 Première étude

6.1.1 L’énoncé

Pour fixer les idées, partons d’un problème très simple :

Ecrire une fonction cube_sum_square qui,
à toute liste de nombres, associe
le cube de la somme des carrés de ses éléments.

L’écriture d’une fonction f peut nécessiter le recours à des fonctions auxiliaires, qui
doivent alors être spécifiées.

6.1.2 Analyse, structuration, solution

Bien étudier l’énoncé fourni, ainsi que soigner la rédaction des énoncés spécifiant les
fonctions auxiliaires éventuelles, n’est jamais une perte de temps. Dans notre exemple, le
fragment

. . . le cube de la somme des carrés . . .

suscite “naturellement” l’idée de recourir à des fonctions auxiliaires cube et sum_square,
dont les spécifications sont les suivantes :

Ecrire une fonction cube qui à tout nombre
associe le cube de ce nombre.

Ecrire une fonction sum_square qui, à toute liste de nombres,
associe la somme des carrés de ses éléments.

Vu le caractère élémentaire du problème posé, donnons sans plus attendre le code de
la fonction principale et des deux fonctions auxiliaires :

(define cube_sum_square

(lambda (l)

(cube (sum_square l))))

(define cube

(lambda (x) (* x x x)))

(define sum_square

(lambda (l)

(if (null? l)

0

(+ (* (car l) (car l))

(sum_square (cdr l))))))

6.1.3 Variantes

Si “naturelle” que soit cette solution, on peut toujours imaginer d’autres solutions,
correspondant à d’autres décisions concernant le choix des fonctions auxiliaires, ou à
l’utilisation d’autres algorithmes. Il est intéressant ici de considérer explicitement d’autres
politiques en ce qui concerne le choix des fonctions auxiliaires. Observons d’abord que, a

priori, la décision de créer une fonction auxiliaire cube ne s’imposait pas vraiment, parce
que son code est vraiment “trop simple”. Soulignons quand même un fait important :
définir une fonction est une chose, la nommer en est une autre. Il aurait été parfaitement
raisonnable de renoncer à nommer la fonction cube, et donc d’utiliser une fonction
anonyme. La solution serait alors

(define cube_sum_square

(lambda (l)

((lambda (x) (* x x x)) (sum_square l))))

(define sum_square ...

Par contre, renoncer aussi à cette fonction anonyme et donc écrire

(define cube_sum_square

(lambda (l)

(* (sum_square l) (sum_square l) (sum_square l))))

aurait été un gaspillage de ressources, conduisant à calculer la valeur de l’expression
(sum_square l) trois fois. Le même raisonnement ne s’applique pas à l’élévation au
carré présente à l’avant-dernière ligne du code de sum_square car l’évaluation de (car l)

est très peu coûteuse.71

Remarque. Lors de l’application de la fonction [[cube_sum_square]] définie par

(define cube_sum_square

(lambda (l)

((lambda (x) (* x x x)) (sum_square l))))

71Remplacer (* (car l) (car l)) par ((lambda (x) (* x x)) (car l)) ne serait donc pas utile.

à un argument approprié [[l]], on doit évaluer d’abord [[(sum_square l)]], créer un
environnement étendu par une liaison de x à cette valeur, puis évaluer (* x x x) dans cet
environnement étendu. Cela correspond à dire, “soit v la valeur de (sum_square l); on
renvoie la valeur de (* x x x), où [[x]] = v”. Il se fait que Scheme permet de traduire
littéralement ce procédé :

(define cube_sum_square

(lambda (l)

(let ((x (sum_square l))) (* x x x))))

La forme spéciale let sera vue au paragraphe 9.1.

6.1.4 Eliminer une fonction auxiliaire ?

On pourrait aussi envisager d’éliminer le nom sum_square, voire la fonction elle-même.
Notons d’emblée que, contrairement à cube, la fonction sum_square a été définie
récursivement . . . ce qui impose de lui donner un nom. Si on souhaite “cacher”
ce nom, il faut utiliser une technique particulière, introduite au chapitre suivant.72

Plus radicalement, le programmeur aurait pu envisager de programmer directement
cube_sum_square, sans recourir à des fonctions auxiliaires (nommées ou anonymes), et en
particulier sans utiliser sum_square. L’idée est a priori défendable, mais sera en principe
rejetée après la courte étude suivante. Pour exprimer

[[(cube_sum_square l)]] = (x2
0 + x2

1 + · · · + x2
n)3

en termes de

[[(cube_sum_square (cdr l))]] = (x2
1 + · · · + x2

n)3 ,

il faut utiliser la règle
(a + b)3 = a3 + 3a2b + 3ab2 + b3 ,

où l’on constate que b est précisément [[(sum_square (cdr l))]], ce qui montre que la
fonction sum_square est indispensable.

6.1.5 Généralisation et réutilisation

Le fait que le recours à une fonction auxiliaire ait été reconnu nécessaire, comme dans le
cas de sum_square, ne signifie pas obligatoirement que la fonction doit être programmée
immédiatement. Il convient de se demander d’abord si cette fonction n’apparâıt pas
“naturellement” comme un cas particulier d’une fonction plus générale. Si c’est le cas,
mieux vaut programmer la fonction plus générale; la probabilité que l’on puisse réutiliser

72Supposons que la fonction principale cube sum square ne soit qu’un petit fragment d’un gros logiciel,
développé en équipe. Il faudra alors que chaque programmeur obtienne du chef d’équipe l’autorisation de
nommer toute fonction autre que la fonction principale qu’il est chargé de construire, sinon il y aurait risque
d’interférence avec les fonctions programmées et nommées par d’autres membres de l’équipe. En pratique,
le programmeur “cachera” les noms des fonctions auxiliaires qu’il aura jugé opportun de développer.

une fonction auxiliaire, dans un contexte différent de celui qui a amené sa spécification et
sa programmation, est d’autant plus élevée que cette fonction est générale. Dans le cas
présent, il est naturel de généraliser la fonction

(x1, . . . , xn) 7→
n

∑

i=1

x2
i

en la fonction

(f, (x1, . . . , xn)) 7→
n

∑

i=1

f(xi) ,

ce qui donne lieu au code suivant :

(define cube_sum_square

(lambda (l) (cube (+_map square l))))

(define cube (lambda (x) (* x x x)))

(define square (lambda (x) (* x x)))

(define +_map

(lambda (f l)

(if (null? l)

0

(+ (f (car l))

(+_map f (cdr l))))))

On peut aller plus loin dans la voie de la généralisation; en effet, l’opérateur somme
apparâıt ici comme la généralisation à un nombre quelconque d’arguments de l’opérateur
binaire d’addition. Tout opérateur binaire admettant un élément neutre est susceptible
d’être étendu de la sorte; nous connaissons déjà la multiplication, dont le neutre est 1, et la
concaténation, dont le neutre est (). (Rappelons que la fonction prédéfinie append calcule
la concaténation d’un nombre quelconque de listes.) Cela justifie que l’on généralise la
fonction

(f, (x1, . . . , xn)) 7→
n

∑

i=1

f(xi) ,

en la fonction
(ω, ν, (x1, . . . , xn)) 7→ Ωn

i=1f(xi) .

Dans cette notation, ω est un opérateur binaire et Ω est la généralisation de ω définie
comme suit :

Ω(()) = ν ,
Ω((x1)) = x1 ,
Ω((x1, x2)) = x1 ω x2 ,
Ω((x1, x2, x3)) = x1 ω (x2 ω x3) ,
. . .

Avec cette nouvelle généralisation, la solution du problème original devient :

(define cube_sum_square

(lambda (l) (cube (gen_map + 0 square l))))

(define cube (lambda (x) (* x x x)))

(define square (lambda (x) (* x x)))

(define gen_map

(lambda (omega nu f l)

(if (null? l)

nu

(omega (f (car l))

(gen_map omega nu f (cdr l))))))

On peut voir tout de suite l’intérêt de la généralisation. Supposons que l’on veuille définir la
fonction qui à toute liste de nombres associe le carré du produit de ces nombres augmentés
de 5; la réponse est maintenant immédiate :

(define square_product_add5

(lambda (l) (square (gen_map * 1 add5 l))))

(define add5 (lambda (n) (+ n 5)))

On peut aussi définir aisément les fonctions reverse_concat_duplic et duplic_concat_reverse
telles que, par exemple

(reverse_concat_duplic ’((a b c) (1 2) (x y)))

==> (y x y x 2 1 2 1 c b a c b a)

(duplic_concat_reverse ’((a b c) (1 2) (x y)))

==> (c b a 2 1 y x c b a 2 1 y x)

Le code est

(define reverse_concat_duplic

(lambda (l) (reverse (gen_map append ’() duplic l))))

(define duplic_concat_reverse

(lambda (l) (duplic (gen_map append ’() reverse l))))

(define duplic (lambda (l) (append l l)))

Il est fréquent d’utiliser gen_map avec comme premier argument append; cela justifie
la définition

(define append_map

(lambda (f l) (gen_map append ’() f l)))

On peut aussi définir append_map directement :

(define append_map

(lambda (f l)

(if (null? l)

’()

(append (f (car l)) (append_map f (cdr l))))))

ou encore en utilisant apply :

(define append_map

(lambda (f l) (apply append (map f l))))

Cette fonction sera réutilisée dans la suite; notons aussi la variante

(define union_map

(lambda (f l)

(if (null? l)

’()

(union (f (car l)) (union_map f (cdr l))))))

D’autres cas particuliers sont fréquents. Notons d’abord que les expressions
(gen_map cons ’() f l) et (map f l) ont même valeur; on a déjà mentionné +_map, et
on définit aussi *_map de telle sorte que (gen_map * 1 f l) et (*_map f l) aient même
valeur.

Deux cas intéressants sont and_map et or_map mais, and et or étant des formes spéciales
et non des fonctions susceptibles d’être passées en arguments, on doit définir ces opérateurs
directement :

(define and_map

(lambda (f l)

(if (null? l)

#t

(and (f (car l)) (and_map f (cdr l))))))

(define or_map

(lambda (f l)

(if (null? l)

#f

(or (f (car l)) (or_map f (cdr l))))))

ou encore

(define and_map

(lambda (f l)

(or (null? l)

(and (f (car l)) (and_map f (cdr l))))))

(define or_map

(lambda (f l)

(and (not (null? l))

(or (f (car l)) (or_map f (cdr l))))))

6.1.6 Filtrage et transformation

La fonction gen_map est un opérateur très général pour traiter les listes élément par
élément. Il est parfois nécessaire de filtrer une liste avant l’application de gen_map, pour
éliminer les éléments auxquels le traitement ne peut pas ou ne doit pas s’appliquer. On
utilise dans ce but la fonction filter définie comme suit :

(define filter

(lambda (p? l)

(cond ((null? l) ’())

((p? (car l)) (cons (car l) (filter p? (cdr l))))

(else (filter p? (cdr l))))))

Le prédicat unaire p? est le filtre; les éléments de l qui ne le vérifient pas sont omis. On
peut alors intégrer le filtrage à la fonction gen_map :

(define gen_map_filter

(lambda (omega nu f p? l)

(gen_map omega nu f (filter p? l))))

ou encore :

(define gen_map_filter

(lambda (omega nu f p? l)

(cond ((null? l) nu)

((p? (car l))

(omega (f (car l))

(gen_map_filter omega nu f p? (cdr l))))

(else (gen_map_filter omega nu f p? (cdr l))))))

Remarque. En utilisant déjà la construction let évoquée plus haut, on a aussi :

(define gen_map_filter

(lambda (omega nu f p? l)

(if (null? l)

nu

(let ((rec (gen_map_filter omega nu f p? (cdr l))))

(if (p? (car l))

(omega (f (car l)) rec)

rec)))))

D’autres généralisations sont possibles. Le filtrage introduit ici est un préfiltrage, puisque
les éléments sont testés avant l’application de f. On pourrait ajouter un postfiltrage,
qui s’exercerait après cette application. A l’opposé, une particularisation intéressante est
map_filter :

(define map_filter

(lambda (f p? l)

(gen_map_filter cons ’() f p? l)))

que l’on pouvait aussi définir directement :

(define map_filter

(lambda (f p? l)

(cond ((null? l) ’())

((p? (car l))

(cons (f (car l)) (map_filter f p? (cdr l))))

(else (map_filter f p? (cdr l))))))

Remarque. Les expressions (map_filter id p? l) et (filter p? l) sont équivalentes
si [[id]] est la fonction identique.

Remarque. La facilité avec laquelle les procédures écrites en Scheme se généralisent et se
particularisent est un atout important du langage.

6.2 Deuxième étude

6.2.1 L’énoncé

On a vu au paragraphe 2.7.6 que la valeur renvoyée par une procédure, ainsi que
les arguments d’une procédure, pouvaient eux-mêmes être des procédures. L’exemple
classique est celui de la procédure compose, qui prend comme arguments deux procédures
et renvoie leur composition. La récursivité permet de définir d’autres procédures d’ordre
supérieur. On peut notamment généraliser compose en une fonction compose_list,
spécifiée comme suit :

Ecrire une fonction compose_list qui, à toute liste de fonctions unaires
composables, associe leur composée.73

Par exemple, [[(compose_list (list f g h))]] sera [[f]]◦[[g]]◦[[h]], la fonction résultant
de la composition de [[f]], [[g]] et [[h]].
Remarque. Rappelons que (f ◦ g)(x) est par définition f(g(x)); on applique d’abord g,
puis f .

73Rappelons que l’opérateur de composition de fonctions est associatif mais pas commutatif. La fonction
identité est neutre pour cet opérateur.

6.2.2 Solution directe, solution par réutilisation

Le schéma de récursion superficielle sur les listes fournit immédiatement une solution :

(define compose_list

(lambda (f_list)

(if (null? f_list)

(lambda (x) x)

(compose (car f_list) (compose_list (cdr f_list))))))

On peut éviter le recours explicite à compose :

(define compose_list

(lambda (f_list)

(if (null? f_list)

(lambda (x) x)

(lambda (x)

((car f_list) ((compose_list (cdr f_list)) x))))))

On a par exemple

((compose_list (list car cdr cdr)) ’(1 2 3 4 5)) ==> 3

On peut aussi noter que le problème posé, quoiqu’entièrement différent du problème
posé au paragraphe précédent, se résout immédiatement grâce à la fonction auxiliaire très
générale introduite dans ce paragraphe précédent :

(define compose

(lambda (f g) (lambda (x) (f (g x)))))

(define id (lambda (x) x))

(define compose_list

(lambda (f_list)

(gen_map compose

(lambda (x) x)

(lambda (x) x)

f_list)))

6.2.3 L’itérateur

Un cas particulier intéressant est celui où toutes les fonctions à composer sont égales. On
appelle nième itérée de la fonction f de D dans D la composée de n fonctions égales à f .

Ecrire une fonction iter tel que pour tout naturel n,
(iter n) soit la fonction qui à toute fonction f

composable avec elle-même associe la nième itérée de f.

La solution est immédiate. On peut écrire

(define iter

(lambda (n)

(lambda (f)

(if (zero? n)

(lambda (x) x)

(compose f ((iter (- n 1)) f))))))

Remarque. Il pourrait parâıtre abusif (ou prétentieux . . . ) de qualifier cette solution
d’immédiate. Avec un peu d’habitude, elle l’est néanmoins, dans la mesure où elle se
commente très simplement. Dans le texte suivant, chaque ligne correspond à une ligne du
programme précédent.

On définit iter,
une fonction à un argument n;
la fonction (iter n) associe à f

si n vaut 0,
la fonction identité,
sinon la composée de f et de f(n−1) .74

Une variante de cette solution est

(define iter

(lambda (n)

(lambda (f)

(lambda (x)

(if (zero? n)

x

(f (((iter (- n 1)) f) x)))))))

On peut aussi réutiliser une fonction précédente :

(define iter

(lambda (n) (lambda (f) (compose_list (n_list n f)))))

(define n_list

(lambda (n x) (if (= n 0) ’() (cons x (n_list (- n 1) x)))))

Voici quelques exemples d’utilisation :

(define add1 (lambda (x) (+ x 1)))

(define add7 ((iter 7) add1))

(add7 8) ==> 15

74La notation f (n) désigne ici la nième itérée de f . On améliorerait l’efficacité en utilisant aussi l’égalité
f (2n) = f (n) ◦ f (n).

(define square (lambda (x) (* x x)))

(define power8 ((iter 3) square))

(power8 2) ==> 256

6.3 Troisième étude

Le lecteur est sans doute déjà convaincu de l’intérêt des schémas récursifs, dont une
surprenante variété de programmes sont de simples instances. On soupçonne néanmoins, à
juste titre d’ailleurs, que de nombreux problèmes demandent une démarche plus créative
qu’une simple instantiation de schéma. Nous verrons d’ailleurs dans la suite du livre
quelques exemples de tels problèmes. Toutefois, c’est une erreur méthodologique de croire
trop tôt qu’un problème ne peut être résolu simplement, ce que nous illustrons par deux
exemples classiques.

6.3.1 Deux énoncés classiques

Ecrire une fonction subsets

qui calcule la liste des sous-ensembles d’un ensemble donné.

Ecrire une fonction partitions

qui calcule la liste des partitions d’un ensemble donné.

6.3.2 Solution du premier problème

Il n’est pas très difficile d’énumérer la liste des sous-ensembles d’un ensemble donné. On
a par exemple, pour l’ensemble {a, b, c}, les sous-ensembles suivants :

{ } , {a} , {b} , {c} , {a, b} , {a, c} , {b, c} , {a, b, c} .

Sur base de cet exemple, et guidé par l’ordre (pourtant non imposé) dans lequel les sous-
ensembles ont été énumérés, on pourrait juger opportun de générer séparément les sous-
ensembles de 0, de 1, de 2 et de 3 éléments. C’est parfaitement possible, mais il y a mieux
—c’est-à-dire plus simple— à faire : envisager l’utilisation directe d’un schéma récursif.
On représentera naturellement un ensemble par une liste sans répétition.75 Pour ce faire,
on se demande comment la liste des sous-ensembles de {a, b, c}, par exemple, pourrait être
construite au départ de la liste des sous-ensembles de {b, c}, c’est-à-dire

{ } , {b} , {c} , {b, c} .

On observe d’abord que les sous-ensembles de {b, c} sont aussi des sous-ensembles de
{a, b, c}, mais que la réciproque n’est pas vraie; les sous-ensembles manquants sont

{a} , {a, b} , {a, c} , {a, b, c} .

75Les ensembles {a, b} et {b, a} sont en fait le même ensemble, tandis que les deux listes (a b) et (b a)

sont distinctes.

On observe ensuite que les sous-ensembles nouveaux sont les anciens dans lesquels on a
inséré l’élément nouveau a.

Soulignons avec insistance que ces deux observations elles-mêmes sont des évidences;
le seul moyen de “passer à côté” est d’omettre d’envisager l’usage des schémas récursifs.
Notons enfin que le cas de base est évident : l’ensemble vide admet un seul sous-ensemble,
lui-même. On obtient immédiatement le code suivant :

(define subsets

(lambda (e)

(if (null? e)

’(())

(append (subsets (cdr e))

(insert_in_all (car e) (subsets (cdr e)))))))

(define insert_in_all

(lambda (x le)

(if (null? le)

’()

(cons (cons x (car le))

(insert_in_all x (cdr le))))))

La fonction auxiliaire insert_in_all prend comme arguments un objet x et une liste de
listes ll. Elle renvoie une liste de listes, dont les éléments sont ceux de ll préfixés de x.

Cette “solution” est correcte mais très inefficace : l’appel (subsets e), quand e n’est
pas vide, provoque deux appels récursifs à (subsets (cdr e)). On y remédie comme
suit :

(define subsets

(lambda (e)

(if (null? e)

’(())

((lambda (le)

(append le (insert_in_all (car e) le)))

(subsets (cdr e))))))

Pour évaluer (subsets e), on évalue une fois (subsets (cdr e)) et on lui applique
la fonction anonyme définie par la λ-forme. On pourrait choisir de donner un nom,
par exemple expand, à cette fonction, à condition d’ajouter en argument supplémentaire
l’élément nouveau, valeur de (car e). On obtient alors la variante suivante :

(define subsets

(lambda (e)

(if (null? e)

’(())

(expand (car e) (subsets (cdr e))))))

(define expand

(lambda (x le)

(append le (insert_in_all x le))))

Remarque. Cette technique consistant à introduire une fonction auxiliaire, souvent
anonyme, dans le seul but d’utiliser plus d’une fois un résultat intermédiaire, est d’usage
très fréquent. Comme nous l’avons déjà signalé au paragraphe 6.1.3, la forme spéciale de
mot-clef let a été introduite à cet usage dans le langage Scheme; dans le cas présent, on
aurait pu écrire

(define subsets

(lambda (e)

(if (null? e)

’(())

(let ((le (subsets (cdr e))))

(append le (insert_in_all (car e) le))))))

6.3.3 Solution du second problème

La fonction partitions est a priori plus difficile à écrire parce que la structure de
l’ensemble des partitions d’un ensemble donné est moins apparente que celle de l’ensemble
des sous-ensembles. Rappelons qu’une partition d’un ensemble E est simplement un
partage de cet ensemble, c’est-à-dire une famille de sous-ensembles non vides de E, disjoints
deux à deux et dont la réunion forme E. Les cinq partitions de {a, b, c} sont

{{a}, {b}, {c}} , {{a}, {b, c}} , {{b}, {a, c}} , {{c}, {a, b}} , {{a, b, c}} .

La bonne question est, à nouveau, comment obtenir les partitions de {a, b, c} au départ
de celles de {b, c}. Il faut penser à se poser cette question;76 la résoudre est simple, en
principe du moins. Les partitions de {b, c} sont

{{b}, {c}} , {{b, c}} .

On observe qu’une partition de {a, b, c} est obtenue au départ d’une partition de {b, c}
selon deux techniques :

1. En insérant le singleton {a} comme partie supplémentaire;
{{b}, {c}} donne {{a}, {b}, {c}};
{{b, c}} donne {{a}, {b, c}}.

2. En insérant l’élément a dans une partie existante;
{{b}, {c}} donne {{a, b}, {c}} et {{b}, {a, c}};
{{b, c}} donne {{a, b, c}}.

76Du moins au début; pour l’habitué, se poser cette question est un réflexe !

On obtient aisément le code suivant :

(define partitions

(lambda (e)

(if (null? e)

’(())

(append (procede_1 (car e) (partitions (cdr e)))

(procede_2 (car e) (partitions (cdr e)))))))

qui, pour la même raison d’efficacité que précédemment, est remplacé par

(define partitions

(lambda (e)

(if (null? e)

’(())

((lambda (lp)

(append (procede_1 (car e) lp)

(procede_2 (car e) lp)))

(partitions (cdr e))))))

Observons que l’ensemble vide admet une seule partition; cette partition ne comporte
aucune partie et est donc l’ensemble vide lui-même.77

La première fonction auxiliaire est immédiate car il suffit de réutiliser la fonction
insert_in_all introduite plus haut; on a

(define procede_1

(lambda (x lp)

(insert_in_all (list x) lp)))

Pour la seconde fonction auxiliaire, il y a aussi possibilité de réutilisation. En effet, on doit,
pour chaque partition de lp, appliquer le procédé 2, ce qui donne une liste de partitions,
puis concaténer toutes ces listes. On a donc

(define procede_2

(lambda (x lp)

(append_map (lambda (p) (split x p)) lp)))

La fonction append_map apparâıt dans la première étude. La fonction split réalise le
procédé 2 proprement dit, pour une partition. Comme toujours lorsque l’on spécifie une
fonction auxiliaire, il convient de le faire de la manière la plus générale possible. Le
second argument ne sera donc pas nécessairement une partition, mais une quelconque liste
de listes. On aura non seulement

(split ’a ’((b) (c))) ==> (((a b) (c)) ((b) (a c)))

(split ’a ’((b c))) ==> (((a b c)))

77La définition d’une partition spécifie que les parties éventuelles qui la composent sont des ensembles
non vides, mais ne spécifie pas qu’il doit y avoir au moins une partie.

mais aussi, par exemple,

(split ’2 ’((1 2) () (3)))

==> (((2 1 2) () (3)) ((1 2) (2) (3)) ((1 2) () (2 3)))

La construction de la fonction split n’est pas immédiate . . . mais elle le devient si
nous employons la tactique habituelle : comment obtient-on (split x ll) à partir de
(split x (cdr ll)) ? Un exemple est toujours éclairant :

(split ’0 ’(() (3))) ;; (split x (cdr ll))

==> (((0) (3)) (() (0 3)))

(split ’0 ’((1 2) () (3))) ;; (split x ll)

==> (((0 1 2) () (3)) ((1 2) (0) (3)) ((1 2) () (0 3)))

Le premier élément du résultat est à créer de toutes pièces; c’est [[(cons (cons x (car ll)) (cdr ll))]].
Le reste s’obtient en remplaçant dans [[(split x (cdr ll))]] (liste de listes de listes)
chaque élément [[ss]] (liste de listes) par [[(cons (car ll) ss)]]. Le code est maintenant
immédiat :

(define split

(lambda (x ll)

(if (null? ll)

’()

(cons (cons (cons x (car ll)) (cdr ll))

(map (lambda (ss) (cons (car ll) ss))

(split x (cdr ll)))))))

On peut à présent utiliser la fonction partition :

(partitions ’(a b c)) ==>

(((a) (b) (c)) ((a) (b c)) ((a b) (c)) ((b) (a c)) ((a b c)))

Réduire le cas d’une liste non vide l au cas de (cdr l) est l’essentiel du travail
d’application du schéma de récursion, mais résoudre le cas de la liste vide est tout aussi
important. “Si la base s’effondre, le sommet ne restera pas inébranlable”. Ce morceau
de sagesse populaire est d’application ici; le lecteur peut s’en rendre compte en essayant
la “variante” de partitions dans laquelle le résultat pour le cas vide — l’expression
’(())— aurait été remplacé par l’expression ’(). Une erreur de ce type peut anéantir
tout un développement.

6.3.4 Approche descendante, approche ascendante

Nous avons fait usage du style de développement “de haut en bas” (ou “top-down”).
On a écrit d’abord la fonction principale partitions; cela nous a amené à spécifier et
à utiliser deux sous-fonctions (fonctions auxiliaires), nommées procede_1 et procede_2.
On a ensuite écrit ces fonctions, ce qui nous a amené à spécifier et à utiliser d’autres

sous-sous-fonctions. Nous privilégions ce style dans tout l’ouvrage, ce qui revient à écrire
une fonction principale avant les fonctions auxiliaires que la fonction principale utilise.
L’intérêt du style “top-down” est clair : les fonctions auxiliaires sont introduites en cas de
besoin seulement, et spécifiées en fonction de ces besoins.

Le style opposé, “de bas en haut” (ou “bottom-up”) expose au risque d’écrire
des fonctions auxiliaires inutiles ou mal adaptées, sur lesquelles on devrait revenir
ultérieurement, après avoir écrit des fonctions les utilisant. Ce style présente néanmoins
deux avantages. Quand le développeur n’est pas directement influencé par un besoin précis,
il écrira souvent des fonctions auxiliaires plus générales, voire toute une bibliothèque de
fonctions auxiliaires qui seront réutilisées intensivement.

Le compromis que nous adoptons est le suivant. On utilise le style “top-down”,
mais en veillant à généraliser les fonctions auxiliaires et à bien les documenter, ce qui
favorise la réutilisation de ces fonctions. Dans le programme partition, nous récupérons
ainsi append_map et insert_in_all, qui avaient été spécifiées et écrites dans un autre
contexte. Nous avons veillé aussi à fournir de split une version plus générale que
strictement nécessaire (surtout au point de vue de sa spécification), de manière à favoriser
une réutilisation éventuelle.

6.4 Quatrième étude

Les trois études que nous avons présentées sont éclairantes mais peu représentatives
des problèmes réels, c’est-à-dire des problèmes auxquels l’informaticien est généralement
confronté. Considérons à présent un problème réel, induit par le petit discours suivant.

Je souhaite emprunter de l’argent, pour acheter une maison et une voiture. J’ai
contacté divers organismes prêteurs qui m’ont proposé différentes combinaisons
de délais et de taux; d’autres n’annoncent pas de taux mais directement le
montant de la mensualité. Dans les rares cas où le taux et la mensualité
étaient annoncés, j’ai recalculé la mensualité moi-même . . . et abouti à un
montant inférieur à celui exigé. Curieusement, la différence tend à être plus
importante pour les taux “voiture” que pour les taux “maisons”. D’où viennent
ces divergences systématiquement en ma défaveur ? Comment puis-je vérifier,
et comparer différentes propositions ?

6.4.1 Clarifier le problème

L’informaticien doit d’abord transformer ce discours78 en l’énoncé d’un problème clair.
Pour cela, il doit d’abord connâıtre les différentes méthodes de calcul; il pourra ensuite
les programmer, puis créer des programmes de comparaison permettant de confronter les
conditions de deux organismes utilisant des méthodes différentes. Le point de départ de
notre démarche est un rappel succinct de la théorie classique des intérêts composés.

78qui a été effectivement tenu à l’auteur, et qui a donné lieu au petit travail décrit dans la suite de ce
paragraphe.

La notion de taux d’intérêt est centrale entre prêteurs et emprunteurs. Le système
habituellement admis est simple. L’emprunteur reçoit du prêteur une somme S et s’engage
à la rembourser, accrue des intérêts, sous forme de “périodicités” (annuités ou mensualités)
en général constantes. On fixe un taux t (souvent exprimé en pourcentage; 0.5% par mois
signifie un taux mensuel de 0.005) et un nombre de périodes n. Supposons pour fixer les
idées que la période est le mois et que le taux fixé est mensuel. Déterminer le montant M
de la mensualité constante n’est pas très difficile. On observe d’abord qu’une somme S,
au terme d’un nombre n de mois, vaudra

S′ = S(1 + t)n ; (ic1)

c’est la formule classique des intérêts composés, qui donne la somme à rembourser par
l’emprunteur si le remboursement s’opère en un seul versement, à l’échéance. Dans le cas
de remboursements mensuels constants, la formule devient

S′ = M(1 + t)n−1 + M(1 + t)n−2 + · · · + M(1 + t) + M ; (ic2)

le ième terme M(1 + t)n−i représente la valeur à l’échéance (au terme du nième mois) de
la mensualité M payée au terme du ième mois, et qui s’est donc valorisée pendant (n−i)
mois. On utilise la formule bien connue

n−1
∑

i=0

bi =
bn − 1

b − 1
,

où b = 1+t, pour effectuer la somme des valorisations des n mensualités et, par élimination
de S′ entre ic1 et ic2, on tire

S(1 + t)n = M
(1 + t)n − 1

t
(ic3)

ou encore

M =
St(1 + t)n

(1 + t)n − 1
(ic4)

qui permet le calcul de la mensualité constante. Le programme Scheme correspondant à
l’égalité ic4 est élémentaire; il s’écrit

(define periodicite

(lambda (S t n)

(/ (* S t (expt (+ 1 t) n))

(- (expt (+ 1 t) n) 1))))

La valeur [[(periodicite S t n)]] donne le montant périodique constant à rembourser,
pour les données S, t et n.79

79Dans certains systèmes anciens, l’identificateur t est assimilé au booléen #t; il est alors nécessaire de
choisir un autre nom de variable.

Remarque. Ce programme comporte un facteur d’inefficacité : il prévoit que l’expression
(1+t)n sera calculée deux fois. Cela n’a rien de catastrophique, mais on pourrait y remédier
en utilisant une fonction auxiliaire anonyme. On aurait alors la variante de gauche ou, en
anticipant sur l’introduction de la forme spéciale let, celle de droite :

(define periodicite (define periodicite

(lambda (S t n) (lambda (S t n)

((lambda (aux) (let ((aux

(/ (* S t aux) (expt (+ 1 t) n)))

(- aux 1))) (/ (* S t aux)

(expt (+ 1 t) n)))) (- aux 1)))))

La fonction periodicite permet de calculer la mensualité en fonction de la somme à
emprunter, du taux d’intérêt mensuel et de la durée du prêt en mois; elle ne permet pas
de calculer directement, par exemple, le taux en fonction de la somme empruntée, de la
mensualité et du nombre de mois. En pratique, l’emprunteur qui connâıt la mensualité
requise, ou sa capacité maximale de remboursement, peut se poser les trois questions
suivantes :

Etant donné que ma capacité de remboursement mensuel est de M euros,

• à quel taux mensuel maximal t puis-je emprunter la somme S, remboursable en n
mois ?

• quelle somme maximale puis-je emprunter au taux mensuel t, pour n mois ?
• en combien de mois minimum puis-je rembourser la somme S empruntée au taux

mensuel t ?

Quoique ces questions ne couvrent pas l’intégralité du discours perplexe qui nous a été
soumis, elles constituent un bon point de départ et notre premier objectif sera de les
programmer.

6.4.2 Inversion d’une fonction réelle

Les trois fonctions à programmer sont visiblement les inverses de la fonction periodicite

par rapport à chacun de ses trois arguments. On voit la nécessité de pouvoir inverser
une fonction de plusieurs variables et il sera donc indiqué de spécifier ce sous-problème
d’inversion puis d’en programmer la solution. Il sera clairement plus économique de poser
le problème dans un cadre général; cela permettra de développer un seul programme, qui
sera utilisé pour répondre aux trois questions ci-dessus, et peut-être à beaucoup d’autres.

Le problème de l’inversion d’une fonction réelle de plusieurs variables réelles suscite
d’emblée deux questions : comment inverse-t-on une fonction et comment tient-on compte
des arguments non concernés par l’inversion. Inverser la fonction

f : R
3 → R : (x1, x2, x3) 7→ f(x1, x2, x3)

par rapport à x2 consiste à construire une fonction

g : R
3 → R : (x1, u, x3) 7→ g(x1, u, x3)

telle que
g(x1, f(x1, x2, x3), x3) = x2 et f(x1, g(x1, u, x3), x3) = u ,

pour toutes valeurs adéquates de x2 et u. Le problème est mathématiquement difficile,
puisque l’inverse n’existe pas toujours, mais devient plus simple dans le cas où la fonction
à inverser est continue et strictement monotone (croissante ou décroissante) par rapport
à l’argument (ici x2) sur lequel porte l’inversion. On va aussi admettre que, comme dans
l’exemple qui nous occupe, tous les arguments sont strictement positifs.

On devrait en principe écrire un opérateur d’inversion distinct pour chaque nombre n
d’arguments et pour chaque rang i ∈ {1, . . . , n} de l’argument sur lequel porte l’inversion,
ce qui semble tout à fait impraticable. Rien ne nous empêche, d’une part, de permuter
les arguments de sorte que l’argument x sur lequel porte l’inversion soit toujours le
premier et, d’autre part, de regrouper les autres arguments en une liste ℓ. Supposons
croissante en x la fonction (x, ℓ) 7→ f(x, ℓ). Comment déterminer la fonction inverse
(u, ℓ) 7→ g(u, ℓ) ? La méthode näıve, mais raisonnablement efficace, consiste à calculer
g(u, ℓ) par approximations successives : on crée une suite (x0, x1, . . .) de nombres telle
que chaque terme approxime g(u, ℓ), l’approximation devenant satisfaisante à partir d’un
certain rang. L’idée de base est que, si f(xn, ℓ) excède u, alors xn excède g(u, ℓ) et on
choisira xn+1 < xn; si u excède f(xn, ℓ), on choisira xn+1 > xn. Cette idée demande à
être précisée. D’une part, il faut préciser la manière dont on choisit les xi successifs et,
d’autre part, il faut choisir une condition d’arrêt. La technique de la bissection consiste à
maintenir un intervalle [m,M ] dans lequel la valeur recherchée se trouve, et à rétrécir cet
intervalle à chaque itération. Au départ, l’intervalle est très grand, par exemple m = 10−6

et M = 107. A chaque étape, on calcule la moyenne µ des bornes de l’intervalle puis
la valeur f(µ, ℓ). En fonction de cette valeur, on décide de s’arrêter (si f(µ, ℓ) = u)80

ou de continuer soit avec l’intervalle [m,µ], soit avec l’intervalle [µ,M ]. Chaque étape
a pour effet de réduire de moitié la longueur de l’intervalle et on peut s’arrêter dès que
cette longueur devient suffisamment petite. On introduit une constante ε telle que l’arrêt
survient quand la longueur de l’intervalle descend en dessous de ε.
Remarque. La moyenne la plus fréquemment utilisée est la moyenne arithmétique mais on
a parfois intérêt, lorsque tous les nombres impliqués sont strictement positifs, à préférer
la moyenne géométrique; c’est le cas ici.81

On définit d’abord les trois variables globales spécifiant respectivement la borne
inférieure de l’intervalle initial, sa borne supérieure et le seuil de tolérance; on définit
aussi une fonction calculant la moyenne (géométrique) de deux nombres :

80Tester une égalité de ce type n’a guère de sens; il serait préférable de tester une “proximité”; nous
n’entrons pas ici dans des considérations du ressort de l’analyse numérique.

81Rappelons que la moyenne arithmétique de deux nombres a et b est a+b

2
; si a et b sont strictement

positifs, leur moyenne géométrique est
√

ab.



6 CONCEPTION DE PROGRAMME 120

(define *min* 1.e-6)

(define *max* 1.e+7)

(define *eps* 1.e-12)

(define mu (lambda (a b) (sqrt (* a b))))

On peut alors programmer l’inversion proprement dite. Nous considérons ici le cas de
fonctions croissantes :

(define inv+

(lambda (f)

(lambda (u l) (i+ f u l *min* *max* *eps*))))

(define i+

(lambda (f u l x0 x1 eps)

(cond ((prox x0 x1 eps) (mu x0 x1))

((prox (f (mu x0 x1) l) u eps) (mu x0 x1))

((< (f (mu x0 x1) l) u) (i+ f u l (mu x0 x1) x1 eps))

(else (i+ f u l x0 (mu x0 x1) eps)))))

(define prox (lambda (a b eps) (< (abs (- a b)) eps)))

L’opérateur inv+ prend comme argument une fonction f croissante en son premier
argument et renvoie la fonction inverse. Elle utilise les fonctions auxiliaires prox et i+,
qu’il convient de spécifier. Pour simplifier les écritures, on notera la valeur d’un objet
Scheme par le nom de cet objet écrit en italique; pour la valeur d’une combinaison, on
utilisera la notation mathématique habituelle.82 Le prédicat prox prend comme arguments
deux réels a et b et un réel positif ε; il renvoie vrai si |a − b| < ε. En ce qui concerne i+,
on a

Si la fonction f : (R × R
k) → R est croissante en son premier argument et si

l’unique solution x de l’équation f(x, ℓ) = u appartient à l’intervalle [x0 : x1],
alors i+(f, u, ℓ, x0, x1, ε) est un nombre x′ proche de x, au sens que x′ ne s’écarte
pas de x de plus de ε/2 ou que f(x′, ℓ) ne s’écarte pas de u de plus de ε.

On définit de manière analogue un opérateur inv- pour inverser les fonctions décroissantes,
qui fera appel à l’opérateur auxiliaire i-; ce dernier ne diffère de i+ que par la permutation
des comparateurs < et > dans les conditions des clauses contenant les appels récursifs. Un
dernier point intéressant consiste à modifier i+ de manière à éviter l’évaluation multiple des
expressions (mu x0 x1) et (f (mu x0 x1) l). Nous utilisons la technique déjà employée
pour la variante de la fonction periodicite; cela permet d’abord d’éviter l’évaluation
multiple de (mu x0 x1) :

(define i+

(lambda (f u l x0 x1 eps)

82On écrira donc, par exemple, h(a, (x + y)/2) au lieu de [[(h a (/ (+ x y) 2))]].

((lambda (aux1)

(cond ((prox x0 x1 eps) aux1)

((prox (f aux1 l) u eps) aux1)

((< (f aux1 l) u) (i+ f u l aux1 x1 eps))

(else (i+ f u l x0 aux1 eps))))

(mu x0 x1))))

On peut réutiliser la même technique pour éviter la double évaluation de l’expression
(f aux1 l); on obtient ainsi :

(define i+

(lambda (f u l x0 x1 eps)

((lambda (aux1)

((lambda (aux2)

(cond ((prox x0 x1 eps) aux1)

((prox aux2 u eps) aux1)

((< aux2 u) (i+ f u l aux1 x1 eps))

(else (i+ f u l x0 aux1 eps))))

(f aux1 l)))

(mu x0 x1))))

Ici aussi, la construction let pourra être utilisée.

6.4.3 Solution du problème simplifié

Nous pouvons maintenant répondre aux trois questions que peut se poser l’emprunteur
connaissant sa capacité mensuelle de remboursement, sous l’hypothèse simplificatrice que
la méthode de calcul correcte est d’application; cela revient à programmer les inverses
somme, taux et nombre_periodes de la fonction periodicite par rapport à ses arguments
S, t et n, respectivement. On rappelle d’abord le code de la fonction periodicite :

(define periodicite

(lambda (S t n)

(/ (* S t (expt (+ 1 t) n))

(- (expt (+ 1 t) n) 1))))

Les trois programmes d’inversion sont construits de la même manière :

(define periodicite<-somme

(lambda (S tn) (periodicite S (car tn) (cadr tn))))

(define somme<-periodicite (inv+ periodicite<-somme))

(define somme

(lambda (M t n) (somme<-periodicite M (list t n))))

(define periodicite<-taux

(lambda (t Sn) (periodicite (car Sn) t (cadr Sn))))

(define taux<-periodicite (inv+ periodicite<-taux))

(define taux

(lambda (S M n) (taux<-periodicite M (list S n))))

(define periodicite<-nombre_periodes

(lambda (n St) (periodicite (car St) (cadr St) n)))

(define nombre_periodes<-periodicite

(inv- periodicite<-nombre_periodes))

(define nombre_periodes

(lambda (S t M) (nombre_periodes<-periodicite M (list S t))))

On notera la technique très simple utilisée pour regrouper en un seul argument les deux
arguments de periodicite qui ne sont pas concernés par l’inversion. On observera aussi
que la fonction periodicite est croissante en ses deux premiers arguments et décroissante
dans le troisième; on a donc utilisé inv+ dans les deux premiers cas et inv- dans le
troisième.

6.4.4 Première cause de divergence

Les programmes proposés devraient en principe résoudre le problème : ils permettent de
calculer la mensualité en fonction du taux mensuel et réciproquement. Néanmoins, ces
programmes élémentaires n’expliquent en rien les nombreuses divergences défavorables
entre les résultats fournis par ces programmes et les conditions proposées par les organismes
prêteurs.

Pour expliquer ces divergences, il a fallu connâıtre les techniques de calcul utilisées.
Elles consistent, d’une part, à “simplifier” le rapport existant entre les données annuelles
et les données mensuelles (technique dite “mois / année”)83 et, d’autre part, à “simplifier”
la formule de calcul des mensualités (technique dite “du taux de chargement”); seule la
première sera étudiée dans ce paragraphe. La technique “mois / année” se pratique de deux
manières. La première consiste à considérer que le taux mensuel (utilisé) est le douzième
du taux annuel (annoncé); c’est systématiquement défavorable à l’emprunteur car on a

(1 + ta) = (1 + tm)12 > 1 + 12tm ,

d’où l’on conclut immédiatement que le taux mensuel correspondant à un taux annuel
donné est moindre que le douzième de ce taux annoncé. Les passages du taux annuel au
taux mensuel (exact) et réciproquement se programment aisément :

(define tm<-ta (lambda (ta) (- (expt (+ ta 1) 1/12) 1)))

(define ta<-tm (lambda (tm) (- (expt (+ tm 1) 12) 1)))

83En profitant du fait que les remboursements sont souvent mensuels alors que le taux annoncé est
annuel.

L’organisme prêteur qui applique un taux mensuel effectif égal au douzième de son taux
annuel nominal utilise en fait un taux annuel effectif supérieur au taux nominal annoncé;
pour savoir quel taux annuel lui sera réellement appliqué, le client pourra utiliser le
programme suivant :

(define ta<-ta_1 (lambda (t) (ta<-tm (/ t 12))))

(ta<-ta_1 0.06) ==> .0616778

On voit par exemple qu’un taux nominal de 6% correspond à un taux réel de 6.168 %.

La seconde manière d’appliquer la technique “mois / année” consiste à calculer l’annuité
sur base du taux annuel annoncé, puis à considérer que la mensualité est le douzième de
l’annuité. A nouveau, ceci est clairement désavantageux pour l’emprunteur, qui rembourse
chaque mensualité avec une avance de 1 à 11 mois, c’est-à-dire avec, en moyenne, cinq
mois et demi d’avance. Ici aussi, le client souhaitera connâıtre le taux réel correspondant
au taux nominal annoncé par l’organisme pratiquant cette méthode. On peut obtenir
le programme adéquat ta<-ta_2 en calculant successivement, au départ du taux annuel
nominal, l’annuité, la mensualité, le taux mensuel réel et le taux annuel réel; autrement
dit, ta<-ta_2 pourrait se réduire à la composition des fonctions que l’on appellerait
naturellement annu<-ta_2, mensu<-annu, tm<-mensu et ta<-tm. Il est tentant de réaliser
ces compositions au moyen de la fonction compose_list introduite au paragraphe 6.2.2
mais un problème technique apparâıt : certaines fonctions à composer comportent plus
d’un arguments, les valeurs retournées dépendant non seulement du taux mais aussi de la
somme empruntée et de la durée du prêt.

Une première solution consiste à supprimer artificiellement ces arguments
surnuméraires en les remplaçant par des variables globales *S* (somme empruntée) et *n*
(durée du prêt en années).84 En réutilisant diverses fonctions précédemment introduites,
on a :

(define annu<-ta_2 (lambda (t) (periodicite *S* t *n*)))

(define mensu<-annu (lambda (a) (/ a 12.0)))

(define tm<-mensu (lambda (M) (taux *S* M (* 12 *n*))))

(define ta<-tm (lambda (tm) (- (expt (+ tm 1) 12) 1)))

(define ta<-ta_2

(compose_list (list ta<-tm tm<-mensu mensu<-annu annu<-ta_2)))

A titre d’illustration, nous considérons le cas d’un taux annuel nominal de 6% et d’une
somme d’un million, à rembourser en 20, 10 ou 3 ans.

84Rappelons ici la convention introduite au paragraphe 2.7.4 : les identificateurs correspondant à des
variables globales définies par l’utilisateur commencent et se terminent par le caractère “*”.

(define *S* 1000000) ==> ...

(define *n* 20) ==> ...

(ta<-ta_2 .06) ==> .063523

(define *n* 10) ==> ...

(ta<-ta_2 .06) ==> .066294

(define *n* 3) ==> ...

(ta<-ta_2 .06) ==> .079255

On constate immédiatement que cette deuxième manière d’appliquer la méthode
“mois / année” est encore plus défavorable à l’emprunteur que la première manière, surtout
si la durée du prêt est faible.

L’usage des variables globales est à éviter car elles nuisent à la modularité et à
la transparence du code. Il est tout à fait contraire à l’esprit de la programmation
fonctionnelle de permettre que la valeur [[(ta<-ta_2 t)]] d’une combinaison ne soit pas
entièrement déterminée par les valeurs [[ta<-ta_2]] et [[t]] de l’opérateur et de l’opérande.
Il est donc impératif d’éliminer *n* mais on peut se permettre de conserver *S* car la
valeur [[(ta<-ta_2 t)]] est en fait indépendante de *S*. Par contre, cette dernière valeur
influe sur les résultats retournés par certaines fonctions auxiliaires; cela sera toutefois sans
inconvénient si ces fonctions sont “cachées”. Contrairement à ce que l’on pourrait croire,
le fait de réintroduire la durée du prêt comme argument explicite de certaines fonctions
ne nous empêche pas de continuer à utiliser compose_list. On a le code suivant :

(define make_ta<-ta_2

(lambda (n)

(compose_list

(list (lambda (tm) (- (expt (+ tm 1) 12) 1)) ; ta<-tm

(lambda (M) (taux *S* M (* 12 n))) ; tm<-mensu

(lambda (a) (/ a 12.0)) ; mensu<-annu

(lambda (t) (periodicite *S* t n)))))) ; annu<-ta_2

On notera que make_ta<-ta_2 est une fonction qui à tout nombre positif (représentant
la durée du prêt) associe une fonction équivalente à ta<-ta_2 donnée plus haut. En
reprenant les mêmes exemples que précédemment, on obtient

(define *S* 1234567) ==> ... ;; somme quelconque

((make_ta<-ta_2 20) .06) ==> .063523

((make_ta<-ta_2 10) .06) ==> .066294

((make_ta<-ta_2 3) .06) ==> .079255

La technique consistant à écrire un “générateur de fonction” permet d’éviter les variables
globales gênantes sans que l’on doive modifier le nombre d’arguments des fonctions
impliquées. Nous aurons encore l’occasion d’écrire des générateurs de fonctions.

6.4.5 Deuxième cause de divergence

Le développement qui précède rend compte des divergences observées dans le cas des prêts
immobiliers mais pas de celles, généralement plus graves, observées dans le cas des prêts à
la consommation. A nouveau, une petite enquête permet de mettre en évidence le moyen
astucieux qu’utilisent certains organismes prêteurs pour annoncer des taux avantageux. Il
est bien clair que l’emprunteur d’une somme S remboursera, le plus souvent en plusieurs
fois, une somme totale S′ supérieure à S. La différence S′ − S est une mesure du prix
du crédit. On définit le taux de chargement comme le quotient tc =def (S′ − S)/nS.85

L’emprunteur d’une somme S remboursable en n mois au taux de chargement mensuel
tc rembourse chaque mois la somme M = S′/n = (1/n + tc)S. La notion de taux de
chargement ressemble à celle de taux d’intérêt. Plus le crédit est cher, plus le taux de
chargement est élevé et un crédit gratuit correspond à un taux de chargement nul. En
outre, le taux de chargement se confond avec le taux d’intérêt dans deux cas extrêmes.
Le premier est celui d’un prêt de durée unitaire : il y a donc un seul remboursement de
S′ = (1+tm)S = (1+tc)S; le second est celui d’un prêt perpétuel, dans lequel l’emprunteur
paie uniquement les intérêts, soit un versement mensuel permanent, égal à Stm ou Stc.
Par contre, pour toute durée n bornée plus grande que 1, un taux de chargement donné
correspond à un taux d’intérêt mensuel supérieur, la différence dépendant de la durée
du prêt. En réutilisant des fonctions antérieurement définies on obtient aisément les
fonctions permettant de calculer la mensualité à partir de la somme à emprunter, du
taux de chargement ou du taux d’intérêt et de la durée du prêt en mois :

(define mensualite<-tm periodicite)

(define tm<-mensualite taux)

(define mensualite<-tc

(lambda (S tc n) (* (+ (/ 1 n) tc) S)))

(define tc<-mensualite

(lambda (S M n) (- (/ M S) (/ 1 n))))

On en tire immédiatement les fonctions de conversion :

(define tc<-tm

(lambda (tm n)

(tc<-mensualite *S* (mensualite<-tm *S* tm n) n)))

(define tm<-tc

(lambda (tc n)

(tm<-mensualite *S* (mensualite<-tc *S* tc n) n)))

85Seul le taux de chargement mensuel semble être utilisé en pratique.

On calcule d’abord quel taux d’intérêt mensuel correspond à un taux de chargement de
0.5 %, pour des durées de prêt de 1, 12, 30 et 240 mois; on obtient

(tm<-tc 0.005 1) ==> 0.005000

(tm<-tc 0.005 12) ==> 0.009080

(tm<-tc 0.005 30) ==> 0.009265

(tm<-tc 0.005 240) ==> 0.007719

On calcule de même quel taux de chargement correspond à un taux d’intérêt mensuel de
0.5 % :

(tc<-tm 0.005 1) ==> 0.005000

(tc<-tm 0.005 12) ==> 0.002733

(tc<-tm 0.005 30) ==> 0.002646

(tc<-tm 0.005 240) ==> 0.002998

On voit que la confusion entre les deux notions de taux peut coûter cher !

7 Accumulateurs et processus itératifs

On a montré au paragraphe 4.5 comment le programme fib, inspiré directement de la
définition récursive de la suite de Fibonacci, était trop inefficace en pratique; le programme
fib_a, introduit dans le même paragraphe, permettait d’éliminer cet inconvénient, grâce
à deux arguments supplémentaires, destinés à mémoriser temporairement des résultats
intermédiaires utiles. Nous avions aussi noté que, dans le programme fib_a, la récursivité
intervenait sous forme dite dégénérée ou terminale. Le processus de calcul présente dans
le cas de la récursivité dégénérée une structure très simple; on dit qu’il est itératif. Dans
ce chapitre, nous étudions ces notions de manière plus approfondie.

7.1 Le principe de l’accumulateur

Un accumulateur est un argument supplémentaire, lié à des résultats intermédiaires à
mémoriser. La notion de résultat intermédiaire est à prendre au sens le plus large : il
s’agit de toute valeur dont la connaissance est mise à profit par le processus de calcul pour
déterminer le résultat final.

Considérons une fois de plus le processus de calcul lié à la fonction classique fact

(§ 4.3), ou plutôt sa simulation par le modèle de substitution :

(fact 4)(* 4 (fat 3))(* 4 (* 3 (fat 2)))(* 4 (* 3 (* 2 (fat 1))))(* 4 (* 3 (* 2 (* 1 (fat 0)))))(* 4 (* 3 (* 2 (* 1 1))))(* 4 (* 3 2))(* 4 6)
==> 24

On voit que ce processus comporte une première phase d’expansion, pendant laquelle
aucune multiplication n’est effectuée, suivie d’une phase de réduction, consistant
essentiellement en l’évaluation des multiplications. Vu l’associativité de la multiplication,
le processus

(fact 4)(* 4 (fat 3))(* (* 4 3) (fat 2))(* (* 4 3 2) (fat 1))(* (* 4 3 2 1) (fat 0))(* 4 3 2 1)
==> 24

serait équivalent. Il peut se simplifier en(* 1 (fat 4))(* 4 (fat 3))

L’intérêt est que chaque état (ou étape) du processus est caractérisé par deux paramètres
seulement. Cela représente, en espace mémoire, une nette économie par rapport au
processus initial. Etant donné un programme, il est simple, en principe, d’étudier le
processus de calcul associé : il suffit de bien connâıtre l’ensemble des règles du langage de
programmation utilisé, ce que l’on appelle la sémantique opérationnelle du langage. La
démarche inverse, qui consiste à reconstituer le programme sur base du processus associé,
reste facile dans le cas présent. On va définir une fonction récursive à deux arguments,
nommée fact_a, telle que les appels successifs se feront avec des arguments égaux aux
valeurs successives des deux paramètres du processus. Cette idée est “naturelle” dans la
mesure où le second paramètre (qui deviendra le premier argument) évoque immédiatement
notre schéma habituel de récursion sur les nombres naturels. Le premier paramètre (qui
deviendra le second argument) répond à notre définition de l’accumulateur : ses valeurs
successives sont clairement des résultats intermédiaires. Par le modèle de substitution, on
a :

(fact_a 4 1)(fat a 3 4)(fat a 2 12)(fat a 1 24)(fat a 0 24) ==> 24

La définition de fact_a est maintenant évidente :86

(define fact_a

(lambda (n a)

(if (zero? n) a (fact_a (- n 1) (* a n)))))

On peut à présent redéfinir la procédure fact comme suit :

(define fact

(lambda (n) (fact_a n 1)))

La fonction fact_a est devenue une fonction auxiliaire; à ce titre, il est indispensable de
la spécifier, ce que l’on fera de la manière suivante :

La fonction fact_a prend comme arguments
un entier naturel n et un nombre a;
elle renvoie comme résultat le produit n!a.

86Observons que le processus associé à la boucle while n > 0 do (a,n) := (a*n,n-1) (construction
classique de nombreux langages de programmation) est analogue à celui associé à l’évaluation de la forme
(fact a n a).

Signalons d’emblée qu’une phrase telle que “fact_a est une version accumulante de fact”
est trop vague pour être d’une quelconque utilité.87

7.2 Autres exemples numériques

La solution “accumulante” que nous venons de commenter n’est pas la seule possible
pour le calcul de la factorielle. Plutôt que de considérer des résultats partiels du type
n ∗ (n−1) ∗ · · · , on pourrait accumuler des produits du type 1 ∗ 2 ∗ · · · . Une solution
alternative consiste en la fonction fact_b, spécifiée comme suit :

Si les valeurs de n, i et b sont n, i (1 ≤ i ≤ n + 1) et (i − 1)! ,
alors la valeur de (fact_b n i b) est n! .

On peut écrire

(define fact_b

(lambda (n i b)

(if (> i n) b (fact_b n (+ i 1) (* b i)))))

(define fact

(lambda (n) (fact_b n 1 1)))

Remarque. Notre spécification pourrait parâıtre incomplète, puisqu’elle ne précise pas
la valeur de (fact_b n i b) quand [[b]] n’est pas ([[i]]− 1)!. L’objection n’est pas
valable parce qu’en usage normal ce cas ne se produit jamais, ni au premier niveau, ni
lors d’appels récursifs subséquents. Par contre on ne pourrait se contenter de dire que
[[(fact_b n 1 1)]] = [[n]]!.

L’emploi d’un accumulateur permet de diminuer significativement l’espace mémoire
requis par l’exécution du processus de calcul de la factorielle. Un gain nettement
plus significatif est possible dans le cas de la suite de Fibonacci, que nous avons déjà
rencontré. Le code “näıf”, directement inspiré de la définition mathématique (§ 1.2.6) a
pu être nettement amélioré, en fait par l’adjonction de deux accumulateurs permettant
de mémoriser les deux derniers termes calculés de la suite de Fibonacci. Le code de la
fonction fib_a a été donné au paragraphe 4.5, ainsi que sa spécification et la description
du processus de calcul associé.88

De manière analogue, la fonction “exponentielle rapide”, dont le code a été donné au
paragraphe 5.6, admet une variante accumulante :

87Il serait encore plus vain d’écrire “La fonction fact a prend comme arguments un entier naturel n et le
nombre 1; elle renvoie comme résultat le nombre n!”. Tout d’abord, cette phrase n’est qu’une paraphrase
du code de fact : elle est donc inutile. En outre, il va de soi que seul le premier appel se fait avec a = 1;
lors des appels récursifs, on a en général a 6= 1. Enfin, le lecteur d’une telle phrase est censé en admettre la
véracité comme un acte de foi; ce n’est pas le cas pour la spécification correcte, qui se démontre aisément
par récurrence sur n. Nous aurons encore l’occasion dans la suite de montrer comment il convient de
spécifier les fonctions accumulantes.

88Ce processus est itératif et analogue à celui associé à la boucle
while n > 0 do (n,a,b) := (n-1,b,a+b).

(define exp

(lambda (m n) (exp_a m n 1)))

(define exp_a

(lambda (m n a)

(cond ((zero? n) a)

((even? n) (exp_a (* m m) (/ n 2) a))

((odd? n) (exp_a m (- n 1) (* m a))))))

On a [[(exp_a m n a)]] = [[a]][[m]][[n]] , d’où [[(exp_a m n 1)]] = [[m]][[n]] en particulier.
Considérons encore l’exemple du coefficient binomial :

(define cbin

(lambda (n u)

(if (zero? n)

1

(/ (* (cbin (- n 1) (- u 1)) u) n))))

dont voici la version accumulante :

(define cbin

(lambda (n u) (cbin_a n u 1)))

(define cbin_a

(lambda (n u a)

(if (zero? n)

a

(cbin_a (- n 1) (- u 1) (/ (* u a) n)))))

Voici un exemple d’exécution :

(cbin 4 6) ==> 15

(cbin_a 4 6 1) ==> 15

La spécification de cbin_a est laissée au lecteur.

7.3 Accumulateurs et traitement de listes

Nous avons déjà rencontré (§ 5.3) la fonction reverse :

(define reverse

(lambda (u)

(if (null? u)

’()

(append (reverse (cdr u))

(list (car u))))))

A cause de l’emploi de la fonction append (d’efficacité linéaire en son premier argument),
la fonction reverse est d’efficacité quadratique. Donnons un exemple d’application, en
utilisant le modèle de substitution :

(reverse ’(0 1 2))(append (reverse '(1 2)) '(0))(append (append (reverse '(2)) '(1)) '(0))(append (append (append (reverse '()) '(2)) '(1)) '(0))(append (append (append '() '(2)) '(1)) '(0))(append (append '(2) '(1)) '(0))(append '(2 1) '(0)) ==> (2 1 0)

On peut aisément obtenir une version accumulante linéaire :

(define reverse_a

(lambda (u a)

(if (null? u)

a

(reverse_a (cdr u) (cons (car u) a)))))

Le modèle de substitution donne :

(reverse_a ’(0 1 2) ’())(reverse a '(1 2) '(0))(reverse a '(2) '(1 0))(reverse a '() '(2 1 0)) ==> (2 1 0)

On a la spécification suivante :

Pour toutes listes u et a,
(append (reverse u) a) équivaut à (reverse_a u a) .

En particulier, (reverse u) équivaut à (reverse_a u ’()) .

Un autre exemple, déjà rencontré (§§ 5.4-5.5), est celui du calcul de la liste des feuilles
d’une liste littérale ou d’un arbre. On construit immédiatement un programme simple et
correct pour ce calcul :

(define flat_list

(lambda (l)

(cond ((null? l) ’())

((atom? (car l))

(if (null? (car l))

(flat_list (cdr l))

(cons (car l) (flat_list (cdr l)))))

(else

(append (flat_list (car l)) (flat_list (cdr l)))))))

On a par exemple

(flat_list ’(a (b c) ((((d))) e) b)) ==> (a b c d e b)

Ce programme donne cependant lieu à un processus de calcul relativement complexe, suite
à l’utilisation de la fonction auxiliaire append. Pour y remédier, on définit une fonction
auxiliaire flat_list_a, plus générale. Elle prend un second argument a, un accumulateur
destiné à contenir un résultat intermédiaire (la valeur de a sera donc une liste littérale
plate, c’est-à-dire une liste de lettres). Cette fonction auxiliaire est spécifiée par l’égalité

[[(flat_list_a l a)]] = [[(append (flat_list l) a)]] ,

valable pour toute liste littérale l et pour toute liste a. On a le code suivant :

(define flat_list_a

(lambda (l a)

(cond

((null? l) a)

((atom? (car l))

(if (null? (car l))

(flat_list_a (cdr l) a)

(cons (car l) (flat_list_a (cdr l) a))))

(else

(flat_list_a (car l) (flat_list_a (cdr l) a))))))

On a par exemple

(flat_list_a ’(a (b c) ((((d))) e) b) ’()) ==> (a b c d e b)

(flat_list_a ’(a (b (c) d)) ’(1 2 3)) ==> (a b c d 1 2 3)

De la spécification générale donnée plus haut, on tire

(flat_list l) équivaut à (flat_list_a l ’()) .

L’usage de la fonction auxiliaire flat_list_a procure un gain de temps et d’espace à
l’exécution, quoique le processus associé ne soit pas itératif.

7.4 Conception de programme, cinquième étude

Nos premières études de cas (§ 6) avaient pour but de montrer comment l’énoncé du
problème à résoudre suggérait, de manière plus ou moins directe, divers moyens de solution;
on a vu aussi comment choisir entre ces moyens, et comment organiser le travail de
conception du programme, notamment par la spécification et la définition de quelques
fonctions auxiliaires appropriées. Dans les trois premiers cas, l’usage direct ou indirect
des schémas de programme s’est révélé utile. Nous traitons ici un problème très simple
mais néanmoins susceptible de recevoir des solutions variées. On verra en particulier que
la notion de schéma de programme doit s’utiliser avec souplesse et discernement, de même
que celle d’accumulateur.

7.4.1 L’énoncé

Ecrire une fonction lpref prenant comme argument une liste l

et retournant la liste des préfixes de l.

Cette spécification est claire; il est cependant nécessaire de préciser la notion de préfixe,
en décidant par exemple que la liste vide et la liste u elle-même sont des préfixes de u.89

7.4.2 Première solution

Notre troisième étude de cas avait montré que l’usage direct du schéma de programme
concernant les listes (ici, les listes plates) pouvait conduire à une solution simple, claire et
raisonnablement efficace. Nous décidons donc d’emblée que le calcul de (pref l) passera
par celui de (pref (cdr l)). On a donc immédiatement un “squelette” de solution :

(define lpref1

(lambda (l)

(if (null? l)

C ;; cas de base

(F (lpref1 (cdr l)) l)))) ;; cas inductif

Clairement, la liste vide admet un seul préfixe, elle-même. En outre, les préfixes d’une
liste l non vide sont, d’une part, la liste vide et, d’autre part, les préfixes de (cdr l),
chacun d’eux étant préfixé de (car l). On obtient immédiatement le code suivant :

(define lpref1

(lambda (l)

(if (null? l)

(list l)

(cons ’() (insert_in_all (car l) (lpref1 (cdr l)))))))

La fonction auxiliaire insert_in_all a été introduite au paragraphe 6.3.2. Elle prend
comme arguments un objet x et une liste de listes ll; elle renvoie une liste de listes, dont
les éléments sont ceux de ll préfixés de x. On observe que les fonctions insert_in_all

et lpref1 sont des instances du schéma habituel.90 Voici un exemple d’utilisation :

(lpref1 ’(a b c)) ==> (() (a) (a b) (a b c))

Le processus associé à lpref1 n’est pas itératif, pas plus d’ailleurs que celui associé à
insert_in_all. A titre d’illustration, on va étudier la possibilité de produire une version
accumulante de ces fonctions. On rappelle d’abord le code de insert_in_all :

89Un autre choix serait également acceptable; le point important est d’expliciter ce choix.
90On aurait pu définir insert in all via la primitive map :

(define put (lambda (x ll) (map (lambda (l) (cons x l)) ll)))

(define insert_in_all

(lambda (x ll)

(if (null? ll)

’()

(cons (cons x (car ll))

(insert_in_all x (cdr ll))))))

Par analogie avec les exemples traités au début de ce chapitre, on obtient la version
suivante :

(define insert_in_all_a

(lambda (x ll a)

(if (null? ll)

a

(insert_in_all_a x

(cdr ll)

(cons (cons x (car ll)) a)))))

Voici des exemples d’exécution :

(insert_in_all ’x ’((a b) () (c)))

==> ((x a b) (x) (x c))

(insert_in_all_a ’x ’((a b) () (c)) ’())

==> ((x c) (x) (x a b))

(insert_in_all_a ’x ’((a b) () (c)) ’(1 (2) 3))

==> ((x c) (x) (x a b) 1 (2) 3)

On observe que l’ordre des éléments dans la liste est inversé. Dans le cas présent, cela
n’est pas gênant; il est cependant indispensable de mentionner ce fait dans la spécification
de insert_in_all_a, que nous laissons au lecteur comme exercice. On peut maintenant
essayer d’obtenir une variante accumulante de lpref1, en appliquant la technique déjà
utilisée pour insert_in_all. On observe d’abord que lpref1 peut se récrire en utilisant
insert_in_all_a au lieu de insert_in_all. La version näıve

(define lpref1bis

(lambda (l)

(if (null? l)

(list l)

(cons ’()

(insert_in_all_a (car l)

(lpref1bis (cdr l))

’())))))

n’est pas très satisfaisante parce qu’elle produit les préfixes dans un ordre peu
conventionnel :

(lpref1bis ’(a b c)) ==> (() (a b) (a b c) (a))

On pourrait remédier à cet inconvénient comme suit :

(define lpref1ter

(lambda (l)

(if (null? l)

(list l)

(cons ’()

(insert_in_all_b (car l)

(lpref1ter (cdr l))

’())))))

(define insert_in_all_b

(lambda (x ll a)

(if (null? ll)

a

(insert_in_all_b x

(cdr ll)

(append a (list (cons x (car ll))))))))

La fonction insert_in_all_b renvoie maintenant une liste non inversée. On a :

(insert_in_all_b ’x ’((a) () (b c)) ’())

==> ((x a) (x) (x b c))

(lpref1ter ’(a b c)) ==> (() (a) (a b) (a b c))

Cette solution n’est pas satisfaisante car la fonction insert_in_all_b est inefficace (à
cause de l’intervention de append). De plus, le processus associé à l’application de
lpref1ter à ses arguments n’est pas itératif. A ce stade, il semble préférable d’abandonner
la recherche d’une version accumulante et/ou itérative de lpref1.

7.4.3 Deuxième solution

Le lecteur aura sans doute remarqué que la liste des suffixes est plus facile à calculer que
celle des préfixes. On a immédiatement

(define lsuff

(lambda (l)

(if (null? l)

(list l)

(cons l (lsuff (cdr l))))))

(lsuff ’(a b c)) ==> ((a b c) (b c) (c) ())

Cela est dû au fait que le plus grand suffixe propre d’une liste [[l]] non vide est tout
simplement [[(cdr l)]]; ceci suggère d’écrire et d’utiliser une fonction auxiliaire butlast,
renvoyant le plus grand préfixe propre d’une liste non vide. On a alors

(define lpref2

(lambda (l)

(if (null? l)

(list l)

(cons l (lpref2 (butlast l))))))

(define butlast

(lambda (l)

(if (null? (cdr l))

’()

(cons (car l) (butlast (cdr l))))))

Ici, lpref2 n’est pas une instance du schéma habituel, car la fonction de réduction cdr

est remplacée par la fonction butlast. On notera cependant que butlast, comme cdr,
renvoie comme résultat une liste plus courte que la liste-argument; c’est ce qui détermine
le succès de l’approche récursive. Notons aussi que la fonction butlast est une instance
du schéma habituel. L’usage de butlast est un exemple de la souplesse nécessaire lors de
l’emploi d’un schéma de récursion. Un autre avantage de cette solution est la possibilité
d’obtenir immédiatement une version accumulante :

(define lpref2_a

(lambda (l a)

(if (null? l)

(cons ’() a)

(lpref2_a (butlast l) (cons l a)))))

(lpref2 ’(a b c)) ==> ((a b c) (a b) (a) ())

(lpref2_a ’(a b c) ’()) ==> (() (a) (a b) (a b c))

7.4.4 Troisième solution

On peut directement mettre à profit le fait que la liste des suffixes est facile à calculer. Il
suffit d’observer que la liste des préfixes d’une liste est la liste des inverses des suffixes de
la liste inverse. Cela donne lieu au code suivant :

(define lpref3

(lambda (l)

(lreverse (lsuff (reverse l)))))

(define lreverse

(lambda (ll)

(if (null? ll)

’()

(cons (reverse (car ll)) (lreverse (cdr ll))))))

(lpref3 ’(a b c)) ==> ((a b c) (a b) (a) ())

On laisse au lecteur le soin de spécifier la fonction lreverse; la fonction reverse est déjà
connue.91

Remarque. Les trois solutions que l’on vient d’exposer sont des exemples typiques du style
de programmation fonctionnel. Dans chaque cas, le problème est résolu par deux ou trois
fonctions très simples, le plus souvent définies récursivement, selon un schéma élémentaire.
L’utilisation de (butlast l) au lieu de l’habituel (cdr l) est une variante dictée par les
particularités du problème posé, mais on reste dans un cadre strictement fonctionnel, qui
se caractérise par trois points :

• Approche abstraite : le raisonnement n’évoque pas les étapes intermédiaires du
calcul.

• Approche procédurale, de type “top-down” : toute fonction dont la programmation
n’est pas immédiate suscite l’introduction (puis la programmation) d’une ou
plusieurs fonctions auxiliaires.

• Usage systématique de la récursivité, inséparable de l’approche fonctionnelle.

7.5 Simplification syntaxique d’une récursion

La technique des accumulateurs permet parfois de transformer un schéma récursif difficile
et/ou inhabituel en un schéma plus simple, voire dégénéré. Nous donnons ici quelques
exemples typiques.

7.5.1 La fonction 91

Considérons d’abord l’exemple classique de la “fonction 91”. Le code est

(define M

(lambda (x)

(if (> x 100) (- x 10) (M (M (+ x 11))))))

On pourrait croire que l’étude de la fonction M serait facilitée si on pouvait éliminer la
récursivité imbriquée. Syntaxiquement, c’est très simple. Il suffit d’imaginer une fonction
auxiliaire M_c, comportant un argument supplémentaire destiné à “contrôler” l’itération
et employé comme suit :

(M_c x 0) équivaut à x;

91Notons aussi que lreverse peut se définir au moyen de la primitive map :
(define lreverse (lambda (ll) (map reverse ll)))

(M_c x 1) équivaut à (M x);
(M_c x 2) équivaut à (M (M x));
(M_c x 3) équivaut à (M (M (M x)));
. . . . . .
Il est alors immédiat d’écrire le code suivant :

(define M_c

(lambda (x c)

(cond ((= c 0) x)

((> x 100) (M_c (- x 10) (- c 1)))

(else (M_c (+ x 11) (+ c 1))))))

Syntaxiquement, la récursivité est dégénérée . . . mais sémantiquement, la fonction M_c est
aussi “bizarre” que la fonction M.

Remarque. On peut démontrer que la valeur de (M x) existe quelle que soit la valeur de
l’entier (relatif) x et, en outre, que cette valeur cöıncide avec celle de (M91 x), si on pose

(define M91

(lambda (x)

(if (> x 100) (- x 10) 91)))

Cela explique le nom donné à la fonction M et montre aussi que cette fonction n’a pas
d’intérêt intrinsèque.

7.5.2 La fonction d’Ackermann

Un exemple du même genre mais plus significatif est celui de la fonction d’Ackermann. Il
s’agit d’une fonction de N × N dans N définie récursivement comme suit :

(define ack

(lambda (m n)

(cond

((zero? m) (+ n 1))

((zero? n) (ack (- m 1) 1))

(else (ack (- m 1) (ack m (- n 1)))))))

Ici aussi, on peut éliminer les appels récursifs imbriqués, mais compter le nombre
d’imbrications ne suffit plus; il faut en outre, pour chacune d’elle, noter le premier
argument de l’appel. Vu la sémantique opérationnelle de Scheme, les appels les plus
internes sont réduits avant les plus externes. On gérera donc une liste d’arguments telle
que ceux des appels internes soient en tête. Concrètement, on prévoit une fonction acl

prenant comme argument une liste non vide de naturels et telle que

(acl ’(n)) équivaut à n;
(acl ’(n m)) équivaut à (ack m n);
(acl ’(n m p)) équivaut à (ack p (ack m n));

. . . . . .
Le code s’écrit alors facilement :

(define acl

(lambda (l)

(cond

((null? (cdr l))

(car l))

((zero? (cadr l))

(acl (cons (+ (car l) 1)

(cddr l))))

((zero? (car l))

(acl (cons 1

(cons (- (cadr l) 1)

(cddr l)))))

(else

(acl (cons (- (car l) 1)

(cons (cadr l)

(cons (- (cadr l) 1)

(cddr l)))))))))

Remarque. On peut démontrer que la fonction ack est totale et crôıt plus rapidement que
toute fonction primitive récursive.

Cet exemple illustre à la fois la puissance de la technique des accumulateurs et ses
limites. Dans les deux cas on a obtenu des programmes à récursivité terminale mais cela
n’a pas significativement simplifié l’étude des procédures concernées ni réellement amélioré
leur efficacité.

7.6 Base fonctionnelle de l’accumulateur, style CPS

Le concept même d’accumulateur semble de nature opérationnelle. Sa motivation réside
dans la structure d’un processus de calcul. Pour le voir, nous reconsidérons deux “états
homologues” des processus de calcul associés aux évaluations des formes (fact 4) et
(fact_a 4 1), par exemple

(* 4 (* 3 (fact 2))) (fact_a 2 12)

On a trois paires de constituants homologues :

fact fact_a

2 2

(* 4 (* 3 ...)) 12

On passe de la version de gauche à celle de droite en créant un argument supplémentaire,
dont la valeur “représente” celle de l’expression (* 4 (* 3 ...)) , c’est-à-dire une

fonction à un argument (qui remplace le “trou” noté par les points de suspension). On
peut donc récrire l’expression en (lambda (k) (* 4 (* 3 k))), dont la valeur est bien
une fonction à un argument. Vu l’associativité de la multiplication, on peut simplifier cette
λ-forme en (lambda (k) (* 12 k)). Enfin, comme toute fonction linéaire x 7→ ax est
connue dès que a est connu, on peut représenter cette dernière λ-forme par son coefficient,
ici 12. Si la multiplication n’était pas associative, on aurait dû conserver la λ-forme
(lambda (k) (* 4 (* 3 k))) , en la représentant éventuellement par la liste (4 3) .

Il ressort de ceci que l’accumulateur est une représentation non fonctionnelle d’un
“argument implicite” fonctionnel. Il est d’ailleurs intéressant d’étudier ce qui se passe
quand on se contente d’expliciter l’argument fonctionnel, sans chercher à le représenter de
manière non fonctionnelle. Dans le cas de la factorielle, la version accumulante fact_a

devient la version fact_c ci-dessous :

(define fact_c

(lambda (n c)

(if (zero? n)

(c 1)

(fact_c (- n 1)

(lambda (k) (c (* n k)))))))

La factorielle se définit alors aisément :

(define fact

(lambda (n) (fact_c n (lambda (k) k))))

Avant de spécifier fact_c, étudions le processus de calcul associé ou, plus précisément, un
processus voisin. On a par exemple

(fact_c 4 (lambda (k) k))(fat  3 (lambda (k) ((lambda (k) k) (* 4 k))))(fat  3 (lambda (k) (* 4 k)))(fat  2 (lambda (k) ((lambda (k) (* 4 k)) (* 3 k))))(fat  2 (lambda (k) (* 4 (* 3 k))))...(fat  0 (lambda (k) (* 4 (* 3 (* 2 (* 1 k))))))((lambda (k) (* 4 (* 3 (* 2 (* 1 k))))) 1)(* 4 (* 3 (* 2 (* 1 1))))
==> 24

Remarque. Scheme n’effectue les réductions qu’à la fin du processus. La dernière étape
n’est donc pas l’évaluation de

((lambda (k) (* 4 (* 3 (* 2 (* 1 k))))) 1)

mais celle de

((lambda (k)

((lambda (k)

((lambda (k)

((lambda (k)

((lambda (k) k) (* 4 k)))

(* 3 k)))

(* 2 k)))

(* 1 k)))

1)

c’est pourquoi nous avons évoqué un processus voisin. Sur le plan conceptuel, cette
différence n’est pas importante ici. Le point crucial est d’observer que fact_c admet
la spécification suivante.

Si n est un naturel et si c est une fonction de N dans N,
alors (fact_c n c ) vaut c(n!).

L’argument fonctionnel auxiliaire c est une continuation. Cette fonction, appliquée à un
résultat intermédiaire du calcul en cours, fournit le résultat final.

Dans le cas de fact_c, l’exécution construit progressivement une fonction de plus en
plus complexe, représentant l’enchâınement des multiplications à faire. A chaque étape,
l’argument continuation est transformé en une fonction impliquant une multiplication
supplémentaire; cette fonction est l’argument de l’appel suivant. Les calculs numériques
n’ont lieu que quand l’enchâınement complet des multiplications a été formé. Cette
technique est le Continuation-Passing Style (CPS). Elle ressemble à la technique de
séparation fonctionnelle vue précédemment mais ne s’identifie pas à elle.

Le style CPS et la notion de continuation permettent de réifier , c’est-à-dire de
transformer en un objet (en fait, une procédure) le concept opérationnel de processus
de calcul. Cette transformation est utile parce qu’elle permet de séparer la “fabrication”
d’un processus de calcul et la mise en œuvre de ce processus. Les exemples élémentaires
vus au paragraphe 5.7 consacré à la séparation fonctionnelle ont déjà montré l’intérêt
potentiel de ce type de transformation. Nous revoyons maintenant ces exemples pour
illustrer le style CPS.

7.6.1 Le produit d’une liste de nombres

Rappelons ici les versions classiques de la fonction calculant le produit d’une liste de
nombres (cf. § 5.7.2) :

(define prodlist0

(lambda (l)

(if (null? l) 1 (* (car l) (prodlist0 (cdr l))))))



7 ACCUMULATEURS ET PROCESSUS ITÉRATIFS 142

(define prodlist1

(lambda (l)

(cond ((null? l) 1)

((zero? (car l)) 0)

(else (* (car l) (prodlist1 (cdr l)))))))

On a, pour la liste (3 5 0 2 4), le processus suivant :

(prodlist1 ’(3 5 0 2 4))(* 3 (pl '(5 0 2 4)))(* 3 (* 5 (pl '(0 2 4))))(* 3 (* 5 0))(* 3 0)
==> 0

Il y a autant de multiplications (par 0) que de facteurs précédant le premier 0. Utiliser
une version accumulante ne résout pas le problème. On peut écrire

(define prodlist1_a

(lambda (l a)

(cond ((null? l) a)

((zero? (car l)) 0)

(else (prodlist1_a (cdr l) (* a (car l)))))))

On a [[(prodlist1_a l a)]] = [[(prodlist1 l)]] ∗ [[a]]. Voici un exemple d’exécution :

(prodlist1_a ’(3 5 0 2 4) 1)(prodlist1 a '(5 0 2 4) (* 1 3))(prodlist1 a '(5 0 2 4) 3)(prodlist1 a '(0 2 4) (* 3 5))(prodlist1 a '(0 2 4) 15)
==> 0

Il y a toujours le même nombre de multiplications; en outre, les facteurs impliqués ne
sont pas nuls. Pour éviter à coup sûr toutes les multiplications au cas où l’un des facteurs
est nul, il existe une solution élémentaire : on parcourt d’abord la liste pour rechercher
un zéro éventuel. L’inconvénient est que, si aucun zéro n’est présent, on devra parcourir
la liste une deuxième fois. La technique de séparation fonctionnelle a donné lieu à une
solution, la fonction prodlist2 (cf. § 5.7.2). Voilà maintenant une solution voisine, basée
sur le CPS :

(define pl3_c

(lambda (l c)

(cond ((null? l) (c 1))

((zero? (car l)) 0)

(else (pl3_c (cdr l)

(lambda (k) (c (* (car l) k))))))))

(define prodlist3

(lambda (l) (pl3_c l (lambda (k) k))))

La fonction auxiliaire [[pl3_c]] peut être spécifiée comme suit :

Si [[l]] est une liste dont les éléments sont les nombres entiers (ou réels)
x1, . . . , xn et si [[f]] est une fonction f de domaine N (ou R), alors
[[(pl3_c l f)]] est f(Πn

i=1xi).

En particulier, cette la fonction [[(pl3_c)]] calcule le produit des éléments de son
premier argument (une liste de nombre) si son second argument est la fonction identique
[[(lambda (k) k)]]. A titre d’exemple, on a

(pl3_c (enum 1 3) (lambda (q) (+ 2 q))) ==> 8

puisque 2 + (1 ∗ 2 ∗ 3) = 8. De même, on a

(pl3_c ’(1 2 1.5 1)

(lambda (k)

(pl3_c (enum 1 (+ k 1))

(lambda (q) (+ 22 q))))) ==> 46

puisque 1 ∗ 2 ∗ 1.5 ∗ 1 = 3, Π([[(enum 1 (+ k 1))]]) = 24 si [[k]] = 3 et enfin [[(+ 22 q)]]
= 46 si [[q ]] = 24.

Observons aussi que l’on retombe sur la solution élémentaire si on remplace la
continuation par un argument non fonctionnel :

(define pl4_a

(lambda (l a)

(cond ((null? l) (prodlist0 a))

((zero? (car l)) 0)

(else (pl4_a (cdr l)

(cons (car l) a))))))

(define prodlist4

(lambda (l) (pl4_a l ’())))

On note que l’argument supplémentaire a n’est pas un nombre, mais une liste de nombres
non nuls; pour multiplier ces nombres, on utilise donc la procédure prodlist0 qui ne teste
pas les nombres à multiplier. La spécification est

Si [[l]] et [[a]] sont des listes dont les éléments sont les nombres entiers
(ou réels) x1, . . . , xn et y1, . . . , ym, respectivement, alors [[(pl4_c l a)]] est
(Πn

i=1xi)(Π
m
i=1yi).

7.6.2 Le double comptage

La séparation fonctionnelle permettait aussi d’éviter efficacement le double parcours de la
liste pour le problème du double comptage. Nous rappelons ici la solution présentée au
paragraphe 5.7.1 :

(define c2

(lambda (l s c)

(if (null? l)

c

(if (eq? (car l) s)

(c2 (cdr l)

s

(lambda (u)

(c (1+ (car u)) (cadr u))))

(c2 (cdr l)

s

(lambda (u)

(c (car u) (1+ (cadr u)))))))))

(define id (lambda (v) v))

(define count2 (lambda (l s) ((c2 l s id) 0 0)))

La solution en CPS est semblable :

(define c3

(lambda (l s c)

(if (null? l)

(c 0 0)

(if (eq? (car l) s)

(c3 (cdr l)

s

(lambda (u v) (c (1+ u) v)))

(c3 (cdr l)

s

(lambda (u v) (c u (1+ v))))))))

(define count3

(lambda (l s) (c3 l s cons)))

L’argument fonctionnel de la fonction c3 évolue simplement; si sa valeur avant un appel
récursif est celle de c, soit

(lambda (x y) (c x y))

sa nouvelle valeur sera celle d’une des expression

(lambda (x y) (c (1+ x) y))

(lambda (x y) (c x (1+ y)))

On peut remplacer l’argument fonctionnel par deux arguments numériques. Cela donne :

(define c4

(lambda (l s a1 a2)

(if (null? l)

(cons a1 a2)

(if (eq? (car l) s)

(c4 (cdr l) s (1+ a1) a2)

(c4 (cdr l) s a1 (1+ a2))))))

(define count4

(lambda (l s) (c4 l s 0 0)))

Cette solution est simple et optimale. On voit que la séparation fonctionnelle et le CPS sont
des techniques utiles, donnant lieu à des solutions efficaces. Parfois, le remplacement de
l’argument fonctionnel par un ou plusieurs accumulateur(s) améliore encore le programme.

8 Expressions symboliques

Au paragraphe 2.5, nous avions déjà brièvement évoqué la “notation pointée”, caractérisée
par l’égalité

[[(cons α β)]] = ([[α]] . [[β]]) .

Les deux notations représentent des listes si et seulement si [[β]] est une liste, mais l’égalité
garde un sens et reste vraie quels que soient α et β. Tout objet résultant de l’application
de cons à deux arguments est, par définition, une paire pointée. Toute liste non vide
est donc une paire pointée. Dans ce chapitre, nous étudions les expressions symboliques

c’est-à-dire, essentiellement, les paires pointées.

Définition. Une expression symbolique est un atome ou une paire formée d’expressions
symboliques.
Remarque. Au sens le plus large, un atome est tout objet qui n’est pas une paire, c’est-
à-dire tout objet qui ne peut être créé avec cons. En ce sens, tout objet Scheme est
une expression symbolique. Le plus souvent, on restreindra l’ensemble des atomes,92 ce
qui reviendra à restreindre l’ensemble des expressions symboliques. Notre but ici étant
d’étudier les propriétés structurelles des expressions symboliques, il est inutile d’être plus
précis à ce stade.

8.1 Arbres binaires

L’ensemble des atomes étant fixé, le domaine des expressions symboliques peut être vu
comme un type abstrait de donnée. Les constructeur et accesseurs sont cons, car, cdr.
On a aussi le reconnaisseur pair? qui reconnâıt les paires pointées, et éventuellement
un reconnaisseur spécifique pour l’ensemble des atomes admis, que nous pouvons noter
s-atom?. Le reconnaisseur du type sera alors défini comme suit :

(define s-exp?

(lambda (x)

(if (atom? x)

(s-atom? x)

(and (s-exp? (car x)) (s-exp? (cdr x))))))

On peut aussi définir

(define s-pair?

(lambda (x)

(and (pair? x) (s-exp? (car x)) (s-exp? (cdr x)))))

Si on note E l’ensemble des atomes admis, l’ensemble des expressions symboliques
apparâıt comme la réalisation en Scheme du domaine des E-arbres binaires introduit
au paragraphe 1.3.2. Par exemple, si E est l’alphabet, on écrira :

92Il est courant de se limiter aux symboles atomiques; on leur ajoute parfois la liste vide, les nombres
et/ou les booléens.

(define *alphabet*

’(a b c d e f g h i j k l m n o p q r s t u v w x y z))

(define s-atom? (lambda (x) (member x *alphabet*)))

8.2 Représentation en mémoire

La zone mémoire utilisée pour la représentation des paires pointées est organisée en cellules,
qui sont composées d’un pointeur gauche et d’un pointeur droit. La représentation en
mémoire de [[(cons α β)]] est un pointeur vers une cellule dont le pointeur gauche est la
représentation de α et le pointeur droit est la représentation de β. La représentation des
atomes en mémoire est unique, ce qui signifie que les deux pointeurs de la cellule utilisée
pour représenter (a . a) sont égaux. Par contre, les deux pointeurs de la cellule utilisée
pour représenter ((a . b) . (a . b)) ne sont pas nécessairement égaux.

La notation pointée est un moyen simple et concis de mettre en évidence la structure
de pointeurs qui représente en mémoire un objet Scheme non atomique; une notation
graphique, plus explicite, sera introduite plus loin. Dans le tableau ci-dessous, les formes
de gauche ont pour valeurs des expressions symboliques, représentées à droite de la flèche
en notation pointée. Comme nous le verrons plus loin, le système affiche en général cette
valeur sous une forme plus concise que la notation pointée.

’a --> a

(cons ’a ’b) --> (a . b)

(cons ’a ’()) --> (a . ())

(cons ’a (cons ’b ’c))) --> (a . (b . c))

(list a) --> (a . ())

(list a b) --> (a . (b . ()))

’(a b c d) --> (a . (b . (c . (d . ()))))

’((a b) (c)) --> ((a . (b . ())) . ((c . ()) . ()))

Le point (entouré d’espaces) et les parenthèses représentent l’appariement. Le
point sépare donc les deux composants d’une paire pointée. On peut, de manière
plus explicite mais moins commode, visualiser la structure de pointeurs. La
représentation en mémoire de la situation induite par l’évaluation en séquence des
formes (define x (cons ’a (cons ’b ’c))), (define y x), (define z (cdr y)) et
(define u (car x)) est donnée à la figure 15.

Vu son encombrement, nous n’utiliserons pas dans la suite ce mode de représentation
appelé “box-notation”; en fait, la notation pointée est appropriée, pour peu que l’on soit
conscient de sa nature arborescente. Les expressions symboliques s’identifient en effet à
des arbres binaires dont les feuilles sont des atomes. Par exemple, l’expression symbolique

(a . ((b . (c . (d . w))) . (((e . (f . x)) . (g . y)) . z))) ,

qui est la valeur de la forme

x

@@Ry

u

@
@@R

-

z

@@R

?

a

-

?

b

?

c

Figure 15: Paires pointées en mémoire

(cons a

(cons (cons b (cons c (cons d w)))

(cons (cons (cons e (cons f x))

(cons g y))

z)))

correspond à l’arbre binaire de la figure 16.

Remarque. Seules les feuilles de l’arbre de la figure 16 sont explicitement étiquetées.
Les étiquettes des nœuds internes sont synthétisées à partir des étiquettes des feuilles et
n’existent donc pas en mémoire. Ces étiquettes ont été écrites ici uniquement pour montrer
la correspondance entre sous-arbre de gauche et car d’une part, entre sous-arbre de droite
et cdr d’autre part. Il est facile de reconstituer la “box-notation” correspondante.

8.3 Notation pointée et notation usuelle

La notation pointée met en évidence la structure d’arbre binaire décoré des expressions
symboliques : chaque nœud a zéro fils (feuille) ou deux fils (nœud interne), auxquels on
accède par car et cdr. Chaque feuille est un atome.93 Tout ceci justifie l’intérêt et l’usage
de la notation pointée, mais cette notation reste néanmoins encombrante, notamment dans
le cas des listes :

() ()

(0) (0 . ())

(0 1) (0 . (1 . ()))

(0 1 2) (0 . (1 . (2 . ())))

((0 1) 2) ((0 . (1 . ())) . (2 . ()))

On passe très facilement de la notation pointée à la notation usuelle, par le procédé suivant :

93Plus exactement, chaque feuille est étiquetée par la représentation d’un atome.

a

b

c

d e g

(a .((b .(c .(d.w))).(((e .(f.x)).(g.y)).z)))

((b .(c .(d.w))).(((e .(f.x)).(g.y)).z))

(b .(c .(d.w)))

(c .(d.w))

(d.w)

w

(((e .(f.x)).(g.y)).z)

((e .(f.x)).(g.y)) z

(e .(f.x)) (g.y)

(f.x)

f x

y

Figure 16: Représentation arborescente d’une paire pointée

Tout point suivi d’une parenthèse ouverte est supprimé,
ainsi que la parenthèse ouverte qui suit
et la parenthèse fermée correspondante.
L’ordre des suppressions est quelconque.

Un point non suivi d’une parenthèse ouverte ne peut pas être supprimé.
Les exemples qui suivent illustrent le procédé de conversion.

((0 . (1 . ())) . (2 . ()))

((0 . (1 . ())) . (2 ))

((0 1 . () ) . (2 ))

((0 1 . () ) 2 )

((0 1 ) 2 )

((a . (b . ())) . ((c . (d . ())) . ()))

((a . (b . ())) . ((c . (d . ()))))

((a . (b . ())) . ((c . (d))))

((a . (b . ())) . ((c d)))

((a . (b . ())) (c d))

((a . (b)) (c d))

((a b) (c d))

L’élimination des points peut n’être que partielle : un point non suivi d’une parenthèse
ouverte n’est pas supprimable. Par exemple, ((a . b) . (c . d)) se simplifie en
((a . b) c . d) ; c’est sous cette forme que l’évaluateur affichera cette expression
symbolique.

8.4 Représentation des listes

Nous avons vu précédemment qu’une liste est conceptuellement un arbre. Nous venons
de voir qu’en machine, une liste est, comme toute expression symbolique, représentée par
une structure de pointeurs qui est aussi un arbre. Il convient de souligner que ces deux
arbres sont bien distincts. En particulier, les nœuds de l’arbre “conceptuel” sont de degré
variable tandis que tout nœud interne de l’arbre “machine” est de degré deux.94 A titre
d’exemple, les deux arbres associés à la liste (a (b c d) ((e f) g)) sont repris à la
figure 17.
Rappel. L’information attachée à un nœud interne se déduit de celle attachée à ses
descendants; en conséquence, seule l’information attachée aux feuilles doit être explicite.
Remarque. En ce qui concerne la représentation des données en Scheme, les listes sont
représentées par des expressions symboliques. Ce choix, guidé par des raisons d’efficacité,
justifie l’emploi de constructeur et accesseurs communs pour ces deux types concrets de
données, ainsi que pour les types abstraits (arbres à degré variable et arbres binaires) qui
leur correspondent.

8.5 Récursivité structurelle et expressions symboliques

Le schéma de base pour les expressions symboliques découle immédiatement de la structure
de l’expression. Une expression symbolique est un atome, ou une paire dont les deux
composants sont des expressions symboliques. On a :

(define F

(lambda (s u)

(if (atom? s)

(G s u)

(H (F (car s) (Ka s u))

(F (cdr s) (Kd s u))

s

u))))

Dans ce code, u représente une liste de zéro, un ou plusieurs arguments supplémentaires.95

Le cas où il n’y a pas d’argument supplémentaire se récrit plus simplement comme suit :

(define F

(lambda (s)

94Un degré constant permet une économie de mémoire.
95Il y a là un léger abus de notation, d’ailleurs évitable (cf. § 5.3).

()f

(a .((b .(c .(d.()))).(((e .(f.())).(g.())).())))

((b .(c .(d.()))).(((e .(f.())).(g.())).()))a

b

(b .(c .(d.()))) (((e .(f.())).(g.())).())

((e .(f.())).(g.())) ()(c .(d.()))

(e .(f.()))c (d.())

()d e (f.())

(g.())

()g

cb d g

e f

(a (b c d) ((e f) g))

(b c d) ((e f) g)a

(e f)

Figure 17: Arbre conceptuel (en haut), arbre machine (en bas)

(if (atom? s)

(G s)

(H (F (car s)) (F (cdr s)) s))))

On peut évaluer la taille d’une expression symbolique par le nombre d’atomes qu’elle
contient, ou par le nombre de nœuds internes, ou encore par le nombre total de nœuds,
au moyen des fonctions suivantes :

(define size_a

(lambda (s)

(if (atom? s)

1

(+ (size_a (car s)) (size_a (cdr s))))))

(define size_i

(lambda (s)

(if (atom? s)

0

(+ 1 (size_i (car s)) (size_i (cdr s))))))

(define size_t

(lambda (s)

(if (atom? s)

1

(+ 1 (size_t (car s)) (size_t (cdr s))))))

Ces programmes sont des instances du schéma structurel, de même que le programme qui
renvoie la liste des atomes d’une expression symbolique :

(define flatten

(lambda (s)

(if (atom? s)

(list s)

(append (flatten (car s))

(flatten (cdr s))))))

La technique de l’accumulateur permet d’améliorer l’efficacité de ce programme, par
élimination de append :

(define flatten_a

(lambda (s a)

(if (atom? s)

(cons s a)

(flatten_a (car s)

(flatten_a (cdr s) a)))))

On a [[(flatten_a s a)]] = [[(append (flatten s) a)]] quelles que soient l’expression
symbolique [[s]] et la liste [[a]]. En particulier, on a aussi [[(flatten_a s ’())]] =
[[(flatten s)]]. On observe que le code de flatten_a est basé sur la récursivité
structurelle mais n’est pas une instance du schéma (qui n’autorise pas les appels récursifs
imbriqués).

L’aplatissement est une opération qui s’applique à tout type d’arbre. A une liste
sont associés un arbre machine que l’on peut aplatir par flatten, et un arbre conceptuel
que l’on peut aplatir par flat_list (§ 5.4). Les deux frondaisons sont naturellement
différentes. Soit l une liste telle que

((a . (b . ())) . ((c . (d . ())) . ())) = [[l]] = ((a b) (c d))

On a

(flatten l) ==> (a b () c d () ())

(flat_list l) ==> (a b c d)

Un problème intéressant consiste à déterminer si une expression symbolique peut
s’écrire sans point ou non. Le schéma structurel permet d’écrire :

(define point-free?

(lambda (s)

(if (atom? s)

#t

(and (point-free? (car s))

(list? (cdr s))

(point-free? (cdr s))))))

Cette solution n’est pas optimale parce que [[(cdr s)]] est parcouru deux fois, mais
on y remédie en utilisant pair? plutôt que list?. On peut aussi utiliser l’égalité
[[(if c #t e)]] = [[(or c e)]] pour obtenir

(define point-free

(lambda (s)

(or (atom? s)

(and (point-free (car s))

(or (null? (cdr s)) (pair? (cdr s)))

(point-free (cdr s))))))

(point-free ’a) ==> #t

(point-free ’((a . b))) ==> #f

(point-free ’(a . (b . ()))) ==> #t

(point-free ’(a b . c)) ==> #f

8.6 Egalité, identité

Divers prédicats permettent de tester l’égalité de deux valeurs d’un type donné. On utilise
eq? pour les symboles atomiques, = pour les nombres,96 et eqv? pour des valeurs atomiques
(symboles, nombres, booléens). Pour les expressions symboliques, on utilise le prédicat
prédéfini equal?, qui aurait pu être défini comme suit :

(define equal?

(lambda (x y)

(cond ((pair? x)

(and (pair? y)

(equal? (car x) (car y))

96Le nombre d’arguments de = et des autres prédicats de comparaison numérique est quelconque

(equal? (cdr x) (cdr y))))

((pair? y) #f)

(else (eqv? x y)))))

On a par exemple

(equal? ’(a . (b . c)) ’(a b . c)) ==> #t

L’usage du prédicat eq? s’étend aux expressions symboliques mais il teste alors
l’identité de deux objets : (eq? x y) sera vrai si x et y désignent la même structure
en mémoire. Concrètement, eq? teste l’égalité de deux pointeurs. La session suivante
montre bien la distinction entre les deux prédicats.

(eq? ’a ’a) ==> #t

(define x ’(a . b)) ==> ...

(define y ’(a . b)) ==> ...

(define z x) ==> ...

(equal? x y) ==> #t

(eq? x y) ==> #f

(eq? x z) ==> #t

Utiliser eq? plutôt que equal? pour des objets non atomiques peut permettre
d’améliorer l’efficacité . . . mais cela exige une bonne connaissance du système. Notons
aussi que les symboles atomiques n’ont jamais qu’une seule représentation en mémoire;
sur ce domaine, eq? et equal? sont toujours logiquement équivalents.

8.7 Déconstruction des expressions symboliques

Toute expression symbolique s’obtient en combinant des atomes au moyen du seul
constructeur cons et ce, d’une manière unique. Décrire, ou “déconstruire”, une expression
symbolique est reconstituer l’unique combinaison (basée sur cons et des atomes) dont la
valeur est cette expression. On peut aisément produire une fonction describe produisant
cette reconstitution. On aura par exemple

(cons 1 (cons (cons (cons ’a ’()) (cons 2 ’())) ’()))

==> (1 ((a) 2)))

(describe ’(1 ((a) 2)))

==> (cons 1 (cons (cons (cons ’a ’()) (cons 2 ’())) ’()))

Le schéma habituel fonctionne immédiatement :

(define describe

(lambda (s)

(if (atom? s)

(cond ((number? s) s)

((boolean? s) s)

((null? s) (quote ’()))

((symbol? s) (list ’quote s))

(else s))

(list ’cons

(describe (car s))

(describe (cdr s))))))

On peut réorganiser le code en

(define describe

(lambda (s)

(cond ((null? s) (quote ’()))

((symbol? s) (list ’quote s))

((pair? s) (list ’cons

(describe (car s))

(describe (cdr s))))

(else s))))

Le langage Scheme, comme tout langage, comporte des mécanismes d’entrée-sortie.
Nous ne les avons guère évoqués jusqu’ici parce que la boucle “read-eval-print” du système
prévoit la lecture de l’expression à évaluer et l’écriture de sa valeur (et/ou de messages
éventuels). On peut cependant prévoir, au cours du processus de calcul, des lectures et
des écritures supplémentaires. Nous avons vu que lors de l’évaluation d’une combinaison,
l’ordre dans lequel les constituants de cette combinaison sont évalués n’est pas imposé. Un
moyen d’imposer un ordre séquentiel, utilisé notamment pour les commandes explicites
d’entrée-sortie, est fourni par la forme spéciale begin. Le processus de calcul associé à
(begin e1 . . . en) consiste en l’évaluation en séquence de e1, . . . , en; la valeur de en (si
elle existe) est la valeur de la forme. Par exemple l’évaluation de

(begin (display "tra")

(newline)

(display "la la")

’inutile

’utile)

provoquera l’affichage

tra

la la

==> utile

Les formes spéciales begin, newline et display (acceptant comme argument une châıne de
caractères),97 permettent d’écrire une variante “pretty-printing” de la fonction describe :

97Une châıne de caractères commence et finit par le délimiteur " (guillemet). La forme spéciale display

permet l’affichage d’une châıne de caractères, sans les délimiteurs. Les châınes de caractères constituent

(define lcons 6) ;; longueur de "(cons "

(define dis ;; (dis n) ecrit n blancs

(lambda (n)

(if (zero? n)

(display "")

(begin (display " ") (dis (- n 1))))))

(define disp

(lambda (n m) (dis (+ n (* m lcons)))))

(define pretty

(lambda (d n m)

(if (not (pair? d))

(begin (disp n 0)

(if (symbol? d) (display "’"))

(display d))

(begin (display "(cons ")

(pretty (car d) 0 (+ m 1))

(newline)

(disp n (+ m 1))

(pretty (cdr d) 0 (+ m 1))

(display ")")))))

(define print

(lambda (d) (begin (newline)

(pretty d 0 0))))

La fonction principale print écrit son argument avec une indentation correcte, au départ
d’une nouvelle ligne. L’évaluation de (pretty d n m) provoque l’affichage correctement
indenté de la valeur de d, sans passage initial à une nouvelle ligne et avec un décalage de
6[[m]]+[[n]] espaces. Ces fonctions ne retournent pas de valeur; seul leur effet d’affichage
est intéressant. Voici quelques exemples d’utilisation.

(print ’a)

’a

==> ...

(print ’(a . (b . 1)))

(cons ’a

(cons ’b

1))

==> ...

un type de données Scheme, avec constructeurs, accesseurs, etc. Nous n’étudierons pas ce type plus avant,
ni d’ailleurs les diverses autres formes spéciales d’entrée-sortie. Quelques autres cas seront évoqués plus
loin, dans des exemples.

L’exemple suivant illustre la structure d’une liste, vue comme un cas particulier
d’expression symbolique :

(print ’(a b c d))

(cons ’a

(cons ’b

(cons ’c

(cons ’d

()))))

==> ...

(print ’((a . (b . 1)) . ((c . 2) . ((d . 3) . e))))

(cons (cons ’a

(cons ’b

1))

(cons (cons ’c

2)

(cons (cons ’d

3)

’e)))

==> ...

Remarque. A titre de facilité syntaxique, le système Scheme permet, dans certaines
circonstances, l’emploi “implicite” de la forme begin. Nous précisons cet usage implicite
dans deux cas seulement. On peut utiliser cette facilité dans les clauses de la forme spéciale
cond (cf. § 3.5.1); par exemple,

(cond (pred (begin e1 e2 e3))

(else (begin e4 e5)))

peut s’écrire

(cond (pred e1 e2 e3)

(else e4 e5))

De même, on peut utiliser le begin implicite dans le corps d’une λ-forme; l’expression

(lambda (x y) (begin e1 e2 e3))

peut s’écrire

(lambda (x y) e1 e2 e3)

9 Abstraction et blocs

Nous avons vu, notamment au chapitre 6, que la résolution d’un problème consiste en
la conception d’un certain nombre de procédures. L’ensemble de ces procédures forme
un programme. Syntaxiquement, cet ensemble n’est pas structuré mais nous avons déjà
évoqué une structure hiérarchique implicite sur laquelle il est utile de revenir brièvement.
Une procédure f appelle une procédure g si la forme définissant f contient une occurrence
de g. Dans l’approche descendante, on écrit la fonction principale avant les fonctions
auxiliaires et, plus généralement, les fonctions appelantes avant les fonctions appelées. On
procède en sens inverse dans l’approche ascendante (cf. § 6.3.4).

La hiérarchie des appels constitue l’architecture du programme, qu’il est essentiel de
mâıtriser pour comprendre le programme. Elle devient naturellement complexe lorsque le
programme résout un problème complexe et comporte de nombreuses fonctions. Cette
complexité est difficile à gérer tant pour les utilisateurs du programme que pour ses
concepteurs. L’utilisateur ne doit pas nécessairement connâıtre le détail de cette hiérarchie;
dans le cas extrême, il lui suffit de connâıtre la fonction principale et il peut même
être opportun de lui interdire l’accès direct aux fonctions auxiliaires. Si les concepteurs
travaillent en équipe, il peut être opportun que chacun n’ait accès qu’à certaines fonctions
définies par les autres; si un programmeur écrit une fonction auxiliaire f, il est essentiel
qu’une éventuelle fonction du même nom, complètement différente et écrite par un autre
programmeur, soit “cachée” et ne provoque donc aucune interférence.

Dans ce chapitre, nous montrons comment le système Scheme permet de rendre
“invisibles” certains objets, fonctionnels ou non, en dehors du cadre où ils doivent être
vus. Nous commençons par le cas élémentaire où l’objet n’est pas une fonction mais une
simple valeur intermédiaire, dont la liaison à une variable n’est utile que localement.

9.1 La forme spéciale let

Supposons que l’on veuille calculer la valeur de l’expression arithmétique

2(a + b)2 + (a + b)(a − c)2 + (a − c)3

pour certaines valeurs de a, b et c. On peut naturellement évaluer les trois termes et faire
la somme des trois résultats. En Scheme, on évaluera la combinaison

(+ (* 2 (+ a b) (+ a b))

(* (+ a b) (- a c) (- a c))

(* (- a c) (- a c) (- a c)))

Néanmoins il est probable que, avec ou sans l’aide d’une calculatrice ou du système
Scheme, on préférera une approche du type

Soient x et y les valeurs de a + b et a − c.
La valeur demandée est celle de 2x2 + xy2 + y3.

Cette démarche a l’avantage évident d’éviter les recalculs des sous-expressions a + b et
a − c. Le langage Scheme permet cette économie, au moyen de la forme spéciale let.
Dans l’exemple qui nous occupe, la combinaison à évaluer sera

(let ((x (+ a b)) (y (- a c)))

(+ (* 2 x x) (* x y y) (* y y y)))

Il convient de souligner un second avantage de cette démarche : le code obtenu est
plus concis et plus lisible; cela contribue à une meilleure fiabilité des programmes. Ce
gain est marginal pour un exemple aussi élémentaire que celui-ci mais, en pratique,
l’avantage de lisibilité est au moins aussi important que l’avantage d’économie : le temps du
programmeur est plus précieux et plus coûteux que celui de l’ordinateur ! Il ne s’agit pas ici
seulement du temps consacré au codage proprement dit par le concepteur-programmeur,
mais aussi du temps consacré à la documentation, à la maintenance, à la modification du
code, à son intégration dans un logiciel plus important, etc. C’est spécialement vrai dans
le cas où la personne chargée de ces tâches n’est pas l’auteur du programme original.

On observe immédiatement que calculer la valeur de l’expression

2(a + b)2 + (a + b)(a − c)2 + (a − c)3

en calculant d’abord
x = a + b et y = a − c

consiste simplement à appliquer la fonction

(x, y) 7→ 2x2 + xy2 + y3

aux arguments
x = a + b et y = a − c .

En fait, le let de Scheme, comme le “soit” du français, cache la définition et l’application
d’une certaine fonction. Cela motive la définition suivante :

La forme spéciale (let ((x α) (y β)) γ)
constitue une variante syntaxique
de la combinaison ((lambda (x y) γ) α β) .

Cette définition, explicitée ici pour le cas de deux liaisons, se généralise immédiatement
à un nombre quelconque de liaisons. Les expressions α et β sont quelconques, de même
que le corps γ de la λ-forme.

Appliquée à notre exemple, cette définition exprime que la forme

(let ((x (+ a b)) (y (- a c)))

(+ (* 2 x x) (* x y y) (* y y y)))

est strictement équivalente à la combinaison

((lambda (x y) (+ (* 2 x x) (* x y y) (* y y y)))

(+ a b)

(- a c))

Comme toujours en Scheme, on ne restreint pas le type des valeurs susceptibles d’être
liées aux paramètres d’une forme let. Ces valeurs peuvent notamment être fonctionnelles;
considérons par exemple les deux fonctions suivantes :

(define hypo (lambda (x y) (sqrt (+ (square x) (square y)))))

(define square (lambda (x) (* x x)))

La définition de la fonction hypo comporte des appels à la fonction auxiliaire square. Si,
pour les raisons évoquées plus haut, on souhaite que l’accès direct à square soit impossible,
on peut utiliser la forme spéciale let et remplacer hypo par

(define hypo1

(let ((square (lambda (x) (* x x))))

(lambda (x y) (sqrt (+ (square x) (square y))))))

ou encore par

(define hypo2

(lambda (x y)

(let ((square (lambda (x) (* x x))))

(sqrt (+ (square x) (square y))))))

Avec l’une de ces définitions, seuls les appels auxiliaires à square, via un appel à hypo1 (ou
hypo2) restent possibles. La variable square reste non liée dans l’environnement global
quand l’une de ces deux définitions y est évaluée. Pour s’en rendre compte, il suffit de
remplacer let par lambda, conformément à la définition de let. Cela donne

(define hypo1

((lambda (square)

(lambda (x y) (sqrt (+ (square x) (square y)))))

(lambda (x) (* x x))))

ou encore

(define hypo2

(lambda (x y)

((lambda (square) (sqrt (+ (square x) (square y))))

(lambda (x) (* x x)))))

Remarque. Soulignons encore, à cette occasion, que la règle des portées en Scheme est
stricte et sans ambigüıté; cela permet au système de distinguer le “x” introduit comme
premier paramètre dans (lambda (x y) ... du “x” introduit comme unique paramètre

dans (lambda (x) ...; bien souvent, des noms distincts (et évocateurs) sont donnés à
des variables pour l’unique bénéfice des auteurs et lecteurs du code.

Dans le cas où la construction let est utilisée pour introduire des fonctions auxiliaires
locales, il n’y a généralement pas une économie significative de ressources; le but visé
est alors exclusivement une meilleure lisibilité du code. De toute manière, l’économie de
ressources peut être réalisée directement au moyen de la forme lambda; la fonction i+

introduite au paragraphe 6.4.2 illustrait bien ce point. Le code initial était

(define i+

(lambda (f u l x0 x1 eps)

(cond ((prox x0 x1 eps) (mu x0 x1))

((prox (f (mu x0 x1) l) u eps) (mu x0 x1))

((< (f (mu x0 x1) l) u) (i+ f u l (mu x0 x1) x1 eps))

(else (i+ f u l x0 (mu x0 x1) eps)))))

Pour éviter un éventuel recalcul de l’expression (mu x0 x1), on avait introduit au
paragraphe 6.4.2 une combinaison du type

((lambda (aux1) ...) (mu x0 x1))

mais l’usage de let donnera une version plus lisible :

(define i+

(lambda (f u l x0 x1 eps)

(let ((aux1 (mu x0 x1)))

(cond ((prox x0 x1 eps) aux1)

((prox (f aux1 l) u eps) aux1)

((< (f aux1 l) u) (i+ f u l aux1 x1 eps))

(else (i+ f u l x0 aux1 eps))))))

On peut de même éviter un éventuel recalcul de l’expression (f aux1 l) :

(define i+

(lambda (f u l x0 x1 eps)

(let ((aux1 (mu x0 x1)))

(let ((aux2 (f aux1 l)))

(cond ((prox x0 x1 eps) aux1)

((prox aux2 u eps) aux1)

((< aux2 u) (i+ f u l aux1 x1 eps))

(else (i+ f u l x0 aux1 eps)))))))

9.2 Portée

Les principes de portée et de renommage relatifs à la λ-forme restent valables pour let

et pour les variantes que nous introduirons plus loin. La session suivante illustre ce fait
dans un cas élémentaire :

(define a 5) ==> ...

(add1 a) ==> 6

(let ((a 3)) (add1 a)) ==> 4

(let ((c 3)) (add1 c)) ==> 4

(add1 3) ==> 4

(add1 a) ==> 6

Voici un exemple plus intéressant :

(define f (let ((b 100)) (lambda (x) (+ x b)))) ==> ...

(let ((b 10)) (f 25)) ==> 125

Il n’est pas nécessaire de connâıtre la définition de f pour déterminer que la forme
(let ((b 10)) (f 25)) est équivalente à la forme plus simple (f 25); le fait que la
variable b n’a pas d’occurrence dans le texte “(f 25)” suffit à conclure.

La plupart des confusions éventuelles dans l’application de la règle de portée
proviennent d’un télescopage de noms. Il suffit d’appliquer méthodiquement les règles
d’évaluation pour éviter les erreurs. On peut aussi renommer (mentalement) certaines
variables liées. Tout cela est très simple mais requiert de l’attention. Nous détaillons un
exemple supplémentaire :

(let ((a 5))

(let ((fun (lambda (x) (max x a))))

(let ((a 10) (x 20))

(fun 1))))==(let (( 5))(let ((fun (lambda (y) (max y ))))(let ((a 10) (x 20))(fun 1))))==(let ((fun (lambda (x) (max x 5))))(let ((a 10) (x 20))(fun 1)))==(let ((fun (lambda (x) (max x 5))))(fun 1))==((lambda (x) (max x 5)) 1)==(max 1 5)
==> 5

Nous avons utilisé le modèle de substitution. Nous allons voir que le modèle des
environnements donne le même résultat. On observe d’abord que l’expression à évaluer
est une variante syntaxique de

((lambda (a)

((lambda (fun)

((lambda (a x) (fun 1))

10

20))

(lambda (x) (max x a))))

5)

La structure d’environnements créée lors de l’évaluation est représentée à la figure 18.
Evaluer l’expression complète dans E0 revient à évaluer

((lambda (fun)

((lambda (a x) (fun 1))

10

20))

(lambda (x) (max x a)))

dans E1, ou encore à évaluer

((lambda (a x) (fun 1))

10

20))

dans E2, ou encore à évaluer

(fun 1)

dans E3, ou encore à évaluer

(max x a)

dans E4.

C0

max : ...

C1

a : 5

C2

fun :[..E1]

C3

a : 10

x : 20

C4

x : 1
- - - -

E0E1E2

E3
E4

Environnement : E0 E1 E2 E3 E4

Environnement d’accès : E0 E0 E1 E2 E1

Figure 18: Structure d’environnements

On observe notamment que l’environnement d’accès de E4 est E1, parce que la création de
E4 est due à l’application d’une fermeture contenant l’environnement E1. Cela explique
pourquoi la valeur de (max x a) dans E4 est 5 et non 10.

9.3 La forme let*

Supposons que l’on veuille calculer la valeur de l’expression arithmétique

2(a + b)2(a − c)2 + (a + b)(a − c) + (a + b)2(a − c)3

pour certaines valeurs de a, b et c. On pourra adopter l’approche suivante :

Soient x et y les valeurs de a + b et a − c.
La valeur demandée est celle de 2x2y2 + xy + x2y3.

On poussera peut-être plus loin la recherche de la concision et de l’économie en écrivant

Soient x et y les valeurs de a + b et a − c;
soit z la valeur de xy.
La valeur demandée est celle de 2z2 + z + z2y.

Cela se traduit immédiatement en

(let ((x (+ a b)) (y (- a b)))

(let ((z (* x y)))

(+ (* 2 z z) z (* z z y))))

Cette expression a une valeur dans tout environnement E où les variables a et b sont
liées à des valeurs numériques. Que les variables x, y et z soient ou non liées dans
l’environnement E n’a aucune incidence sur la valeur de l’expression. Une expression
équivalente est

(let ((x (+ a b)))

(let ((y (- a b)))

(let ((z (* x y)))

(+ (* 2 z z) z (* z z y)))))

Par contre, des expressions telles que

(let ((x (+ a b)) (y (- a b)) (z (* x y)))

(+ (* 2 z z) z (* z z y)))

et

(let ((z (* x y)))

(let ((x (+ a b)) (y (- a b)))

(+ (* 2 z z) z (* z z y))))

ne sont visiblement pas équivalentes à la précédente.98 En particulier, elles n’ont de valeur
que dans un environnement où les quatre variables a, b, x et y sont liées à des valeurs
numériques. Cet exemple montre qu’il est parfois utile d’imbriquer des constructions let
et que l’ordre des imbrications est important. Cela motive la définition suivante :

98Elles sont cependant équivalentes entre elles.

La forme spéciale (let* ((x α) (y β)) γ)
constitue une variante syntaxique
de la forme spéciale (let ((x α)) (let ((y β)) γ)) .

A titre d’exemple, on peut récrire avec let* la dernière variante de la fonction i+ introduite
au paragraphe 9.1 :

(define i+

(lambda (f u l x0 x1 eps)

(let* ((aux1 (mu x0 x1))

(aux2 (f aux1 l)))

(cond ((prox x0 x1 eps) aux1)

((prox aux2 u eps) aux1)

((< aux2 u) (i+ f u l aux1 x1 eps))

(else (i+ f u l x0 aux1 eps))))))

Comme la définition de let, celle de let* se généralise immédiatement à un nombre
quelconque de liaisons.

9.4 La forme spéciale letrec

Nous avons vu comment la forme let pouvait être utilisée pour introduire des
fonctions auxiliaires locales. Cependant, de telles fonctions ne peuvent pas être définies
récursivement. Plus précisément, lors de l’évaluation de la forme

(let ((f exp)) α) ,

toute occurrence de f dans exp fera référence à la valeur éventuelle de f dans
l’environnement courant. En effet, si on récrit la forme let en

((lambda (f) α) exp) ,

on voit immédiatement que exp (au contraire de α) n’est pas dans la portée de la λ-
forme. Cela ne convient pas à un utilisateur souhaitant définir récursivement une fonction
auxiliaire qui sera localement liée à f. La forme letrec permet de régler ce problème et
on pourra donc écrire, par exemple,

(define fact

(lambda (n)

(letrec ((fact-it

(lambda (k acc)

(if (zero? k)

acc

(fact-it (-1+ k) (* k acc))))))

(fact-it n 1))))

On peut simuler le processus d’exécution comme suit :(fat 2)(fat-it 2 1)((lambda (k a) ...) 2 1)(if (zero? 2) 1 (fat-it 1 2))(fat-it 1 2)...2
Avec un let simple, le passage de la cinquième ligne à la sixième ne serait pas possible.

Le letrec est donc une variante sémantique du let. L’exemple ci-dessus suffit
à montrer l’importance de cette différence. Cette différence nous empêche, plus
généralement, de voir letrec comme une construction accessoire, qu’il serait possible
d’éliminer.99

La forme spéciale letrec permet la définition de fonctions auxiliaires mutuellement
récursives. On a par exemple

(letrec ((even?

(lambda (n) (if (zero? n) ’#t (odd? (- n 1)))))

(odd?

(lambda (n) (if (zero? n) ’#f (even? (- n 1))))))

(odd? 4))

==> #f

On peut simuler le processus de calcul comme suit :

(odd? 4)

((lambda (n) (if (zero? n) ’#f (even? (- n 1)))) 4)

(if (zero? 4) ’#f (even? (- 4 1)))

(even? 3)

...

#f

Si on avait utilisé let au lieu de letrec, cette évaluation n’aurait pas été possible, la
variable locale even? étant non liée dans l’environnement où l’expression (even? 3) est
évaluée.

Ces exemples suffisent à clarifier la sémantique de toute expression du type

(letrec ((f1 (lambda (...) (...)))

...

(fn (lambda (...) (...))))

EXP)

99Nous reviendrons plus loin sur des moyens de simuler, et donc d’éliminer, la construction letrec.

où EXP désigne une expression. En particulier, nous n’essayons pas de donner une
sémantique à une forme letrec dans laquelle la valeur à lier à un paramètre ne serait
pas obtenue par l’évaluation d’une λ-forme.

Un usage fréquent de letrec est la définition d’une fonction auxiliaire locale récursive
mais d’autres usages sont peut-être plus importants. On a déjà rencontré plusieurs fois le
cas d’une fonction retournant comme résultat une fonction. Un exemple simple est celui
de la fonction offset qui, à tout nombre entier n, associe la fonction numérique ajoutant
n à son argument. Sa définition est

(define offset

(lambda (n)

(lambda (y) (+ n y))))

Si la fonction à retourner doit se définir récursivement, il sera commode d’utiliser un
letrec. Au paragraphe 5.3.5, nous avions défini et commenté les fonctions d’extension
*_ext et +_ext comme suit :

(define *_ext

(lambda (f e)

(lambda v

(if (null? v)

e

(f (car v) (apply (*_ext f e) (cdr v)))))))

(define +_ext

(lambda (f)

(lambda (x . v)

(if (null? v)

x

(f x (apply (+_ext f) v))))))

Nous avions noté que ces fonctions, sans être elles-mêmes récursives, renvoyaient des
fonctions récursives, à savoir les valeurs de (*_ext f e) et (+_ext f), respectivement.
Cela suggère de les récrire en utilisant letrec :

(define *_ext

(lambda (f e)

(letrec ((valrec

(lambda v

(if (null? v)

e

(f (car v) (apply valrec (cdr v)))))))

valrec)))

(define +_ext

(lambda (f)

(letrec ((valrec

(lambda (x . v)

(if (null? v)

x

(f x (apply valrec v))))))

valrec)))

Un autre exemple d’usage de la forme spéciale letrec est fourni par les itérateurs.
On a déjà évoqué la fonction iter, telle que l’application de l’opérateur [[(iter n)]]
à une fonction unaire f produise la composition de [[n]] exemplaires de cette fonction.
Nous avons donné au paragraphe 6.2.3 deux programmes simples utilisant l’opérateur de
composition et aussi une solution plus directe :

(define iter

(lambda (n)

(lambda (f)

(lambda (x)

(if (zero? n)

x

(f (((iter (- n 1)) f) x)))))))

On définit de manière analogue iter_bis, de telle sorte que les fonctions [[((iter n) f)]]
et [[((iter_bis f) n)]] soient égales :

(define iter_bis

(lambda (f)

(lambda (n)

(lambda (x)

(if (zero? n)

x

(f (((iter_bis f) (- n 1)) x)))))))

La fonction [[(iter_bis f)]] fait l’objet d’une définition récursive, mieux mise en évidence
dans la variante suivante :

(define iter_bis

(lambda (f)

(letrec ((frec

(lambda (n)

(if (zero? n)

(lambda (x) x)

(lambda (x) (f ((frec (- n 1)) x)))))))

frec)))

On pourrait aussi écrire

(define iter_bis

(lambda (f)

(letrec ((frec

(lambda (n)

(lambda (x)

(if (zero? n)

x

(f ((frec (- n 1)) x)))))))

frec)))

9.5 Schémas récursifs avec let

La forme spéciale let est souvent utile dans le cadre d’une définition récursive. Par
exemple, le schéma

(define F

(lambda (n)

(if (zero? n)

c

(H (F (- n 1)) n))))

permet de définir autant de fonctions qu’il existe d’instances pour la constante c et pour la
fonction auxiliaire H. Supposons par exemple que l’expression (H x y) soit (+ (* 2 x) y)

ou encore (+ x x y). A priori on pourrait croire que les deux choix sont équivalents
puisque les expressions (+ (* 2 x) y) et (+ x x y) sont équivalentes. D’un point de
vue pratique pourtant, l’usage de l’instance (+ x x y) est à éviter parce qu’elle comporte
deux occurrences de x; cette variable sera liée à la valeur de l’appel récursif (F (- n 1)),
valeur qui sera donc calculée deux fois. On évite cet éventuel facteur d’inefficacité en
introduisant une forme let pour le calcul de cette valeur. Cela donne lieu au schéma
modifié suivant :

(define F

(lambda (n)

(if (zero? n)

c

(let ((rec (F (- n 1)))) (H rec n)))))

Cette technique est d’application pour tous les schémas de récursion. Notons aussi que
les programmes instances des schémas modifiés sont parfois plus lisibles que les instances
des schémas originaux. Par exemple, le programme

(define f

(lambda (n u)

(if (zero? n)

u

(let ((v (f (- n 1) u)))

(/ (+ v (/ (* v v) u)) 3)))))

est (un peu) plus lisible que le programme

(define f

(lambda (n u)

(if (zero? n)

u

(/ (+ (f (- n 1) u)

(/ (* (f (- n 1) u) (f (- n 1) u)) u))

3))))

En outre, le premier programme est d’efficacité linéaire en n, tandis que le second est
exponentiel.

Le problème du double comptage, abordé au paragraphe 5.7.1, illustre le même
phénomène. Les fonctions

(define count1

(lambda (l s)

(if (null? l)

(cons 0 0)

(if (eq? (car l) s)

(cons (1+ (car (count1 (cdr l) s)))

(cdr (count1 (cdr l) s)))

(cons (car (count1 (cdr l) s))

(1+ (cdr (count1 (cdr l) s))))))))

et

(define count5

(lambda (l s)

(if (null? l)

(cons 0 0)

(let ((rec (count5 (cdr l) s)))

(if (eq? (car l) s)

(cons (1+ (car rec)) (cdr rec))

(cons (car rec) (1+ (cdr rec))))))))

permettent toutes deux de calculer, par exemple, le nombre des “a” et le nombre des
“non-a” dans une liste; on a

(count1 ’(a b a c d a c) ’a) ==> (3 . 4)

(count5 ’(a b a c d a c) ’a) ==> (3 . 4)

Les résultats sont les mêmes mais les temps de calcul sont très différents. L’évaluation
de (count1 ℓ) devient laborieuse dès que la longueur de la liste ℓ est de l’ordre de 20;
par contre, l’évaluation de (count1 ℓ) est quasi instantanée, même si ℓ comporte 2 000
éléments. La fonction count5 peut se récrire

(define count5

(lambda (l s)

(if (null? l)

(cons 0 0)

((lambda (rec)

(if (eq? (car l) s)

(cons (1+ (car rec)) (cdr rec))

(cons (car rec) (1+ (cdr rec)))))

(count5 (cdr l) s)))))

On comparera cette solution aux précédentes (les fonctions count0 à count4 introduites
au paragraphe 5.7.1) et on notera l’utilité de l’omniprésent lambda.

9.6 Un exemple de structuration

Le problème des partitions, abordé au paragraphe 6, permet d’illustrer concrètement le
contenu du présent chapitre. La fonction principale partitions est une instance du
schéma superficiel sur les listes qui requiert l’usage de let pour nommer localement le
résultat de l’appel récursif. Les deux fonctions auxiliaires de premier niveau, procede_1
et procede_2, sont d’intérêt local et ne seront documentées qu’à l’usage des concepteurs;
il est naturel de les “cacher”, par un simple let puisqu’elles ne sont pas récursives. La
fonction auxiliaire insert_in_all, de deuxième niveau, pourrait raisonnablement être
utilisée en dehors du problème des partitions; on a donc choisi de ne pas la cacher et de
la documenter aussi pour l’utilisateur (cf. § 6.3.2). Il en va de même pour la fonction
append_map (cf. § 6.1.5), mais pas pour la fonction split, introduite par un letrec. On
peut donc présenter le code comme suit :

(define partitions
(lambda (e)
(let ((procede_1

(lambda (x lp)
(insert_in_all (list x) lp)))

(procede_2
(lambda (x lp)

(letrec ((split
(lambda (x ll)

(if (null? ll)
’()
(cons (cons (cons x (car ll)) (cdr ll))

(map (lambda (ss)
(cons (car ll) ss))

(split x (cdr ll))))))))

(append_map (lambda (p) (split x p)) lp)))))
(if (null? e)

’(())
(let ((rec (partitions (cdr e))))
(append (procede_1 (car e) rec)

(procede_2 (car e) rec)))))))

(define insert_in_all ...)

(define append_map ...)

Pour conclure, notons encore un “truc” parfois utile en cas de panne d’inspiration.
Supposons que l’on définisse une certaine fonction prenant comme argument une liste
(ou un autre type structuré pour lequel un schéma de récursion existe) et qu’une idée
algorithmique ne soit pas immédiatement apparente. Il n’est pas rare que le simple fait
d’écrire

(define mystery

(lambda (l)

(if (null? l)

c

(let ((rec (mystery (cdr l))))

(...)))))

et donc de nommer localement ce que l’on espère être un résultat intermédiaire crucial,
donne un sérieux coup de pousse à l’imagination ! La technique des schémas avec let est
réellement très efficace.

9.7 Le problème des Cavaliers

Cette variante d’un problème classique montre comment l’usage systématique de la forme
let conduit à des programmes clairs, concis et raisonnablement efficaces.

Deux Cavaliers blancs et deux Cavaliers noirs évoluent sur un quart d’échiquier
(un carré de quatre cases de côté) dont six cases sont inaccessibles et dont les
dix cases disponibles sont numérotées de 0 à 9; la situation initiale est illustrée
à la figure 19. On respecte la règle des Echecs concernant le mouvement du
Cavalier.

On s’intéresse aux longueurs minimales des séquences de mouvements
conduisant à une permutation des quatre pièces, aucune ne retrouvant sa place
initiale. On traitera aussi le cas particulier où les Cavaliers blancs prennent la
place des Cavaliers noirs et réciproquement. Enfin, on déterminera s’il existe
une position inaccessible pour les quatre Cavaliers, quel que soit le nombre de
mouvements.

Nous représentons une situation par une liste de quatre nombres donnant la position
des Cavaliers; la liste (a b c d) correspond à la situation dans laquelle

Figure 19: La situation initiale

• le Cavalier noir se trouvant initialement en 1 se trouve en a;
• le Cavalier noir se trouvant initialement en 3 se trouve en b;
• le Cavalier blanc se trouvant initialement en 6 se trouve en c;
• le Cavalier blanc se trouvant initialement en 0 se trouve en d.

Un mouvement possible est représenté par une paire pointée; la paire (a . b) représente
un mouvement de la position a vers la position b. Deux variables globales représentent la
situation initiale et la liste des mouvements possibles :

(define *init* ’(1 3 6 0))

(define *moves*

’((1 . 6) (1 . 8) (2 . 7) (2 . 9) (3 . 8) (4 . 5)

(4 . 9) (5 . 4) (6 . 1) (6 . 0) (7 . 2) (7 . 8)

(8 . 1) (8 . 3) (8 . 7) (9 . 2) (9 . 4) (0 . 6)))

On voit par exemple qu’un Cavalier se trouvant en position 8 peut aller en un coup à la
position 1, 3 ou 7 (si cette position est libre).

Les situations finales acceptables sont peu nombreuses; en effet, la liste (1 3 6 0)

admet 4! = 24 permutations, dont 9 sont des dérangements.100 Il est donc commode de
représenter ces situations finales acceptables, sous forme d’une liste explicite :

(define *final* ’(

(0 1 3 6) (0 6 1 3) (0 6 3 1)

100Rappelons qu’une permutation d’un ensemble E est une fonction p de E dans E telle que, pour tout
x ∈ E, il existe un et un seul y ∈ E tel que p(y) = x. Dans ce livre, l’ensemble E sera toujours fini; on
peut alors dire que la fonction p est une permutation de E si et seulement si p(E) = E. Un point fixe de p
est une solution de l’équation p(x) = x. Un dérangement de E est une permutation sans points fixe.

(3 0 1 6) (3 1 0 6) (3 6 0 1)

(6 0 1 3) (6 0 3 1) (6 1 0 3)))

Si on se limite aux permutations dans lesquelles les Cavaliers blancs prennent la place des
Cavaliers noirs (et réciproquement), on a :

(define *best* ’((0 6 1 3) (0 6 3 1) (6 0 1 3) (6 0 3 1)))

Vu l’énoncé du problème, les prises sont interdites. Une case-destination n’est envisageable
que si elle est libre, donc distincte des cases occupées. Le prédicat d’égalité arithmétique
permet de tester si deux cases sont identiques; il est commode de définir en outre

(define n= (lambda (x y) (not (= x y))))

On peut améliorer la lisibilité en rebaptisant les accesseurs pour les listes-situations et
pour les paires-mouvements, composées d’une case-origine et d’une case-destination :

(define k1 car) (define k2 cadr)

(define k3 caddr) (define k4 cadddr)

(define org car) (define dst cdr)

La fonction centrale du problème est la fonction succ. Elle prend comme arguments une
situation et un mouvement; si ce mouvement est possible dans la situation donnée, la
fonction renvoie la situation résultante, sinon elle renvoie #f. On a :

(define succ

(lambda (sit mv)

(let ((a (k1 sit)) (b (k2 sit)) (c (k3 sit)) (d (k4 sit))

(e (org mv)) (f (dst mv)))

(cond

((and (= a e) (n= b f) (n= c f) (n= d f))

(list f b c d))

((and (= b e) (n= a f) (n= c f) (n= d f))

(list a f c d))

((and (= c e) (n= a f) (n= b f) (n= d f))

(list a b f d))

((and (= d e) (n= a f) (n= b f) (n= c f))

(list a b c f))

(else #f)))))

Les cinq clauses de la forme cond correspondent aux cinq cas possibles : l’origine du
mouvement correspond à l’une des quatre positions de la situation donnée ou à aucune.

Sur base de la fonction succ, on peut définir des fonctions analogues prenant comme
arguments une situation et une liste de mouvements, ou une liste de situations et une liste
de mouvements, et renvoyant la liste des situations résultantes; on a

(define succs

(lambda (sit mvs)

(if (null? mvs)

’()

(let ((s1 (succ sit (car mvs)))

(s (succs sit (cdr mvs))))

(if s1 (insert_sit s1 s) s)))))

Une situation σ appartient à [[(succs sit mvs)]] s’il existe un mouvement µ appartenant
à [[mvs]] qui permette de passer de [[sit]] à σ.

Remarques. Nous utilisons insert_sit plutôt que cons parce qu’il est commode de gérer
des listes triées de situations. Notons aussi que la dernière ligne du code de succs utilise le
fait que la condition d’un if est assimilée à vrai dès qu’elle n’est pas fausse; c’est pourquoi
il était intéressant que la fonction précédente succ renvoie #f pour signaler l’absence de
situation résultante.

On écrit de même

(define succss

(lambda (sits mvs)

(if (null? sits)

’()

(let ((l1 (succs (car sits) mvs))

(l (succss (cdr sits) mvs)))

(merge_sits l1 l)))))

Une situation σ appartient à [[(succs sits mvs)]] s’il existe une situation ρ appartenant
à [[sits]] et un mouvement µ appartenant à [[mvs]] qui permette de passer de ρ à σ.
Remarque. La fonction merge_sits renvoie la fusion de deux listes triées. Nous
laissons au lecteur le soin de spécifier et de programmer les fonctions insert_sit et
merge_sits. Une solution simple consiste à adapter les fonctions vues au paragraphe 5.3.4
(tri lexicographique); on utilisera l’ordre numérique pour obtenir un ordre lexicographique
sur les situations. Le lecteur pourra également programmer les fonctions diff et inter,
calculant respectivement la différence et l’intersection de deux ensembles modélisés par
des listes triées.

La fonction succss permet en principe de résoudre le problème posé. En effet, par
applications successives, on peut construire l’ensemble des situations accessibles depuis la
situation initiale en un coup, en deux coups, etc. Cette approche näıve est peu efficace car
elle ne tient pas compte des cycles et, comme tout mouvement est réversible, les cycles sont
nombreux. Par exemple, la situation initiale est accessible à partir d’elle-même en zéro
coup, mais aussi en deux, en quatre, en six, etc., par exemple en répétant les mouvements
1-8, 8-1.

On définit donc une fonction new_succss qui ne donne que les “nouveaux” successeurs,
c’est-à-dire les situations non encore rencontrées. Les situations déjà rencontrées, et donc
dorénavant interdites, sont groupées dans un troisième argument. On a :

(define new_succss

(lambda (sits mvs forbid)

(let ((ss (succss sits mvs)))

(diff ss forbid))))

Une situation σ appartient à [[(new_succs sits mvs forbid)]] si elle appartient à
[[(succs sits mvs)]] sans appartenir à [[forbid]].

Après ces préliminaires, il est maintenant facile d’écrire une fonction qui, étant donnés
un nombre [[n]] de coups, une liste [[inits]] de situations initiales et une liste [[finals]]
de situations finales, détermine quelles situations finales sont accessibles en [[n]] coups ou
en moins de [[n]] coups. On a par exemple

(define gen

(lambda (n inits finals)

(if (= n 0)

(list 0 ’() inits 0 1 ’())

(let* ((rec1 (gen (- n 1) inits finals))

(rec (cdr rec1)))

(let ((old (car rec)) (new (cadr rec))

(lold (caddr rec)) (lnew (cadddr rec)))

(let ((old1 (merge old new))

(new1 (new_succss new *moves* old)))

(let ((lold1 (+ lold lnew)) (lnew1 (length new1))

(int (inter new1 finals)))

(newline)

(display (list n lold1 lnew1 int))

(list n old1 new1 lold1 lnew1 int))))))))

Les listes de situations pouvant être longues, la fonction gen donne aussi leur longueur;
de plus, elle imprime des résultats intermédiaires intéressants. Par exemple, on a

(gen 4 (list *init*) *final*)

(1 1 2 ())

(2 3 3 ())

(3 6 6 ())

(4 12 11 ())

==> (4

((1 2 6 0) (1 3 6 0) (1 7 6 0) (1 8 6 0) (2 3 6 0)

(7 3 1 0) (7 3 6 0) (7 8 6 0) (8 3 1 0) (8 3 1 6)

(8 3 6 0) (8 7 6 0))

((1 9 6 0) (2 3 1 0) (2 8 6 0) (3 7 6 0) (7 1 6 0)

(7 3 1 6) (7 3 8 0) (7 8 1 0) (8 2 6 0) (8 7 1 0)

(9 3 6 0))

12

11

())

La valeur renvoyée est une liste de six résultats. Le premier rappelle que quatre coups
consécutifs ont été joués. Le second résultat et le quatrième donnent la liste des situations
accessibles en moins de quatre coups et leur nombre (12). Le troisième résultat et le
cinquième donnent la liste des situations accessibles en quatre coups mais pas moins, et
leur nombre (11). Le sixième résultat est la liste des situations finales accessibles en quatre
coups mais pas moins; on constate que cette liste est vide. De plus, en cours d’exécution,
des résultats s’affichent pour un, deux, trois et quatre coups; ces résultats sont élagués :
ils comprennent les longueurs des listes de situations générées, mais pas ces listes elles-
mêmes, sauf pour la liste des situations finales accessibles.
Remarque. Il faut éviter la construction d’objets trop gros, et pour cela évaluer a priori la
taille des objets construits. Dans le cas présent, la taille maximale d’une liste de situations
est le nombre de quadruplets ordonnés de nombres distincts compris entre 0 et 9; ce nombre
est 10 ∗ 9 ∗ 8 ∗ 7 = 5040.

Pour résoudre le problème posé, l’utilisateur doit appliquer gen avec un premier
argument suffisamment grand. Des essais montrent que la valeur minimale permettant
de répondre à toutes les questions de l’énoncé est 49. Pour gagner de la place, on a omis
certaines lignes et naturellement abrégé la réponse finale.

(gen 49 (list *init*) *final*)

(1 1 2 ())

...

(25 2035 191 ())

(26 2226 171 ((3 0 1 6) (6 1 0 3)))

(27 2397 145 ())

(28 2542 128 ((0 1 3 6) (3 1 0 6) (3 6 0 1)))

(29 2670 120 ())

...

(39 4166 181 ())

(40 4347 173 ((0 6 1 3) (6 0 1 3) (6 0 3 1)))

(41 4520 152 ())

(42 4672 120 ((0 6 3 1)))

(43 4792 94 ())

...

(49 5037 3 ())

==> (49

((0 1 2 3) (0 1 2 4) ... (9 8 7 5) (9 8 7 6))

((2 4 9 5) (2 9 4 5) (9 2 4 5))

5037

3

())

Les conclusions sont les suivantes :

• Le nombre de coups conduisant à une situation finale est de 26 au minimum.
• Le nombre de coups conduisant à une situation finale permutant les couleurs est de

40 au minimum.
• Toute situation est accessible en moins de 50 coups.
• Les trois situations les moins accessibles sont (2 4 9 5), (2 9 4 5) et (9 2 4 5);

49 coups sont nécessaires.

Le problème des Cavaliers n’est que l’un des nombreux représentants de la classe des
problèmes dits “de recherche” ou “d’exploration dans un espace d’état”. Les problèmes
de cette classe comportent tous un certain ensemble structuré (ici, les 5 040 situations
possibles) et un chemin à déterminer dans cette ensemble (ici, une suite de mouvements).
Fondamentalement, il n’y a aucune difficulté si ce n’est la taille de l’espace et le nombre
potentiellement très élevé de chemins possibles.

Dans le cas présent, aucune tactique n’est requise pour réduire le travail de
recherche, car 5 040 situations correspondent à un espace très réduit. Néanmoins, à
titre d’illustration, nous présentons une tactique dont l’emploi est fréquemment nécessaire
en pratique. La fonction gen construit les chemins depuis la situation initiale jusqu’à
la situation finale. Faire l’inverse ne serait ni plus ni moins efficace puisque chaque
mouvement est réversible.101 Par contre, on peut attendre une amélioration de l’efficacité
en combinant une recherche vers l’avant et une recherche vers l’arrière. La variante gen2

construit les chemins à partir de leurs deux extrémités, de manière symétrique. On a :

(define (gen2 n inits finals)

(if (= n 0)

(list 0

(list ’() inits 0 1)

(list ’() finals 0 9)

’()

’())

(let* ((rec1 (gen2 (- n 1) inits finals))

(recf (cadr rec1)) (recb (caddr rec1)))

(let ((fold (car recf)) (fnew (cadr recf))

(flold (caddr recf)) (flnew (cadddr recf))

(bold (car recb)) (bnew (cadr recb))

(blold (caddr recb)) (blnew (cadddr recb)))

(let ((fold1 (merge fold fnew))

(fnew1 (new_succss fnew *moves* fold))

(bold1 (merge bold bnew))

(bnew1 (new_succss bnew *moves* bold)))

(let ((flold1 (+ flold flnew)) (flnew1 (length fnew1))

(blold1 (+ blold blnew)) (blnew1 (length bnew1))

(i_a (inter fold1 bnew1))

101Ce n’est pas le cas pour tous les problèmes de recherche et, parfois, enchâıner les mouvements vers
l’arrière plutôt que vers l’avant accrôıt très significativement l’efficacité de la recherche.

(i_b (inter fnew1 bnew1)))

(newline)

(display (list n i_a i_b))

(list n

(list fold1 fnew1 flold1 flnew1)

(list bold1 bnew1 blold1 blnew1)

i_a

i_b)))))))

La liste renvoyée comporte le nombre n = [[n]] d’itérations, un quadruplet concernant
la progression vers l’avant (“f” signifie “forward”), un quadruplet analogue concernant
la progression vers l’arrière (“b” signifie “backward”) et deux listes [[i_a]] et [[i_b]]
comportant les situations apparaissant à l’intersection des fronts avant et arrière. Le
quadruplet “avant” comporte les listes des situations accessibles par l’avant en moins de
n coups et en exactement n coups, et les longueurs de ces listes. La liste [[i_a]] contient
les situations accessibles par l’avant en moins de n pas et, simultanément, accessibles
par l’arrière en exactement n pas. De même, la liste [[i_b]] contient les situations
simultanément accessibles par l’avant et par l’arrière, en exactement n pas dans les deux
cas. Ces deux listes, si elles ne sont pas vides, témoignent de l’existence d’un ou plusieurs
chemin(s) de longueur moindre que 2n, ou égale à 2n, respectivement.

Puisqu’avec la fonction gen, 49 itérations suffisaient pour répondre à toutes les
questions de l’énoncé, il n’en faudra plus que 25 avec la fonction gen2, comme l’indique
l’exécution suivante :

(gen2 25 (list *init*) *final*)

(1 () ())

...

(12 () ())

(13 () ((2 9 7 8) (9 8 2 7)))

(14 ((2 9 7 1) (9 3 2 7)) ((2 4 7 8) ... (9 7 2 3)))

...

(25 ((1 6 8 0) ... (8 6 3 0)) ())

==> (25

(((0 2 1 3) (0 2 3 1) ... (9 8 7 6))

((0 2 1 6) ... (9 8 4 0))

2035

191)

(((0 1 2 3) ... (9 8 7 6))

((1 6 8 0) ... (8 6 3 0))

5001

24)

((1 6 8 0) ... (8 6 3 0))

())

On a vu précédemment que 26 étapes suffisaient pour passer de la situation initiale
(1 3 6 0) à l’une des situations finales (3 0 1 6) et (6 1 0 3); la présente exécution

montre que les chemins réalisant ce transfert ont nécessairement pour étape médiane la
situation (2 9 7 8) ou la situation (9 8 2 7).

Remarques. Dans le cas présent, aborder le problème “des deux côtés”, c’est-à-dire depuis
la situation initiale et depuis la situation finale, n’abrège pas la recherche, puisque tout
l’espace est exploré. Notons déjà que dans beaucoup de problèmes analogues, l’espace
à explorer est trop grand pour être construit en entier, ce qui peut rendre nécessaire
l’emploi d’une tactique plus fine que la recherche exhaustive et systématique. Un défaut
potentiel des programmes présentés ici est qu’ils ne fournissent pas la suite de mouvements
permettant de passer de la situation initiale à l’une des situations finales. On peut
facilement adapter les programmes pour remédier à cette lacune, mais au prix d’une
consommation de ressources nettement plus élevées. La raison en est que dans une
situation donnée, plusieurs mouvements sont en général possibles, ce qui donne lieu à une
explosion combinatoire du nombre de chemins à construire et à explorer. Nous reviendrons
sur ces points au chapitre 11 mais auparavant il convient de souligner que, parfois, cette
explosion combinatoire peut être évitée par un raisonnement simple. Un cas extrême de
cette situation favorable est abordé au paragraphe suivant.

9.8 Le problème des cruches

Ce problème classique montre que, dans certains cas, l’explosion combinatoire peut être
entièrement évitée. Plus précisément, une situation intermédiaire semble avoir plusieurs
successeurs possibles, mais un seul d’entre eux mérite d’être considéré.

On dispose de deux cruches dont les contenances en litres sont respectivement
a et b et d’une source inépuisable d’eau. On demande de prélever exactement
c litres d’eau en un nombre minimum d’opérations. Les nombres a, b et c sont
des entiers naturels distincts tels que a < b et c < a + b. On admet qu’une
cruche ne permet de prélever avec précision que sa contenance nominale.

On remarque d’emblée qu’il existe six manipulations possibles :

1. Remplir la petite cruche.
2. Remplir la grande cruche.
3. Transvaser de la petite cruche vers la grande.
4. Transvaser de la grande cruche vers la petite.
5. Vider la petite cruche.
6. Vider la grande cruche.

Après remplissage, une cruche contient exactement sa contenance nominale.
Transvaser une cruche dans l’autre ne modifie pas le contenu global des deux cruches. Un
point important est qu’avant et après chaque opération, au plus une des deux cruches peut
avoir un contenu autre que nul ou maximal, car toute opération, y compris le transvasement
d’une cruche dans l’autre, a pour effet de remplir ou de vider une cruche.

Il est clair qu’à la première étape on a deux possibilités : remplir la petite cruche,
ou remplir la grande. On observe aussi qu’à chaque étape ultérieure on n’a qu’une seule
possibilité intelligente. En effet, exactement une étape sur deux est un transvasement (en
provenance de la cruche pleine, s’il y en a une, ou à destination de la cruche vide sinon);
ce transvasement a pour effet, soit de vider la cruche de départ, soit de remplir la cruche
d’arrivée. Dans le premier cas, l’étape suivante consiste à remplir la cruche vide; dans le
second cas, l’étape suivante consiste à vider la cruche pleine.

On représente une situation par une paire pointée dont les composants sont les contenus
de la petite et de la grande cruche, respectivement. Une situation est dite initiale si l’une
des cruches est pleine et l’autre vide. Une situation est dite intéressante si l’une des
cruches est pleine ou vide tandis que l’autre n’est ni pleine ni vide. Une situation est
convenable si elle est initiale ou intéressante.

On définit une fonction next_sit à quatre arguments : les contenances p et g de la
petite cruche et de la grande cruche, une situation intéressante sit, et l’un des symboles
trans et in/out; cette fonction renvoie la seule situation convenable atteignable en un
transvasement (si le quatrième argument est trans) ou en remplissant la cruche vide ou
en vidant la cruche pleine (si le quatrième argument est in/out) :

(define next_sit

(lambda (p g sit op)

(let ((p1 (car sit)) (g1 (cdr sit)))

(if (eq? op ’trans)

(cond ((= p1 p)

(let ((dg (- g g1)))

(if (> p dg)

(cons (- p dg) g)

(cons 0 (+ g1 p)))))

((= p1 0)

(if (> p g1)

(cons g1 0)

(cons p (- g1 p))))

((= g1 g) (cons p (- g1 (- p p1))))

((= g1 0) (cons 0 p1)))

(cond ((= p1 p) (cons 0 g1))

((= p1 0) (cons p g1))

((= g1 g) (cons p1 0))

((= g1 0) (cons p1 g)))))))

Le processus étant déterministe, on peut écrire une fonction seq_sits générant la séquence
de toutes les situations distinctes issues d’une situation convenable donnée :

(define seq_sits

(lambda (p g sit op past)

(if (member sit past)

past

(let ((new_sit (next_sit p g sit op))

(op1 (if (eq? op ’trans) ’in/out ’trans)))

(seq_sits p g new_sit op1 (cons sit past))))))

La valeur de (seq_sits p g sit op past) est la liste des situations que l’on peut
atteindre au départ de la situation initiale sit en alternant les opérations trans et in/out
et sans passer par une situation élément de [[past]]; la première opération effectuée est
[[op]] et les contenances des cruches sont [[p]] et [[g]].

A titre d’exemple, considérons le cas de cruches de huit et onze litres. Les deux
situations initiales convenables sont (8 . 0) et (0 . 11). La fonction seq_sits permet
de construire toutes les situations que l’on peut obtenir à partir des situations initiales;
on peut utiliser reverse pour que ces situations soient énumérées dans l’ordre naturel :

(reverse (seq_sits 8 11 ’(8 . 0) ’trans ’()))

==>

((8 . 0) (0 . 8) (8 . 8) (5 . 11) (5 . 0) (0 . 5)

(8 . 5) (2 . 11) (2 . 0) (0 . 2) (8 . 2) (0 . 10)

(8 . 10) (7 . 11) (7 . 0) (0 . 7) (8 . 7) (4 . 11)

(4 . 0) (0 . 4) (8 . 4) (1 . 11) (1 . 0) (0 . 1)

(8 . 1) (0 . 9) (8 . 9) (6 . 11) (6 . 0) (0 . 6)

(8 . 6) (3 . 11) (3 . 0) (0 . 3) (8 . 3) (0 . 11) (8 . 11))

(reverse (seq_sits 8 11 ’(0 . 11) ’trans ’()))

==>

((0 . 11) (8 . 3) (0 . 3) (3 . 0) (3 . 11) (8 . 6)

(0 . 6) (6 . 0) (6 . 11) (8 . 9) (0 . 9) (8 . 1)

(0 . 1) (1 . 0) (1 . 11) (8 . 4) (0 . 4) (4 . 0)

(4 . 11) (8 . 7) (0 . 7) (7 . 0) (7 . 11) (8 . 10)

(0 . 10) (8 . 2) (0 . 2) (2 . 0) (2 . 11) (8 . 5)

(0 . 5) (5 . 0) (5 . 11) (8 . 8) (0 . 8) (8 . 0) (0 . 0))

Si on souhaite prélever quinze litres, les situations finales adéquates sont (8 . 7) et
(4 . 11); on observe dans les listes ci-dessus que 17 étapes sont nécessaires, puisque
la paire (8 . 7) est le 17ième élément de la première liste; la paire (4 . 11) n’apparâıt
qu’en 19ième position, dans la seconde liste.

Si on observe plus attentivement ces deux listes, il apparâıt que, abstraction faite
du dernier élément de chacune d’elles (qui correspondent aux situations particulières où
les deux cruches sont vides, et où les deux cruches sont pleines, ces listes sont inverses
l’une de l’autre. La longueur commune de ces listes est 36. Deux situations séparées
par un transvasement ne sont pas essentiellement différentes, en ce sens que le contenu
total des deux cruches est le même; il y a donc 18 situations essentiellement différentes,
correspondant à tous les contenus totaux possibles, de 1 à 18 litres. A posteriori, il
est évident que, si les contenances des deux cruches sont a et b, il y a au maximum

a+ b− 1 situations intéressantes essentiellement différentes. On peut démontrer qu’il y en
a exactement a + b− 1 si a et b sont premiers entre eux. Naturellement, la démonstration
doit résulter d’une démarche mathématique, mais le programme et ses exécutions peuvent
suggérer la conjecture à démontrer.

10 Abstraction : données et algorithmes

La notion d’abstraction introduite au chapitre précédent et concrétisée par le mécanisme
let et ses variantes, facilite la structuration d’un programme, vu comme un ensemble de
fonctions. Ce mécanisme permet de donner des noms locaux à des objets, ce qui facilite la
construction d’autres objets utilisant les premiers; plus précisément, ce mécanisme donne
une vue simplifiée, “abstraite”, d’un objet complexe, dans laquelle des sous-objets ne sont
pas décrits mais simplement mentionnés par leur nom.

Une autre démarche d’abstraction, que nous étudions dans ce chapitre et le suivant,
consiste à repérer, dans un ensemble d’objets distincts, des points communs significatifs.
Ces points communs peuvent être étudiés pour eux-mêmes, ce qui peut faciliter la
construction ultérieure d’objets analogues, par la récupération de ce qui concerne les points
communs.

10.1 Abstraction sur les données

Souvent, les données et résultats d’un même problème peuvent se représenter concrètement
(en machine) de plusieurs manières, chacune pouvant avoir ses avantages. Changer de
structures de données concrètes implique de nombreuses modifications éparses dans les
programmes, ce qui est un inconvénient majeur. Nous montrons dans ce paragraphe
comment on peut éviter ce problème. La meilleure méthode consiste à utiliser des
données abstraites. Même quand cela ne semble pas pratiquement possible, il est toujours
intéressant de découpler les algorithmes “de haut niveau” d’une part, et ceux liés à la
représentation des données en machine d’autre part; seuls ces derniers seront à revoir en
cas de changement de représentation.

10.1.1 Deux exemples classiques

On représente habituellement un nombre rationnel par un couple d’entiers (num,den).
Deux possibilités apparaissent d’emblée : n’admettre que la forme réduite (dénominateur
positif, numérateur et dénominateur premiers entre eux) ou autoriser aussi les formes non
réduites. Dans le premier cas, il faut décider si la réduction a lieu dès que le rationnel
est construit, ou seulement quand il est utilisé et/ou affiché. En ce qui concerne la
représentation concrète en machine, il faut encore décider si un rationnel sera une paire
pointée (un arbre binaire) ou une liste. Le premier choix est plus économique en ce qui
concerne l’utilisation de la mémoire; la paire pointée (2 . 3) occupe une cellule tandis
que la liste (2 3), c’est-à-dire (2 . (3 . ())), en occupe deux. Par contre, à l’affichage,
(2 3) est (un peu) plus lisible que (2 . 3).

La conclusion est qu’il faut laisser toutes les possibilités ouvertes, et minimiser et
localiser au mieux les fragments de programmes dépendant de la représentation adoptée.
Nous savons que la chose est possible . . . puisque nous l’avons pratiquée inconsciemment
depuis le début, en écrivant des algorithmes manipulant, par exemple, des nombres entiers
et des listes. En effet, nous n’avons pas précisé comment le système Scheme représente

les nombres, et ce n’est qu’au paragraphe 8.2 que nous avons donné quelques indications
sur la manière dont Scheme représente les listes. Cela ne nous a pas empêché d’écrire
des programmes parce que nous ne manipulons les données que par le biais de leurs
constructeurs et accesseurs, dont les spécifications peuvent être vues comme un système
d’axiomes.102 Comme les entiers et les listes sont des types primitifs du système Scheme,
nous n’avons pas à écrire les programmes manipulant les représentations concrètes en
mémoire, c’est-à-dire le code lié aux fonctions primitives +, cons, car, etc. Pour les types
non primitifs, nous devrons les écrire, et éventuellement les modifier si nous changeons la
manière dont ces types non primitifs sont réalisés en termes des types primitifs de Scheme.
Dans un premier temps, nous nous efforcerons de définir les types non primitifs de manière
abstraite, avec constructeurs et accesseurs. La partie “changeante” de la programmation
de ces types se limitera alors à la programmation des constructeurs et accesseurs.

On peut imaginer par exemple les rationnels abstraits, basés sur le constructeur
make-ratl et les accesseurs numr et denr. On sait que, si n et d sont des entiers
(d 6= 0), (make-ratl n d) est un rationnel égal à n/d; d’autre part, si r est un rationnel,
alors les valeurs de (numr r) et (denr r) sont les numérateur et dénominateur d’une
fraction (réduite ou non) correspondant à r. On considère séparément les problèmes
d’utilisation des rationnels (via le constructeur make-ratl et les accesseurs numr et denr)
et le problème de la réalisation de ces derniers (en termes de primitives Scheme). Le
programme d’addition de deux rationnels abstraits sera

(define r+

(lambda (x y)

(make-ratl

(+ (* (numr x) (denr y)) (* (numr y) (denr x)))

(* (denr x) (denr y)))))

C’est une simple traduction de l’égalité

a

b
+

c

d
=

ad + bc

bd
.

Ce programme est indépendant de la manière dont les nombres rationnels sont réalisés en
machine; il ne doit donc pas être modifié si on change de mode de réalisation; dans ce cas,
seules les primitives make-ratl, (numr r) et (denr r) devront être récrites. Un choix
raisonnable pour ces primitives est

(define numr (lambda (rtl) (car rtl)))

(define denr (lambda (rtl) (cadr rtl)))

(define make-ratl

(lambda (int1 int2)

102En ce qui concerne les nombres entiers, ce système axiomatique est une variante du système de Peano,
qui est à la base de l’étude de l’arithmétique en logique formelle.

(if (zero? int2)

(error "make-ratl: The denominator cannot be zero.")

(list int1 int2))))

Le rationnel 2/3 admet alors comme représentations des listes telles que (2 3) (-10 -15).
Remarque. La procédure error interrompt le calcul et affiche son argument, qui est une
châıne de caractères.

Un autre choix possible est

(define numr (lambda (rtl) (car rtl)))

(define denr (lambda (rtl) (cdr rtl)))

(define make-ratl

(lambda (int1 int2)

(if (zero? int2)

(error "make-ratl: The denominator cannot be zero.")

(let ((g (gcd int1 int2)) (s (if (> int2 0) 1 -1)))

(cons (/ int1 g s) (/ int2 g s))))))

Ici, la seule représentation adéquate de 2/3 est la paire pointée (2 . 3). Le point
important n’est pas de faire le bon choix de représentation, mais de permettre l’écriture
de programmes d’application indépendants de ce choix.

Un autre exemple classique est le polynôme formel sur l’ensemble des entiers. En
effet, un polynôme tel que 3x5 + 2x3 − 4x − 7 , se représente naturellement par la liste
des coefficients, ici (3 0 2 0 -4 -7), ou encore par la liste des coefficients non nuls
accompagnés du degré du monôme correspondant, ce qui donne ici ((5 . 3) (3 . 2)

(1 . -4) (0 . -7)) . La première solution est préférable pour les polynômes “pleins”,
la seconde pour les polynômes “creux”. Pour ne pas devoir écrire deux jeux (ou plus)
de programmes d’application sur les polynômes, on utilisera une représentation abstraite.
Nous n’en donnons ici que les principes. Un polynôme est soit le polynôme nul, soit un
triplet composé d’un degré, d’un coefficient du terme de plus haut degré et d’un reste.
Pour le polynôme 3x5 + 2x3 − 4x − 7 , les éléments de ce triplet sont respectivement 5,
3 et 2x3 − 4x − 7 . Une réalisation concrète se composera donc du constructeur sans
argument the-zero-poly et du constructeur à trois arguments poly-cons, ainsi que
des trois accesseurs degree, coeff et rest-poly. On aura aussi deux reconnaisseurs,
zero-poly? et poly?. Tout programme d’application s’écrira en termes de ces primitives.
En cas de changement de mode de représentation des polynômes, seules les primitives
devront être récrites.

10.1.2 Arbres binaires complètement étiquetés

Nous avons déjà rencontré le type arbre binaire et son équivalent en Scheme, le domaine
des expressions symboliques. Un arbre binaire est, soit un atome, soit un couple d’arbres

binaires. Intuitivement et graphiquement, les arbres binaires sont des arborescences de
degré 2 dont les feuilles s’identifient à des étiquettes atomiques. On peut imaginer des
arborescences de degré 2 dont chaque nœud porterait une étiquette; en fait, de telles
arborescences interviennent dans des problèmes pratiques variés. Il est intéressant de
créer un domaine structuré pour de tels arbres.

Un K-arbre binaire complètement étiqueté103 est soit l’arbre vide, soit un triplet
comportant une clef (“key”) élément de K, un sous-arbre de gauche et un sous-arbre de
droite. On aura donc un constructeur sans argument pour l’arbre vide et un constructeur
à trois arguments pour les arbres non vides; on aura également trois accesseurs. Nous
donnons aussi une représentation concrète simple : l’arbre vide s’identifie à la liste vide et
un arbre non vide à la liste de ses trois composants; les constructeurs sont :

(define mk_e_tree (lambda () ’()))

(define mk_k_tree (lambda (k l r) (list k l r)))

Les reconnaisseurs sont :

(define e_tree? null?)

(define k_tree?

(lambda (x) ;; x est un objet quelconque

(and (pair? x) (key? (car x))

(pair? (cdr x)) (ek_tree? (cadr x))

(pair? (cddr x)) (ek_tree? (caddr x))

(null? (cdddr x)))))

(define ek_tree?

(lambda (x) (or (e_tree? x) (k_tree? x))))

(define key? (lambda (x) (and (integer? x) (>= x 0))))

Les accesseurs sont :

(define key car)

(define left cadr)

(define right caddr)

Nous donnons ici les versions concrète et abstraite de l’arbre représenté à la figure 20 :

(define conc_tree ’(4 (2 (1 () ()) ()) (5 (3 () ()) (7 () ()))))

(define abst_tree

(mk_k_tree 4

(mk_k_tree 2

(mk_k_tree 1 (mk_e_tree) (mk_e_tree))

(mk_e_tree))

103Dans ce paragraphe, le mot “arbre” désignera le type d’arbre que nous introduisons maintenant. De
plus, K sera l’ensemble N des entiers naturels.

(mk_k_tree 5

(mk_k_tree 3 (mk_e_tree) (mk_e_tree))

(mk_k_tree 7 (mk_e_tree) (mk_e_tree)))))

On laisse au lecteur le soin de définir un schéma de récursion sur les arbres, comme
nous l’avons fait précédemment pour d’autres types de données. De nombreux problèmes
simples seront résolus par des instances de ce schéma. A titre d’exemple, voici le
programme qui détermine si un nombre entier donné apparâıt comme clef dans un arbre
donné (dont toutes les clefs sont supposées numériques) :

(define in_tree?

(lambda (i tr)

(if (e_tree? tr)

#f

(or (= i (key tr))

(in_tree? i (left tr))

(in_tree? i (right tr))))))

Ceci peut se récrire en :

(define in_tree?

(lambda (i tr)

(and (not (e_tree? tr))

(or (= i (key tr))

(in_tree? i (left tr))

(in_tree? i (right tr))))))

Cependant, certains problèmes requièrent une approche un peu moins simple. Un arbre
non vide est dit conditionné ou ordonné si la clef de tout nœud interne est plus grande
ou égale aux clefs de tous ses descendants de gauche, et plus petite ou égale aux clefs de
tous ses descendants de droite. Il n’est pas très difficile d’écrire un prédicat déterminant
si un arbre est conditionné ou non, mais deux écueils sont à éviter. D’une part, l’approche
“näıve” selon laquelle un arbre non vide serait conditionné si la clef de la racine est
comprise entre les clefs des deux fils (s’il existent) et si les deux sous-arbres fils sont eux-
mêmes conditionnés est manifestement incorrecte; la condition est nécessaire mais pas
suffisante, ainsi que le montre le contre-exemple de la figure 20.

Par ailleurs, l’approche “prudente” selon laquelle un arbre non vide serait conditionné
si la clef de la racine est supérieure à tous ses descendants de gauche et inférieure à tous
ses descendants de droite et si les deux sous-arbres fils sont eux-mêmes conditionnés est
manifestement inefficace, parce que les mêmes comparaisons sont répétées plusieurs fois.

En fait, un arbre est conditionné si ses deux fils sont conditionnés et si sa racine est
supérieure à tous les éléments de la branche la plus à droite du fils gauche, et inférieure
à tous les éléments de la branche la plus à gauche du fils droit. Les prédicats auxiliaires
greq? et leeq? testent les deux dernières conditions.



10 ABSTRACTION : DONNÉES ET ALGORITHMES 189

Figure 20: Un arbre non conditionné

(define greq?

(lambda (n tr)

(or (e_tree? tr)

(and (>= n (key tr)) (greq? n (right tr))))))

(define leeq?

(lambda (n tr)

(or (e_tree? entr)

(and (<= n (key tr)) (leeq? n (left tr))))))

(define condit_1?

(lambda (tr)

(or (e_tree? tr)

(and (condit_1? (left tr))

(greq? (key tr) (left tr))

(condit_1? (right tr))

(leeq? (key tr) (right tr))))))

Ce programme n’est pas optimal; une version plus efficace est possible si on dispose d’une
borne supérieure *max* absolue pour les étiquettes des arbres. On peut alors définir

(define condit_2?

(lambda (tr) (or (e_tree? tr) (tree_ok? 0 tr *max*))))

(define tree_ok?

(lambda (min tr max)

(and (<= min (key tr))

(>= max (key tr))

(or (e_tree? (left ntr))

(tree_ok? min (left ntr) (key ntr)))

(or (e_tree? (right ntr))

(tree_ok? (key ntr) (right ntr) max)))))

Si un arbre est conditionné, la liste de ses étiquettes est triée, à condition que dans cette

liste toute étiquette se trouve entre les étiquettes de ses descendants de gauche et celles
de ses descendants de droite. La fonction traversal calcule cette liste :

(define traversal

(lambda (tr)

(if (e_tree? tr)

’()

(append (traversal (left tr))

(cons (key tr) (traversal (right tr)))))))

On peut utiliser la technique des accumulateurs introduite au paragraphe 7.3 pour éviter
l’usage de append, en écrivant une fonction auxiliaire traversal_a telle que

[[(traversal_a tr acc)]] = [[(append (traversal tr) acc)]] .

(define traversal_a

(lambda (tr acc)

(if (e_tree? tr)

acc

(traversal_a (left tr)

(cons (key tr)

(traversal_a (right tr) acc))))))

(define traversal (lambda (tr) (traversal_a tr ’())))

10.1.3 Arbres, tas et tri

Une arbre est un tas si l’étiquette d’un nœud est supérieure aux étiquettes de ses
descendants. La notion de tas est utile dans diverses applications. Le programme
heap? teste si un arbre est un tas (“heap” en anglais); il est analogue au programme
condit_2. On observera, dans les liaisons locales introduites par let, que les règles de
portée empêchent toute confusion entre les liaisons locales et globales de key, left et
right.104

(define heap?

(lambda (tr)

(or (e_tree? tr)

(let ((key (key entr))

(left (left tr)) (right (right entr)))

(and (greq? key left)

(greq? key right)

(heap? left)

(heap? right))))))

104Les liaisons locales sont des arbres, les liaisons globales sont des accesseurs.

(define heapify

(lambda (tr)

(if (e_tree? tr)

tr

(let ((key (key tr))

(left (left tr)) (right (right tr)))

(let ((lh (heapify left)) (rh (heapify right)))

(adjust key lh rh))))))

(define adjust

(lambda (ky lh rh)

(cond ((and (greq? ky lh) (greq? ky rh))

(mk_k_tree ky lh rh))

((greq? ky lh)

(let

((krh (key rh)) (lrh (left rh)) (rrh (right rh)))

(mk_k_tree krh lh (adjust ky lrh rrh))))

((greq? ky rh)

(let

((klh (key lh)) (llh (left lh)) (rlh (right lh)))

(mk_k_tree klh (adjust ky llh rlh) rh)))

(else

(let ((klh (key lh)) (krh (key rh)))

(let ((llh (left lh)) (rlh (right lh))

(lrh (left rh)) (rrh (right rh)))

(if (> klh krh)

(mk_k_tree klh (adjust ky llh rlh) rh)

(mk_k_tree krh lh (adjust ky lrh rrh)))))))))

Si dans la liste des étiquettes d’un tas, l’étiquette d’un nœud vient toujours avant
l’étiquette des descendants de ce nœud, alors la liste est “presque” triée par ordre
décroissant. Une variante du prédicat traversal permet de le vérifier. Nous écrivons
cette variante en utilisant un letrec et un accumulateur :

(define pre_trav

(lambda (tr)

(letrec

((pre_trav_a

(lambda (tr acc)

(if (e_tree? tr)

acc

(cons (key tr)

(pre_trav_a (left tr)

(pre_trav_a (right tr)



10 ABSTRACTION : DONNÉES ET ALGORITHMES 192

acc)))))))

(pre_trav_a tr ’()))))

Considérons l’arbre dont la représentation concrète est

(5

(3 (5 (3 () (5 (4 () (7 () ())) (3 () ()))) (6 () ()))

(4 (5 (3 () (4 () ())) (6 () ())) ()))

(6 () (5 (4 () (7 () ())) (3 () ()))))

Cet arbre n’est pas un tas, mais la fonction heapify le transforme en le tas

(7

(7 (6 (5 () (5 (4 () (3 () ())) (3 () ()))) (3 () ()))

(6 (5 (4 () (3 () ())) (4 () ())) ()))

(6 () (5 (5 () (4 () ())) (3 () ()))))

dont on voit immédiatement qu’il a même structure que l’arbre initial. La fonction
pre_trav, appliquée à l’arbre initial et au tas résultant, fournit respectivement les listes

(3 4 7 5 3 5 6 3 3 4 5 6 4 5 6 4 7 5 3)

(7 7 6 5 5 4 3 3 3 6 5 4 3 4 6 5 5 4 3)

alors que la version triée de ces deux listes est

(7 7 6 6 6 5 5 5 5 5 4 4 4 4 3 3 3 3 3)

Ceci suggère qu’il devrait exister une variante de pre_trav qui, appliquée à un tas,
fournirait la liste triée des étiquettes de ce tas. On peut utiliser le code suivant :

(define hp_sort_trav

(lambda (hp)

(if (e_tree? hp)

’()

(let ((u1 (hp_sort_trav (left hp)))

(u2 (hp_sort_trav (right hp)))

(k (key hp)))

(cons k (merge u1 u2))))))

(define merge

(lambda (u1 u2)

(cond ((null? u1) u2)

((null? u2) u1)

(else (let ((a1 (car u1)) (a2 (car u2)))

(if (> a1 a2)

(cons a1 (merge (cdr u1) u2))

(cons a2 (merge u1 (cdr u2)))))))))

La fonction principale hp_sort_trav prend un tas comme argument et fournit la liste
triée des étiquettes. La fonction auxiliaire merge “fusionne” deux listes triées.
Remarque. Cette technique de tri n’est pas optimale mais elle est raisonnablement efficace;
la fonction merge peut être utile dans des contextes variés, aussi ne l’a-t-on pas cachée
dans un letrec.

Nous terminons ce paragraphe par un programme permettant la visualisation de la
structure des arbres; il est analogue à celui utilisé au paragraphe § 8.7 pour les expressions
symboliques.

(define space ;; (space n) ecrit n blancs

(lambda (n)

(if (zero? n)

(display "")

(begin (display " ") (space (- n 1))))))

(define pr_tree

(lambda (tr d)

(if (e_tree? tr)

(begin (space d) (display " -"))

(let ((d1 (1+ d)))

(begin (space d1) (display (key tr))

(newline) (pr_tree (left tr) d1)

(newline) (pr_tree (right tr) d1))))))

10.1.4 Le type enregistrement

Les K-arbres binaires complètement étiquetés sont des cas particuliers d’enregistrements.
Un enregistrement est une structure de donnée admettant un nombre fixé de composants,
chacun d’eux ayant un type donné. Nous verrons plus loin que le langage Scheme dispose
d’un type de donnée particulier, le vecteur, qui permet une réalisation simple et efficace
des enregistrements. On peut aussi décider de réaliser concrètement les enregistrements au
moyen de listes de taille constante, ainsi que nous l’avons fait aux paragraphes précédents
pour les arbres binaires complètement étiquetés. Nous considérons brièvement ici le cas
général du problème de la réalisation d’un type enregistrement.

Le reconnaisseur record? prend comme arguments un objet [[u]] et une liste de
propriétés [[lp]] et renvoie #t si [[u]] et [[lp]] sont des listes de même longueur ℓ
et si pour tout i = 1, . . . , ℓ, le ième objet de [[u]] satisfait la ième propriété
de [[lp]].

Remarque. Une propriété est ici un prédicat à un argument.

Une solution simple et efficace est



10 ABSTRACTION : DONNÉES ET ALGORITHMES 194

(define record?

(lambda (u lp)

(or (and (null? u) (null? lp))

(and (pair? u) (pair? lp)

((car lp) (car u))

(record? (cdr u) (cdr lp))))))

Le reconnaisseur k_tree? introduit plus haut, à savoir

(define k_tree?

(lambda (x) ;; x est un objet quelconque

(and (pair? x) (key? (car x))

(pair? (cdr x)) (ek_tree? (cadr x))

(pair? (cddr x)) (ek_tree? (caddr x))

(null? (cdddr x)))))

(define e_tree? null?)

(define ek_tree? (lambda (x) (or (e_tree? x) (k_tree? x))))

(define key? (lambda (x) (and (integer? x) (>= x 0))))

peut se définir de manière équivalente, en termes de record? :

(define k_tree?

(lambda (u)

(record? u

(list (lambda (x) (and (integer? x) (>= x 0)))

(lambda (v) (or (null? v) (k_tree? v)))

(lambda (v) (or (null? v) (k_tree? v)))))))

Nous laissons au lecteur le soin de développer d’autres primitives pour le type abstrait
générique enregistrement.

10.2 Les graphes

10.2.1 Deux modes de représentation

Intuitivement, un graphe orienté est une structure formée de nœuds et d’arcs. Les nœuds
sont des objets quelconques et les arcs sont des paires ordonnées de nœuds. On peut
voir les graphes comme des généralisations naturelles des notions de listes et d’arbres.
La figure 21 représente un graphe comportant six nœuds et huit arcs. Il ne semble pas
facile de créer un domaine structuré des graphes de manière abstraite, au moyen d’un
constructeur et d’accesseurs. Par contre, il n’est pas difficile de représenter un graphe au
moyen de structures déjà introduites. Un premier mode de représentation, appelé Mode I

dans la suite, consiste à voir un graphe comme un arbre binaire dont le fils gauche est la
liste des nœuds et le fils droit est la liste des arcs, un arc étant lui-même un arbre binaire

a b

cde

f

Figure 21: Représentation d’un graphe orienté

dont le fils gauche est l’origine et le fils droit l’extrémité. Le graphe de la figure 21 peut
s’écrire

c([a, b, c, d, e, f ], [c(b, a), c(b, d), c(c, b), c(d, c), c(d, f), c(e, b), c(e, f), c(f, a)]) .

Un second mode, appelé Mode II dans la suite, consiste à représenter un graphe par une
liste de listes de nœuds. Chaque liste de nœuds a pour premier élément un nœud x du
graphe et pour reste la liste des successeurs de x. Un nœud y est successeur du nœud x
dans un graphe si l’arc d’origine x et d’extrémité y appartient à ce graphe. Le graphe de
la figure 21 peut s’écrire

[[a], [b, a, d], [c, b], [d, c, f ], [e, b, f ], [f, a]] .

Ces deux techniques de représentation de graphe se réalisent simplement en Scheme.
La représentation en mode I du graphe de la figure 21 est

((a b c d e f) . ((b . a) (b . d) (c . b) (d . c)

(d . f) (e . b) (e . f) (f . a)))

et celle en mode II est

((a) (b a d) (c b) (d c f) (e b f) (f a))

L’existence de plusieurs représentations adéquates suggère l’utilité d’un type abstrait
graphe, mais définir un tel type n’est pas commode, parce qu’un graphe n’apparâıt pas
comme la composition unique d’un ensemble de sous-graphes. De plus, la notion de graphe
admet quantité de variantes et de raffinements. Les arcs peuvent être colorés, c’est-à-dire
répartis en plusieurs catégories, ou encore étiquetés, c’est-à-dire porteurs d’une information
spécifique; on peut aussi parfois utiliser des “hyperarcs”, qui mettent en relation plus de
deux nœuds, etc. A défaut d’un type abstrait, on peut cependant concevoir que tout
programme d’application concernant les graphes soit écrit en termes d’un petit ensemble
de primitives, fixé une fois pour toutes; seules ces primitives dépendraient explicitement
du mode de représentation choisi. Le choix des primitives adéquates n’est pas évident.
Par exemple, on conçoit immédiatement que la fonction qui à tout nœud associe la liste

des successeurs est importante, mais en va-t-il de même pour la liste des prédécesseurs ?
Cela dépend naturellement des applications que l’on vise. Nous n’approfondirons pas cette
question ici, mais développerons quelques programmes simples à titre d’illustration, sur
base des deux modes de représentation introduits plus haut. Au paragraphe § 10.7.1,
nous utiliserons ces programmes pour résoudre un problème classique de la théorie des
automates finis.

10.2.2 Nœuds successeurs

Le premier exercice consiste à écrire trois primitives essentielles : une fonction mk_graph

qui crée un graphe à partir d’une liste de nœuds et une liste d’arcs, une fonction nodes

qui renvoie la liste des nœuds d’un graphe, et une fonction succs qui renvoie la liste des
successeurs d’un nœud d’un graphe. En utilisant le mode I, on a

(define mk_graph cons) (define nodes car) (define arcs cdr)

Un arc comportant une origine et une extrémité, on a aussi

(define mk_arc cons) (define orig car) (define extr cdr)

On définit alors

(define succs

(lambda (nd gr)

(let ((nodes (nodes gr)) (arcs (arcs gr)))

(if (member nd nodes)

(succs_arcs nd arcs)

(error "unknown node" nd)))))

(define succs_arcs

(lambda (nd arcs)

(cond ((null? arcs) ’())

((equal? (orig (car arcs)) nd)

(add_elem (extr (car arcs))

(succs_arcs nd (cdr arcs))))

(else (succs_arcs nd (cdr arcs))))))

La fonction auxiliaire add_elem a été introduite au paragraphe 5.3.3. On a par exemple

(define *g0*

’((a b c d e f) . ((b . a) (b . d) (c . b) (d . c)

(d . f) (e . b) (e . f) (f . a))))

(succs ’e *g0*) ==> (b f)

(succs ’k *g0*) ==> Error - unknown node k

Avec le mode II de représentation des graphes, on conserve les primitives mk_arc, orig
et extr, et on redéfinit les primitives de graphe comme suit :

(define mk_graph

(lambda (node* arc*)

(map (lambda (nd)

(let ((node_succs

(map_filter extr

(lambda (arc)

(equal? (orig arc) nd))

arc*)))

(cons nd node_succs)))

node*)))

(define nodes

(lambda (gr) (map car gr)))

(define arcs

(lambda (gr)

(union_map

(lambda (n+s)

(let ((n (car n+s)) (s+ (cdr n+s)))

(map (lambda (nd) (mk_arc n nd)) s+)))

gr)))

La fonction union_map a été introduite au paragraphe 6.1.5. On peut alors écrire :

(define succs

(lambda (nd gr)

(cond ((null? gr) (error "unknown node" nd))

((equal? (caar gr) nd) (cdar gr))

(else (succs nd (cdr gr))))))

On a par exemple

(define *g0*

’((a) (b a d) (c b) (d c f) (e b f) (f a)))

(arcs *g0*) ==> ((b . a) (b . d) (c . b) (d . c)

(d . f) (e . b) (e . f) (f . a))

(equal? (mk_graph (nodes *g0*) (arcs *g0*)) *g0*) ==> #t

(succs ’e *g0*) ==> (b f)

(succs ’k *g0*) ==> Error - unknown node k

10.2.3 Nœuds accessibles

Le deuxième exercice consiste à déterminer tous les nœuds accessibles à partir d’un nœud
donné. Un nœud y est accessible à partir d’un nœud x si y est x ou est accessible à partir
d’un successeur de x. Cette définition suggère immédiatement une “solution” näıve :

(define naive_offspring

(lambda (nd gr)

(add_elem nd (naive_offspring* (succs nd gr) gr))))

(define naive_offspring*

(lambda (nd* gr)

(if (null? nd*)

’()

(union (naive_offspring (car nd* gr))

(naive_offspring* (cdr nd* gr))))))

La fonction union a été introduite au paragraphe 5.3.3.

Remarque. Le type abstrait ensemble sera développé au paragraphe 10.6; on verra alors
que les fonctions add_elem et union utilisées ici sont des réalisations particulières des
opérateurs ensemblistes correspondants.

On voit immédiatement pourquoi la fonction naive_offspring est inopérante : la
terminaison n’est pas garantie. Pour y remédier, un moyen est de noter que dans un
graphe comportant n nœuds, tout nœud accessible peut être atteint en n − 1 étapes au
plus. On a donc

(define offspring

(lambda (nd gr)

(off nd gr (length (nodes gr)))))

(define off

(lambda (nd gr k)

(if (= k 0)

’()

(add_elem nd (off* (succs nd gr) gr (- k 1))))))

(define off*

(lambda (nd* gr k)

(if (null? nd*)

’()

(union (off (car nd*) gr k) (off* (cdr nd*) gr k)))))

Voici quelques exemples d’application :

(offspring ’a *g0*) ==> (a)

(offspring ’d *g0*) ==> (d c b f a)

(offspring ’f *g0*) ==> (f a)

On notera que ce programme est indépendant de la représentation choisie pour les graphes,
puisqu’il manipule ceux-ci via les primitives succs et nodes.

Le paramètre supplémentaire garantit la terminaison du programme mais non ses
performances. Il est rarement nécessaire d’itérer n − 1 fois la recherche de successeurs
pour atteindre tous les nœuds accessibles à partir d’un nœud donné d’un graphe qui
en comporte n. Un moyen souvent plus efficace consiste à mémoriser tous les nœuds
rencontrés et à ne pas traiter deux fois le même nœud. L’absence de répétition permettra
de remplacer add_elem et union par cons et append, respectivement. On a

(define offspring_bis

(lambda (nd gr)

(off*_bis (list nd) gr ’())))

(define off*_bis

(lambda (nd* gr acc)

(cond ((null? nd*) acc)

((member (car nd*) acc)

(off*_bis (cdr nd*) gr acc))

(else

(off*_bis (append (succs (car nd*) gr) (cdr nd*))

gr

(cons (car nd*) acc))))))

et ainsi

(offspring_bis ’d *g0*) ==> (f a b c d)

(offspring_bis ’e *g0*) ==> (f c d a b e)

(offspring_bis ’f *g0*) ==> (a f)

Une troisième version plus efficace s’obtient en éliminant l’usage de append :

(define offspring_ter

(lambda (nd gr)

(off_ter nd gr ’())))

(define off_ter

(lambda (nd gr acc)

(if (member nd acc)

acc

(off*_ter (succs nd gr) gr (cons nd acc)))))

(define off*_ter

(lambda (nd* gr acc)

(cond ((null? nd*) acc)

((member (car nd*) acc)

(off*_ter (cdr nd*) gr acc))

(else

(off_ter (car nd*)

gr

(off*_ter (cdr nd*)

gr

acc))))))

(offspring_ter ’d *g0*) ==> (b c a f d)

(offspring_ter ’e *g0*) ==> (c d b a f e)

(offspring_ter ’f *g0*) ==> (a f)

10.3 Abstraction et schémas de récursion

A ce stade, le lecteur est sans doute convaincu de l’intérêt de l’approche récursive; grâce
aux quelques schémas de récursion illustrés dans les chapitres précédents, la mise en œuvre
de cette approche est facile, quasi systématique. Il convient cependant de souligner, même
dans un traité introductif, que l’approche par schémas a ses limites. Le programmeur ne
peut se restreindre à une technique qui permet de résoudre “seulement” la majorité des
problèmes habituellement rencontrés. Plus spécifiquement, le programmeur doit pouvoir
adapter à ses besoins les outils dont il dispose, ce qui implique parfois de modifier
légèrement les outils en question. En fait, le schéma de récursion n’est qu’un texte formel
représentant, parfois de manière trop rigide, une idée de récursion particulière. Le point
important est l’idée elle-même, ce que nous illustrons dans les paragraphes suivants.

10.3.1 Sur quel(s) argument(s) faire porter la récursion ?

On peut résumer l’approche récursive en peu de mots : ramener le cas de l’entier naturel
n > 0 au cas n − 1 (ou au cas m, où m < n), ramener le cas de la liste u non vide au
cas du reste de u (ou au cas d’une autre liste plus courte que u), etc. Cependant, quand
la fonction à définir comporte plusieurs arguments, le choix de celui sur lequel portera la
récursion n’est pas indifférent. Nous avons déjà évoqué dans le premier chapitre la fonction
coefficient binomial, à deux arguments entiers naturels. L’approche récursive est, dans son
principe, toujours la même, mais plusieurs variantes existent, dont voici trois exemples :

cb(n, p) =def if (p=n ∨ p=0) then 1 else cb(n−1, p) + cb(n−1, p−1) ;
cb(n, p) =def if p=0 then 1 else [n ∗ cb(n−1, p−1)] / p ;
cb(n, p) =def if p=0 then 1 else [(n−p+1) ∗ cb(n, p−1)] / p .

Ces trois définitions se traduisent aisément en Scheme; les programmes correspondant
aux deux dernières définitions sont efficaces, au contraire du programme correspondant

à la première définition. La raison est évidente; dans la première définition, l’évaluation
de cb(n, p) donne lieu à deux appels récursifs direct (cb(n−1, p) et cb(n−1, p−1)) alors
que dans les deux dernières définitions, l’évaluation de cb(n, p) donne lieu à un seul appel
récursif direct (cb(n−1, p−1) dans le deuxième cas, cb(n, p−1) dans le troisième cas).

L’exemple du coefficient binomial et plusieurs autres du même genre suggèrent que,
quand c’est possible, il vaut mieux que l’application d’une fonction à ses arguments donne
lieu à un seul appel récursif direct. Il n’est cependant pas toujours possible d’obtenir
une définition récursive vérifiant cette condition. Pour le voir, mettons en évidence la
manière dont les définitions récursives sont obtenues. On sait que cb(n, p) est le nombre
de choix possibles de p objets au sein d’une collection de n objets (distincts). Supposons
0 < p < n et soit X l’un des objets de la collection. Il existe deux types de choix :
ceux qui contiennent X (et p − 1 autres objets choisis parmi les n − 1 restants) et ceux
qui ne contiennent pas X (mais p autres objets choisis parmi les n − 1 restants). Cette
simple constatation suffit à établir la première définition récursive de la fonction cb. Des
manipulations numériques supplémentaires, ou des raisonnements moins immédiats, sont
nécessaires pour obtenir les deux autres définitions, plus efficaces.

Bien souvent, on obtiendra aisément une définition “inefficace”, mais il sera difficile ou
impossible d’obtenir une définition plus efficace. Un premier cas est celui des partitions.
Soit p(n, k) le nombre de partages possibles d’une collection de n objets distincts en k lots
non vides; soit X l’un des objets. Si 1 < k < n, on distingue deux types de partages :
ceux dans lesquels X constitue à lui seul un lot (les n − 1 autres objets sont répartis en
k − 1 lots), et les autres, qui résultent de l’adjonction de X à l’un des lots résultant du
partage des n − 1 autres objets en k lots. A nouveau, on déduit immédiatement de cette
simple constatation la définition récursive suivante :

p(n, k) =def if (k=n ∨ k=1) then 1 else p(n−1, k−1) + k ∗ p(n−1, k) .

Cette définition ressemble nettement à la première définition de la fonction cb mais, dans
le cas présent, on ne voit pas comment obtenir une définition équivalente, impliquant un
seul appel récursif direct.

Un autre cas classique est celui des dérangements. Un dérangement est une
permutation sans point fixe, c’est-à-dire une permutation qui ne laisse aucun objet à sa
place. On sait que le nombre de permutations de n objets est n!; parmi celles-ci, certaines
comportent au moins un point fixe et d’autres n’en comportent pas; on note d(n) le
nombre de dérangements à n éléments. Soit n > 1 et un dérangement de {1, . . . , n} tel
que p(n) = i, i ∈ {1, . . . , n − 1}. Deux cas sont possibles : p(i) = n et p(i) 6= n. Dans
le premier cas, les n − 2 éléments restants forment un dérangement; dans le second, la
permutation p′ de domaine {1, . . . , n − 1} telle que p′(j) = i si p(j) = n et p′(j) = p(j)
sinon est un dérangement. En tenant compte de ce que i peut prendre n− 1 valeurs, on a
l’équation d(n) = (n−1)∗[d(n−2)+d(n−1)] d’où on tire aisément une définition récursive
(les cas de base sont d(0) = 1 et d(1) = 0). Il est possible d’obtenir un programme plus
efficace comme suit : soit q(n, i) le nombre de permutations de n éléments comportant au



10 ABSTRACTION : DONNÉES ET ALGORITHMES 202

moins i points fixes. On a

d(n) = q(n, 0) − q(n, 1) + q(n, 2) − . . . + (−1)nq(n, n) ,

ce qui permet un calcul simple de d(n) si on utilise les égalités

q(n, p) = cb(n, p) ∗ (n − p)! = n!/p! .

Ces égalité sont faciles à établir; le facteur cb(n, p) correspond au choix de p points fixes
dans {1, . . . , n} et le facteur (n−p)! correspond au nombre de permutations (avec ou sans
point fixe) des n − p éléments restants; de plus, cb(n, p) = n!/[p! ∗ (n − p)!].105

La conclusion de cette petite étude est que l’on ne peut ignorer en pratique la récursivité
structurelle mixte, que nous avions illustrée au paragraphe 5.6 au moyen de l’algorithme
d’Euclide. En fait, de nombreux problèmes importants se résolvent aisément par cette
technique; la relative inefficacité du programme obtenu n’est pas toujours gênante mais
surtout, elle n’est pas toujours évitable.

La suite de ce chapitre est consacrée à quelques problèmes dans lesquels un bon usage
de la récursivité conditionne la qualité de la solution.

10.4 Le problème du sac à dos

Ce problème est un représentant typique de la classe des problèmes d’optimisation. Nous
le présentons sous une forme simple mais de nombreux problèmes pratiques se ramènent
facilement au problème du sac à dos présenté ici. De plus, les techniques de programmation
qu’il permet d’illustrer simplement (approche “top-down”, récursivité mixte, données
abstraites) sont de première importance.

10.4.1 Enoncé

On a une collection d’objets; chaque objet a un poids qui est un nombre entier strictement
positif et une utilité qui est un nombre réel strictement positif. Un chargement est une
sous-collection d’objets; le poids d’un chargement est naturellement la somme des poids
des objets qu’il contient; son utilité est la somme des utilités des objets. Le problème
consiste à déterminer le chargement d’utilité maximale, dont le poids n’excède pas un
poids maximal donné.

Remarques. Le problème du “sac à dos” (knapsack) est NP-complet, ce qui signifie
concrètement que l’on ne dispose pas pour ce problème d’un algorithme efficace en toute
généralité.106

105La notion de dérangement permet de résoudre le célèbre problème des chapeaux : si n personnes
déposent leurs chapeaux au vestiaire et les récupèrent ensuite au hasard, quelle est la probabilité que
personne ne retrouve le sien ? La réponse est d(n)/n!; ce terme converge rapidement vers 1/e = 0.367879...
quand n augmente. La suite est oscillante, ce qui conduit à un résultat surprenant : la probabilité est
toujours supérieure à 1/e quand n est pair, et inférieure quand n est impair.

106On trouvera une présentation plus précise de ces notions de NP-complétude et d’efficacité dans [10].



10 ABSTRACTION : DONNÉES ET ALGORITHMES 203

10.4.2 Stratégie de résolution, données abstraites

Les cas de base se repèrent aisément. Quand la collection est vide, la solution est le
chargement vide, d’utilité nulle; c’est aussi le cas lorsque le poids maximal est nul. Le cas
inductif est celui où la collection C n’est pas vide et le poids maximal L n’est pas nul.

Nous traiterons le cas inductif comme pour les problèmes du paragraphe précédent.
Etant donné un objet arbitraire X de la collection, il y a deux types de chargements :
ceux qui négligent X et ceux qui contiennent X. Les chargements du premier type sont
relatifs à la collection C \ {X} et au poids maximal L. Les chargements du second type
s’obtiennent en ajoutant X à des chargements relatifs à la collection C \ {X} et au poids
maximal L−p(X); de tels chargements n’existent que si le poids p(X) de l’objet distingué
est inférieur au poids maximal L.

La tactique est donc de calculer, séparément, les solutions optimales, relatives à une
collection amputée d’un élément X, et aux poids maximaux L et L−p(X), respectivement.
On déduit de ces deux chargements (parfois un seul si L < p(X)) le chargement solution
du problème posé.

Le type structuré “collection” est récursif. On a

• La constante de base the-empty-coll;
• Le constructeur add-obj-coll (deux arguments);
• Les reconnaisseurs coll? et empty-coll?;
• Les accesseurs obj-coll et rem-coll.

Pour le type non récursif “objet”, on a

• Le constructeur mk-obj (deux arguments);
• Le reconnaisseur obj?;
• Les accesseurs poids et utilite.

On utilise aussi un type solution; un objet de ce type comporte une collection, ainsi
que le poids total et l’utilité totale de cette collection; on aura notamment le constructeur
mk-sol et les trois accesseurs char, ptot et utot.

10.4.3 Développement du programme

Nous utilisons comme souvent l’approche “top-down”. La première version met en
évidence le double cas de base : poids limite nul ou collection vide. On a immédiatement :

(define knap

(lambda (pm c)

(if (or (= 0 pm) (empty-coll? c))

(mk-sol the-empty-coll 0 0)

(...))))

La partie à préciser concerne le cas inductif. Son traitement requiert la distinction d’un
objet x de c et, au moins, le calcul récursif d’une solution du premier type. On obtient

(define knap

(lambda (pm c)

(if (or (= 0 pm) (null? c))

(mk-sol the-empty-coll 0 0)

(let ((x (obj-coll c)) (rc (rem-coll c)))

(let ((s1 (knap pm rc)))

(...))))))

Pour savoir si on devra aussi considérer une solution du second type, une comparaison de
poids est nécessaire; cela donne :

(define knap

(lambda (pm c)

(if (or (= 0 pm) (null? c))

(mk-sol the-empty-coll 0 0)

(let ((x (obj-coll c)) (rc (rem-coll c)))

(let ((s1 (knap pm rc))

(px (poids x)) (ux (utilite x)))

(if (>= pm px)

(...)

s1)))))

Le cas simple où x excède pm est réglé. L’autre cas requiert un second calcul récursif :

(if (>= pm px)

(let ((s2 (knap (- pm px) rc)))

(...))

s1)))))

Il reste à déterminer si la solution cherchée est s1, ou la solution obtenue en ajoutant x à
s2 :

(define knap

(lambda (pm c)

(if (or (= 0 pm) (null? c))

(mk-sol the-empty-coll 0 0)

(let ((x (obj-coll c)) (rc (rem-coll c)))

(let ((s1 (knap pm rc))

(px (poids x)) (ux (utilite x)))

(if (>= pm px)

(let ((s2 (knap (- pm px) rc)))

(if (> (+ (utot s2) (utilite x))

(utot s1))

(...)

s1))

s1)))))

La version finale est :

(define knap

(lambda (pm c)

(if (or (= 0 pm) (empty-coll? c))

(mk-sol the-empty-coll 0 0)

(let ((x (obj-coll c)) (rc (rem-coll c)))

(let ((s1 (knap pm rc))

(px (poids x)) (ux (utilite x)))

(if (>= pm px)

(let ((s2 (knap (- pm px) rc)))

(if (> (+ (utot s2) (utilite x))

(utot s1))

(mk-sol

(add-obj-coll x (char s2))

(+ px (ptot s2))

(+ ux (utot s2)))

s1))

s1))))))

10.4.4 Données concrètes et essais

On peut réaliser le type abstrait collection par le type concret list, avec les définitions
suivantes :

(define the-empty-coll ’())

(define add-obj-coll cons)

(define coll? list?)

(define empty-coll? null?)

(define obj-coll car)

(define rem-coll cdr)

Le type objet est concrétisé par le type pair :

(define mk-obj cons)

(define obj? pair?)

(define poids car)

(define utilite cdr)

Pour le type solution, on utilise aussi le type pair, restreint au cas où la deuxième
composante est aussi de type pair :

(define mk-sol

(lambda (c p u) (cons c (cons p u))))

(define char car)

(define ptot cadr)

(define utot cddr)

Nous considérons une collection c de 20 objets :

(define c ’((1 . 1) (1 . 2) (1 . 3) (2 . 2) (2 . 2)

(2 . 3) (2 . 3) (2 . 4) (2 . 5) (3 . 3)

(3 . 5) (3 . 8) (4 . 4) (4 . 6) (4 . 7)

(5 . 3) (5 . 6) (6 . 8) (7 . 8) (8 . 9)))

==> ...

Remarque. Pour décrire cette collection, nous avons “court-circuité” les données abstraites,
pour abréger l’écriture.

On cherche ensuite les chargements optimaux pour les poids respectifs de 0, 5, 10, 15,
20 et 25 :

(knap 0 c) ==>

(() 0 . 0)

(knap 5 c) ==>

(((2 . 5) (3 . 8)) 5 . 13)

(knap 10 c) ==>

(((1 . 3) (2 . 5) (3 . 8) (4 . 7)) 10 . 23)

(knap 15 c) ==>

(((1 . 3) (2 . 4) (2 . 5) (3 . 5) (3 . 8) (4 . 7))

15 . 32)

(knap 20 c) ==>

(((1 . 2) (1 . 3) (2 . 4) (2 . 5) (3 . 5) (3 . 8)

(4 . 6) (4 . 7)) 20 . 40)

(knap 25 c) ==>

(((1 . 1) (1 . 2) (1 . 3) (2 . 3) (2 . 3) (2 . 4)

(2 . 5) (3 . 5) (3 . 8) (4 . 6) (4 . 7)) 25 . 47)

(knap 30 c) ==>

(((1 . 2) (1 . 3) (2 . 3) (2 . 3) (2 . 4) (2 . 5)

(3 . 5) (3 . 8) (4 . 6) (4 . 7) (6 . 8)) 30 . 54)

Il est aussi intéressant de mesurer les temps de calcul :

v 5 10 15 20 25 30
Temps (ms) 165 2 150 13 000 47 500 120 000 260 000

Le comportement est typiquement exponentiel, ce qui rend le calcul pratiquement
impossible pour des grandes collections et des poids élevés. Ce fait est mis à profit en

cryptographie; la plupart des techniques de chiffrement sont susceptibles de déchiffrement
illicite, mais l’effort de calcul est tel que, en pratique, certaines méthodes de cryptage sont
très sûres.

10.5 Le problème de la monnaie

Nous présentons dans ce paragraphe et le suivant deux exemples supplémentaires de l’idée
de récursion qu’illustrait déjà le programme du sac à dos. Ces exemples montrent que le
programmeur a intérêt à abstraire les idées algorithmiques sous-jacentes aux programmes
qu’il écrit ou qu’il rencontre, de manière à pouvoir les réutiliser. Le second exemple
montre aussi que beaucoup de problèmes admettent plusieurs approches, conduisant à des
solutions d’efficacité variable; il nous permettra de plus d’introduire et d’utiliser les types
abstraits ensemble et multi-ensemble.

Un problème classique que l’on peut résoudre par la récursivité mixte est celui de la
monnaie. On dispose d’une liste d’espèces disponibles, en quantité illimitée; de combien
de manières différentes peut-on payer une somme S donnée ? Par exemple, au moyen de
pièces de un euro et de billets de cinq euros, on peut payer la somme de 13 euros de trois
manières différentes : 13 pièces de un euro, ou 8 pièces de un euro et 1 billet de cinq euros,
ou 3 pièces de un euro et 2 billets de cinq euros. On doit donc avoir

(money ’(1 5) 13) ==> 3

La stratégie de résolution par récursivité est simple. Si on doit payer une somme de S euros
et que l’on dispose, entre autres, de pièces de x euros, deux types de solutions existent;
celles qui n’utilisent pas cette pièce, et celles qui l’utilisent au moins une fois. On a donc:

money(S, ℓ) = money(S, ℓ\{x}) + money(S − x, ℓ) .

On voit que le cas de base correspond à la liste vide ou à une somme négative ou nulle;
le cas inductif correspond à une liste non vide et une somme strictement positive. Le
programme aura donc la structure

(define money

(lambda (l s)

(if (or (<= s 0) (null? l))

... ;; cas de base

...))) ;; cas inductif

Le cas inductif est immédiat; il suffit de choisir une valeur x dans la liste non vide [[l]];
le choix naturel est [[(car l)]]. Le cas de base est simple . . . mais il faut quand même
être attentif. Il n’y a aucune solution si la somme est négative. Si la somme est nulle, il
y a une solution, qui est la solution vide. Si la somme est strictement positive, la liste est
vide et il n’y a pas de solution. Le programme est donc

(define money

(lambda (l s)

(if (or (<= s 0) (null? l))

(if (= s 0) 1 0)

(+ (money (cdr l) s)

(money l (- s (car l)))))))

Une variante équivalente est

(define money

(lambda (l s)

(cond ((< s 0) 0)

((= s 0) 1)

((null? l) 0)

(else (+ (money (cdr l) s)

(money l (- s (car l))))))))

L’ensemble des espèces est représenté par une liste sans répétition. Il n’est pas requis que
cette liste soit triée, ni que les espèces soient multiples ou sous-multiples l’une de l’autre.

10.6 Ensembles : type abstrait et application

10.6.1 Structures de données ensemblistes

Nous avons déjà évoqué la notion d’ensemble (fini) au paragraphe 6.3.2, où un ensemble
était assimilé à une liste sans répétition. Toutefois, un ensemble diffère d’une liste
sans répétition sur un point important : dans un ensemble, l’ordre des éléments n’a pas
d’importance. Cela signifie notamment que l’ensemble {a, b} pourrait être représenté
par la liste (a b), mais aussi par la liste (b a). Un moyen de garantir l’unicité de la
représentation est d’utiliser des listes sans répétition et triées. Cette solution implique
le choix d’une relation d’ordre. Notons aussi que le choix de l’une ou l’autre des
représentations a une influence sur l’efficacité des programmes réalisant des opérations
ensemblistes. En fonction des opérations que l’on souhaite optimiser, on pourra préférer
telle ou telle représentation. Par exemple, si on choisit de représenter un ensemble par
toute liste contenant les éléments de l’ensemble et aucun autre, l’opération ensembliste de
réunion est réalisée par une simple concaténation. L’inconvénient est que chaque ensemble
admet une infinité de représentations : {a, b} se représente par (a b) ou (b a), mais aussi
par (b a a a b), par exemple; cela complique certaines autres opérations ensemblistes.
Enfin, dans certains cas, il est commode de représenter un ensemble par un arbre, ou par
une autre structure.

Le plus souvent, on préférera définir un type abstrait pour les ensembles et ainsi écrire
des programmes indépendants de la représentation, quitte à accepter des performances
un peu moins bonnes. Les primitives de ce type sont celles de la théorie élémentaire des
ensembles finis. On a les constructeurs the_empty_set (0-aire) et add_elem (adjonction
d’un élément dans un ensemble, binaire), les reconnaisseurs unaires set? et empty_set?

et l’accesseur unaire select qui, appliqué à un ensemble non vide, renvoie d’une part un
élément d’un ensemble non vide et d’autre part l’ensemble des autres éléments. Comme
une fonction ne renvoie qu’un seul résultat, ce résultat sera un objet auquel on pourra
appliquer l’accesseur elem pour obtenir l’élément, et l’accesseur rem_set pour obtenir
l’ensemble des éléments restants (concrètement, l’objet sera une paire pointée, d’où elem

et rem_set seront simplement car et cdr).

10.6.2 Trois réalisations concrètes du type ensemble

Si on adopte la liste sans répétition comme type concret, les primitives abstraites peuvent
se coder comme suit :

(define empty_set? null?)

(define the_empty_set list)

(define select (lambda (s) s))

(define elem car)

(define rem_set cdr)

(define add_elem

(lambda (x s) (if (member x s) s (cons x s))))

La fonction auxiliaire member est prédéfinie en Scheme (cf. § 4.3.2).

Si on adopte la liste triée sans répétition comme type concret, on doit disposer d’un
prédicat de comparaison comp qui réalise un ordre total sur le domaine de tous les éléments
susceptibles d’appartenir à un ensemble. Les primitives abstraites peuvent se coder comme
précédemment, sauf :

(define add_elem

(lambda (x s) (if (member x s) s (insert x s comp)))

Le définition de insert a été donnée au paragraphe 5.3.4.

Si on adopte la liste quelconque comme type concret, on a le codage suivant :

(define empty_set? null?)

(define the_empty_set list)

(define select

(lambda (s)

(letrec ((remove

(lambda (x l)

(if (null? l)

’()

(let ((rec (remove x (cdr l))))

(if (equal? x (car l))

rec

(cons (car l) rec)))))))

(cons (car s) (remove (car s) (cdr s))))))



10 ABSTRACTION : DONNÉES ET ALGORITHMES 210

(define elem car)

(define rem_set cdr)

(define add_elem cons)

10.6.3 Fonctions d’interface

La fonction list->set produit un ensemble au départ d’une liste d’éléments; cette liste
n’est pas supposée triée et peut contenir des répétitions :

(define list->set

(lambda (l)

(if (null? l)

(the_empty_set)

(add_elem (car l) (list->set (cdr l))))))

La fonction set->list produit la liste des éléments d’un ensemble donné.

(define set->list

(lambda (s)

(if (empty_set? s)

’()

(let ((split (select s)))

(cons (elem split) (set->list (rem_set split)))))))

La liste produite est toujours sans répétition; elle est triée si le type abstrait ensemble
est réalisé au moyen de listes triées. Si on souhaite obtenir inconditionnellement une liste
triée, on écrira

(define set->sorted_list

(lambda (s)

(if (empty_set? s)

’()

(let ((split (select s)))

(insert (elem split)

(set->list (rem_set split))

comp)))))

On a par exemple :

(set->list (list->set ’(c a c a b c c a a b))) ==> (c a b)

Ces fonctions illustrent bien la relative perte de performance liée à l’emploi des types
abstraits et, plus précisément, à l’ignorance du type concret sous-jacent. Par exemple,
si on savait que des listes quelconques sont utilisées, la fonction identique serait une
réalisation acceptable de list->set; de même, si on savait que des listes sans répétition
étaient utilisées pour représenter les ensembles, la fonction identique serait une réalisation
acceptable de set->list, et même de set->sorted_list si ces listes étaient triées.

Pour illustrer plus commodément les programmes développés dans la suite de ce
paragraphe, nous définissons cinq ensembles et une liste d’ensembles :

(define *s1* (list->set ’(a c d f h))) ==> ...

(define *s2* (list->set ’(a b e f g))) ==> ...

(define *s3* (list->set ’(a b d h j))) ==> ...

(define *s4* (list->set ’(a c e g k))) ==> ...

(define *s5* (list->set ’(a d g j l))) ==> ...

(define *ls* (list *s1* *s2* *s3* *s4* *s5*)) ==> ...

10.6.4 Version abstraite des opérations de base

Le prédicat binaire d’appartenance in? est le concept de base de la théorie; il se code
facilement :

(define in?

(lambda (x s)

(and (not (empty_set? s))

(let ((split (select s)))

(or (equal? x (elem split))

(in? x (rem_set split)))))))

Remarque. Selon le type des éléments, on peut remplacer equal? par un prédicat d’égalité
plus spécifique (cf. § 2.6).

A partir de ce prédicat, on définit facilement les trois opérations binaires classiques
d’intersection, de réunion et de différence :

(define inter

(lambda (u v)

(if (empty_set? u)

(the_empty_set)

(let ((split (select u)))

(let ((el (elem split)) (rem (rem_set split)))

(let ((rec (inter rem v)))

(if (in? el v) (add_elem el rec) rec)))))))

(define union

(lambda (u v)

(if (empty_set? u)

v

(let ((split (select u)))

(let ((el (elem split)) (rem (rem_set split)))

(let ((rec (union rem v)))

(if (in? el v) rec (add_elem el rec))))))))

(define diff

(lambda (u v)

(if (empty_set? u)

(the_empty_set)

(let ((split (select u)))

(let ((el (elem split)) (rem (rem_set split)))

(let ((rec (diff rem v)))

(if (in? el v) rec (add_elem el rec))))))))

On a par exemple :

(set->list (inter *s2* *s5*)) ==> (a g)

(set->list (union *s2* *s1*)) ==> (b e g a c d f h)

(set->list (diff *s3* *s4*)) ==> (b d h j)

10.6.5 Un opérateur plus général

On définit maintenant une fonction occur_list prenant comme arguments une liste ℓ
d’ensembles et un entier n strictement positif, et renvoyant l’ensemble des éléments
appartenant à exactement n ensembles de ℓ. On définira aussi la variante occur_list*, qui
renvoie l’ensemble des éléments appartenant à au moins n ensembles de ℓ. Ces fonctions
généralisent les opérations d’intersection et de réunion.

La stratégie de construction pour les programmes occur_list et occur_list* est la
même que pour les problèmes du sac à dos et de la monnaie. En dehors des cas de base, un
élément appartient à n ensembles d’une collection c d’ensembles s’il appartient au premier
ensemble de la collection et aussi à n − 1 des autres, ou s’il n’appartient pas au premier
ensemble de la collection mais appartient à n autres ensembles de cette collection. Le cas
de base se présente quand n vaut 1, d’une part, et quand n excède la taille de la collection,
d’autre part.

(define occur_list

(lambda (n ll)

(cond ((= n 1) (once ll))

((< (length ll) n) (the_empty_set))

(else (let ((within (occur_list (- n 1) (cdr ll)))

(without (occur_list n (cdr ll))))

(union (inter (car ll) within)

(diff without (car ll))))))))

(define occur_list*

(lambda (n ll)

(cond ((= n 1) (union_list ll))

((< (length ll) n) ’())

(else (let ((within (occur_list* (- n 1) (cdr ll)))

(without (occur_list* n (cdr ll))))

(union (inter (car ll) within)

(diff without (car ll))))))))

On observe que les deux fonctions diffèrent seulement par leur cas de base. Trois fonctions
auxiliaires doivent être définies :

(define union_list

(lambda (ll)

(if (null? ll)

(the_empty_set)

(union (car ll) (union_list (cdr ll))))))

(define once

(lambda (ll)

(if (null? ll)

(the_empty_set)

(union (diff_list (car ll) (cdr ll))

(diff (once (cdr ll)) (car ll))))))

(define diff_list

(lambda (u ll)

(if (null? ll) u (diff_list (diff u (car ll)) (cdr ll)))))

La fonction diff_list prend pour arguments un ensemble s et une liste d’ensembles ℓ et
renvoie l’ensemble des éléments de s qui n’appartiennent à aucun ensemble de la liste ℓ :

(set->list (diff_list *s1* (list *s2* *s3*))) ==> (c)

La fonction union_list renvoie la réunion des ensembles d’une liste d’ensembles donnée
en argument :

(set->list (union_list *ls*)) ==> (f b h c e k a d g j l)

La fonction once renvoie l’ensemble des éléments qui appartiennent à exactement un
ensemble d’une liste d’ensembles donnée en argument :

(set->list (once *ls*)) ==> (k l)

(set->list (once ’())) ==> ()

(set->list (once (list *s1*))) ==> (a c d f h)

(set->list (once (list *s1* *s1*))) ==> ()

Voici quelques exemples d’utilisation des fonctions occur_list et occur_list* :

(set->list (occur_list 3 *ls*)) ==> (d g)

(set->list (occur_list 4 *ls*)) ==> ()

(set->list (occur_list* 3 *ls*)) ==> (a d g)

10.6.6 Solution alternative

Il est souvent prudent de ne pas s’arrêter à la première idée de résolution venue. On peut
très bien imaginer d’autres moyens de programmer les fonctions précédentes. Une autre
stratégie élémentaire repose sur la notion de multi-ensemble. Formellement, un multi-
ensemble ME de support E = {x1, . . . , xp} est un ensemble de type {(x1, n1), . . . , (xp, np)}
où les ni sont des entiers strictement positifs. Intuitivement, un multi-ensemble est un
ensemble où l’on tient compte des répétitions. L’ensemble {(a, 3), (b, 2)}, par exemple, est
le multi-ensemble de support {a, b} auquel a appartient trois fois et b appartient deux fois.

La stratégie consiste à écrire, d’une part, une fonction associant à une liste d’ensembles
donnée le multi-ensemble formé par la multi-réunion des ensembles de cette liste; d’autre
part, on écrit aussi une fonction associant à un entier strictement positif n et un multi-
ensemble M l’ensemble des éléments appartenant n fois à M .

La première tâche consiste à développer le type multi-ensemble, sur base du type
abstrait ensemble. Un multi-ensemble étant un ensemble de couples, les primitives du
type ensemble sont conservées mais des primitives spécifiques sont ajoutées. Notons aussi
que certains ensembles de couples ne sont pas des multi-ensembles; un contre-exemple est
{(a, 3), (a, 4)}. Nous convenons aussi de représenter les couples par des paires pointées.
On définit les primitives suivantes :

(define empty_m_set? empty_set?)

(define the_empty_m_set the_empty_set)

(define m_select

(lambda (m)

(let ((split (select m)))

(let ((el* (elem split)) (rm (rem_set split)))

(let ((el (car el*)) (rep (cdr el*)))

(if (= rep 1)

(cons el rm)

(cons el (cons (cons el (- rep 1)) rm))))))))

(define m_elem car)

(define rem_m_set cdr)

Ces fonctions permettent d’extraire une occurrence d’un élément d’un multi-ensemble,
tandis les fonctions elem et rem_set permettent d’extraire un couple, donc toutes les
occurrences d’un élément. Dans le même esprit, on définit la fonction m_add_elem qui
insère une occurrence d’un élément dans un multi-ensemble.

(define m_add_elem

(lambda (x m)

(if (empty_m_set? m)

(add_elem (cons x 1) m)

(let ((split (select m)))

(let ((el* (car split)) (rm (cdr split)))

(let ((el (car el*)) (rep (cdr el*)))

(if (equal? el x)

(add_elem (cons el (+ rep 1)) rm)

(add_elem el* (m_add_elem x rm)))))))))

Remarque. L’emploi de la fonction ensembliste add est dangereux dans le contexte des
multi-ensembles, car il peut conduire à la création d’un ensemble de couples qui ne serait
plus un multi-ensemble.

Notons aussi la fonction d’appartenance à un multi-ensemble :

(define m_in

(lambda (x m)

(if (empty_m_set? m)

0

(let ((split (select m)))

(let ((el* (car split)) (rm (cdr split)))

(if (equal? (car el*) x)

(cdr el*)

(m_in x rm)))))))

Cette fonction renvoie le nombre de fois qu’un objet appartient à un multi-ensemble.

La deuxième tâche consiste à créer les fonctions m_s_union et m_s*_union, permettant
respectivement d’adjoindre un élément à un multi-ensemble, de réunir un ensemble à un
multi-ensemble, et de réunir une liste d’ensembles en un multi-ensemble. La réunion d’un
ensemble E à un multi-ensemble M résulte de l’adjonction à M de tous les éléments de
E :

(define m_s_union

(lambda (s m)

(if (empty_set? s)

m

(let ((split (select s)))

(m_add_elem (elem split)

(m_s_union (rem_set split) m))))))

La réunion d’une liste d’ensembles en un multi-ensemble résulte de la réunion au multi-
ensemble vide de chacun des ensembles de la liste :

(define m_s*_union

(lambda (ls)

(if (null? ls)

(the_empty_m_set)

(m_s_union (car ls) (m_s*_union (cdr ls))))))

*ls* ==>

((a c d f h) (a b e f g) (a b d h j) (a c e g k) (a d g j l))

(m_s*_union *ls*) ==>

((l . 1) (j . 2) (g . 3) (d . 3) (a . 5) (k . 1)

(e . 2) (c . 2) (h . 2) (b . 2) (f . 2))

La troisième tâche consiste en l’écriture d’une fonction renvoyant l’ensemble des
éléments répétés un nombre donné de fois, ou au moins un nombre donné de fois, dans un
multi-ensemble :

(define occur_m

(lambda (n ms)

(if (empty_set? ms)

(the_empty_set)

(let ((split (select ms)))

(let ((x (m_elem split))

(rec (occur_m n (rem_set split))))

(if (= (cdr x) n)

(add_elem (car x) rec)

rec))))))

(define occur_m*

(lambda (n ms)

(if (empty_m_set? ms)

(the_empty_m_set)

(let ((split (select ms)))

(let ((x (m_elem split))

(rec (occur_m* n (rem_set split))))

(if (>= (cdr x) n)

(add_elem (car x) rec)

rec))))))

On peut alors récrire les fonctions occur_list et occur_list* comme suit :

(define occur_list_bis

(lambda (n ll)

(occur_m n (m_s*_union ll))))

(define occur_list*_bis

(lambda (n ll)

(occur_m* n (m_s*_union ll))))

Ces fonctions sont équivalentes aux précédentes, sauf que les éléments de l’ensemble-
solution sont énumérés dans l’ordre inverse :

*ls* ==>

((a c d f h) (a b e f g) (a b d h j) (a c e g k) (a d g j l))

(set->list (occur_list_bis 4 *ls*)) ==> ()

(set->list (occur_list_bis 3 *ls*)) ==> (g d)

(set->list (occur_list*_bis 3 *ls*)) ==> (g d a)

(set->list (occur_list*_bis 2 *ls*)) ==> (j g d a e c h b f)

Les deux solutions occur_list et occur_list_bis sont très différentes dans leur
conception et leur réalisation mais sont également satisfaisantes dans la mesure où leur
efficacité est comparable. Ce dernier point n’est pas évident; nous y reviendrons au
chapitre 12.5.4.

10.7 Un exercice à propos des automates finis

L’exercice présenté dans ce paragraphe utilise diverses fonctions introduites antérieurement,
et en particulier les fonctions se rapportant aux ensembles et aux graphes. Il concerne
les automates finis; la présentation des automates est tirée de [10], dont sont extraites les
figures 22 et 23.

10.7.1 Les automates finis

Un automate fini est un objet structuré à cinq éléments :

1. Un espace d’état, c’est-à-dire un ensemble fini Q de symboles appelés états.
2. Un alphabet, c’est-à-dire un ensemble fini Σ de symboles appelés lettres d’entrée. Les

ensembles Q et Σ sont disjoints.
3. Un ensemble fini ∆ de transitions (voir ci-dessous).
4. Un état initial, élément de l’espace d’état.
5. Un sous-ensemble de l’espace d’état, dont les éléments sont appelés états terminaux,

ou états accepteurs.

Un mot est une liste de lettres d’entrée.107 Une transition est un objet structuré à trois
éléments qui sont un état appelé origine de la transition, un mot et un état appelé extrémité

de la transition. Un automate peut être représenté par un graphe dont les nœuds sont les
états et dont les arcs sont les transitions; chaque arc est étiqueté par le mot de la transition
correspondante. Une exécution est un chemin dans ce graphe, c’est-à-dire une suite d’états
s0, . . . , sn telle que s0 est l’état initial et sn est un état terminal; de plus, il existe une suite
de transitions τ1, . . . , τn telle que τi a pour origine si−1 et pour extrémité si; si wi est le
mot contenu dans τi, on dit que l’exécution reconnâıt le mot w, où w est la concaténation
des wi. L’automate de la figure 22 comporte cinq états et douze transitions; son alphabet
est {a, b}. L’état initial est q0, ce qui est indiqué graphiquement par le symbole >. Il y
a un seul état final q4, repéré par un double cercle. Les transitions sur le mot vide sont
étiquetées ε, celles sur le mot a par a et celles sur le mot b par b.

107Les parenthèses de liste sont habituellement omises.

Figure 22: Un automate fini non déterministe

On note Σ∗ l’ensemble des mots que l’on peut construire au moyen de l’alphabet Σ.
Un automate d’alphabet Σ définit (ou accepte) un langage, c’est-à-dire un sous-ensemble
L de Σ∗. Un mot appartient à ce langage s’il existe une exécution qui le reconnâıt. Le
langage défini par l’automate de la figure 22 contient par exemple le mot aaababb, reconnu
par l’exécution q0(ε)q3(a)q3(a)q1(a)q2(b)q1(a)q3(b)q4(ε)q3(b)q4. On dit que la lecture du
mot aaababb fait passer l’automate de l’état q0 à l’état q4.

Remarque. Supposons qu’un automate admette la transition (s,w, s′), où w est la
concaténation de w1 et w2. L’automate obtenu en ajoutant un état s′′ 6∈ Σ et en remplaçant
la transition (s,w, s′) par les deux transitions (s,w1, s

′′) et (s′′, w2, s
′) est équivalent à

l’automate de départ, c’est-à-dire qu’il définit le même langage. On peut donc supposer
que les mots contenus dans les transitions d’un automate comportent au plus une lettre
(le mot vide étant admis) sans restreindre le pouvoir expressif des automates, c’est-à-dire
la classe des langages qu’ils permettent de définir.

10.7.2 Représentation des automates finis

Nous ne ferons ici qu’esquisser la construction d’un type abstrait automate. On donne
d’abord la représentation de l’automate de la figure 22 :

(define *autom*

(list->automaton

’((q0 q1 q2 q3 q4)

(a b)

((q0 () q1) (q0 () q3) (q1 (a) q2) (q1 (a) q3)

(q2 (b) q1) (q3 (a) q1) (q3 (a) q3) (q3 (b) q4)

(q4 (a) q1) (q4 (a) q2) (q4 () q3) (q4 (b) q3))

q0

(q4))))

L’argument de la fonction d’interface [[list->automaton]] est la représentation concrète
de l’automate, le résultat renvoyé étant l’automate lui-même. La représentation concrète
est la liste des représentations des cinq constituants de l’automate. Rien n’empêche de
choisir pour [[list->automaton]] la fonction identité, ce qui suggère le constructeur et les
accesseurs suivants :

(define mk_aut list) (define mk_trans list)

(define states car) (define orig car)

(define alph cadr) (define word cadr)

(define trans caddr) (define extr caddr)

(define init cadddr)

(define finals (lambda (x) (car (cddddr x))))

10.7.3 Les automates finis déterministes

Un automate est déterministe si les mots de ses transitions sont des lettres et si, pour
chaque état q ∈ Q et chaque lettre x ∈ Σ, il existe un et un seul état q′ ∈ Q tel que (q, x, q′)
soit une transition de l’automate.108 L’automate de la figure 22 est non déterministe; par
exemple, au départ de l’état q3, la lecture de la lettre a peut mener en l’état q1 ou en
l’état q3; de plus, il comporte des transitions sur le mot vide.

Les automates déterministes sont très commodes, notamment parce qu’il est immédiat
de déterminer si un mot w est reconnu ou non par un tel automate. Il suffit de construire
l’unique exécution correspondant à w et de détecter si le dernier état de cette exécution
est terminal ou non. Un résultat fondamental de la théorie des automates (voir [10]) est
qu’à tout automate fini on peut associer un automate fini déterministe équivalent, c’est-
à-dire acceptant le même langage. Le but de l’exercice proposé ici est de programmer la
construction de cet automate équivalent.

Un automate déterministe est essentiellement une fonction de transition δ qui, à tout
état q et toute lettre x, associe l’état q′ = δ(q, x) tel que (q, x, q′) est une transition
de l’automate. L’automate non déterministe réalise seulement une relation de transition

entre états. Rendre déterministe un automate correspond donc à rendre fonctionnelle une
relation. Or, on sait que, à toute relation binaire R sur un ensemble Q correspond une
fonction fR de l’ensemble P(Q) sur lui-même, P(Q) étant l’ensemble des sous-ensembles
de Q. Par définition, si E est un sous-ensemble de Q, on a

fR(E) =def {q′ ∈ Q : il existe q ∈ E tel que (q, q′) ∈ R} .

Concrètement, si l’espace d’état de l’automate de départ est {q0, q1, q2, q3, q4}, l’espace
d’état de sa version déterministe comportera 25 = 32 états. La fonction de transition est
alors facile à calculer. Par exemple, pour l’automate de la figure 22,

δ({q2, q4}, a) = {q1, q2, q3}
108Un automate déterministe comportant n états et dont l’alphabet contient p lettres admet donc

exactement np transitions.

En effet, on ne peut pas lire a au départ de q2, mais au départ de q4, on a q4(a)q1,
q4(a)q2 et q4(ε)q3(a)q3. (On observera l’intervention d’une transition sur le mot
vide ε.) La conclusion est que l’une des transition de l’automate déterministe sera
({q2, q4}, a, {q1, q2, q3}) .

Etant donné un état q, il sera nécessaire de connâıtre son extension, notée Ext(q),
définie comme l’ensemble des états accessibles à partir de q en lisant le mot vide,
éventuellement plusieurs fois. On programme facilement la fonction Ext, sur base du
programme offspring introduit au paragraphe 10.2.3 :

(define ext

(lambda (aut q)

(let ((states (states aut)) (trans (trans aut)))

(let ((arcs

(map_filter

(lambda (tr) (mk_arc (orig tr) (extr tr)))

(lambda (tr) (null? (word tr)))

trans)))

(offspring_ter q (mk_graph states arcs))))))

Le seul point délicat est la construction d’un graphe dont les nœuds sont les états et dont
les arcs correspondent aux transitions sur le mot vide. On utilise la fonction map_filter

pour ne retenir que ces transitions et pour les transformer en arcs. La fonction ext se
généralise immédiatement en

(define ext*

(lambda (aut q*)

(union_map (lambda (q) (ext aut q)) q*)))

Cette fonction renvoie la réunion des extensions des états appartenant à son
argument [[q*]]. La fonction auxiliaire [[union_map]] prend comme arguments une
fonction f et un ensemble E et renvoie la réunion des f(u) : u ∈ E; on a

[[union map]](f,E) =
⋃

u∈E

f(u) .

Si q est un état et x est une lettre x, la ℓ-extension de q et x, notée ℓ-ext(q, x), est
l’ensemble des états accessibles à partir de l’état q en lisant la lettre x. Le programme
correspondant est :

(define l_ext

(lambda (aut q x)

(let ((trans (trans aut)))

(map_filter

extr

(lambda (tr)

(and (equal? (orig tr) q)

(equal? (word tr) (list x))))

trans))))

De même, ℓ-ext(q∗, x) est l’ensemble des états accessibles à partir d’un état q ∈ q∗ en
lisant la lettre x. Le programme correspondant est :

(define l_ext*

(lambda (aut q* x)

(union_map (lambda (q) (l_ext aut q x)) q*)))

Ici, la fonction auxiliaire offspring n’est pas utilisée, car la lettre x doit être lue une
seule fois.

On programme alors la fonction de transition δ du nouvel automate; elle prend pour
argument l’ancien automate, un ensemble d’états (qui est donc un seul état du nouvel
automate) et une lettre de l’alphabet (commun aux deux automates) :

(define del

(lambda (aut q* x)

(ext* aut (l_ext* aut q* x))))

La fonction renvoie l’ensemble des états que l’on peut atteindre à partir d’un état q ∈q*
par lecture du mot [[x]] (comportant une seule lettre). La lecture de ce mot comporte une
lecture de la lettre correspondante, et zéro, une ou plusieurs lectures du mot vide.

10.7.4 Le programme de conversion

On peut maintenant écrire le programme determinize, qui produit un automate
déterministe équivalent à l’automate passé en argument. Chacun des cinq composants
de l’automate résultat est construit à partir des cinq composants de l’automate argument.
Le premier composant [[new_states]] de l’automate résultat est l’ensemble des parties de
l’espace d’état de l’automate argument; on le calcule au moyen du programme subsets

dont une version concrète a été vue au paragraphe 6.3.2. Le deuxième composant est
l’alphabet alph, commun aux deux automates. Le troisième composant est la liste des
transitions. L’automate résultat comporte exactement une transition par couple (q, x)
où q est un état de l’automate argument et x une lettre de l’alphabet. Pour chaque
couple, la transition correspondante est calculée par le programme del; les opérateurs
s_map et union_map (versions ensemblistes des opérateurs de liste map et append_map)
sont utilisés pour former l’ensemble de toutes les transitions. Le quatrième composant est
l’état initial; celui de l’automate résultat est l’extension de celui de l’automate argument.
Enfin, le cinquième composant est l’ensemble des états terminaux. Un état de l’automate
résultat est terminal si et seulement si un de ses éléments est un état terminal de l’automate
argument, ce que l’on détermine au moyen de l’opérateur filter (version ensembliste),
et du prédicat inter?, qui détecte si deux ensembles (ici, un état de l’automate résultat

et l’ensemble des états terminaux de l’automate argument) ont au moins un élément en
commun. On obtient ainsi le code suivant :

(define determinize

(lambda (aut)

(let ((states (states aut))

(alph (alph aut))

(trans (trans aut))

(init (init aut))

(finals (finals aut)))

(let ((new_states (subsets states)))

(mk_aut new_states

alph

(union_map

(lambda (x)

(s_map (lambda (ns)

(mk_trans ns x (del aut ns x)))

new_states))

alph)

(ext aut init)

(filter (lambda (ns) (inter? finals ns))

new_states))))))

A titre d’illustration, on peut utiliser ce programme pour convertir en automate
déterministe l’exemple de la figure 22. Nous n’affichons qu’un fragment du résultat :

(define *det* (automaton->list (determinize *autom*))) ==> ...

(states *det*) ==> (() ... (q0 q1 q2 q3 q4))

(alph *det*) ==> (a b)

(trans *det*) ==> ((() a ()) ... ((q0 q1 q2 q3 q4) b (q1 q3 q4)))

(init *det*) ==> (q0 q1 q3)

(finals *det*) ==> ((q4) ... (q0 q1 q2 q3 q4))

On peut vérifier que l’automate comporte 25 = 32 états (dont 16 états terminaux : ceux
qui contiennent q4). L’alphabet comporte deux lettres, donc il y a 64 transitions.

10.7.5 Le programme de réduction

Le fait qu’il ait un grand nombre d’états ne signifie pas que l’automate résultant de la
conversion soit très complexe; en fait, la plupart de ces états ne sont pas accessibles à partir
de l’état initial et sont donc inutiles. On peut à nouveau utiliser le programme offspring
pour construire l’ensemble des états utiles; les autres sont simplement éliminés, de même
que les transitions qui les concernent. On obtient ainsi le programme suivant :

(define minimize

(lambda (aut)

(let ((states (states aut))

(alph (alph aut))

(trans (trans aut))

(init (init aut))

(finals (finals aut)))

(let ((graph

(mk_graph states

(s_map (lambda (tr)

(mk_arc (orig tr) (extr tr)))

trans))))

(let ((m_states (offspring_ter init graph)))

(let ((m_trans

(filter (lambda (tr) (in? (orig tr) m_states))

trans))

(m_finals

(filter (lambda (ns) (in? ns m_states))

finals)))

(mk_aut m_states

alph

m_trans

init

m_finals)))))))

Ce programme construit d’abord le graphe [[graph]] sous-jacent à l’automate à minimiser,
en transformant chaque transition en un arc. On construit ensuite l’ensemble [[m_states]]
des états utiles, accessibles à partir de l’état initial, puis on ne retient que les transitions
dont l’origine est accessible; de même, seuls les états terminaux accessibles sont retenus.
L’automate résultant peut être nettement moins encombrant; dans le cas de notre exemple,
il n’y a plus que quatre états (dont deux terminaux) et donc huit transitions. Cet automate
déterministe est représenté à la figure 23.

Voici la trace correspondante du programme de minimisation :

(define *small*

(automaton->list (minimize (determinize *autom*)))) ==> ...

La liste [[*small*]] est la représentation concrète de l’automate minimisé de la figure 23.

(states *small*) ==> ((q1 q3 q4) (q1 q2 q3) (q3 q4) (q0 q1 q3))

(alph *small*) ==> (a b)

(trans *small*) ==>

(((q3 q4) a (q1 q2 q3)) ((q1 q3 q4) a (q1 q2 q3))

((q1 q2 q3) a (q1 q2 q3)) ((q0 q1 q3) a (q1 q2 q3))

Figure 23: Un automate fini déterministe

((q3 q4) b (q3 q4)) ((q1 q3 q4) b (q3 q4))

((q1 q2 q3) b (q1 q3 q4)) ((q0 q1 q3) b (q3 q4)))

(init *small*) ==> (q0 q1 q3)

(finals *small*) ==> ((q3 q4) (q1 q3 q4)))

Remarque. Rien ne garantit que le programme de minimisation fournisse l’automate le
plus petit possible; il ne fait que supprimer la partie inaccessible de l’automate passé en
argument.

11 Abstraction procédurale

La notion de procédure est la clef de la décomposition d’un problème en sous-problèmes.
Nous avons déjà illustré abondamment l’importance cruciale de cette décomposition,
qui intervient même pour des problèmes relativement élémentaires. Nous avons aussi
mentionné que l’emploi de l’approche “top-down” n’est pas absolu; en particulier,
quand le programmeur est amené à définir une fonction auxiliaire, il trouvera souvent
utile d’envisager la définition d’une fonction plus générale que nécessaire; des fonctions
auxiliaires bien pensées sont souvent réutilisables dans des contextes très différents de
celui de leur introduction. Dans ce chapitre, nous donnons quelques exemples plus
significatifs de l’abstraction procédurale; le fait qu’en Scheme une procédure puisse
accepter des procédures comme données et produire des procédures comme résultats rend
ce langage spécialement adapté à l’abstraction procédurale. On notera aussi que cette
notion perfectionne le concept informel de réutilisation d’idée algorithmique, illustré au
chapitre précédent. En effet, une procédure réutilisable comporte une idée algorithmique
sous-jacente, mais aussi un travail de codage, qu’il est intéressant de pouvoir réutiliser
aussi.

Nous présentons l’abstraction procédurale de manière . . . concrète, en résolvant deux
problèmes classiques, le problème des huit reines et le problème du voyageur de commerce.
Nous verrons qu’en dépit des différences apparentes, ces deux problèmes sont du type
“recherche et optimisation”; de plus, les programmes correspondants sont des instances
d’un schéma très général, qui permet de résoudre le problème de la recherche abstraite.

L’abstraction procédurale et l’abstraction sur les données sont deux concepts bien
distincts et indépendants; nous travaillerons ici en données concrètes, l’avantage de données
abstraites n’existant pas vraiment pour les problèmes considérés.

11.1 Le problème des huit reines

11.1.1 Introduction

Un échiquier est un carré de huit cases de côté. Deux reines sur un échiquier sont en prise
si elles se trouvent sur une même ligne, sur une même colonne ou sur une même diagonale.
Une case de l’échiquier est repérée par un couple (a, b) où a est le numéro de la ligne (1
pour la ligne inférieure, 8 pour la ligne supérieure) et b est le numéro de la colonne (1 pour
la colonne de gauche, 8 pour celle de droite). Si deux reines occupent respectivement les
positions (a, b) et (a+ i, b+ j), ces reines sont en prise si i = 0, j = 0 ou i = ±j. Il est clair
que l’on ne peut placer plus de huit reines sur un échiquier sans que deux d’entre elles
soient en prise. Le problème posé consiste à construire toutes les possibilités de placer
exactement huit reines sur un échiquier sans que deux d’entre elles soient en prise.

Une configuration acceptable (s’il en existe) comportera une reine par colonne, chacune
se trouvant sur une ligne distincte. On pourra la représenter concrètement par une liste
de type (a1, . . . , a8), permutation de la liste (1, . . . , 8). Les positions correspondantes des
reines forment l’ensemble {(ai, i) : i = 1, . . . , 8}. Toute solution est une permutation; une

permutation vérifie naturellement les conditions relatives aux lignes et aux colonnes, et
sera une solution si elle vérifie en outre la condition relative aux diagonales.

11.1.2 Idée algorithmique

Une première idée consiste à généraliser le problème en le dotant d’un paramètre n
représentant la dimension de l’échiquier. Même si l’échiquier classique est de dimension 8,
le problème des reines existe pour tout échiquier de taille n×n. On pense alors à l’approche
récursive habituelle, et à exprimer les solutions du problème des n reines en fonction de
celles du problème des n − 1 reines. Il apparâıt cependant que cette approche n’est pas
facile. On observe d’ailleurs que le problème admet une solution pour n = 1, aucune
solution pour n = 2 et n = 3, et deux solutions pour n = 4.

Une deuxième idée simple consiste à observer qu’il existe 8! permutations des nombres
de 1 à 8, soit 40 320 configurations envisageables. On peut concevoir de construire et
tester toutes ces configurations. Cette technique näıve n’est pas mauvaise en soi; elle
serait même excellente si une proportion appréciable de ces configurations étaient des
solutions au problème posé, mais ce n’est visiblement pas le cas.109 Il suffit d’imaginer
une mise en œuvre concrète, à la main, pour découvrir une optimisation élémentaire.
L’expérimentateur place les reines une à une, dans des colonnes contiguës. S’il place,
par exemple, une reine en colonne 1, ligne 4, puis une seconde en colonne 2, ligne 5, la
condition diagonale est déjà violée, et l’adjonction de nouvelles reines dans les six autres
colonnes n’arrangera rien. Les 6! = 720 candidats-solutions correspondants peuvent être
éliminés d’un coup.

L’un des buts de l’exercice est de montrer comment une idée opératoire (“on fait
ceci, puis on fait cela”) peut s’intégrer méthodiquement dans le paradigme fonctionnel.
D’autre part, la stratégie de construction et de test utilisée ici et ses diverses variantes
sont intéressantes pour elles-mêmes, vu leurs applications nombreuses et importantes,
notamment en intelligence artificielle.

11.1.3 Développement du programme

L’approche näıve de construction et de test consiste ici à générer d’emblée les 40 320
permutations de la liste (1 2 3 4 5 6 7 8), puis à les tester en ne retenant que
les permutations respectant la condition des diagonales. L’approche raisonnable de
construction et de test consiste ici à générer ces permutations progressivement, en ajoutant
un élément à la fois, et en testant l’acceptabilité après chaque adjonction. Nous convenons
qu’une liste de nombres de 1 à 8 ne comportant aucune répétition est interprétée comme
une configuration partielle. Par exemple, la liste (7 4 8) correspond à une situation où
trois reines ont déjà été placées dans les trois colonnes les plus à droite de l’échiquier,110

respectivement dans les cases (7, 6), (4, 7) et (8, 8). Cette configuration s’étend a priori

109Nous verrons que 92 configurations seulement sont des solutions.
110Comme un élément s’ajoute commodément en tête de liste, on considèrera que l’échiquier est rempli

de la droite (colonne 8) vers la gauche (colonne 1).

de cinq manières, en les configurations partielles (1 7 4 8), (2 7 4 8), (3 7 4 8),
(5 7 4 8) et (6 7 4 8). De ces cinq solutions partielles, seules (1 7 4 8) et (3 7 4 8)

respectent la condition des diagonales et pourront être prolongées; les trois autres sont
éliminées à ce stade.

Pour mettre en œuvre cette technique, nous devrons disposer d’une fonction associant
à une configuration partielle la liste de ses prolongements. S’agit-il des prolongements
immédiats (adjonction d’une seule reine), ou des prolongements complets (adjonction de
toutes les reines restantes) ou encore de prolongements de longueur spécifiée au moyen
d’un paramètre additionnel ? Il ne semble pas y avoir de raison de préférer l’une de ces
trois options; nous choisissons donc arbitrairement la première;111 on devra donc avoir,
par exemple,

(extend_par_sol ’(7 4 8)) ==> ((1 7 4 8) (3 7 4 8))

Nous dérogeons ici à l’approche descendante (“top-down”) habituellement préférée, sans
pour autant adopter l’approche montante (“bottom-up”). La fonction extend_par_sol

a un rôle intermédiaire : en amont, une fonction principale (à définir) devra l’utiliser;
en aval, la réalisation de extend_par_sol donnera sans doute lieu à la spécification et
à la réalisation de fonctions auxiliaires. Commencer ainsi “in medias res” est approprié
ici parce que la fonction extend_par_sol semble nécessaire, quels que soit les autres
choix de programmation faits par ailleurs. Toujours arbitrairement, nous poursuivons le
développement par la partie amont, selon l’approche montante, c’est-à-dire par la fonction
qui utilisera extend_par_sol comme fonction auxiliaire.

Pour résoudre le problème posé au moyen de la fonction de base extend_par_sol,
nous devons généraliser celle-ci de manière à obtenir non seulement des prolongements
immédiats mais aussi des prolongements plus complets. Ces derniers sont en fait des
prolongements de prolongements immédiats. Nous définissons donc extend_par_sol*,
dont le premier argument est une liste de configurations partielles (de même longueur)
et dont le second argument est un entier positif; s’il vaut n, la fonction renvoie la liste
des prolongements d’ordre n, c’est-à-dire ceux obtenus par adjonction de n reines aux
configurations appartenant au premier argument. On définira naturellement la nouvelle
fonction par induction, et nous devons décider si cette induction portera sur la liste
premier argument ou sur l’entier naturel second argument, ou encore sur les deux à la
fois. La deuxième technique est ici la plus commode : les prolongements d’ordre n, si n est
strictement positif, sont simplement les prolongements d’ordre n − 1 des prolongements
immédiats, tandis que les seuls prolongements d’ordre 0 de configurations partielles sont
ces configurations elles-mêmes. Ces considérations imposent clairement le schéma suivant :

(define extend_par_sol*

(lambda (psols m)

111Le programmeur est fréquemment confronté à des choix de ce genre; bien souvent le choix lui-même
est peu important, mais il est par contre essentiel qu’il soit clairement explicité. Si la spécification d’une
fonction auxiliaire manque de précision, elle risque d’être réalisée d’une manière . . . et utilisée d’une ou
plusieurs manières différentes !

(if (zero? m)

psols

(extend_par_sol* (...) (- m 1)))))

La partie manquante est une forme dont la valeur doit être la liste des prolongements
immédiats des configurations de la liste [[psols]]. On obtient cette liste en concaténant
les listes des prolongements immédiats de chaque élément de [[psols]. Ces listes sont
obtenues par application de la fonction de base extend_par_sol et leur concaténation est
produite par la fonction append_map, introduite au paragraphe 6.1.5; on a donc le code
complet :

(define extend_par_sol*

(lambda (psols m)

(if (zero? m)

psols

(extend_par_sol* (append_map extend_par_sol psols)

(- m 1)))))

La réponse au problème posé sera [[(extend_par_sol* ’(()) 8)]].

On traite maintenant la partie aval du problème. La procédure extend_par_sol

s’obtient par composition d’une fonction qui construit les huit extensions possibles et
d’une fonction filtrante qui rejette les extensions inacceptables, telles que la reine ajoutée
et l’une des autres sont en prise. On obtient ainsi le code suivant :

(define extend_par_sol

(lambda (psol)

(filter legal?

(map (lambda (n) (cons n psol)) (enum 1 8)))))

Les fonctions auxiliaires enum et filter ont été introduites aux paragraphes 5.6 et 6.1.6,
respectivement. Avant de passer à la construction du prédicat legal?, nous notons que
la technique de composition d’une fonction construisant une liste et d’un filtre éliminant
certains éléments de la liste est commode et élégante, et donc d’emploi fréquent. Dans les
cas où l’on escompte un fort taux d’élimination, il peut être intéressant de combiner la
construction et le filtrage. Dans le cas présent, cela donne la variante suivante :

(define extend_par_sol

(lambda (psol)

(letrec

((e_p_s

(lambda (m)

(if (> m 8)

’()

(let ((mpsol (cons m psol)) (rec (e_p_s (+ m 1))))

(if (legal? mpsol) (cons mpsol rec) rec))))))

(e_p_s 1))))

Les fonctions enum et filter ne sont plus utilisées. On observera la définition d’une
fonction locale récursive.

Il reste à définir le prédicat legal?. Son argument est une liste non vide dont la tête x
est un nombre de 1 à 8 et dont le reste ℓ est une configuration partielle acceptable. Il
vérifie que la nouvelle reine x ne menace aucune des reines de ℓ. Cette vérification se fait
récursivement sur ℓ, ce qui implique l’introduction du prédicat local récursif leg?, dont la
spécification est laissée au lecteur. On obtient le code suivant :

(define legal?

(lambda (l)

(letrec ((leg?

(lambda (x u d)

(or (null? u) (let ((y (car u)) (v (cdr u)))

(and (not (= x (+ y d)))

(not (= x (- y d)))

(leg? x v (+ d 1))))))))

(and (not (member (car l) (cdr l)))

(leg? (car l) (cdr l) 1)))))

Le programme est maintenant terminé. On a :

(extend_par_sol* ’(()) 8) ==>

((4 2 7 3 6 8 5 1) ... (5 7 2 6 3 1 4 8))

Les permutations-solutions sont données dans l’ordre lexicographique des listes inversées;
par exemple, la solution (7 2 6 3 1 4 8 5) précède la solution (3 6 2 7 1 4 8 5)

parce que le nombre 58413627 associé à la première solution est inférieur au nombre
58417263 associé à la seconde solution. On vérifie aussi qu’il y a 92 solutions :

(length (extend_par_sol* ’(()) 8)) ==> 92

11.2 Généralisation

Après avoir considéré et résolu le problème des huit reines, il est quasi inévitable d’observer
que le problème garde tout son sens si la dimension de l’échiquier est un nombre naturel
quelconque. Dans ce paragraphe, nous adaptons le programme précédent de telle sorte
qu’il résolve le problème des n reines, quel que soit n.

Signalons d’emblée que ce travail de généralisation est quelque peu artificiel; on aurait
en fait dû d’emblée considérer le problème général. Nous ne l’avons pas fait, parce que,
dans certains cas moins évidents, il est légitime de penser à généraliser un programme
déjà écrit. Diverses techniques existent pour ce faire, qu’il est plus facile d’introduire sur
un exemple simple (voire simpliste).

11.2.1 Première technique

Dans la mesure où nous n’avons utilisé aucune propriété particulière du nombre 8
pour concevoir le programme, on imagine aisément qu’il suffise de remplacer toutes les
occurrences de 8 par 9, par exemple, pour obtenir un programme résolvant le problème des
neuf reines. On peut donc ainsi obtenir des programmes pour n’importe quelle dimension
n de l’échiquier. Cette technique est simple et rapide, mais encombrante et peu élégante
puisqu’il faut générer une version du programme pour chaque dimension.112 De plus, la
technique est dangereuse; on peut imaginer que dans le texte du programme apparaissent
non seulement des occurrences de 8, mais aussi d’autres nombres, comme 64, le nombre
de cases de l’échiquier. Il faut donc penser à remplacer aussi 64 par 81. Un risque plus
sournois existe. Dans un programme dimensionné pour le nombre 8, certaines occurrences
de 7 pourraient correspondre à 8 − 1 et d’autres, par exemple, à la somme 1 + 2 + 4 des
diviseurs propres de 8. Dans la version dimensionnée pour le nombre 9, les premières
occurrences devraient être remplacées par 8 (c’est-à-dire 9−1) et les secondes par 4 (c’est-
à-dire 1 + 3). En conclusion, cette première technique n’est mentionnée ici qu’à titre de
mise en garde; elle est à proscrire.

11.2.2 Deuxième technique

Il est dangereux de “corriger” un programme pour l’adapter à un cas malencontreusement
ignoré lors du développement, mais on peut recommencer point par point ce développement
en prenant en compte le paramètre n. A chaque spécification de procédure, on envisage
d’introduire n comme argument supplémentaire. Les modifications éventuelles ont
naturellement des conséquences sur la suite du développement. Ce travail de “re-
développement” n’est pas difficile, du moins si la personne qui l’accomplit a parfaitement
compris le premier développement, mais il est fastidieux. Le risque sournois évoqué au
paragraphe précédent existe toujours, mais l’obligation de remplacer chaque nombre par
une expression impliquant (peut-être) le nouveau paramètre favorise la mâıtrise de ce
risque. Le lecteur est invité a généraliser par cette technique le programme donné plus
haut.

11.2.3 Troisième technique

L’avantage de la technique de “re-développement” est que son résultat est, en principe,
le programme qui aurait été produit au départ, si le paramètre supplémentaire n’avait
été ignoré. L’inconvénient est qu’une fraction importante du code a dû être altérée
en profondeur, notamment parce que certaines procédures reçoivent un argument
supplémentaire. Un moyen d’éviter cela consiste à faire du paramètre réintroduit une
variable globale. Ce moyen est aussi expéditif que la première technique, mais quand
même moins dangereux; elle peut être acceptée dans un contexte de “prototypage”, c’est-
à-dire de construction de versions préliminaires et à brève durée de vie d’un programme.

112Notons cependant que cette génération pourrait être automatisée.

Concrètement, dans le programme que nous avons développé, la constante numérique 8 est
à remplacer par la constante symbolique *dim*. Ce remplacement doit intervenir dans le
code de extend_par_sol : “(enum 1 8)” devient “(enum 1 *dim*)” (première variante)
ou “(> m 8)” devient “(> m *dim*)” (deuxième variante). De plus, il est commode
d’introduire une fonction principale d’appel :

(define queens

(lambda () (extend_par_sol* ’(()) *dim*)))

Remarque. Cette définition peut être incluse dans le fichier du programme; ce n’est que
lors de l’appel de cette fonction sans argument que les solutions seront calculées. Par
contre, inclure en fin de fichier la définition

(define queens (extend_par_sol* ’(()) *dim*))

provoquerait le calcul dès le chargement du programme, et inclure cette définition en début
de fichier provoquerait une erreur. D’une manière générale, un fichier Scheme comporte
très peu de définitions non fonctionnelles.

On peut maintenant résoudre le problème général, comme le montre la session suivante :

(define *dim* 8) ==> ...

(queens) ==> ((4 2 7 3 6 8 5 1) ... (5 7 2 6 3 1 4 8))

(length (queens)) ==> 92

(define *dim* 5) ==> ...

(queens) ==> ((4 2 5 3 1) ... (2 4 1 3 5))

(length (queens)) ==> 10

Cette solution a l’avantage de n’introduire que des altérations superficielles et mineures; le
nombre d’arguments de chaque procédure est inchangé. L’inconvénient est le danger lié à
l’emploi des variables globales, qui réduisent la lisibilité et la modularité des programmes.

11.2.4 Quatrième technique

Il est possible de conserver l’avantage de la technique précédente sans introduire de variable
globale. L’idée est de remplacer la variable extend_par_sol par une forme du type
(mk_extend_par_sol dim) qui aurait même valeur. La définition précédente, du type

(define extend_par_sol (lambda (psol) (... *dim* ...)))

devient

(define (mk_extend_par_sol dim) (lambda (psol) (... dim ...)))

que l’on récrit en

(define mk_extend_par_sol

(lambda (dim) (lambda (psol) (... dim ...))))

Cette technique d’élimination de variable globale est très commode, parce qu’elle
est entièrement mécanique.113 La fonction [[mk_extend_par_sol]] est un générateur
de procédure, en ce sens que [[(mk_extend_par_sol dim)]] est une procédure
équivalente à [[extend_par_sol]], pour une taille [[dim]] de l’échiquier. Les
fonctions utilisant la procédure extend_par_sol subissent une modification analogue
et les appels à [[extend_par_sol]] sont simplement remplacés par des appels à
[[(mk_extend_par_sol dim)]].

Voici la version finale du programme de construction de la liste des solutions. elle
réutilise le prédicat legal? défini plus haut.

(define queens

(lambda (dim) ((mk_extend_par_sol* dim) ’(()) dim)))

(define mk_extend_par_sol*

(lambda (dim)

(lambda (psols m)

(if (zero? m)

psols

((mk_extend_par_sol* dim)

(append_map (mk_extend_par_sol dim) psols)

(- m 1))))))

(define mk_extend_par_sol

(lambda (dim)

(lambda (psol)

(letrec ((e_p_s

(lambda (m)

(if (> m dim)

’()

(let ((mpsol (cons m psol))

(rec (e_p_s (+ m 1))))

(if (legal? mpsol)

(cons mpsol rec)

rec))))))

(e_p_s 1)))))

On évalue alors la forme (queens n) pour obtenir la liste des solutions du problème des
[[n]] reines.

113Notons déjà que la seconde réécriture n’est pas indispensable. Nous verrons plus loin que l’écriture
(define (f x y) ...) est une variante syntaxique de l’écriture habituelle (define f (lambda (x y)

...)).

11.3 Arbre de recherche

11.3.1 Introduction

Conceptuellement, les configurations partielles et les solutions du problème des n reines
forment un arbre complètement étiqueté, que nous appelons arbre de recherche. L’étiquette
de la racine est la configuration vide; les étiquettes des nœuds internes sont les
configurations partielles et celles des feuilles sont les solutions. Les fils d’un nœud
interne N sont étiquetés par les prolongements immédiats de la configuration étiquetant N .
Les feuilles de l’arbre de recherche correspondent aux configurations non prolongeables;
seules les feuilles de niveau n donnent lieu à des solutions; les feuilles de niveau moindre
correspondent à des échecs de la recherche. L’arbre de recherche associé au problème
des quatre reines est donné à la figure 24; pour améliorer la clarté, chaque nœud est
étiqueté, non par la configuration correspondante, mais par la première position de cette
configuration (et donc, celle qui a été générée en dernier lieu). On observe que cet arbre n’a
que deux feuilles de niveau 4, auxquelles correspondent les chemins-solutions (2 4 1 3)

et (3 1 4 2).

Figure 24: Arbre de recherche, problème des quatre reines

Nous n’avons pas introduit de type abstrait pour le domaine des arbres à degré variable
et complètement étiquetés.114 Concrètement, nous représentons un tel arbre par une paire
dont le premier composant est l’étiquette de la racine et le second composant est la liste
des fils. La représentation de l’arbre de la figure 24 sera donc

(init (1 (3) (4 (2))) (2 (4 (1 (3)))) (3 (1 (4 (2)))) (4 (1 (3)) (2)))

L’indentation suivante met en évidence les sous-arbres de premier niveau :

(init

(1 (3) (4 (2)))

(2 (4 (1 (3))))

(3 (1 (4 (2))))

(4 (1 (3)) (2)))

114C’est un bon exercice pour le lecteur.

En fait, l’indentation à tous les niveau permettrait de mieux mettre en évidence la structure
arborescente, au prix d’un encombrement certain lors de l’affichage. Voici un fragment
de la version complètement indentée; seuls les deux premiers fils sont complètement
représentés :

(init

(1

(3)

(4

(2)))

(2

(4

(1

(3))))

(3 ...)

(4 ...))

11.3.2 Programmation

Dans la mesure où la structure d’arbre de recherche met bien en évidence la logique du
problème des reines, il est intéressant de pouvoir construire cet arbre par programme. Nous
définissons une fonction q_tree telle que [[(q_tree dim psol p)]] est un sous-arbre de
l’arbre de recherche pour le problème des [[dim]] reines; sa racine correspond à la solution
[[psol]] et la profondeur est limitée à [[p]]. On obtient aisément le code suivant :

(define q_tree

(lambda (dim psol p)

(let ((ps1 (if (pair? psol) (car psol) ’init)))

(if (zero? p)

(list ps1)

(cons ps1 (q_tree* dim psol p))))))

(define q_tree*

(lambda (dim psol p) ;; p > 0

(q_tree*aux dim psol p 1)))

(define q_tree*aux

(lambda (dim psol p q)

(if (> q dim)

’()

(let ((ps (cons q psol))

(rec (q_tree*aux dim psol p (+ q 1))))

(if (legal? ps)

(cons (q_tree dim ps (- p 1)) rec)

rec)))))

A titre d’exemple, considérons le cas [[dim]] = 6. Une solution partielle est (3 1), qui
correspond à une reine en 6ième colonne, 1ère ligne, et une reine en 5ième colonne, 2ième
ligne. L’évaluation

(q_tree 6 ’(3 1) 2) ==> (3 (5 (2)) (6 (2)))

nous montre que, en quatrième colonne, seules les lignes 5 et 6 conviennent et, en troisième
colonne, la ligne 2. On a donc les solutions partielles (2 5 3 1) et (2 6 3 1) L’évaluation

(q_tree 6 ’(3 1) 4) ==> (3 (5 (2 (4))) (6 (2)))

montre qu’aucune de ces deux configurations ne se prolonge en une solution complète. La
première se prolonge seulement en (4 2 5 3 1) et la seconde ne se prolonge pas du tout.

Pour construire l’arbre entier, on définit

(define queens_tree

(lambda (dim) (q_tree dim ’() dim)))

et on a par exemple

(queens_tree 4) ==> (init (1 (3) (4 (2))) ... (4 (1 (3)) (2)))

ce qui est la représentation de l’arbre de la figure 24; on a aussi

(queens_tree 3) ==> (init (1 (3)) (2) (3 (1)))

ce qui montre l’absence de solution pour le problème des trois reines.

11.3.3 Indentation

On peut aussi afficher l’arbre de manière plus lisible, comme on l’avait fait au
paragraphe 8.7. On a par exemple :

(define indent

(lambda (qtree d)

(let ((node (car qtree)) (qtree* (cdr qtree)))

(newline)

(space d)

(write node)

(indent* qtree* (+ d 1)))))

(define indent*

(lambda (qtree* d)

(for-each (lambda (qt) (indent qt d)) qtree*)))

Remarque. La fonction for-each est une primitive de Scheme. Dans le cas particulier
où son premier argument est une fonction unaire, elle pourrait être définie comme suit :

(define for-each

(lambda (f l)

(if (null? l)

’anyvalue

(begin (f (car l)) (for-each f (cdr l))))))

La fonction auxiliaire space a été définie au paragraphe 10.1.3. Voici un fragment de
l’arbre relatif au problème des cinq reines :

(queens_tree 5) ==>

(init (1 (3 (5 (2 (4)))) (4 (2 (5 (3)))) (5 (2))) (2 ...) ... (5 ...))

(indent (queens_tree 5) 0) ==>

1

3

5

2

4

4

2

5

3

5

2

2

...

...

5

...

;No value

11.4 La technique du retour arrière

11.4.1 Un problème de taille

La taille de l’arbre de recherche associé au problème des n reines crôıt très vite avec n.
Les fonctions

(define num_nodes

(lambda (qtree)

(let ((qtree* (cdr qtree)))

(+ 1 (num_nodes* qtree*)))))

(define num_nodes*

(lambda (qtree*)

(apply + (map num_nodes qtree*))))

permettent le calcul du nombre de nœuds de l’arbre de recherche. On a

(num_nodes (queens_tree 4)) ==> 17

(num_nodes (queens_tree 7)) ==> 552

(num_nodes (queens_tree 10)) ==> 35539

(num_nodes (queens_tree 13)) ==> ;Aborting!: out of memory

Ce phénomène d’explosion combinatoire est inévitable; de plus, il survient dans la plupart
des problèmes d’optimisation. Cela ne signifie pas que, par exemple, le problème des treize
reines soit inaccessible, mais plutôt qu’il faudrait le résoudre sans mémoriser l’arbre de
recherche entier.

11.4.2 Fusionner construction et exploitation

Chaque solution du problème des reines correspond à une feuille de profondeur maximale
de l’arbre de recherche. Il est très possible de détecter toutes ces feuilles sans mémoriser
l’arbre entier; il suffit de construire l’arbre “en profondeur d’abord”, c’est-à-dire en visitant
tous les descendants d’un nœud avant de visiter ses frères. On convient aussi de visiter les
frères dans l’ordre gauche-droite. Il est ainsi possible de mener à bien une exploration
exhaustive de l’arbre en ne maintenant en mémoire que la branche reliant le “nœud
courant” à la racine. Le gain est considérable, puisque la taille d’une branche ne peut
excéder la dimension du problème, tandis que le nombre total de nœuds de l’arbre de
recherche est généralement fonction exponentielle de cette taille.

Pour fixer les idées, nous décrivons partiellement à la figure 25 le processus de calcul
correspondant à l’exploration en profondeur d’abord et de gauche à droite de l’arbre de la
figure 24. Cet arbre comporte six branches que nous numérotons de 1 (branche la plus à
gauche) à 6 et qui seront construites et visitées dans cet ordre. Le tableau de la figure 25
est à lire ligne par ligne; chaque ligne comporte trois étapes de calcul. Pour chaque étape,
on donne la “branche courante” précédée de son numéro et suivie de l’action à accomplir.
L’action “avant” prolonge la branche d’un élément; l’action “arrière” enlève le dernier
élément ajouté; l’action “échec” vérifie l’impossibilité de prolonger la branche courante et
l’action “solution” vérifie que la branche courante est une solution, qui sera rangée dans
une liste et/ou affichée.

branche action branche action branche action

1 : [ ] avant 1 : [1] avant 1 : [1, 3] échec
1 : [1, 3] arrière 1 : [1] avant 2 : [1, 4] avant
2 : [1, 4, 2] échec 2 : [1, 4, 2] arrière 2 : [1, 4] arrière
2 : [1] arrière 2 : [ ] avant 3 : [2] avant
3 : [2, 4] avant 3 : [2, 4, 1] avant 3 : [2, 4, 1, 3] solution
3 : [2, 4, 1, 3] arrière . . . . . . 6 : [4, 2] arrière
6 : [4] arrière 6 : [ ] fin

Figure 25: Problème des quatre reines, parcours de l’arbre de recherche

L’action “arrière”, caractéristique de la méthode, lui donne son nom retour arrière

(“backtracking” en anglais). Cette technique est habituelle dans les problèmes
d’optimisation, de recherche opérationnelle et d’intelligence artificielle. Il faut souligner
que le gain de performance ne concerne a priori que l’espace mémoire consommé. Si on
veut déterminer toutes les solutions du problème, il faudra construire et visiter l’arbre
de recherche entier. Cependant, si on se contente d’un sous-ensemble de solutions, voire
d’une seule, la méthode de retour arrière devient nettement plus rapide, comme nous le
verrons après l’avoir programmée.

11.4.3 Programmation de la méthode de retour arrière

La méthode de retour arrière se concrétise en une seule procédure, courte mais subtile.
Dans l’expression (steps next psol dim), dim désigne la dimension du problème et psol
est l’étiquette du nœud courant, une configuration légale mais peut-être incomplète; next
représente la position d’une reine supplémentaire à placer. L’évaluation de l’expression
(steps next psol dim) provoque la recherche de la prochaine solution, à partir du nœud
courant; la valeur est la solution en question, si elle existe, et la liste vide, sinon. La
recherche progresse dans la branche correspondant à next, si elle existe, dans la branche
suivante sinon. La procédure réutilise sans modification le prédicat legal? défini au
paragraphe 11.1.3. On a le code suivant :

(define steps

(lambda (next psol dim)

(newline)

(write psol)

(if (> next dim)

(if (null? psol)

’()

(steps (add1 (car psol)) (cdr psol) dim))

(let ((lp (cons next psol)))

(if (legal? lp)

(if (= (length lp) dim)

lp

(steps 1 lp dim))

(steps (add1 next) psol dim))))))

Considérons à nouveau le problème des quatre reines et l’arbre de recherche de la
figure 24. A titre d’exemple, nous choisissons comme nœud courant l’avant-dernier nœud
de la deuxième branche de l’arbre, correspondant à la solution partielle {(4, 3), (1, 4)},115
représentée par la liste (4 1). La session

(steps 1 ’(4 1) 4) ==> (3 1 4 2)

115Rappelons que le couple (a, b) indique la présence d’une reine en ligne a, colonne b.

montre que la solution suivante (en fait, la première) est fournie par la troisième branche.
La session

(steps 1 ’(3 1 4 2) 4) ==> (2 4 1 3)

fournit la seconde solution; enfin, la session

(steps 1 ’(2 4 1 3) 4) ==> ()

montre qu’il n’y a plus de solution.

Si on cherche toutes les solutions du problème, la méthode du retour arrière semble peu
attrayante, parce qu’il faut “relancer” la recherche après chaque solution. Cet inconvénient
disparâıt si l’on dispose d’une procédure qui automatise cette relance et réunit les solutions
dans une liste. La procédure research accomplit ce travail.

(define research

(lambda (dim)

(letrec ((loop

(lambda (sol)

(if (null? sol)

’()

(cons sol (loop (steps (add1 (car sol))

(cdr sol)

dim)))))))

(loop (steps 1 ’() dim)))))

On a par exemple

(research 4) ==> ((3 1 4 2) (2 4 1 3))

(research 6) ==>

((5 3 1 6 4 2) (4 1 5 2 6 3) (3 6 2 5 1 4) (2 4 6 1 3 5))

La technique du retour arrière est donc temporellement aussi efficace que la technique
de construction complète de l’arbre, dans le cas où on s’intéresse à la liste complète des
solutions. Par contre, si on se contente de quelques solutions, voire d’une seule, la technique
du retour arrière est de loin plus efficace. Sur un système raisonnablement configuré, il
n’est pas possible de construire globalement l’arbre de recherche relatif au problème des
reines dès que la dimension dépasse 10. Avec le retour arrière, il est possible d’aller
nettement plus loin; la session suivante donne les trois premières solutions du problème
des vingt reines :

(steps 1 ’() 20)

==> (11 6 14 7 10 8 19 16 9 17 20 18 12 15 13 4 2 5 3 1)

(steps 1 ’(11 6 14 7 10 8 19 16 9 17 20 18 12 15 13 4 2 5 3 1) 20)

==> (11 7 14 6 8 10 19 16 9 17 20 18 12 15 13 4 2 5 3 1)

(steps 1 ’(11 7 14 6 8 10 19 16 9 17 20 18 12 15 13 4 2 5 3 1) 20)

==> (7 13 11 8 6 18 10 17 9 20 16 19 15 12 14 4 2 5 3 1)

Remarque. Les solutions sont triées dans l’ordre lexicographique croissant des inverses des
listes qui les représentent.

11.5 Le problème du voyageur de commerce

Il s’agit encore d’un problème classique. Les données sont un ensemble de villes, que nous
supposons numérotées de 1 à n, une “carte” dmap des distances (positives) entre villes,
et un nombre positif lmax. Un circuit est un itinéraire partant d’une ville donnée et y
revenant après être passé exactement une fois par chaque autre ville. On demande de
trouver (s’il existe) un circuit de longueur inférieure à lmax.

On voit immédiatement que le problème est très semblable à celui des reines. Une
solution sera représentée par une permutation des nombres de 1 à [[n]] et on construira les
circuits progressivement, à partir de circuits partiels. Il est en fait très simple de reprendre
une version du programme des reines, par exemple celle du paragraphe 11.2.4, et de
l’adapter au problème présent. Cependant, notre but ici n’est pas de reprendre seulement
l’idée du programme, mais aussi la majeure partie du code. Ce code se composait de
deux procédures générales concernant la technique de recherche (mk_extend_par_sol et
mk_extend_par_sol*) et de deux procédures spécifiques, queens et legal?. On remplace
la procédure queens par la procédure travel :

(define travel

(lambda (dim)

((mk_extend_par_sol* dim) ’((1)) (- dim 1))))

Pour gagner du temps, on considère que le point de départ et d’arrivée est toujours la ville
numéro 1.

Le cas du prédicat legal? est plus délicat parce que, pour ne pas devoir modifier (si
peu que ce soit) les deux procédures de recherche, ce prédicat prend un seul argument, la
dimension du problème (c’est-à-dire, ici, le nombre de villes). Pourtant, il est clair que
la légalité d’un circuit partiel ou complet dépend de deux autres paramètres, la matrice
des distances entre villes et la longueur maximale autorisée. On connâıt maintenant la
technique pour résoudre ce problème, qui consiste à construire un générateur de procédure
mk_legal? dont les deux arguments seront la carte des distances et la longueur maximale.

Pour permettre l’expérimentation, nous nous fixons une matrice des distances entre
douze villes numérotées de 1 à 12. Si la dimension du problème est inférieure à 12, seules
les colonnes les plus à gauche sont prises en compte. Cette matrice est :

0 200 316 200 316 282 447 412 423 632 141 223

0 141 282 316 200 400 223 316 600 141 100

0 423 447 316 510 223 400 707 282 100

0 141 200 282 412 316 447 141 360

0 141 141 360 200 316 200 412

0 200 223 141 400 141 300

0 360 141 200 316 500

0 223 538 300 282

0 316 282 412

0 510 700

0 223

0

La matrice est naturellement symétrique et de diagonale nulle, puisque la distance de a
à b est égale à la distance de b à a et que la distance d’une ville à elle-même est nulle.
En Scheme, nous représentons cette matrice par la variable fonctionnelle globale *dmap*

définie comme suit :116

(define *dmap*

(lambda (i j)

(cond ((= i j) 0)

((> i j) (dmap j i))

((= i 1) (cond ((= j 2) 200)

... ...

((= j 12) 316)))

((= i 2) ...)

... ...

((= i 11) (cond ((= j 12) 223))))))

Remarque. Il peut parâıtre curieux de réadmettre une donnée globale juste après l’avoir
éliminée. La raison est pragmatique : nous voulons éviter d’avoir à réintroduire le code de
la matrice des distances à chaque essai.

On définit alors

(define mk_legal?

(lambda (dmap lmax)

(lambda (npcirc)

(letrec

((circ?

(lambda (t l-c)

(cond

((null? l-c) #t)

(else (and (not (= t (car l-c)))

(circ? t (cdr l-c)))))))

(size-c

(lambda (l-c)

(cond

((= (car l-c) 1) 0)

(else (+ (size-c (cdr l-c))

116Nous verrons au chapitre 12 un autre moyen de représenter les matrices.

(dmap (car l-c) 1)

(dmap (car l-c) (cadr l-c))

(- (dmap (cadr l-c) 1))))))))

(and (circ? (car npcirc) (cdr npcirc))

(<= (size-c npcirc) lmax))))))

A titre de premier exemple, nous cherchons le plus court circuit impliquant les cinq
premières villes. Cela se fait par approximations successives. On cherche d’abord tous les
circuits de longueur maximale inférieure à 1200 :

(define legal? (mk_legal? *dmap* 1200)) ==> ...

(travel 5) ==> ((4 5 3 2 1) (5 4 2 3 1) (4 5 2 3 1)

(3 2 5 4 1) (2 3 5 4 1) (3 2 4 5 1))

Il y a trois solutions (chaque circuit est représenté dans les deux sens). Par contre, la
session

(define legal? (mk_legal? *dmap* 1100)) ==> ...

(travel 5) ==> ()

montre que la longueur de ces circuits dépasse 1100. Par essais successifs ou par mesure
directe, on trouve que la meilleure solution est le circuit (4 5 2 3 1), ou le circuit inverse
(3 2 5 4 1), dont la longueur est 1114.

Cette version du programme est correcte mais présente deux inconvénients
évidents. Le premier est l’obligation de redéfinir legal?, et donc d’évaluer
(define legal? (mk_legal? *dmap* ...)), chaque fois que l’on souhaite changer la
longueur maximale autorisée et le second est qu’il n’est pas possible de trouver d’emblée
les circuits de longueur optimale. Signalons d’emblée, en ce qui concerne le premier point,
que le “remède” näıf consistant à intégrer simplement la définition de legal? dans travel
est incorrecte. On ne peut donc pas écrire

(define travel ;; programme incorrect !!

(lambda (dim dmap lmax)

(let ((legal? (mk_legal? *dmap* lmax)))

((mk_extend_par_sol* dim) ’((1)) (- dim 1)))))

En effet, d’après le principe de portée lexicale, la définition locale de legal?

est sans effet, la variable legal? n’ayant aucune occurrence dans l’expression
((mk_extend_par_sol* dim) ’((1)) (- dim 1)). Ce problème sera réglé dans le cadre
du programme générique de recherche développé au paragraphe suivant; on y verra aussi
comment automatiser la découverte du circuit de longueur minimale.

11.6 Programme générique de recherche

On peut pousser un peu plus loin la réutilisation des procédures en définissant un
programme générique de recherche de solutions optimales, à partir duquel il serait facile

de programmer des procédures spécifiques pour des problèmes d’optimisation particuliers,
tels ceux des reines et du voyageur de commerce.

Nous nous intéressons à tous les problèmes d’optimisation pour lesquels l’espace des
solutions est inclus dans l’ensemble des permutations des nombres de 1 à n, n étant la
dimension du problème. Cette classe est relativement vaste, et on indiquera brièvement
comment on pourrait encore l’élargir. Le programme générique comporte notamment le
code correspondant à la génération des permutations.

(define solver

(lambda (legal? dim initlist depth)

(letrec

((extend_par_sol* (lambda (psols m) (...)))

(extend_par_sol (lambda (psol) (...))))

(extend_par_sol* initlist depth))))

Les codes des fonctions auxiliaires extend_par_sol* et extend_par_sol ont été donnés
au paragraphe 11.1.3.117 On notera que les variables dim et legal? ne sont pas locales
aux procédures auxiliaires, mais ne sont pas non plus globales,118 puisqu’elles sont des
arguments de la procédure principale solver. Cela rend ici inutile la technique du
générateur de procédure.

(define solve_queens

(lambda (dim)

(let ((legal_queens? ...))

(solver legal_queens? dim ’(()) dim))))

(define solve_travel

(lambda (dim dmap lmax)

(let ((mk_legal_travel? ...))

(solver (mk_legal_travel? dmap lmax) dim ’((1)) (- dim 1)))))

Le code des prédicats auxiliaires legal_queens? et mk_legal_travel? sont
respectivement ceux des prédicats legal? (cf. § 11.1.3) et mk_legal? (cf. § 11.5).

11.7 Processus d’approximations successives

Un moyen très simple d’automatiser la recherche du circuit de taille optimale est de coupler
la méthode de bissection introduite au paragraphe 6.4.2 avec la procédure solve_travel.
Si on suppose qu’un circuit de longueur inférieure à 220 existe, on peut utiliser le code
suivant :

(define best_travel

(lambda (dim dmap)

117Dans le code de la seconde fonction, la constante numérique 8 est remplacée par la variable dim.
118Par rapport aux procédures auxiliaires, ces variables sont dites libres.

(letrec

((iter

(lambda (i j)

(let* ((k (quotient (+ i j 1) 2))

(sols (solve_travel dim dmap k)))

(if (null? sols)

(iter k j)

(if (= (+ i 1) j)

(cons j sols)

(iter i k)))))))

(iter -1 (expt 2 20)))))

On a notamment

(best_travel 1 *dmap*) ==> (0 (1))

(best_travel 2 *dmap*) ==> (400 (2 1))

(best_travel 3 *dmap*) ==> (657 (3 2 1) (2 3 1))

(best_travel 4 *dmap*) ==> (939 (4 2 3 1) (3 2 4 1))

(best_travel 5 *dmap*) ==> (1114 (4 5 2 3 1) (3 2 5 4 1))

(best_travel 6 *dmap*) ==>

(1139 (4 5 6 3 2 1) (4 5 6 2 3 1) (3 2 6 5 4 1) (2 3 6 5 4 1))

(best_travel 7 *dmap*) ==>

(1339 (4 5 7 6 3 2 1) (4 5 7 6 2 3 1)

(3 2 6 7 5 4 1) (2 3 6 7 5 4 1))

(best_travel 8 *dmap*) ==>

(1469 (4 5 7 6 8 3 2 1) (2 3 8 6 7 5 4 1))

Le premier élément de la liste-résultat est la longueur commune des circuits composant le
reste de cette liste.

Cette solution est relativement lente, parce que jusqu’à une vingtaine d’appels à la
fonction solve_travel sont nécessaires. Une meilleure approche consiste à adapter les
procédures steps et research introduites au paragraphe 11.4.3 pour n’explorer qu’une
seule fois l’arbre de recherche. Le pas initial consiste en la recherche du premier circuit;
soit ℓ0 sa longueur. Ensuite, on continue la recherche en ignorant tous les circuits de
longueur supérieure à ℓ0, jusqu’à en trouver un de longueur inférieure ℓ1, et ainsi de suite.
Le travail de programmation est laissé au lecteur.

12 Instructions altérantes et vecteurs

12.1 Introduction

Au terme de ce parcours, nous avons appris à apprécier la simplicité et la flexibilité de la
structure de liste. Une raison de cette simplicité est le caractère séquentiel de la liste : on
accède aux éléments un par un, dans l’ordre, et, la plupart du temps, on ne procède pas à
des comptages. Ceci contraste avec les structures à accès direct, par exemple les tableaux,
largement utilisés dans beaucoup de langages de programmation; dans un tableau, le temps
d’accès au nième élément est constant, indépendant de n. Cependant, le lecteur familier
des langages à tableaux comme Fortran, Pascal et C n’ignore pas que le calcul d’index119

dans le maniement des tableaux est à la source de nombreuses erreurs de programmation.
Les structures de listes sont commodes aussi pour une autre raison : elles sont construites
mais jamais modifiées (jusqu’ici); elles ne peuvent être détruites que si elles sont devenues
inutiles, c’est-à-dire inaccessibles.120

Cette simplicité a un prix, que nous illustrons par un exemple simple :

(define u ’(a b)) ==> ...

(define v ’(c d)) ==> ...

(define w (append u v)) ==> ...

L’évaluation de (append u v) requiert, via l’application de la fonction [[cons]], autant de
cellules que d’éléments dans u. De plus, comme nous l’avons déjà mentionné, le temps
d’evaluation de (append u v) est proportionnel à la longueur de [[u]]. Avant l’évaluation
du troisième define, la situation en mémoire est celle de la figure 26.

u
R

?
a

-

?
b

•
v
R

?
c

-

?
d

•

Figure 26: Situation avant utilisation de append

Le symbole • représente un pointeur vers la liste vide, qui est un atome (non représenté
sur la figure). La situation après le troisième define est représentée à la figure 27; deux
nouvelles cellules ont été consommées.

On pourrait souhaiter ne pas consommer de nouvelles cellules et simplement altérer

le pointeur vers la liste vide présent dans la représentation de u, de manière à obtenir la

119L’index, ou indice d’un élément de tableau est son numéro, le nombre entier qui permet d’accéder à
cet élément.

120Le système dispose d’un “récupérateur de cellules”, un processus qui détecte les cellules devenues
inaccessibles et rend à nouveau disponible la mémoire correspondante. Le fonctionnement du récupérateur
est géré par le système et l’utilisateur n’a le plus souvent pas à s’en préoccuper.

u

Figure 27: Situation après utilisation de append

situation illustrée à la figure 28. La commande append! permet exactement cela, ainsi
que l’indique la session suivante :

(define u ’(a b)) ==> ...

(define v ’(c d)) ==> ...

(append! u v) ==> (a b c d)

u ==> (a b c d)

u
R

?
a

-

?
b

-
v
R

?
c

-

?
d

•

Figure 28: Situation après utilisation de append!

Le danger de ce mode de concaténation est que le premier argument a été altéré;
après évaluation de l’expression (append! u v) la valeur de u est devenue (a b c d),
comme l’indiquent la session et la figure. Le risque lié à l’emploi d’une commande comme
append! (on ne peut guère appeler ce type d’objet une fonction) est double. D’une part,
il y a l’altération elle-même. Jusqu’ici, nous n’avons jamais évoqué la possibilité d’altérer
un objet déjà construit, et on peut penser que cette politique de non-altération est une
des raisons de la (relative) simplicité de la programmation en Scheme. D’autre part, les
conséquences de l’altération de pointeur qui survient lors de l’évaluation de l’expression
(append! u v) ne se limitent pas nécessairement au changement de la valeur de u, comme
le montre la session suivante :

(define u ’(a b)) ==> ...

(define v ’(c d)) ==> ...

(define w ’(e f)) ==> ...

(define x (append w u)) ==> ...

x ==> (e f a b)

(define y (append! u v)) ==> ...

x ==> (e f a b c d)

L’évaluation de l’expression (define y (append! u v)) a modifié la valeur de x, alors
que cette variable n’apparaissait pas dans l’expression. On conçoit que ce phénomène,
appelé effet de bord, puisse provoquer des erreurs de programmation.

Dans la suite de ce chapitre, nous voyons comment réaliser des programmes altérants
comme append!, et aussi comment les utiliser en mâıtrisant le phénomène d’effet de bord.

Nous voyons aussi comment le type vecteur permet de structurer des données en
permettant l’accès direct à celles-ci. L’utilité de l’accès direct serait limitée s’il ne
s’accompagnait de la possibilité d’altérer les données. En fait, le type vecteur permet
l’accès direct en lecture et en écriture.

12.2 L’instruction d’affectation

Les effets de bord apparaissent déjà avec la plus simple des instructions altérantes, set!.121

L’évaluation dans un environnement E de l’expression

(set! x e)

ne produit pas de valeur utile, mais a pour effet d’altérer dans l’environnement E la liaison
relative à x, dont la nouvelle valeur est celle de l’expression e. (Si x n’est pas lié dans E,
un message d’erreur est produit.)

Voici un exemple simple illustrant le comportement de set! :

(define *i* 0) ==> ...

(define counter

(lambda () (begin (set! *i* (+ i 1)) *i*))) ==> ...

(counter) ==> 1

(counter) ==> 2

(counter) ==> 3

(set! *i* 0) ==> ...

(counter) ==> 1

Chaque évaluation de (counter) modifie par effet de bord la variable globale *i*.

Une application plus amusante de la commande set! est la simulation de la forme
spéciale letrec à partir de la forme spéciale let. Considérons l’exemple classique de la
factorielle :

121En anglais, cela se prononce généralement “set-bang” . . . ce qui n’est pas un mauvais moyen de rappeler
le danger lié à l’usage de ce type d’instruction.

(define fact

(letrec ((fact_it

(lambda (n a)

(if (= n 0) a (fact_it (- n 1) (* n a))))))

(lambda (n) (fact_it n 1))))

On a vu que l’usage de let à la place de letrec rendrait la définition inopérante. Par
contre, cette définition est entièrement équivalente à

(define fact

(let ((fact_it ’any_value))

(let ((body

(lambda (n a)

(if (= n 0) a (fact_it (- n 1) (* n a))))))

(set! fact_it body))

(lambda (n) (fact_it n 1))))

C’est le set! qui permet de lier, dans un même environnement, la variable fact à la
valeur d’une expression qui contient cette variable, en réalisant la récursion. Notons aussi
la présence d’un begin implicite dans le let externe; d’une manière générale, la forme

(let (...) e1 . . . en e)

est équivalente à la forme

(let (...) (begin e1 . . . en e)) .

Les expressions ei sont évaluées, mais leurs valeurs sont ignorées; seuls leurs effets sont
pris en compte. (Voir aussi la remarque à la fin du paragraphe 8.7.)

Remarque. A titre de curiosité, signalons qu’il est possible de mettre en œuvre la récursion
en utilisant seulement la forme lambda. L’exemple suivant est tiré de [1] :

((lambda (n)

((lambda (fact) (fact fact n))

(lambda (ft k)

(if (= k 0) 1 (* k (ft ft (- k 1)))))))

4)

==> 24

12.3 Altération de structures

Les commandes set-car! et set-cdr! permettent d’altérer une paire pointée, ainsi que
le montre la session suivante :

(define x ’(a . b)) ==> ...

(define y x) ==> ...

(set-car! x ’c) ==> ...

(set-cdr! y ’d) ==> ...

x ==> (c . d)

y ==> (c . d)

On notera que, x et y étant le même objet, toute altération de x est aussi une altération
de y et réciproquement.

La session suivante produit une liste infinie et une liste de profondeur infinie :

(define x ’(a)) ==> ...

(set-cdr! x x) ==> ...

x ==> (a a a a a a a a a a a a a a ...

(define y ’(a)) ==> ...

(set-car! y y) ==> ...

y ==> (((((((((((((((((((((((((((((...

On peut produire aisément la fonction altérante append! déjà mentionnée en utilisant
set-cdr! :

(define append!

(lambda (u v)

(set-cdr! (last-pair u) v)

u))

(define last-pair

(lambda (u)

(cond ((null? u) (error "liste vide" u))

((null? (cdr u)) u)

(last-pair (cdr u)))))

Remarque. On notera la présence d’un begin implicite dans la définition de append!

12.4 Les vecteurs

12.4.1 Pourquoi des vecteurs?

La fonction prédéfinie list-ref renvoie un élément d’une liste spécifié par son rang, le
rang du premier élément étant 0 :

(list-ref ’(a b c) 0) ==> a

(list-ref ’(a b c) 2) ==> c

(list-ref ’(a b c) 3) ==> Error - incorrect range

L’accès étant séquentiel, le temps d’exécution de (list-ref l i) est proportionnel à [[i]].
Considérons la fonction write-ref définie comme suit :

(define write-ref

(lambda (l i e)

(cond ((null? l) (error "liste trop courte" l))

((zero? i) (cons e (cdr l)))

(else (cons (car l) (write-ref (cdr l) (- i 1) e))))))

On a

(write-ref ’(10 11 12 13) 2 ’aa) ==> (10 11 aa 13)

Le temps d’exécution de (write-ref l i e) est proportionnel à [[i]], et le nombre de
cellules consommées est égal à [[i]] + 1.

Les vecteurs permettront l’accès direct, le temps de lecture et d’écriture (altérante)
d’un élément étant indépendant du rang (de l’indice) de cet élément.

On peut remédier à la consommation de mémoire en utilisant la commande
write-ref! :

(define write-ref!

(lambda (l i e)

(cond ((null? l) (error "liste trop courte" l))

((zero? i) (set-car! l e))

(else (write-ref! (cdr l) (- i 1) e)))))

On a

(define l ’(10 11 12 13)) ==> ...

(write-ref! l 2 ’aa) ==> ...

l ==> (10 11 aa 13)

Le temps d’exécution reste proportionnel au rang spécifié en argument.

12.4.2 Les primitives vectorielles

Le constructeur vector et l’accesseur vector-ref sont les analogues pour les vecteurs du
constructeur list et de l’accesseur list-ref. On a donc

(list 4 ’x 3 ’(7 . a)) ==> (4 x 3 (7 . a))

(vector 4 ’x 3 ’(7 . a)) ==> #(4 x 3 (7 . a))

(list-ref ’(4 x 3 (7 . a)) 2) ==> 3

(vector-ref ’#(4 x 3 (7 . a)) 2) ==> 3

Comme déjà mentionné, la différence est que l’accès aux éléments d’une liste est séquentiel,
tandis que l’accès aux éléments d’un vecteur est direct. On observe qu’à l’affichage, le
symbole # permet de distinguer un vecteur d’une liste.

On a aussi la primitive vector-length qui donne la longueur d’un vecteur, c’est-à-dire
le nombre d’éléments qu’il contient :

(vector-length ’#(4 x 3 (7 . a))) ==> 4

Les indices des éléments d’un vecteur de longueur n sont 0, 1, . . . , n − 1. La primitive
make-vector produit un vecteur d’une longueur spécifiée en argument; le contenu de ce
vecteur peut être non spécifié, ou spécifié par un second argument qui sera la valeur de
chacun des éléments du vecteur :

(make-vector 4) ==> #(() () () ())

(make-vector 4 ’a) ==> #(a a a a)

L’accesseur subvector permet d’accéder à un sous-vecteur :

(subvector ’#(a b c d) 1 4) ==> #(b c d)

(subvector ’#(a b c d) 0 1) ==> #(a)

(subvector ’#(a b c d) 4 4) ==> #()

(subvector ’#(a b c d) 3 5) ==> Error - 5 out of range

On a également les fonctions de conversion vector->list et list->vector :

(vector->list ’#(a b c d)) ==> (a b c d)

(list->vector ’(a b c d)) ==> #(a b c d)

Voici enfin une illustration de la primitive d’écriture altérante des vecteurs :

(define v ’#(0 2 4 6 8 10 12 14 16 18)) ==> ...

(define w v) ==> ...

(vector-set! v 3 -99) ==> ...

v ==> #(0 2 4 -99 8 10 12 14 16 18)

w ==> #(0 2 4 -99 8 10 12 14 16 18)

En dépit du danger lié aux effets de bord, l’altération est une notion utile, qui peut accélérer
notablement l’exécution de certains programmes, tout en restreignant la consommation
de mémoire.

12.4.3 Tri par insertion

La technique du tri de liste par insertion est bien connue (cf. § 5.3.4); dans le cas des listes
de nombres, on la réalise au moyen des deux procédures suivantes :

(define insertsort

(lambda (ls)

(if (or (null? ls) (null? (cdr ls)))

ls

(insert (car ls)

(insertsort (cdr ls))))))

(define insert

(lambda (a ls)

(cond

((null? ls) (cons a ’()))

((< a (car ls)) (cons a ls))

(else (cons (car ls)

(insert a (cdr ls)))))))

Le même principe permet de trier un vecteur in situ, c’est-à-dire sans consommation de
mémoire auxiliaire. La fonction principale est :

(define vector-insertsort!

(lambda (v)

(let ((size (vector-length v)))

(letrec ((loop

(lambda (k)

(if (< k size)

(begin (vector-insert! k v)

(loop (+ k 1)))))))

(loop 1)))))

On observe immédiatement que, pour un vecteur [[v]] de longueur n + 1 (dont les indices
sont nécessairement 0, . . . , n), la commande

vector-insertsort! v

est équivalente à la séquence

(vector-insert! 1 v)

(vector-insert! 2 v)
...

...
...

...
(vector-insert! n v)

La fonction altérante auxiliaire, qui fait réellement le travail, est

(define vector-insert!

(lambda (k vec)

(let ((val (vector-ref vec k)))

(letrec

((insert-h

(lambda (m)

(if (zero? m)

(vector-set! vec 0 val)

(let ((c (vector-ref vec (- m 1))))

(if (< val c)

(begin (vector-set! vec m c)

(insert-h (- m 1)))

(vector-set! vec m val)))))))

(insert-h k)))))

On a les spécifications suivantes :

Si v est (lié à) un vecteur dans l’environnement courant avant l’exécution de
(vector-insertsort! v), alors v est (lié à) la version triée de ce vecteur après
l’exécution.

Cette spécification est une conséquence immédiate de la suivante :

Si le préfixe v[0 :k−1] est trié avant l’exécution de la forme (vector-insert!

k v), alors cette exécution a pour effet d’insérer v[k] à sa place. Le suffixe
v[k+1:n] n’est pas altéré.

Un essai permet de visualiser le mode de fonctionnement de la commande
(vector-insert! k v) :

(define v ’#(9 3 7 0 5 1)) ==> ...

v ==> #(9 3 7 0 5 1)

(vector-insert! 1 v) ==> ...

v ==> #(3 9 7 0 5 1)

(vector-insert! 2 v) ==> ...

v ==> #(3 7 9 0 5 1)

(vector-insert! 3 v) ==> ...

v ==> #(0 3 7 9 5 1)

(vector-insert! 4 v) ==> ...

v ==> #(0 3 5 7 9 1)

(vector-insert! 5 v) ==> ...

v ==> #(0 1 3 5 7 9)

A chaque étape, la longueur du préfixe trié du vecteur augmente d’une unité. Plus
précisément, l’appel de (vector-insert! k v) crée la liaison val: v[k] puis provoque
le nombre adéquat de “décalages”, suivi de l’écrasement de la dernière case recopiée par
val. Considérons le fragment important du programme :

... ...

;; (insert-h m)

(let ((c (vector-ref vec (- m 1))))

(if (< val c)

(begin

(vector-set! vec m c)

(insert-h (- m 1)))

(vector-set! vec m val)))

... ...

Voici un fragment de la liste des états successifs du vecteur v et des variables val et c;
il montre la manière dont un élément de la partie non triée du vecteur est inséré dans la
partie triée :

k: 4 v: #(0 3 7 9 5 1)) val: 5

init v: #(0 3 7 9 5 1)) val: 5 c: 9

then v: #(0 3 7 9 9 1)) val: 5 c: 7

then v: #(0 3 7 7 9 1)) val: 5 c: 3

else v: #(0 3 5 7 9 1))

12.4.4 Retournement d’une liste et d’un vecteur

Le programme de retournement d’une liste, dans sa version accumulante, évite l’usage
coûteux de append :

(define rev-it

(lambda (l)

(letrec ((r0 (lambda (u v)

(if (null? u)

v

(r0 (cdr u) (cons (car u) v))))))

(r0 l ’()))))

Ce programme n’est pas altérant : le résultat est une liste distincte de l’argument. On
pourrait construire un programme altérant reverse!, en “renversant” les pointeurs de la
liste. Le retournement in situ d’un vecteur v[0 : n−1] se fait plus aisément, en permutant
successivement v[0] et v[n − 1], v[1] et v[n − 2], etc. On a le programme suivant :

(define vector-reverse!

(lambda (v)

(let ((n (vector-length v)) (t 0))

(letrec

((switch (lambda (i j)

(begin (set! t (vector-ref v i))

(vector-set! v i (vector-ref v j))

(vector-set! v j t))))

(loop (lambda (i j)

(if (< i j)

(begin (switch i j)

(loop (+ i 1) (- j 1)))))))

(loop 0 (- n 1))))))

La boucle loop organise la séquence de permutations (il y en a n/2 si n est pair, (n−1)/2
si n est impair. La permutation de v[i] et v[j] est réalisée par la commande (switch i j),
qui elle-même est une séquence de trois affectations, utilisant une variable locale t. On a
par exemple :

(define v ’#(1 2 3 4)) ==> ...

(vector-reverse! v) ==> ...

v ==> ’#(4 3 2 1)

12.4.5 Vecteurs aléatoires

Les systèmes Scheme possèdent en général une primitive de génération de nombres
aléatoires ou pseudo-aléatoires. La valeur de (random r) est un nombre (pseudo-)aléatoire
compris entre 0 (inclus) et [[r]] (exclu). Ce nombre est entier si l’argument est entier, réel
si l’argument est réel. Les primitives random et make-vector permettent de créer des
vecteurs aléatoires :

(make-vector 4) ==> #(() () () ())

(random 1000) ==> 624 ;; [0...999]

(define random-vector

(lambda (n)

(let ((v (make-vector n)))

(letrec ((fill

(lambda (i)

(if (< i n)

(begin (vector-set! v i (random 1000))

(fill (+ i 1)))))))

(fill 0))

v)))

(random-vector 5) ==> #(90 933 656 240 587)

(random-vector 5) ==> #(666 943 203 632 512)

12.5 Efficacité expérimentale des programmes

Nous avons eu plusieurs fois l’occasion de signaler l’importance de la notion d’efficacité d’un
programme, même si ce concept n’a pas été — et ne sera pas — formellement défini. Dans
ce paragraphe, nous proposons deux techniques permettant d’évaluer expérimentalement
cette efficacité. La première se base sur la chronométrie, les systèmes de programmation
fournissant le plus souvent un moyen de mesurer le temps. La seconde approche consiste à
compter lors d’une exécution le nombre de fois qu’il est fait appel aux primitives de base,
telles les fonctions + et cons. Les deux techniques utilisent les instructions altérantes et
les effets de bord.

12.5.1 Un chronomètre

Les systèmes Scheme comportent souvent la procédure sans argument runtime qui renvoie
la durée (mesurée, par exemple, en microsecondes) pendant laquelle le système a utilisé

l’unité centrale de la machine sur laquelle il fonctionne. A partir de cette procédure
il est possible d’écrire un programme timer qui estime le coût temporel d’un appel de
procédure :

(define timer

(lambda (proc arg) ;; arg : argument unique

(let* ((t0 (runtime))

(val (proc arg))

(t1 (runtime))

(del (- t1 t0)))

(newline) (display "Time = ") (write del)

val)))

La fonction apply permet d’obtenir facilement une version adaptée aux procédures à
plusieurs arguments :

(define timer*

(lambda (proc . args) ;; args: liste d’arguments

(let* ((t0 (runtime))

(val (apply proc args))

(t1 (runtime))

(del (- t1 t0)))

(newline) (display "Time = ") (write del)

val)))

A titre d’exemple, on peut comparer les vitesses d’exécution des deux programmes de
calcul des coefficients binomiaux donnés aux paragraphes 1.2.1 et 4.3.1 :

(timer* cb1 10 5) ==> Time = 0.01

252

(timer* cb1 16 8) ==> Time = 0.59

12870

(timer* cb1 20 10) ==> Time = 8.41

184756

(timer* cb2 20 10) ==> Time = 0.00

184756

(timer* cb2 200 100) ==> Time = 0.01

90548514656103281165404177077484163874504589675413336841320

(timer* cb2 2000 1000) ==> Time = 0.10

... (601 chiffres !)

On voit que le programme cb1 est d’une lenteur inacceptable !

12.5.2 Vérification expérimentale : la suite de Fibonacci

Avec le chronomètre, nous comparons trois versions d’efficacité très différente du calcul
de la suite de Fibonacci :



12 INSTRUCTIONS ALTÉRANTES ET VECTEURS 257

(define fib

(lambda (n)

(if (< n 2) n (+ (fib (- n 1)) (fib (- n 2))))))

(define fib_it

(lambda (n)

(letrec

((fib_a

(lambda (n a b)

(if (= n 0) a (fib_a (- n 1) b (+ a b))))))

(fib_a n 0 1))))

(define fib_fast

(lambda (n)

(letrec

((sqr (lambda (n) (* n n)))

(fd

(lambda (n)

(cond ((= n 0) ’(0 . 1))

((= n 1) ’(1 . 1))

((even? n)

(let* ((p (/ n 2))

(df (fd p))

(d1 (car df)) (d2 (cdr df)))

(cons (* d1 (- (* 2 d2) d1))

(+ (sqr d1) (sqr d2)))))

((odd? n)

(let* ((p (/ (- n 1) 2))

(df (fd p))

(d1 (car df)) (d2 (cdr df)))

(cons (+ (sqr d1) (sqr d2))

(* d2 (+ (* 2 d1) d2)))))))))

(car (fd n)))))

On vérifie d’abord que le temps nécessaire au calcul de [[(fib n)]] est proportionnel au
résultat :

(time fib 21) ==> Time = 0.57

10946

(time fib 22) ==> Time = 0.94

17711

(time fib 23) ==> Time = 1.52

28657

(time fib 24) ==> Time = 2.39

46368

C’est non satisfaisant puisque [[(fib n)]] ≃ 0.447 ∗ 1.618n.
Ensuite, on vérifie que le temps nécessaire au calcul de [[(fib_it n)]] est quasi

proportionnel à l’argument :

(time fib_it 2000) ==> Time = 0.05

(time fib_it 4000) ==> Time = 0.11

(time fib_it 6000) ==> Time = 0.17

(time fib_it 8000) ==> Time = 0.24

(time fib_it 10000) ==> Time = 0.32

Enfin, on vérifie que le temps nécessaire au calcul de [[(fib_fast n)]] est encore plus
rapide :

(time fib_it 100000) ==> Time = 8.75

(time fib_fast 100000) ==> Time = 2.00

12.5.3 Comptage d’instructions

Supposons que l’on veuille compter le nombre de cellules consommées lors de l’évaluation
d’une expression. Il suffit de compter le nombre de fois que la primitive cons est appelée.
Considérons les définitions suivantes :

(define *cons* 0)

(define kons

(lambda (x y)

(set! *cons* (1+ *cons*))

(cons x y)))

On voit aisément que le programme kons est fonctionnellement équivalent à la primitive
cons, tout en provoquant le comptage souhaité par effet de bord. A titre de premier
exemple, nous allons comparer trois versions du programme de calcul des préfixes d’une
liste donnée (cf. § 7.4). Nous reproduisons ici ces programmes, dans lesquels cons est
remplacé par kons.

(define lpref1
(lambda (l) (if (null? l)

(kons l ’())
(kons ’() (put (car l) (lpref1 (cdr l)))))))

(define put
(lambda (x ll)
(if (null? ll) ’() (kons (kons x (car ll)) (put x (cdr ll))))))

(define lpref2
(lambda (l)
(if (null? l) (kons l ’()) (kons l (lpref2 (butlast l))))))

(define butlast
(lambda (l)
(if (null? (cdr l)) ’() (kons (car l) (butlast (cdr l))))))

(define lpref3 (lambda (l) (lreverse (lsuff (reverse l)))))

(define reverse (lambda (l) (rev l ’())))

(define lsuff
(lambda (l) (if (null? l) (kons l ’()) (kons l (lsuff (cdr l))))))

(define lreverse
(lambda (ll) (if (null? ll) ’() (kons (rev (car ll) ’())

(lreverse (cdr ll))))))

(define rev
(lambda (l a) (if (null? l) a (rev (cdr l) (kons (car l) a)))))

Remarque. L’expression (list l), utilisée dans les versions du paragraphe 7.4, a
été remplacée par (kons x ’()), pour tenir compte de la cellule consommée lors de
l’évaluation de l’expression.

Pour tester la version lpref1, on utilisera la définition suivante :

(define count_1

(lambda (n) (set! *cons* 0) (lpref1 (enum 1 n)) *cons*))

On définit de même count_2 et count_3. Rappelons que enum, appliqué aux entiers a et
b, construit la liste des entiers de a à b :

(define enum

(lambda (p q) (if (> p q) ’() (cons p (enum (+ p 1) q)))))

Les fonctions de comptage consistent en une séquence de trois opérations : remise du
compteur à 0, évaluation de l’expression à tester pour une liste de n éléments, affichage
du compteur, qui est la valeur renvoyée par la fonction de comptage. La session suivante
nous donne le nombre de cellules consommées pour quelques longueurs de liste.

(define *list_length* ’(1 2 5 10 50 100)) ==> ...

*list_length* ==> (1 2 5 10 50 100)

(map count_1 *list_length*) => (4 9 36 121 2601 10201)

(map count_2 *list_length*) => (2 4 16 56 1276 5051)

(map count_3 *list_length*) => (6 11 32 87 1427 5352)

On voit que l’efficacité spatiale (et temporelle) des trois fonctions est quadratique, c’est-
à-dire proportionnelle au carré de la taille de l’argument. On ne peut pas faire mieux,
puisque la taille du résultat à fournir est elle-même quadratique. On observe cependant
que la première version est deux fois plus gourmande que les deux dernières.

12.5.4 Ensembles ou multi-ensembles ?

L’exemple des variantes de lpref illustre bien la technique de comptage mais, comme
nous le verrons dans l’appendice, une étude expérimentale n’était pas nécessaire pour ce
cas simple. Par contre, les deux variantes occur_list et occur_list_bis de l’opérateur
général ensembliste introduit au paragraphe 10.6.5 constituent un banc d’essai plus
intéressant, parce que l’efficacité des programmes est moins facilement prévisible. On
compare les deux versions sur base du nombre d’appels aux primitives null?, cons, car
et cdr; on ne tiendra pas compte des types abstraits ici : les ensembles sont représentés par
des listes sans répétition (non triées). On construit d’abord le programme permettant de
créer des données de test. La valeur de (make-data p q r) est une liste de [[p]] ensembles
(listes sans répétitions) contenant chacun [[q]] entiers naturels compris entre 0 et [[r]]− 1.
On a

(define make-data

(lambda (p q r) (map (lambda (x) (make-set q r)) (enum 1 p)))

(define make-set

(lambda (q r)

(if (zero? q)

’()

(let ((rec (make-set (- q 1) r)))

(cons (ps-random r rec) rec)))))

Remarques. On doit avoir [[q]] ≤ [[r]]. Une exécution infinie est possible, mais avec une
probabilité nulle. La fonction n’est raisonnablement efficace que si [[q]] n’est pas trop
proche de [[r]].

(define ps-random

(lambda (r l)

(let ((try (random r)))

(if (member try l) (ps-random r l) try))))

On considère d’abord une liste de 20 ensembles de 10 nombres naturels strictement
inférieurs à 100 :

(define *data1* (make-data 20 10 100)) ==> ...

On peut alors observer un phénomène intéressant; l’évaluation de (occur_list n *data1*)

est relativement rapide si [[n]] est extrême, c’est-à-dire proche de sa valeur minimale 1 ou de
sa valeur maximale 20, et très lent sinon. Par contre, l’évaluation de (occur_list_bis n *data1*)

requiert un temps moyen, indépendant de [[n]]. En fait, cela était prévisible en regardant
la structure des deux programmes. En particulier, dans le cas du premier programme, les
valeurs extrêmes de [[n]] conduisent plus rapidement que les valeurs médianes aux cas de
base de la récursion (1 et [[n]]+1). L’expérimentation permet de quantifier le phénomène :

le premier programme est plus efficace pour les valeurs 1,18,19 et 20 et d’efficacité com-
parable pour les valeurs 2, 16 et 17. Pour les valeurs de 3 à 15, le second programme est
meilleur; ses performances ont de plus l’avantage de ne pas dépendre de [[n]].

Considérons à présent des données de taille plus réduite :

(define *data2* (make-data 8 4 20))

Les conclusions sont analogues aux précédentes, mais on observe que pour les valeurs
médianes la supériorité du second programme sur le premier est moins marquée. Cela
suggère que les performances du premier programme se dégradent plus vite que celles
du second quand la taille des données augmente. Le lecteur pourra confirmer ce fait en
procédant à des essais supplémentaires.

12.6 Tabulation

12.6.1 Introduction

Il n’est pas difficile de trouver un algorithme récursif efficace pour le calcul des coefficients
binomiaux ou pour les nombres de Fibonacci mais, dans de nombreux cas, la tâche est
moins évidente. Un exemple est celui du nombre de partitions d’un ensemble donné; nous
avons vu au paragraphe 10.3.1 que si p(n, k) est le nombre de partages possibles d’une
collection de n objets distincts en k lots non vides (on suppose 1 ≤ k ≤ n), on a la
définition récursive suivante :

p(n, k) =def if (k=n ∨ k=1) then 1 else p(n−1, k−1) + k ∗ p(n−1, k) .

Cette définition donne immédiatement lieu à un programme récursif simple, mais
inefficace :

(define num_part

(lambda (n k)

(if (or (= k n) (= k 1))

1

(+ (num_part (- n 1) (- k 1))

(* k (num_part (- n 1) k))))))

Les recalculs systématiques de résultats intermédiaires font que le temps augmente très
vite en fonction des arguments :

(time* num_part 10 5) ==> Time = 0.01

(time* num_part 20 10) ==> Time = 4.87

L’idée de la tabulation est de maintenir une table dans laquelle ces résultats intermédiaires
sont mémorisés, en vue d’une réutilisation possible. Cette table sera un vecteur organisé
comme suit :

indice 0 1 2 3 4 5 6 . . .
(n, k) (1,1) (2,1) (2,2) (3,1) (3,2) (3,3) (4,1) . . .

La fonction (n, k) 7→ n(n−1)
2 + k − 1 calcule l’indice associé à des arguments donnés.

Une mise en œuvre simple et efficace de cette idée consiste à utiliser une table globale.
On a

(define index

(lambda (n k)

(+ (/ (* n (- n 1)) 2) k -1)))

(define *MAX* 300) ;; valeur maximale de n

(define *MEMO* (make-vector (index *MAX* *MAX*) #f))

(define num_part_glob

(lambda (n k)

(let ((w (vector-ref *MEMO* (index n k))))

(if w

w

(let ((x (if (or (= k n) (= k 1))

1

(+ (num_part_glob (- n 1) (- k 1))

(* k (num_part_glob (- n 1) k))))))

(vector-set! *MEMO* (index n k) x)

x)))))

Si on refuse l’usage des variables globales, on peut écrire

(define num_part_loc

(lambda (n k)

(let ((table (make-vector (index n n) #f))) ;; table locale

(letrec

((aux

(lambda (m q)

(let ((w (vector-ref table (index m q))))

(if w

w

(let ((x (if (or (= q 1) (= q m))

1

(+ (aux (- m 1) (- q 1))

(* q (aux (- m 1) q))))))

(vector-set! table (index m q) x)

x))))))

(aux n k)))))

Remarque. Il est intéressant d’observer la manière dont la table se remplit en cours
d’exécution; cela peut se faire en intercalant dans le code un ordre d’impression après
chaque modification de la table :

(num_part_loc 5 3) ==>

#(#f #f #f #f #f 1 #f #f #f #f #f #f #f #f)

#(#f #f 1 #f #f 1 #f #f #f #f #f #f #f #f)

#(#f 1 1 #f #f 1 #f #f #f #f #f #f #f #f)

#(#f 1 1 #f 3 1 #f #f #f #f #f #f #f #f)

#(#f 1 1 #f 3 1 #f #f 6 #f #f #f #f #f)

#(#f 1 1 1 3 1 #f #f 6 #f #f #f #f #f)

#(#f 1 1 1 3 1 #f 7 6 #f #f #f #f #f)

#(#f 1 1 1 3 1 #f 7 6 #f #f #f 25 #f)

25

L’inconvénient d’une table locale est que chaque appel à num_part_loc provoque la
création d’une nouvelle table. On peut éviter cet inconvénient en utilisant une variable
libre :

(define num_part_free

(let ((table (make-vector (index *MAX* *MAX*) #f)))

(letrec

((aux

(lambda (m q)

(let ((w (vector-ref table (index m q))))

(if w

w

(let ((x (if (or (= q 1) (= q m))

1

(+ (aux (- m 1) (- q 1))

(* q (aux (- m 1) q))))))

(vector-set! table (index m q) x)

x))))))

aux)))

La table est créée lors de l’évaluation de la forme define et réutilisée à chaque appel,
comme le montre la session suivante :

(timer* num_part_loc 200 100) ==> Time = 1.73

(timer* num_part_loc 200 100) ==> Time = 1.73

(timer* num_part_free 200 100) ==> Time = 1.73

(timer* num_part_free 200 100) ==> Time = 0.00

Avec la table locale, deux calculs successifs de la même valeur (non affichée ici car il s’agit
d’un nombre de 235 chiffres) prennent le même temps; avec la table non locale, le second
“calcul” est instantané, car la valeur est trouvée dans la table.

12.6.2 Une solution générale

La technique de tabulation est potentiellement utile dans de nombreux cas, ce qui suggère
la conception d’un mécanisme automatique de mise en œuvre. Cependant, si on ne connâıt
pas à l’avance les valeurs possibles des arguments, l’emploi d’une structure à accès direct
peut devenir difficile. Pour simplifier le problème, nous utilisons une liste associative,
c’est-à-dire une liste de paires pointées (variable – valeur). La fonction memoize prend
comme argument une procédure (quelconque) et renvoie une procédure fonctionnellement
équivalente mais utilisant une table pour éviter les recalculs. On a

(define memoize

(lambda (proc)

(let ((table ’()))

(lambda args

(let ((v (assoc args table)))

(if v

(cdr v)

(let ((val (apply proc args)))

(set! table

(cons (cons args val) table))

val)))))))

Remarque. Le nombre et la nature des arguments de la procédure à tabuler n’étant pas
connus, on utilise la fonction assoc plutôt que assq ou assv (cf. § 4.3.2).

12.6.3 Quelques exemples

La définition

(define memo_num_part (memoize num_part))

n’est pas optimale, parce que seules les valeurs dont le calcul a été demandé par l’utilisateur
sont placées dans la table, ainsi que le montre la session suivante (les valeurs des appels
ont été omises) :

(timer* memo_num_part 18 7) ==> Time = 0.66 [...]

(timer* memo_num_part 18 7) ==> Time = 0. [...]

(timer* memo_num_part 17 6) ==> Time = 0.25 [...]

On voit que l’appel pour (18,6) n’a pas provoqué la tabulation de l’appel récursif (17,7).
La variante suivante remédie à ce défaut :

(define memo_num_part_bis

(memoize (lambda (n k)

(if (or (= k n) (= k 1))

1

(+ (memo_num_part_bis (- n 1) (- k 1))

(* k (memo_num_part_bis (- n 1) k)))))))

On a par exemple :

(timer* memo_num_part 18 7) ==> Time = 0.02 [...]

(timer* memo_num_part 18 7) ==> Time = 0. [...]

(timer* memo_num_part 17 6) ==> Time = 0. [...]

Le gain d’efficacité est ici considérable. Il faut quand même tenir compte de
l’encombrement de la table et aussi du temps passé à l’explorer; dans certains cas, le gain
de temps est plus limité. Un exemple typique est celui du programme money introduit au
paragraphe 10.5. On définit

(define memoney

(memoize (lambda (l s)

(cond ((< s 0) 0)

((= s 0) 1)

((null? l) 0)

(else (+ (memoney (cdr l) s)

(memoney l (- s (car l)))))))))

On a par exemple

(timer* money ’(1 5 20 50) 333) ==> Time = 10.60

1680

(timer* memoney ’(1 5 20 50) 333) ==> Time = 4.80

1680

On ne gagne qu’un facteur 2, mais cela reste intéressant si on effectue plusieurs applications
du programme, dans la mesure où la table est réutilisée d’une fois à l’autre :

(timer* money ’(1 5 20 50) 366) ==> Time = 14.63

2204

(timer* memoney ’(1 5 20 50) 366) ==> Time = 1.24

2204

(timer* money ’(1 5 20 50) 355) ==> Time = 13.17

2044

(timer* memoney ’(1 5 20 50) 355) ==> Time = 0.

2044

A Appendice

A.1 Vérification formelle des programmes

Nous avons parfois évoqué dans le texte la possibilité de prouver, au sens mathématique
du terme, certaines propriétés de procédures définies par l’utilisateur. Plusieurs techniques
existent mais la plus simple consiste à utiliser le principe d’induction relatif aux données
que traitent les procédures étudiées.

A.1.1 Programmes numériques

Pour fixer les idées, nous traitons d’abord le cas des programmes de calcul de la suite de
Fibonacci fib, fib_it et fib_fast (cf. § 12.5.2). La première propriété à prouver est

Pour tout n ∈ N, si [[n]] = n,
alors [[(fib n)]] = [[(fib_it n)]] = [[(fib_fast n)]] .

ou encore, en se référant à la définition de la fonction fib et aux codes de fib, fib_it et
fib_fast :122

Pour tout n ∈ N, si [[n]] = n,
alors fib(n) = [[(fib_a n 0 1)]] = [[(car (fd n))]] .

Pour la première égalité, la difficulté est qu’il n’est pas possible de la prouver par
récurrence directe. Il faut d’abord construire et démontrer un lemme, dont l’égalité
sera un simple corollaire. Typiquement, ce lemme traduit l’idée algorithmique sous-
jacente au programme étudié, c’est-à-dire fib_it. Cette idée est que, lors des appels
récursifs successifs, les deux derniers arguments de fib_it sont deux nombres de Fibonacci
consécutifs; cela donne le lemme déjà énoncé au paragraphe 4.5 :

Pour tous i, n ∈ N,
si [[n]] = n, [[a]] = fib(i) et [[b]] = fib(i+1),
alors [[(fib_a n a b)]] = fib(n+i).

Construire cet énoncé exige une bonne compréhension de la fonction auxiliaire fib_a; il
serait étonnant que l’on puisse vérifier formellement un programme sans l’avoir compris.
Par contre, il est immédiat de voir que la propriété initiale est un corollaire de ce
lemme; plus précisément, cette propriété correspond au cas particulier i = 0. Enfin,
la démonstration du lemme, par récurrence sur n, ne pose aucune difficulté. On démontre
d’abord que le lemme est vrai pour n = 0. En effet, il devient alors

Pour tout i ∈ N, si [[a]] = fib(i) et [[b]] = fib(i+1),
alors [[(fib_a 0 a b)]] = fib(i).

Il suffit d’exploiter le code du programme. On a

122On admet évidemment que le programme fib, traduction immédiate de la définition de la fonction fib,
est correct; on a donc [[(fib n)]] = fib(n) si [[n]] = n.

[[(fib_a 0 a b)]]
= [[(if (= 0 0) a (fib_a (- 0 1) b (+ a b)))]]
= [[a]]
= fib(i)

Supposons que le lemme soit vrai pour un certain n et démontrons qu’il reste vrai pour
n + 1. On suppose donc vrai l’énoncé suivant :

Hypothèse :
Pour tout j ∈ N, si [[n]] = n, [[a’]] = fib(j) et [[b’]] = fib(j+1),
alors [[(fib_a n a’ b’)]] = fib(n+j).

On doit en déduire, pour tout i ∈ N, l’énoncé suivant :

Thèse :
Si [[n]] = n, [[a]] = fib(i) et [[b]] = fib(i+1),
alors [[(fib_a (+ n 1) a b)]] = fib(n+1+i).

On fixe dans l’hypothèse la valeur j = i + 1, ce qui donne

Si [[n]] = n, [[a’]] = fib(i+1) et [[b’]] = fib(i+2),
alors [[(fib_a n a’ b’)]] = fib(n+i+1).

On a fib(i+2) = fib(i+1) + fib(i), par définition de fib; les égalités [[a]] = fib(i) et
[[b]] = fib(i+1) sont donc équivalentes aux égalités [[b]] = fib(i+1) et [[(+ a b)]] = fib(i+2).
De plus, on a n+i+1 = n+1+i. L’énoncé précédent se récrit donc en

Si [[n]] = n, [[a]] = fib(i) et [[b]] = fib(i+1),
alors [[(fib_a n b (+ a b))]] = fib(n+1+i).

On note alors que, vu le code de la fonction auxiliaire fib_a, on a

Si [[n]] = n,
alors [[(fib_a (+ n 1) a b)]] = [[(fib_a n b (+ a b))]] .

Le dernier énoncé devient

Si [[n]] = n, [[a]] = fib(i) et [[b]] = fib(i+1),
alors [[(fib_a (+ n 1) a b)]] = fib(n +1+i).

c’est-à-dire la thèse; ceci achève la démonstration de l’égalité concernant fib_it.

L’égalité concernant la fonction auxiliaire fd se démontre de manière analogue à celle
concernant fib_it. La première étape consiste à découvrir le lemme adéquat, qui est :

Pour tout n ∈ N, si [[n]] = n, alors [[(fd n)]] est la paire pointée de composants
fib(n) et fib(n+1).

A nouveau, le théorème à démontrer est un corollaire immédiat du lemme; si [[(fd n)]] est
la paire pointée de composants fib(n) et fib(n+1), alors [[(car (fd n))]] est bien fib(n).
La démonstration du lemme n’est pas difficile, du moins si on perçoit bien les propriétés
mathématiques de fib sur lesquelles se base le programme fd :

Si n = 2p, fib(p) = d1 et fib(p+1) = d2,
alors fib(n) = d1(2d2 − d1) et fib(n+1) = d2

1 + d2
2;

Si n = 2p+1, fib(p) = d1 et fib(p+1) = d2,
alors fib(n) = d2

1 + d2
2 et fib(n+1) = d2(2d1 + d2).

Par exemple, en posant p = 6, de fib(6) = 8 et fib(7) = 13, on déduit fib(12) = 8(26−8) =
144, fib(13) = 82 + 132 = 233 et fib(14) = 13(16+13) = 377. Ces propriétés se démontrent
simplement par récurrence sur p.

Le principe d’induction utilisé pour démontrer les propriétés des programmes
numériques est le plus souvent la récurrence, simple ou complète, mais des exceptions
existent. Un exemple connu est celui de la fonction d’Ackermann introduite au
paragraphe 7.5.2. On voudrait d’abord démontrer qu’il existe une et une seule fonction
mathématique Ack, dont le domaine de définition est exactement N × N et qui vérifie sur
ce domaine les propriétés

∀n ≥ 0 : Ack(0, n) = n + 1 ,
∀m > 0 : Ack(m, 0) = Ack(m − 1, 1) ,
∀m,n > 0 : Ack(m,n) = Ack(m − 1,Ack(m,n − 1)) ;

on voudrait prouver aussi que le programme ack est une réalisation de la fonction Ack ;
en particulier, l’évaluation de (ack m n) se termine toujours si [[m]], [[n]] ∈ N.

Pour étudier cette question, il faut utiliser un principe d’induction sur l’ensemble
N × N, basé sur l’ordre lexicographique (§ 5.3.4). Rappelons que (a, b) ≺ (c, d) est vrai si
a < c ou si a = c et b < d. Le principe d’induction est

Si pour tous x, y ∈ N, dès que la propriété P est vraie de tout couple (a, b)
tel que (a, b) ≺ (x, y), elle est vraie de (x, y), alors P est vraie de tout couple
(x, y).

Cela se note
∀x∀y [∀a∀b [(a, b) ≺ (x, y) ⇒ P (a, b)] ⇒ P (x, y)]

∀x∀y P (x, y)

On prouve par l’absurde la validité de ce principe. Supposons que, pour une certaine
propriété P , l’hypothèse de la règle soit vérifiée mais pas la conclusion. L’ensemble
des (x, y) falsifiant P n’est pas vide; soit x0 le plus petit naturel tel que pour un y
au moins, P (x0, y) est faux; soit aussi y0 le plus petit naturel tel que P (x0, y0) est
faux (le couple (x0, y0) est le ≺-minimum de l’ensemble des couples falsifiant P ). On
a ∀a∀b [(a, b) ≺ (x0, y0) ⇒ P (a, b)] sans avoir P (x0, y0), d’où une contradiction.

En ce qui concerne la fonction Ack, soit la propriété P (x, y) exprimant que Ack(x, y)
est univoquement défini. Il suffit de montrer que cette propriété vérifie l’hypothèse du
principe d’induction. Pour ce faire, on suppose donc que P (a, b) est vrai pour tous a, b
tels que (a, b) ≺ (x, y) et on démontre P (x, y). On distingue trois cas :

• Si x = 0, on a Ack(x, y) = y + 1, d’où P (x, y) est vrai.

• Si x > 0 et y = 0, on a Ack(x, y) = Ack(x − 1, 1). Comme (x − 1, 1) ≺ (x, y), on a
P (x− 1, 1), d’où Ack(x− 1, 1) et Ack(x, y) sont univoquement définis, et P (x, y) est
vrai.

• Soit x > 0 et y > 0. On a (x, y−1) ≺ (x, y), donc P (x, y−1) est vrai et Ack(x, y−1)
est univoquement défini; soit z sa valeur. On a (x− 1, z) ≺ (x, y) donc Ack(x− 1, z)
est univoquement défini; soit u sa valeur. On a nécessairement Ack(x, y) = u, d’où
P (x, y) est vrai.

Le même raisonnement permet d’établir la propriété

∀m,n ∈ N [([[m]] = m ∧ [[n]] = n) ⇒ [[(ack m n)]] = Ack(m,n)] .

A.1.2 Fonctions de listes

Les propriétés des programmes de listes se traitent comme celles des programmes
numériques, sauf que le principe d’induction sur les listes est utilisé si nécesssaire. Un
exemple très simple, sans induction, est le lemme suivant :

Pour tout objet x et pour toutes listes [[u]] et [[v]], on a
[[(cons x (append u v))]] = [[(append (cons x u) v))]]

C’est une conséquence immédiate du code de append et des liens existant entre les
primitives cons, car et cdr (cf. § 2.5). Ce lemme est utilisé dans la preuve par induction
de la propriété d’associativité de la fonction [[append]] (cf. § 5.3.2) :

Pour toutes listes [[u]], [[v]] et [[w]], on a
[[(append (append u v) w)]] = [[(append u (append v w))]]

On voit immédiatement que la propriété est vraie si [[u]] est la liste vide : les deux membres
de l’égalité se réduisent à [[(append v w)]]. Si [[u]] n’est pas vide, on suppose que la
propriété est vraie pour [[(cdr u)]] (hypothèse d’induction). Le membre de gauche du
théorème se réduit à

[[(append (cons (car u) (append (cdr u) v)) w)]]

qui, en vertu du lemme, se récrit en

[[(cons (car u) (append (append (cdr u) v) w))]]

puis, par l’hypothèse inductive, en

[[(cons (car u) (append (cdr u) (append v w)))]]

et enfin, vu le code des append, en

[[(append u (append v w))]],

qui est le membre de droite.

Enfin, il est intéressant de revenir sur la propriété des programmes flat_list et
flat_list_a mentionnée sans justification au paragraphe 7.3 :

Si [[l]] et [[a]] sont des listes, on a
[[(flat_list_a l a)]] = [[(append (flat_list l) a)]] .

Si [[l]] est la liste vide, les deux membres de l’égalité se réduisent à [[a]]. Sinon, on utilise le
principe d’induction profonde pour les listes, et on suppose que la propriété est vraie pour
[[(cdr l)]], et aussi pour [[(car l)]] si cette valeur est une liste (hypothèses inductives).
Le membre de gauche devient

[[(flat_list_a (car l) (flat_list_a (cdr l) a))]]

et, par hypothèse inductive

[[(flat_list_a (car l) (append (flat_list (cdr l)) a))]] .

A ce stade, on distingue trois cas. Si [[(car l)]] est la liste vide, la valeur précédente se
récrit en

[[(append (flat_list (cdr l)) a)]]

qui, vu le code de flat_list, se récrit en

[[(append (flat_list l) a)]] .

Si [[(car l)]] est un atome, la valeur précédente se récrit en

[[(cons (car l) (append (flat_list (cdr l)) a))]]

et, en tenant compte du lemme du début du paragraphe, en

[[(append (cons (car l) (flat_list (cdr l))) a)]]

et, vu le code de flat_list, en

[[(append (flat_list l) a)]] .

Si enfin [[(car l)]] est une liste, l’hypothèse inductive permet de récrire

[[(flat_list_a (car l) (append (flat_list (cdr l)) a))]]

en

[[(append (flat_list (car l)) (append (flat_list (cdr l)) a))]]

puis, vu l’associativité de [[append]], en

[[(append (append (flat_list (car l)) (flat_list (cdr l))) a)]]

qui, vu le code de flat_list, se récrit enfin en

[[(append (flat_list l) a)]] .

A.2 Etude formelle de l’efficacité des programmes

Les techniques introduites aux paragraphes 12.5.1 et 12.5.3 permettent de déterminer
expérimentalement l’efficacité des programmes, c’est-à-dire la quantité de ressources que
leur usage requiert. Il est parfois possible d’obtenir des conclusions par une étude formelle
basée à nouveau sur les principes d’induction. Nous illustrons cette technique avec les
programmes lpref1, lpref2 et lpref3 introduits au paragraphe 7.4, en nous référant
aux versions du paragraphe 12.5.3 (avec “cons” au lieu de “kons”).

A.2.1 Première version

Lemme. Si [[ll]] est une liste de longueur n, alors l’évaluation de (put x ll) requiert 2n
appels à cons.
Démonstration. En désignant par p(n) le nombre d’appels, on observe immédiatement que
p(0) = 0 et, si n > 0, p(n) = 2 + p(n − 1) puisque l’évaluation de (put x ll) requiert
deux appels à cons et l’évaluation de (put x (cdr ll)).

Théorème. Si [[l]] est une liste de longueur n, alors l’évaluation de (lpref1 l) requiert
(n + 1)2 appels à cons.
Démonstration. En désignant par L(n) le nombre d’appels, on observe immédiatement
que L(0) = 1 et, si n > 0, L(n) = 1 + p(n) + L(n − 1) = 1 + 2n + L(n −
1) puisque l’évaluation de (lpref1 l) requiert un appel à cons et l’évaluation de
(put (car l) (lpref1 (cdr l))), [[(lpref1 (cdr l))]] étant une liste de longueur n.
On montre facilement que la seule fonction L de domaine N qui vérifie L(0) = 1 et
L(n) = 1 + 2n + L(n − 1) si n > 0 vérifie aussi L(n) = (n + 1)2.

A.2.2 Deuxième version

Lemme. Si [[l]] est une liste de longueur n > 0, alors l’évaluation de la forme (butlast l)

requiert n − 1 appels à cons.
Démonstration. En désignant par b(n) le nombre d’appels, on observe immédiatement que
b(1) = 0 et, si n > 1, b(n) = 1 + b(n− 1) puisque l’évaluation de (butlast l) requiert un
appel à cons et l’évaluation de (butlast (cdr l)).

Théorème. Si [[l]] est une liste de longueur n, alors l’évaluation de (lpref2 l) requiert
1 + n(n+1)/2 appels à cons.
Démonstration. En désignant par L(n) le nombre d’appels, on observe que L(0) = 1 et,
si n > 0, L(n) = 1 + b(n) + L(n − 1) = n + L(n − 1) puisque l’évaluation de (lpref2 l)

requiert un appel à cons et l’évaluation de (lpref2 (butlast l)). On montre facilement
que la seule fonction L de domaine N qui vérifie L(0) = 1 et L(n) = n + L(n− 1) si n > 0
vérifie aussi L(n) = 1 + n(n+1)/2 .

A.2.3 Troisième version

Lemme. Si [[l]] est une liste de longueur n et si [[a]] est une liste, alors l’évaluation de
(rev l a) requiert n appels à cons.

Démonstration. En désignant par r(n) le nombre d’appels, on observe que r(0) = 0 et,
si n > 0, r(n) = 1 + r(n − 1) puisque l’évaluation de (rev l a) requiert l’évaluation de
(rev (cdr l) (cons (car l) a)).
Remarque. Si [[l]] est une liste de longueur n, alors l’évaluation de (reverse l) requiert
n appels à cons, puisque cette forme se réduit à (rev l ’()) .

Lemme. Si [[l]] est une liste de longueur n, alors l’évaluation de (lsuff l) requiert n + 1
appels à cons.
Démonstration. En désignant par s(n) le nombre d’appels, on observe immédiatement que
s(0) = 1 et, si n > 1, s(n) = 1 + s(n − 1) puisque l’évaluation de (lsuff l) requiert un
appel à cons et l’évaluation de (lsuff (cdr l)).
Remarque. Si [[l]] est une liste de longueur n, alors [[(lsuff l)]] est la liste des n + 1
suffixes de [[l]], classés par ordre de longueur décroissante.

Lemme. Si [[ll]] est une liste de n listes dont les longueurs respectives sont n − 1, n −
2, . . . , 1, 0, alors l’évaluation de (lreverse ll) requiert n(n + 1)/2 appels à cons.
Démonstration. En désignant par lr(n) le nombre d’appels, on observe immédiatement
que lr(0) = 0 et, si n > 0, lr(n) = 1+r(n−1)+lr(n−1) = n+lr(n−1) puisque l’évaluation
de (lreverse ll) requiert un appel à cons, l’évaluation de (rev (car ll) ’()) et
l’évaluation de (lreverse (cdr ll)).

Théorème. Si [[l]] est une liste de longueur n, alors l’évaluation de (lpref3 l) requiert
(n + 3)(n + 4)/2 appels à cons.
Démonstration. En désignant par L(n) le nombre d’appels, on observe immédiatement
que L(n) = r(n) + s(n) + lr(n + 1), d’où

L(n) = n + (n + 1) +
(n + 1)(n + 2)

2
=

(n + 3)(n + 4)

2
− 4 .

Remarque. Ces résultats obtenus par l’analyse du code confirment les résultats obtenus
par comptage d’instructions lors de l’exécution (cf. § 12.5.3).

A.3 Compléments sur le langage Scheme

Nous n’avons pas présenté ni même évoqué toutes les primitives du langage Scheme. Dans
ce paragraphe nous introduisons quelques constructions supplémentaires utiles.

A.3.1 Quasi-citation et macros

Nous avons vu que la forme spéciale quote (caractère ’) permettait de “bloquer”
l’évaluation d’une expression. La variante backquote (caractère ‘), couplée avec les
formes spéciales unquote (caractère ,) et unquote-splicing (caractère ,@) donne plus
de souplesse et permet le déblocage partiel, ainsi que le montre la session suivante :

‘(1 2 ,(sqrt 9)) ==> (1 2 3)

‘(4 ,(map sqrt ’(25 36))) ==> (4 (5 6))

‘(4 ,@(map sqrt ’(25 36))) ==> (4 5 6)

Le langage Scheme permet à l’utilisateur de définir ses propres mécanismes.
Supposons par exemple que la forme spéciale or ne soit pas prédéfinie. La fonction my_or,
définie par l’expression

(define my_or

(lambda args

(if (null? args)

#f

(let ((test (car args)))

(if test test (apply my_or (cdr args)))))))

ne convient pas, précisément parce qu’il s’agit d’une fonction : tous les arguments doivent
être évalués avant que la fonction soit appliquée. On peut par contre définir my_or comme
une forme spéciale, par une définition de macro :

(define-syntax my_or

(syntax-rules ()

((my_or) #f)

((my_or e) e)

((my_or e1 e2 ...)

(let ((x e1))

(if x x (my_or e2 ...))))))

La notion de macro ne sera pas définie ici; notons seulement que l’effet de la définition
de macro est de provoquer une réécriture. Par exemple, lors de l’évaluation de
(my_or a b c), cette expression sera réécrite en l’expression

(let ((x a))

(if x x (let ((x b)) (if x x c))))

qui sera alors évaluée. La nouvelle forme spéciale my_or est entièrement équivalente à la
forme spéciale or.

Un autre exemple de macro est

(define-syntax let

(syntax-rules ()

((let ((var val) ...) e1 e2 ...)

((lambda (var ...) e1 e2 ...) val ...))))

On notera le rôle de l’identificateur “...” qui permet une réécriture correcte de la forme
let en la combinaison correspondante.

A.3.2 Autres constructions utiles

Le define est essentiellement utilisé pour définir des fonctions, au moyen de la forme
lambda. Il est permis d’utiliser une notation plus concise; on peut par exemple abréger

(define fact

(lambda (n)

(if (zero? n) 1 (* n (fact (- n 1))))))

en

(define (fact n)

(if (zero? n) 1 (* n (fact (- n 1)))))

Cette notation est commode mais nous l’avons évitée dans ce livre pour souligner
que l’objet défini est bien la fonction [[fact]] et non une hypothétique “expression
paramétrique” [[(fact n)]].

On peut aussi utiliser define pour des définitions locales, à la place de let ou letrec.
Le code

(define (fact n)

(define (fact_it n a)

(if (zero? n)

a

(fact_it (- n 1) (* n a))))

(fact_it n 1))

est équivalent au code

(define fact

(lambda (n)

(letrec ((fact_it

(lambda (n a)

(if (zero? n)

a

(fact_it (- n 1) (* n a))))))

(fact_it n 1))))

La construction do est parfois utilisée pour construire des itérations; en voici un
exemple :

(do ((vec (make-vector 5))

(i 0 (+ i 1)))

((= i 5) vec)

(vector-set! vec i i)) ==> #(0 1 2 3 4)

Les variables vec et i sont initialisées à un vecteur de longueur 5 et à 0, respectivement.
A chaque étape, la valeur [[i]] est affectée à la composante d’indice [[i]] de vec et la
valeur [[(+ i 1)]] est affectée à i. L’itération se termine quand [[(= i 5)]] est vrai et la
valeur [[vec]] est renvoyée.

En pratique, on préfère une construction plus riche, le named-let, qui correspond à un
usage particulier mais fréquent du letrec : la définition d’une fonction auxiliaire récursive
(éventuellement terminale) et son utilisation immédiate. On peut écrire par exemple

((letrec ((fact

(lambda (x)

(if (= x 0) 1 (* x (fact (- x 1)))))))

fact)

5)

ou encore

(letrec ((fact

(lambda (x)

(if (= x 0) 1 (* x (fact (- x 1)))))))

(fact 5))

Une variante syntaxique plus concise de ce qui précède est

(let fact ((x 5))

(if (= x 0) 1 (* x (fact (- x 1)))))

Le nom, qui justifie l’appellation “let nommé” est ici la variable fact. Formellement, cette
construction est définie par une macro, dans laquelle le code pour le let simple apparâıt
aussi :

(define-syntax let

(syntax-rules ()

;; code pour le let simple (voir plus haut)

((let name ((var val) ...) e1 e2 ...)

((letrec ((name

(lambda (var ...) e1 e2 ...)))

name)

val ...))))

